



BASIC Programs

for Small Computers

Things To Do in 4K or Less

Over 30 programs using less than 4K of memory.

For the VIC-20™, TRS-80 Model I® and Color
Computer®, TI-99/4A®, and other computers.

C. Regena

BASIC Programs for Small Computers

Things To Do in 4K or Less

C. Regena

COMPUTE! Publications, Inc. 

One of the ABC Publishing Companies

Greensboro, North Carolina

VIC-20 is a trademark of Commodore Electronics Limited. TI-99/4 and TI-99/4A are trademarks of Texas Instruments, Inc. TRS-80 Color Computer, TRS-80 Model I, and MC-10 are trademarks of Tandy, Inc.

Copyright 1984, COMPUTE! Publications, Inc. All rights reserved.

Reproduction or translation of any part of this work beyond that permitted by Sections 107 and 108 of the United States Copyright Act without the permission of the copyright owner is unlawful.

Printed in the United States of America

ISBN 0-942386-38-8

10 9 8 7 6 5 4 3 2 1

COMPUTE! Publications, Inc., Post Office Box 5406, Greensboro, NC 27403, (919) 275-9809, is one of the ABC publishing companies, and is not associated with any manufacturer of personal computers. VIC-20 is a trademark of Commodore Electronics Limited. TI-99/4 and TI-99/4A are trademarks of Texas Instruments, Inc. TRS-80 Color Computer, TRS-80 Model I, and MC-10 are trademarks of Tandy, Inc.

Contents

Acknowledgments	v
Foreword	vii
Introduction	ix
Chapter 1: Conserving Memory	1
Chapter 2: Translating BASIC	11
Chapter 3: Quick Calculations	21
Chapter 4: Graphics	45
Chapter 5: Sound	87
Chapter 6: Math Competency	111
Chapter 7: Homework Helpers	147
Chapter 8: Computer Sorting	193
Chapter 9: Fun in Less Than 4K	207
Appendix A: A Beginner's Guide to Typing In Programs ...	259
Appendix B: How to Type In Programs	261
Appendix C: BASIC Keyword Abbreviations	264
Index	266



Acknowledgments


I want to offer special thanks to the staff of South Elementary School in Cedar City, Utah, who gave me the idea for this book. The school had purchased three computers, then received six hand-me-down computers from the high school. Although the teachers were excited, they asked, "What can we do with these computers?" This school, like many others, is anxious to give its students quality education, which includes computer experience.

In this book I offer some suggestions for what you *can* do with computers that have limited memory — and what you *can* do without having to spend any more money on peripherals.

I appreciate the help of my five children, Chery, Richard, Cindy, Bob, and Randy Whitelaw, who helped type the programs, then tested them to make sure they worked on several computers. My children and their teachers have continued to give me ideas for possible computer programs. I also acknowledge my husband, Chandler Whitelaw. He never complained as I outgrew my computer room and put computers in the kitchen, dining room, and bedroom.

Without the help of the COMPUTE! Book editors, this book would not be possible. I appreciate their suggestions and encouragement.

C. Regena



Foreword

If you own a VIC-20, a TRS-80 Color Computer, or another computer with limited memory, you probably know that it can be difficult to make it do complicated things. Memory restrictions simply prevent you from taking advantage of many popular commercial programs. Many useful programs seem to require more memory than your computer has.


However, there's a lot you can do with limited memory, and some of those huge software applications can be trimmed down or cut into miniprograms that run quite nicely on the computer you already have.

C. Regena is the popular author of the best-selling *Programmer's Reference Guide to the TI-99/4A* and *Programmer's Reference Guide to the Color Computer*; she also writes a column about the VIC-20 and Commodore 64 in *COMPUTE!'s Gazette* and one about the TI-99/4A in *COMPUTE! Magazine*. She has written and collected some of her best and most powerful programs for this book. All the programs take up less than 4K of memory in your computer—some of them considerably less. But they're all useful, from those which teach typing to those which test your knowledge of music or geography.

You'll see how to conserve memory in programs, letting you make the most of the memory your computer has. You'll even see how to translate programs from one computer's version of BASIC to another. Best of all, though, you'll have more than 100 programs to type in and use.

In fact, to make this book as useful as possible, many of the programs are accompanied by explanations of how the program works. Studying the programs and explanations will show you techniques that you can use as you write your own programs.

The programs range from games to loan and interest calculators, from teaching you the notes of a keyboard to plotting points on a graph. There are programs which turn your computer into a powerful calculator and programs which test your skill at solving mathematical word problems. You'll find uses for your computer again and again as you expand your library of programs.



Introduction

Many microcomputers used in homes and schools today have limited memory. Some companies started with a 4K model, then later added more memory and more capabilities. For example, Radio Shack attracted the new home consumer market with the TRS-80 Model I, a 4K black-and-white computer. Most dealers are now selling the Model IV (and others). Radio Shack's color computer line started with a 4K model and a 16K Extended BASIC model. That 4K Color Computer (affectionately called CoCo) has been replaced by the TRS-80 Micro Color Computer, Model MC-10, a smaller, compact microcomputer in the \$100 range.

Commodore swept the home market with the 5K VIC-20 (actually 3583 bytes of free memory), an easy-to-use color computer with a full-size keyboard which now sells for under \$100. Early Atari 400 and 800 owners probably purchased their computers with 8K. Sinclair owners could purchase a 1K or 2K computer or could make one from a kit.

Many people use their first computers to learn about computing, then upgrade to a more expensive computer with greater memory capacity. A lot of these early models are being shoved aside, given as hand-me-downs to another family member, or donated to schools. Many new computer owners are purchasing limited-memory computers because the price is attractive, and they figure *any* computer is better than *no* computer. Schools are purchasing low-end computers so they can get *more* computers and more students can have hands-on experience in class.

But you may have found that owning a computer is much like buying a house — you need to keep buying peripherals to go with it. In most cases, the computer peripherals cost more than the computer, and you may not be willing to add on those expenses. You want to make do with what you have.

But What Can You Do With 4K?

There is much you *can* do within 4K, and without buying extras. You still have a very powerful machine without having to add anything on. You *can* enjoy your computer just as it is.

I consider working within the memory limitations of a computer one of the challenges of that computer, just like having eight colors, a certain size screen, or particular graphics commands. You can't expect your microcomputer to do everything a mainframe computer can do — you didn't have to pay as much for your little computer. But it can do a few things that a big computer cannot do.

I hope these programs and ideas will be helpful to you and will whet your appetite so you can develop your own programs within the limitations of your computer. Although most of these programs are educational, I have tried to include a variety of subjects so you can get an idea of the many different applications for home computers.


I assume you have a computer, a television or monitor to see what you're doing, and a cassette recorder with blank tapes to save your programs, plus all the necessary cables to connect everything. You don't need to buy anything else to run the programs in this book.

The first section (Chapters 1-3) consists of programs that contain no graphics or sound statements. They are written in translatable BASIC, so you should be able to use them with any computer that uses BASIC. It was impossible to test the programs with *all* microcomputers, so explanations are included in case you need to make minor translations. Computers with larger memory capabilities may also use these programs.

The second section (Chapters 4-9) consists of programs for specific computers — TRS-80 Model I, TRS-80 Color Computer, MC-10 Micro Color Computer, unexpanded VIC-20, TI-99/4, and TI-99/4A. (Although the TI computers do not have the memory limitations of the other machines, program versions have been included. They can usually be entered in a limited amount of time, and are useful to any computer owner.)

The listings can even give you an idea of how to make these programs work on other computers.

Chapter 1
**Conserving
Memory**



Chapter 1

Conserving Memory

If you program efficiently, you can really fit quite a bit into 4K or less. Here are some suggestions for memory-saving programming techniques.

- **Delete REMark statements.** As you are programming, REMark statements help to document your program so you can keep track of various sections of the program. However, if you need to conserve memory, the first and easiest step could be to remove all the REM statements.

- **Use short variable names.** Many computers recognize only the first two letters of the name anyway, so the rest take up valuable space. If you can use variables with one letter, you conserve even more memory than by using two-letter names. For example, REDSCORE and BLUESCORE may be shortened to *R* and *B*.

- **Repeat variable names where possible.** For example, in the beginning of the program you may read in ten variables:

```
10 FOR I=1 TO 10
20 READ V
30 PRINT V,V*V
40 NEXT I
```

Later in the program you have another counter loop. Use the variable *I* again:

```
600 FOR I=1 TO 1000
610 NEXT I
```

Many computers allow you to leave off the index in a NEXT statement:

```
50 FOR J=1 TO 50
60 NEXT
```

- **Use FOR-NEXT or GOSUB.** Anytime you have repetitious code, see if a FOR-NEXT loop or a GOSUB procedure will work.

- **Plan all GOTO logic.** If you have many GOTO statements in your program, it may be much more efficient to rearrange your program.

- **Use the DRAW command, if your computer allows, to create graphic displays.** Subroutines or FOR-NEXT loops can save lines when used with the DRAW command.

- **DIMension wisely.** If you need to use an array with 25 names, you waste valuable memory if you arbitrarily DIMension for 40 names. Many computers automatically DIMension 11 elements whenever you use a subscripted number without a previous

Conserving Memory

DIMension statement. If you really need fewer subscripts, be sure to use the DIMension statement to save memory. Also remember to use the zero element — you don't need to start with A(1).

- **READ and DATA.** If you have several assignment statements, a DATA and READ procedure may be more efficient:

```
10 A=1
20 B=3
30 C=2
40 D=5
50 E=7
```

is equivalent to

```
10 READ A,B,C,D,E
20 DATA 1,3,2,5,7
```

- **Use ON-GOTO or ON-GOSUB instead of several logical IF-THEN statements.** For example:

```
10 IF I=1 THEN 640
20 IF I=2 THEN 730
30 IF I=3 THEN 870
40 IF I=4 THEN 930
```

is equivalent to

```
10 ON I GOTO 640,730,870,930
```

You may first need to make sure that variable I is a positive integer less than 5.

- **Leave out spaces if your computer allows it.** The statements are harder to read, but use less space. If a BASIC word follows a variable name, a space may be required between the variable and the next word.

- **If your computer allows it, use multistatement lines.** The most common statement separator is the colon. Some computers use the slash. Use abbreviations, if your computer allows them, to crowd more on each line. For example, many computers use ? for PRINT and ! or ' (apostrophe) for REMark. Abbreviations alone will not save memory, but by using them you can put more statements and commands on a line, saving a small number of bytes otherwise used for additional line numbers. You need to be careful when you're combining statements if an IF-THEN statement is involved or if any program transfer statements go to one of the statements to be combined.

- **Use low line numbers.** Often during program development it is wise to number your statements by tens so you can

later insert statements. I often plan different sections of the program starting with the thousands. After a program is complete, I save memory by renumbering the lines — perhaps using 10, 20, 30, 40, and so on, or, even better, 1, 2, 3, 4, etc.

Before you begin entering any of the programs in this book, take a look at Appendix A, "A Beginner's Guide to Typing In Programs," and Appendix B, "How to Type In Programs." These will help you understand how to enter the listings in this book into your own computer.

Nouns

Computers: VIC-20; TRS-80 Model I; TRS-80 16K Color Computer; MC-10; TI-99/4; TI-99/4A

Following is an example of how to conserve memory by combining lines. The first program, "Nouns," works as is on a VIC-20. To add another section of DATA or to add graphics and sound, this program needs to take up less memory.

Combining commands for one statement number usually saves five bytes per line (depending on your computer). In conserving memory for Nouns, I went through the listing and marked every line that was referred to by another command — such as GOTO, GOSUB, or IF-THEN, plus every statement after an IF-THEN statement. I had to make sure those lines contained the correctly referenced line numbers.

The second listing of Nouns shows how the lines were combined. For instance, notice that lines 4 through 30 in the VIC program were combined into lines 2 and 3 in the TRS-80 version. Lines 32 through 48 in the VIC version were combined into two lines, lines 4 and 5, in the TRS-80 program. Commands such as GOSUB were carefully modified to reflect the new line numbers in the second, crunched version of the program. Line 48 in the VIC program, for example, read GOSUB200. This is the last expression in line 5 of the TRS-80 program, but it was changed to GOSUB 22 because of the new line numbering of this shorter listing. Combining lines in this way saves memory, but you need to be careful to prevent errors in line numbers.

The second listing works on the TRS-80 Model I, TRS-80 16K Color Computer, and MC-10 computer. The third listing is the TI-99/4 or TI-99/4A version. Note that TI BASIC does not allow combining commands on one line, but you can combine PRINT statements.

Nouns presents a quiz to identify nouns. A word is chosen randomly and printed on the screen. The user presses 1 if the

Conserving Memory

word is a person, 2 if the word is a place, 3 if the word is a thing, or 4 if the word is not a noun. After ten words a score is given. Since the answers are multiple-choice, the program does not continue until the correct number is chosen. The score is based on correct answers for the first response.

Program 1-1. Nouns

VIC-20

```
4 PRINT"{CLR}"
6 PRINTTAB(5);"*****"
8 PRINTTAB(5);"* NOUNS *"
10 PRINTTAB(5);"*****"
12 PRINT
14 PRINT"A NOUN IS A WORD THAT"
16 PRINT"IS A PERSON,{2 SPACES}PLACE,"
18 PRINT"OR THING."
20 PRINT
22 PRINT"YOU WILL BE SHOWN A"
24 PRINT"WORD.{2 SPACES}PRESS THE"
26 PRINT"RIGHT NUMBER FOR"
28 PRINT"PERSON, PLACE, THING,"
30 PRINT"OR NOT A NOUN."
32 DIMW$(100),W(100)
34 FORC=0TO100
36 READW$(C),W(C)
38 NEXT C
40 N$(1)="PERSON"
42 N$(2)="PLACE"
44 N$(3)="THING"
46 N$(4)="NOT A NOUN"
48 GOSUB200
49 D=0
50 SC=0
52 FORI=1TO10
54 PRINT"{CLR}"
56 FL=0
58 R=INT(RND(0)*101)
60 IFW(R)=0THEN58
62 PRINT"{2 SPACES}** ";W$(R);" **"
64 PRINT
66 FORC=1TO4
68 PRINTC;N$(C)
70 NEXTC
72 PRINT
74 PRINT"?"
76 GETES
78 IFE$=""THEN76
80 E=ASC(E$)-48
```

Conserving Memory

```
82 IF(E<1)+(E>4)THEN76
84 PRINTN$(E)
86 IFE=W(R)THEN96
88 PRINT"SORRY, TRY AGAIN."
90 PRINT
92 FL=1
94 GOTO76
96 PRINT"CORRECT!"
98 IF FL=1 THEN 104
100 SC=SC+1
102 W(R)=0
104 GOSUB200
106 NEXT I
107 PRINT"{CLR}"
108 PRINTTAB(5); "NOUNS"
110 PRINT
112 PRINT"SCORE: "
114 PRINTSC; "CORRECT"
116 PRINTTAB(5); "OUT OF 10 WORDS"
118 PRINT
120 PRINT
121 D=D+1:IFD>8THEN300
122 PRINT"PRESS 1 TRY AGAIN"
124 PRINTTAB(6); "2 END"
126 GETES$
128 IF E$="1"THEN 50
130 IF E$="2"THEN300
132 GOTO126
200 PRINT
202 PRINT"PRESS RETURN"
204 GETES$
206 IFE$=""THEN204
208 IFASC(E$)<>13 THEN 204
210 RETURN
220 DATAMAN, 1, BOY, 1, DOG, 3, TEXAS, 2, ROGER, 1, WALL, 3, W
AS, 4, UTAH, 2, IDAHO, 2, CHAIR, 3, HAT, 3
222 DATAHAND, 3, GIRL, 1, LADY, 1, JILL, 1, CINDY, 1, JOE, 1,
BOB, 1, COWBOY, 1, TOWN, 2, SKY, 2, HOME, 2
224 DATARIVER, 3, HILL, 3, GLASS, 3, CUP, 3, DISH, 3, WENT, 4
, BEFORE, 4, AFTER, 4, WHEN, 4, NOW, 4, THEN, 4
226 DATADRESS, 3, PANTS, 3, SHIRT, 3, SOCKS, 3, SHOES, 3, MA
P, 3, FLAG, 3, MOTHER, 1, FATHER, 1, DAD, 1
228 DATASISTER, 1, BROTHER, 1, BABY, 1, DESK, 3, TABLE, 3, W
ATER, 3, FLOWER, 3, OHIO, 2, IOWA, 2, MAINE, 2
230 DATABOOK, 3, PAPER, 3, PENCIL, 3, BOX, 3, TREE, 3, BEAR,
3, FOX, 3, BALL, 3, QUIET, 4, QUICK, 4
232 DATABLUE, 4, GREEN, 4, BEADS, 3, ANNE, 1, ABLE, 4, ABOUT
, 4, ABOVE, 4, ACROSS, 4, ACTOR, 1, AGAIN, 4
234 DATABACON, 3, BAKE, 4, BIRD, 3, BANANA, 3, BEACH, 2, PAR
IS, 2, LONDON, 2, JAPAN, 2, MEXICO, 2
```

Conserving Memory

```
236 DATA AFRICA, 2, CANADA, 2, CAR, 3, NEVER, 4, WALKED, 4, J
    UMPED, 4, OF, 4, ARE, 4, BY, 4, ATE, 4, BIG, 4
238 DATA INTO, 4, HAPPY, 4, SAD, 4, ALICE, 1, SMALL, 4, EGYPT
    , 2, TEACHER, 1, IS, 4
300 END
```

Program 1-2. Nouns

TRS-80 Model I and Color Computer; MC-10

```
2 CLS:PRINTTAB(12);"*****":PRINTTAB(12);"* NOU
  NS *":PRINTTAB(12);"*****":PRINT:PRINT"A NOU
  N IS A WORD THAT IS A":PRINT"PERSON, PLACE, OR T
  HING."
3 PRINT:PRINT"YOU WILL BE SHOWN A WORD.":PRINT"PRE
  SS THE RIGHT NUMBER FOR":PRINT"PERSON, PLACE, TH
  ING, OR":PRINT"NOT A NOUN."
4 DIM W$(100),W(100)
5 FORC=0TO100:READW$(C),W(C):NEXTC:N$(1)="PERSON":
  N$(2)="PLACE":N$(3)="THING":N$(4)="NOT A NOUN":G
  OSUB22
6 S=0:FORI=1TO10
7 CLS:F=0
8 R=RND(100):IF W(R)=0 THEN 8
9 PRINT" ** ";W$(R);" *":PRINT:FORC=1TO4:PRINTC;
  N$(C):NEXT:PRINT:PRINT"?"
10 E$=INKEY$:IFE$=""THEN10
11 E=ASC(E$)-48:IF(E<1)+(E>4)THEN10
12 PRINTN$(E):IFE=W(R)THEN14
13 PRINT"SORRY, TRY AGAIN.":PRINT:F=1:GOTO10
14 PRINT"CORRECT!":IFF=1THEN16
15 S=S+1:W(R)=0
16 GOSUB22:NEXT
17 CLS:PRINTTAB(5);"NOUNS":PRINT:PRINT"SCORE:":PRI
  NTS;"CORRECT":PRINTTAB(5);"OUT OF 10 WORDS":PRI
  NT:PRINT:D=D+1:IFD>8THEN33
18 PRINT"PRESS 1 TRY AGAIN":PRINTTAB(6);"2 END"
19 E$=INKEY$:IFE$="1"THEN6
20 IFE$="2"THEN33
21 GOTO19
22 PRINT:PRINT"PRESS ENTER"
23 E$=INKEY$:IFE$=""THEN23
24 IF ASC(E$)<>13THEN23
25 RETURN
26 DATAMAN, 1, BOY, 1, DOG, 3, TEXAS, 2, ROGER, 1, WALL, 3, WA
  S, 4, UTAH, 2, IDAHO, 2, CHAIR, 3, HAT, 3, HAND, 3, GIRL, 1,
  LADY, 1, JILL, 1, CINDY, 1, JOE, 1
27 DATABOB, 1, COWBOY, 1, TOWN, 2, SKY, 2, HOME, 2, RIVER, 3,
  HILL, 3, GLASS, 3, CUP, 3, DISH, 3, WENT, 4, BEFORE, 4, AFT
  ER, 4, WHEN, 4, NOW, 4, THEN, 4
```

- 28 DATADDRESS, 3, PANTS, 3, SHIRT, 3, SOCKS, 3, SHOES, 3, MAP
3, FLAG, 3, MOTHER, 1, FATHER, 1, DAD, 1, SISTER, 1, BROTH
ER, 1, BABY, 1, DESK, 3, TABLE, 3
- 29 DATAWATER, 3, FLOWER, 3, OHIO, 2, IOWA, 2, MAINE, 2, BOOK
3, PAPER, 3, PENCIL, 3, BOX, 3, TREE, 3, BEAR, 3, FOX, 3, B
ALL, 3, QUIET, 4, QUICK, 4
- 30 DATABLUE, 4, GREEN, 4, BEADS, 3, ANNE, 1, ABLE, 4, ABOUT,
4, ABOVE, 4, ACROSS, 4, ACTOR, 1, AGAIN, 4, BACON, 3, BAKE
4, BIRD, 3, BANANA, 3, BEACH, 2
- 31 DATAPARIS, 2, LONDON, 2, JAPAN, 2, MEXICO, 2, AFRICA, 2,
CANADA, 2, CAR, 3, NEVER, 4, WALKED, 4, JUMPED, 4, OF, 4, A
RE, 4, BY, 4, ATE, 4, BIG, 4
- 32 DATAINTO, 4, HAPPY, 4, SAD, 4, ALICE, 1, SMALL, 4, EGYPT,
2, TEACHER, 1, IS, 4
- 33 END

Program 1-3. Nouns

TI-99/4A

```
110 CALL CLEAR
120 PRINT TAB(5); "*****":TAB(5); "* NOUNS *":TA
B(5); "*****"
130 PRINT : "YOU WILL BE SHOWN A WORD. {3 SPACES}PRE
SS THE RIGHT NUMBER FOR PERSON, PLACE, THING,
OR {4 SPACES}NOT A NOUN."
140 DIM W$(100),W(100)
150 FOR C=0 TO 100
160 READ W$(C),W(C)
170 NEXT C
180 N$(1)="PERSON"
190 N$(2)="PLACE"
200 N$(3)="THING"
210 N$(4)="NOT A NOUN"
220 GOSUB 570
230 D=0
240 SC=0
250 FOR I=1 TO 10
260 CALL CLEAR
270 FL=0
280 RANDOMIZE
290 R=INT(101*RND)
300 IF W(R)=0 THEN 290
310 PRINT " ** ";W$(R); " ***":
320 FOR C=1 TO 4
330 PRINT C;N$(C)
340 NEXT C
350 PRINT : "?"
360 CALL KEY(0,K,S)
370 IF (K<49)+(K>52)THEN 360
380 PRINT :N$(K-48)
```

Conserving Memory

```
390 IF K=48=W(R)THEN 430
400 PRINT "SORRY, TRY AGAIN."::
410 FL=1
420 GOTO 360
430 PRINT "CORRECT!"
440 IF FL=1 THEN 470
450 SC=SC+1
460 W(R)=0
470 GOSUB 570
480 NEXT I
490 CALL CLEAR
500 PRINT TAB(5);"NOUNS":;"SCORE":;SC;"CORRECT":TAB
  B(5);"OUT OF 10 WORDS":;:
510 D=D+1
520 IF D>8 THEN 690
530 PRINT "PRESS 1 TRY AGAIN":TAB(7);"2 END"
540 CALL KEY(0,K,S)
550 IF K=49 THEN 240
560 IF K=50 THEN 690 ELSE 540
570 PRINT "PRESS ENTER"
580 CALL KEY(0,K,S)
590 IF K<>13 THEN 580
600 RETURN
610 DATA MAN,1,BOY,1,DOG,3,TEXAS,2,ROGER,1,WALL,3,
  WAS,4,UTAH,2,IDAHO,2,CHAIR,3,HAT,3,HAND,3,GIRL
  ,1,LADY,1
620 DATA JILL,1,CINDY,1,JOE,1,BOB,1,COWBOY,1,TOWN,
  2,SKY,2,HOME,2,RIVER,3,HILL,3,GLASS,3,CUP,3,DI
  SH,3,WENT,4
630 DATA BEFORE,4,AFTER,4,WHEN,4,NOW,4,THEN,4,DRES
  S,3,PANTS,3,SHIRT,3,SOCKS,3,SHOES,3,MAP,3,FLAG
  ,3,MOTHER,1
640 DATA FATHER,1,DAD,1,SISTER,1,BROTHER,1,BABY,1,
  DESK,3,TABLE,3,WATER,3,FLOWER,3,OHIO,2,IOWA,2,
  MAINE,2
650 DATA BOOK,3,PAPER,3,PENCIL,3,BOX,3,TREE,3,BEAR
  ,3,FOX,3,BALL,3,QUIET,4,QUICK,4,BLUE,4,GREEN,4
  ,BEADS,3
660 DATA ANNE,1,ABLE,4,ABOUT,4,ABOVE,4,ACROSS,4,AC
  TOR,1,AGAIN,4,BACON,3,BAKE,4,BIRD,3,BANANA,3,B
  EACH,2
670 DATA PARIS,2,LONDON,2,JAPAN,2,MEXICO,2,AFRICA,
  2,CANADA,2,CAR,3,NEVER,4,WALKED,4,JUMPED,4,OF,
  4,ARE,4
680 DATA BY,4,ATE,4,BIG,4,INTO,4,HAPPY,4,SAD,4,ALI
  CE,1,SMALL,4,EGYPT,2,TEACHER,1,IS,4
690 END
```

Chapter 2
**Translating
BASIC**

Grade 5
Translating
BASIC

Chapter 2

Translating BASIC

Translating programs from one version of BASIC to another computer's BASIC can become a skill of its own. Some BASIC programs are translatable, but others aren't. It all depends on how complex the program is and what special features of a computer it uses.

Many of the programs in this book are computer-specific. In other words, the program will work on only one computer, using that machine's version of BASIC. Others, however, can be translated from one computer to another. The short programs listed in this chapter, and in Chapter 3, are good examples of "translatable" programs. All of these programs can be entered into almost any home computer which uses BASIC. This is possible because the programs are simple and use no special graphic or sound features peculiar to only one or two computers.

Before each program in this book is a note indicating which computers can use the program as it's listed. Sometimes this note will list various computers. Other times it will simply read *translatable*. When you see this word, you should be able to type the program in and RUN it on your own computer, no matter what model it is. All you have to do is look the program over carefully and make a few simple changes. The most important commands that must be changed from one computer to another are listed below:

To clear the screen:

TRS-80, MC-10: CLS

VIC-20, 64, PET: PRINT "{CLR}" (SHIFT and CLR/HOME)

TI-99/4, TI-99/4A: CALL CLEAR

Apple: HOME

Atari: PRINT " " (ESC CTRL CLR)

Timex/Sinclair: CLS

Random number from 1 to A:

TRS-80, MC-10: RND(A)

TRS-80, Extended BASIC: R=RND(-TIMER):R=RND(A)

VIC-20, 64, PET: INT(A*RND(0)+1)

TI-99/4, TI-99/4A: RANDOMIZE
INT(A*RND+1)

Apple: RND(A)+1

Apple IIe: INT(A*RND(1)+1)

Atari: INT(A*RND(1)+1)

Timex/Sinclair: INT(A*RND+1)

Translating BASIC

To detect a key being pressed:

TRS-80, MC-10: E\$ = INKEY\$

VIC-20, 64, PET: GET E\$

TI-99/4, TI-99/4A: CALL KEY(0,K,S)

Apple: GET E\$

Atari: E = PEEK(764)

Timex/Sinclair: LET E\$ = INKEY\$

Not all of the computers referred to here have programs specific to them in the book, but I've included the commands to help you as you translate the programs in this chapter and in Chapter 3 to your own computer.

Besides these changes, you may want to add graphics and sound to enhance the program as it is printed here. Commands involving string variables may need to be changed as well. You will probably also want to adjust the PRINT statements to reflect the number of columns your own monitor screen has.

If the note listing the computers a program can run on indicates it works for the TRS-80 Color Computer, it will work for both the 4K and 16K models, as well as with the 16K Extended BASIC computer.

Some of the VIC-20 programs in this book have a short line before the listing, stating that lines must be entered using abbreviations. If you don't abbreviate the keywords in these lines, the program will not work. Be sure to use abbreviations for all BASIC keywords when you type in the lines marked with an asterisk (*). If you don't, those programs probably won't run; if they do, they won't work correctly. Most of the abbreviations are simply the first letter, then SHIFT, and then the second letter of the keyword. Refer to Appendix C, "How to Abbreviate Keywords," for a complete list of BASIC keywords and their abbreviations on the VIC-20.

If you need to edit a line, be sure the line also contains abbreviations when you are through. For example, if you LIST a line that you previously typed with abbreviations, the full keyword is displayed. If you then edit that line on the screen, part of it may be cut off, since the VIC-20 will show a maximum of only 87 characters on a line. Your program may not run if this happens. You need to either retype the entire line or replace all the keywords in the line you LISTed with abbreviations. This is especially important with all lines marked with an asterisk (*) in the VIC-20 program listings.

Roman Numerals

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; MC-10;
TI-99/4; TI-99/4A

"Roman Numerals" illustrates how a standard BASIC program can be adapted for other computers. The first listing works exactly as is on the TRS-80 Model I, TRS-80 Color Computer, and MC-10. The second listing shows how the program was changed for the VIC-20. The commands for clearing the screen, choosing a random number, and detecting a key pressed were changed, and some of the PRINT statements were adapted for the 22-column screen and color capabilities of the VIC-20. The third listing is the TI-99/4 or TI-99/4A version.

Here's how Roman Numerals works:

Line	Function
110-150	Reads from DATA the Roman numeral values for the Arabic numbers in hundreds H\$, tens T\$, and ones S\$.
160	Clear the screen.
170	Initialize R\$ to be the null string.
180-190	Choose a random number from 1 to 1999 and set NN equal to N.
200-220	If the number starts with 1000, set R\$ to "M" and subtract 1000 from N.
230-260	Calculate the hundreds value for the Roman numeral.
270-300	Calculate the tens value for the Roman numeral.
310-320	Calculate the ones value for the Roman numeral.
330-340	PRINT the Roman numeral, R\$.
350-380	Ask for the corresponding number as INPUT.
390	Test user's answer.
400	PRINT correct answer if user's answer is incorrect.
410-420	If user's answer is incorrect, PRINT "PRESS ENTER" message, then return for another problem.
430-440	PRINT message for correct answer.
450-490	PRINT option for another problem and branch appropriately.
500	END.

Program 2-1. Roman Numerals

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
110 FOR I=1 TO 9
120 READ H$(I), T$(I), S$(I)
130 NEXT I
```

Translating BASIC

```
140 DATA C,X,I,CC,XX,II,CCC,XXX,III,CD,XL,IV,D,L,V
150 DATA DC,LX,VI,DCC,LXX,VII,DCCC,LXXX,VIII,CM,XC
    ,IX
160 CLS
170 R$=""
180 N=RND(1999)
190 NN=N
200 IF N<1000 THEN 230
210 R$="M"
220 N=N-1000
230 IF N<100 THEN 270
240 NR=INT(N/100)
250 R$=R$+H$(NR)
260 N=N-NR*100
270 IF N<10 THEN 310
280 NR=INT(N/10)
290 R$=R$+T$(NR)
300 N=N-NR*10
310 IF N=0 THEN 330
320 R$=R$+S$(N)
330 PRINT "GIVEN THE ROMAN NUMERAL"
340 PRINT R$
350 PRINT "WHAT IS THE CORRESPONDING"
360 PRINT "NUMBER?"
370 INPUT ANS
380 PRINT
390 IF ANS=NN THEN 430
400 PRINT "THE NUMBER IS";NN
410 INPUT "PRESS <ENTER>";E$
420 GOTO 160
430 PRINT "CORRECT!"
440 PRINT
450 PRINT "ANOTHER PROBLEM?"
460 PRINT "(Y/N)"
470 E$=INKEY$
480 IF E$="Y" THEN 160
490 IF E$<>"N" THEN 470
500 END
```

Program 2-2. Roman Numerals

VIC-20

```
110 FOR I=1 TO 9
120 READ H$(I),T$(I),S$(I)
130 NEXT I
140 DATA C,X,I,CC,XX,II,CCC,XXX,III,CD,XL,IV,D,L,V
150 DATA DC,LX,VI,DCC,LXX,VII,DCCC,LXXX,VIII,CM,XC
    ,IX
160 PRINT "{CLR}"
170 R$=""
```

```
180 N=INT(1999*RND(0))+1
190 NN=N
200 IF N<1000 THEN 230
210 R$="M"
220 N=N-1000
230 IF N<100 THEN 270
240 NR=INT(N/100)
250 R$=R$+H$(NR)
260 N=N-NR*100
270 IF N<10 THEN 310
280 NR=INT(N/10)
290 R$=R$+T$(NR)
300 N=N-NR*10
310 IF N=0 THEN 330
320 R$=R$+S$(N)
330 PRINT"{BLU}GIVEN THE ROMAN":PRINT"NUMERAL{RED}
"
340 PRINT R$
350 PRINT:PRINT"{BLU}WHAT IS THE"
360 PRINT"CORRESPONDING NUMBER?{BLK}"
370 INPUT ANS
380 PRINT
390 IF ANS=NN THEN 430
400 PRINT"{BLU}THE NUMBER IS";NN
410 PRINT:INPUT"{GRN}PRESS <RETURN>";E$
420 GOTO 160
430 PRINT"{BLU}CORRECT!"
440 PRINT
450 PRINT "ANOTHER PROBLEM?"
460 PRINT "(Y/N)"
470 GET E$
480 IF E$="Y" THEN 160
490 IF E$<>"N" THEN 470
500 END
```

Program 2-3. Roman Numerals

TI-99/4; TI-99/4A

```
110 FOR I=1 TO 9
120 READ H$(I),T$(I),S$(I)
130 NEXT I
140 DATA C,X,I,CC,XX,II,CCC,XXX,III,CD,XL,IV,D,L,V
150 DATA DC,LX,VI,DCC,LXX,VII,DCCC,LXXX,VIII,CM,XC
,IX
160 CALL CLEAR
170 R$=""
175 RANDOMIZE
180 N=INT(1999*RND)+1
190 NN=N
200 IF N<1000 THEN 230
```

Translating BASIC

```
210 R$="M"
220 N=N-1000
230 IF N<100 THEN 270
240 NR=INT(N/100)
250 R$=R$&H$(NR)
260 N=N-NR*100
270 IF N<10 THEN 310
280 NR=INT(N/10)
290 R$=R$&T$(NR)
300 N=N-NR*10
310 IF N=0 THEN 330
320 R$=R$&S$(N)
330 PRINT "GIVEN THE ROMAN NUMERAL"
340 PRINT R$
350 PRINT : "WHAT IS THE CORRESPONDING"
360 PRINT "NUMBER?"
370 INPUT ANS
380 PRINT
390 IF ANS=NN THEN 430
400 PRINT "THE NUMBER IS";NN
410 INPUT "PRESS <ENTER>":E$
420 GOTO 160
430 PRINT "CORRECT!"
440 PRINT
450 PRINT "ANOTHER PROBLEM?"
460 PRINT "(Y/N)"
470 CALL KEY(0,K,S)
480 IF K=89 THEN 160
490 IF K<>78 THEN 470
500 END
```

Adverbs

Computers: Translatable

Here is a program written in portable or translatable BASIC. You should be able to translate this program to the version of BASIC your computer uses without much difficulty. You can probably tell which lines to change for your particular computer. The listing works as is on the TRS-80 Model I, TRS-80 Color Computer, and MC-10.

Here's how Adverbs works:

Line	Function
110-200	Clear screen; PRINT instructions.
210-260	READ words in as data (article, noun, verb, and adverb for nine of each type of word).
270-320	Wait for user to press ENTER.

330	Clear screen.
340	Initialize score.
350	Perform quiz ten times.
360	Clear screen.
370-400	Randomly choose words for sentence.
410-470	Randomly choose order for sentence and print sentence.
480-500	Enter answer.
510-530	If answer is incorrect, PRINT correct answer.
540-550	If answer is correct, PRINT message and increment score.
560-610	Wait for user to press ENTER for next problem.
620-680	Clear screen and PRINT score.
690-720	PRINT option to try again and branch appropriately.
730-740	Clear screen and END.

Program 2-4. Adverbs


Translatable

```
110 CLS
120 PRINT TAB(8); "*****"
130 PRINT TAB(8); "* ADVERBS *"
140 PRINT TAB(8); "*****"
150 PRINT
160 PRINT
170 PRINT "YOU WILL BE SHOWN A SENTENCE."
180 PRINT
190 PRINT "TYPE THE ADVERB"
200 PRINT "THEN PRESS <ENTER>."
210 FOR C=0 TO 9
220 READ A$(C), B$(C), C$(C), D$(C)
230 NEXT C
240 DATA THE, CAT, CRAWLED, QUICKLY, A, DOG, JUMPED, QUIE
    TLY, MY, DEER, RAN, HAPPILY, YOUR, COW, LOPED, SLYLY
250 DATA HIS, FOX, WIGGLED, SLOWLY, HER, WOLF, GALLOPED,
    JOYFULLY, ITS, BOY, SPED, RAPIDLY, OUR, GIRL, CREEPED
    , SILENTLY
260 DATA THAT, BUG, HURRIED, CALMLY, ONE, BEAR, MOVED, SW
    IFTLY
270 PRINT
280 PRINT
290 PRINT "PRESS <ENTER> TO START."
300 E$=INKEY$
310 IF E$="" THEN 300
320 IF ASC(E$)<>13 THEN 300
330 CLS
340 S=0
```

Translating BASIC

```
350 FOR T=1 TO 10
360 CLS
370 A=RND(10)-1
380 B=RND(10)-1
390 C=RND(10)-1
400 D=RND(10)-1
410 I=RND(3)
420 ON I GOTO 430,450,470
430 PRINT A$(A);" ";B$(B);" ";C$(C);" ";D$(D);"."
440 GOTO 480
450 PRINT A$(A);" ";B$(B);" ";D$(D);" ";C$(C);"."
460 GOTO 480
470 PRINT D$(D);" ";A$(A);" ";B$(B);" ";C$(C);"."
480 PRINT
490 INPUT "ADVERB ";V$
500 PRINT
510 IF V$=D$(D) THEN 540
520 PRINT "THE ADVERB IS: ";D$(D)
530 GOTO 560
540 PRINT "CORRECT!"
550 S=S+1
560 PRINT
570 PRINT "PRESS <ENTER>."
580 E$=INKEY$
590 IF E$="" THEN 580
600 IF ASC(E$)<>13 THEN 580
610 NEXT T
620 CLS
630 PRINT "YOUR SCORE IS"
640 PRINT TAB(5);"RIGHT",S
650 PRINT TAB(5);"WRONG",10-S
660 PRINT
670 PRINT
680 PRINT
690 PRINT "TRY AGAIN? (Y/N)"
700 E$=INKEY$
710 IF E$="Y" THEN 330
720 IF E$<>"N" THEN 700
730 CLS
740 END
```

Chapter 3
Quick
Calculations



Chapter 3

Quick Calculations

You can use your computer to perform important calculations. In fact, I know a bank that bought a microcomputer system (for about \$6000). When asked what they did with their computer, they said it was used for amortization. You can use your own \$100 computer to do the same thing.

You can use your computer as a glorified calculator — figuring conversions and formulas. The difference between a calculator and a computer is that with a computer you can put the formula in a program once and then not worry about it again. You can RUN the program and input known values. The program can give you the final answer almost instantly. With a calculator, you need to remember the equation or formula for each problem.

Let's look at some examples. The simple program below converts an angle in degrees to an angle in radians. The general conversion formula is $180 \text{ degrees} = \pi \text{ radians}$.

```
110 PRINT
120 INPUT "NUMBER OF DEGREES = ";D
130 PRINT
140 PRINT "RADIANS: ";D*3.14159/180
150 GOTO 110
```

Type in the program, then RUN. The computer will ask for the number of degrees. Enter a number, such as 45. The equivalent number of radians in decimal form will be printed. The program continues until you press RUN/STOP or BREAK.

Many microcomputers allow you to *define* a function at the beginning of the program, calling it a variable name. Later, any calculations using that variable will evaluate the function. There are several restrictions, depending on the type of computer you have. One of the requirements is that the computer must process the line defining the function before the function is actually used. In some computers the definition statement must have a lower line number than any line in which the function is used. To play it safe, put definitions at the beginning of the program.

There are several forms of the definition statement. The Commodore, Apple, and TRS-80 use DEF FN name (variable), where a variable is specified in parentheses and the name is often a letter. Some computers allow functions to be redefined anywhere in the program. An example is:

```
100 DEF FN A(R)=3.14159*R*R
200 PRINT FN A(5)
```

Quick Calculations

The formula with the value of 5 for R is evaluated and printed.

In TI BASIC, leave out FN. More than one variable is permitted, and you don't have to have a dependent variable:

```
100 DEF A(R)=3.14159*R*R
200 PRINT A(5)
```

or

```
100 DEF R16=INT(16*RND+1)
200 CALL COLOR(5,R16,R16)
```

Each time R16 is used, a random number from 1 to 16 is chosen. Within one command you can use the function several times:

```
210 CALL HCHAR(R16,R16,R16+63,R16)
```

Suppose you want to evaluate the expression $10x^2 + 7x + 1$ for values of x equal to 0, 1, 3, 5, -1, and -5. Here is one way it could be done:

```
100 DEF FNA(X)=10*X*X+7*X+1
110 PRINT
120 FOR I=1 TO 6
130 READ X
140 PRINT "X = ";X,"A(X) = ";FNA(X)
150 NEXT I
160 PRINT
170 DATA 0,1,3,5,-1,-5
180 END
```

Line 100 defines the function A(X). The space between FN and A(X) is optional, so to save memory you can leave it out, as above. Line 130 reads the values for X from the DATA statement in line 170. Line 140 prints the X value and the evaluated function.

The following programs illustrate how you can use your computer to take the drudgery out of calculations. These programs can be translated for your computer quite easily. If you want, you can combine lines as shown in Chapter 1 to make the listings shorter. Remember to use your own computer's commands for such things as clearing the screen, creating random numbers, or detecting a key pressed.

Conversions

Computers: Translatable

This program includes conversion factors for linear measurements. First, choose which measurement you need to convert, then enter a number. The equivalent measurement is printed. You can

use this general form to write your own program for the conversions you often need.

Program 3-1. Conversions

Translatable

```
10 CLS
20 PRINT "*** CONVERSIONS ***"
30 PRINT
40 PRINT "CHOOSE:"
50 PRINT " 1 CENTIMETER TO FOOT"
60 PRINT " 2 FOOT TO CENTIMETER"
70 PRINT " 3 CENTIMETER TO INCH"
80 PRINT " 4 INCH TO CENTIMETER"
90 PRINT " 5 INCH TO FOOT"
100 PRINT " 6 FOOT TO INCH"
110 PRINT " 7 METER TO FOOT"
120 PRINT " 8 FOOT TO METER"
130 PRINT " 9 FEET TO MILE"
140 PRINT "10 MILE TO FEET"
150 PRINT "11 MILE TO KILOMETER"
160 PRINT "12 KILOMETER TO MILE"
170 PRINT "13 METER TO YARD"
180 PRINT "14 YARD TO METER"
190 INPUT A
200 IF A>14 THEN 10
210 IF A<1 THEN 10
220 PRINT
230 PRINT
240 ON A GOTO 250,300,350,400,450,500,550,600,650,
    700,750,800,850,900
250 PRINT "1 CENTIMETER = .032808 FOOT"
260 PRINT
270 INPUT "CENTIMETERS = ";N
280 PRINT ".032808*N;" FOOT"
290 GOTO 940
300 PRINT "1 FOOT = 30.48006 CENTIMETER"
310 PRINT
320 INPUT "NUMBER OF FEET = ";N
330 PRINT 30.48006*N;" CENTIMETERS"
340 GOTO 940
350 PRINT "1 CENTIMETER = .39370 INCH"
360 PRINT
370 INPUT "NUMBER OF CENTIMETERS = ";N
380 PRINT N*.39370;" INCH"
390 GOTO 940
400 PRINT "1 INCH = 2.540005 CENTIMETER"
410 PRINT
420 INPUT "NUMBER OF INCHES = ";N
430 PRINT N*2.540005;" CENTIMETERS"
```

Quick Calculations

```
440 GOTO 940
450 PRINT "1 INCH = 1/12 FOOT"
460 PRINT
470 INPUT "NUMBER OF INCHES = ";N
480 PRINT N/12;" FOOT"
490 GOTO 940
500 PRINT "1 FOOT = 12 INCHES"
510 PRINT
520 INPUT "NUMBER OF FEET = ";N
530 PRINT 12*N;" INCHES"
540 GOTO 940
550 PRINT "1 METER = 3.280833 FEET"
560 PRINT
570 INPUT "METERS = ";N
580 PRINT 3.280833*N;" FEET"
590 GOTO 940
600 PRINT "1 FOOT = .3048006 METER"
610 PRINT
620 INPUT "NUMBER OF FEET = ";N
630 PRINT .3048006*N;" METER"
640 GOTO 940
650 PRINT "1 FOOT = 1/5280 MILE"
660 PRINT
670 INPUT "NUMBER OF FEET = ";N
680 PRINT N/5280;" MILE"
690 GOTO 940
700 PRINT "1 MILE = 5280 FEET"
710 PRINT
720 INPUT "MILES = ";N
730 PRINT 5280*N;" FEET"
740 GOTO 940
750 PRINT "1 MILE = 1.60935 KILOMETERS"
760 PRINT
770 INPUT "MILES = ";N
780 PRINT 1.60935*N;" KILOMETERS"
790 GOTO 940
800 PRINT "1 KILOMETER = .62137 MILE"
810 PRINT
820 INPUT "KILOMETERS = ";N
830 PRINT N*.62137;" MILES"
840 GOTO 940
850 PRINT "1 METER = 1.093611 YARDS"
860 PRINT
870 INPUT "METERS = ";N
880 PRINT N*1.093611;" YARDS"
890 GOTO 940
900 PRINT "1 YARD = .91440183 METER"
910 PRINT
920 INPUT "YARDS = ";N
```

```
930 PRINT N*.91440183;" METERS"  
940 PRINT  
950 PRINT "CHOOSE:"  
960 PRINT " 1 ANOTHER PROBLEM"  
970 PRINT " 2 END PROGRAM"  
980 A$=INKEY$  
990 IF A$="1" THEN 10  
1000 IF A$<>"2" THEN 980  
1010 PRINT  
1020 END
```

Distance

Computers: Translatable

Any type of formula can be put into a simple program. This program uses the formula $\text{Distance} = \text{Rate} * \text{Time}$. It allows you to choose which item becomes the unknown. You input the known values, and the unknown value will be printed.

Program 3-2. Distance

Translatable

```
10 CLS  
20 PRINT "***DISTANCE=RATE*TIME**"  
30 PRINT  
40 PRINT  
50 PRINT "WHICH IS THE UNKNOWN?"  
60 PRINT " 1 DISTANCE"  
70 PRINT " 2 RATE"  
80 PRINT " 3 TIME"  
90 A$=INKEY$  
100 IF A$="2" THEN 210  
110 IF A$="3" THEN 290  
120 IF A$<>"1" THEN 90  
130 PRINT  
140 PRINT  
150 PRINT "DISTANCE = RATE * TIME"  
160 PRINT  
170 INPUT "RATE = ";R  
180 INPUT "TIME = ";T  
190 PRINT "DISTANCE =";R*T  
200 GOTO 360  
210 PRINT  
220 PRINT  
230 PRINT "RATE = DISTANCE/TIME"  
240 PRINT  
250 INPUT "DISTANCE = ";D  
260 INPUT "TIME = ";T  
270 PRINT "RATE =";D/T  
280 GOTO 360
```

Quick Calculations

```
290 PRINT
300 PRINT
310 PRINT "TIME = DISTANCE/RATE"
320 PRINT
330 INPUT "DISTANCE = ";D
340 INPUT "RATE = ";R
350 PRINT "TIME =";D/R
360 PRINT
370 PRINT
380 PRINT "CHOOSE:"
390 PRINT " 1 ANOTHER PROBLEM"
400 PRINT " 2 END PROGRAM"
410 A$=INKEY$
420 IF A$="1" THEN 30
430 IF A$<>"2" THEN 410
440 PRINT
450 END
```

Squares

Computers: Translatable

"Squares" calculates the area, side, or perimeter of a square if you know one of these quantities. Line 10 clears the screen. Lines 90, 260, and 520 detect a key pressed. You may need to change these lines to GET A\$ or CALL KEY(0,K,S) for your particular computer.

Program 3-3. Squares

Translatable

```
10 CLS
20 PRINT "*** SQUARES ***"
30 PRINT
40 PRINT
50 PRINT "WHICH IS UNKNOWN?"
60 PRINT " 1 AREA"
70 PRINT " 2 SIDE"
80 PRINT " 3 PERIMETER"
90 A$=INKEY$
100 IF A$="2" THEN 200
110 IF A$="3" THEN 410
120 IF A$<>"1" THEN 90
130 PRINT
140 PRINT
150 PRINT "AREA = SIDE * SIDE"
160 PRINT
170 INPUT "SIDE = ";S
180 PRINT "AREA =";S*S
190 GOTO 470
```

```
200 PRINT
210 PRINT
220 PRINT "NEED TO FIND SIDE"
230 PRINT "WHAT IS GIVEN?"
240 PRINT " 1 AREA"
250 PRINT " 2 PERIMETER"
260 A$=INKEY$
270 IF A$="1" THEN 350
280 IF A$<>"2" THEN 260
290 PRINT
300 PRINT "SIDE = PERIMETER/4"
310 PRINT
320 INPUT "PERIMETER = ";P
330 PRINT "SIDE =";P/4
340 GOTO 470
350 PRINT
360 PRINT "SIDE = SQR ROOT OF AREA"
370 PRINT
380 INPUT "AREA = ";A
390 PRINT "SIDE =";SQR(A)
400 GOTO 470
410 PRINT
420 PRINT
430 PRINT "PERIMETER = 4*SIDE"
440 PRINT
450 INPUT "SIDE = ";S
460 PRINT "PERIMETER =";4*S
470 PRINT
480 PRINT
490 PRINT "CHOOSE: "
500 PRINT " 1 ANOTHER PROBLEM"
510 PRINT " 2 END PROGRAM"
520 A$=INKEY$
530 IF A$="1" THEN 10
540 IF A$<>"2" THEN 520
550 PRINT
560 END
```

Rectangles

Computers: Translatable

"Rectangles" also illustrates plane geometry principles. The computer will calculate area, perimeter, or the length of the unknown side if all other necessary quantities are given. Lines 90, 340, and 500 detect a key pressed. Change these to fit your computer by using GET A\$ or CALL KEY(0,K,S).

Quick Calculations

Program 3-4. Rectangles

Translatable

```
10 CLS
20 PRINT "*** RECTANGLES ***"
30 PRINT
40 PRINT
50 PRINT "WHAT IS UNKNOWN?"
60 PRINT " 1 AREA"
70 PRINT " 2 PERIMETER"
80 PRINT " 3 ONE SIDE"
90 A$=INKEY$
100 IF A$="2" THEN 210
110 IF A$="3" THEN 290
120 IF A$<>"1" THEN 90
130 PRINT
140 PRINT
150 PRINT "AREA = LENGTH * WIDTH"
160 PRINT
170 INPUT "LENGTH = ";L
180 INPUT "WIDTH = ";W
190 PRINT "AREA =";L*W
200 GOTO 450
210 PRINT
220 PRINT
230 PRINT "PERIMETER = 2*LENGTH + 2*WIDTH"
240 PRINT
250 INPUT "LENGTH = ";L
260 INPUT "WIDTH = ";W
270 PRINT "PERIMETER =";2*L+2*W
280 GOTO 450
290 PRINT
300 PRINT "WHAT IS GIVEN?"
310 PRINT " 1 AREA"
320 PRINT " 2 PERIMETER"
330 PRINT
340 A$=INKEY$
350 IF A$="2" THEN 420
360 IF A$<>"1" THEN 340
370 INPUT "AREA = ";A
380 INPUT "GIVEN SIDE = ";S
390 IF A$<>"1" THEN 370
400 PRINT "SIDE =";A/S
410 GOTO 450
420 INPUT "PERIMETER = ";P
430 INPUT "GIVEN SIDE = ";S
440 PRINT "SIDE =";(P-2*S)/2
450 PRINT
460 PRINT
470 PRINT "CHOOSE:"
```

```
480 PRINT " 1 ANOTHER PROBLEM"  
490 PRINT " 2 END PROGRAM"  
500 A$=INKEY$  
510 IF A$="1" THEN 10  
520 IF A$<>"2" THEN 500  
530 PRINT  
540 END
```

Circles

Computers: Translatable

"Circles" uses the formulas for the radius, diameter, area, or circumference of a circle. First, decide which is the unknown, or the answer you'll want. Second, determine the given quantity and input that value. The answer will then be printed.

The program uses INPUT, rather than detecting a key press, to receive a menu choice.

(Note: This program will not work on the TRS-80 4K Color Computer because the program uses the square root function, which is not available in 4K Color BASIC. You may, however, adapt the program by leaving out all sections involving the square root and by combining lines and leaving out all spaces. For a program which calculates a number's square root on the Color Computer, refer to Program 3-6.)

Program 3-5. Circles

Translatable

```
10 PRINT "** CIRCLES **"  
20 PRINT  
30 PRINT "WHICH IS UNKNOWN?"  
40 PRINT " 1 RADIUS"  
50 PRINT " 2 DIAMETER"  
60 PRINT " 3 AREA"  
70 PRINT " 4 CIRCUMFERENCE"  
80 INPUT A  
90 IF A>=1 THEN 120  
100 PRINT "ENTER 1, 2, 3, OR 4"  
110 GOTO 80  
120 IF A>4 THEN 100  
130 PRINT  
140 ON A GOTO 150,410,670,930  
150 PRINT "WHAT IS GIVEN?"  
160 PRINT " 1 DIAMETER"  
170 PRINT " 2 AREA"  
180 PRINT " 3 CIRCUMFERENCE"  
190 INPUT A  
200 IF A>=1 THEN 230
```

Quick Calculations

```
210 PRINT"ENTER 1, 2, OR 3"
220 GOTO 190
230 IF A>3 THEN 210
240 PRINT
250 ON A GOTO 260,310,360
260 PRINT"RADIUS = DIAMETER/2"
270 PRINT
280 INPUT "DIAMETER = ";D
290 PRINT"RADIUS = ";D/2
300 GOTO 1180
310 PRINT"RADIUS = SQR(AREA/PI)"
320 PRINT
330 INPUT "AREA = ";A
340 PRINT"RADIUS = ";SQR(A/3.14159)
350 GOTO 1180
360 PRINT"RADIUS = CIRCUM/2 PI"
370 PRINT
380 INPUT "CIRCUMFERENCE = ";C
390 PRINT"RADIUS = ";C/(2*3.14159)
400 GOTO 1180
410 PRINT"WHAT IS GIVEN?"
420 PRINT" 1 RADIUS"
430 PRINT" 2 AREA"
440 PRINT" 3 CIRCUMFERENCE"
450 INPUT A
460 IF A>=1 THEN 490
470 PRINT"ENTER 1, 2, OR 3"
480 GOTO 450
490 IF A>3 THEN 470
500 PRINT
510 ON A GOTO 520,570,620
520 PRINT"DIAMETER = 2*RADIUS"
530 PRINT
540 INPUT "RADIUS = ";R
550 PRINT"DIAMETER = ";2*R
560 GOTO 1180
570 PRINT"DIAMETER = 2*SQR(AREA/PI)"
580 PRINT
590 INPUT "AREA = ";A
600 PRINT"DIAMETER = ";2*SQR(A/3.14159)
610 GOTO 1180
620 PRINT"DIAMETER = CIRCUM/PI"
630 PRINT
640 INPUT "CIRCUMFERENCE = ";C
650 PRINT"DIAMETER = ";C/3.14159
660 GOTO 1180
670 PRINT"WHICH IS GIVEN?"
680 PRINT" 1 RADIUS"
690 PRINT" 2 DIAMETER"
```

Quick Calculations

```
700 PRINT " 3 CIRCUMFERENCE"
710 INPUT A
720 IF A>=1 THEN 750
730 PRINT"ENTER 1, 2, OR 3"
740 GOTO 710
750 IF A>3 THEN 730
760 PRINT
770 ON A GOTO 780,830,880
780 PRINT"AREA = PI*RADIUS SQUARED"
790 PRINT
800 INPUT "RADIUS = ";R
810 PRINT"AREA = ";3.14159*R*R
820 GOTO 1180
830 PRINT"AREA = PI * DIAM SQUARED/4"
840 PRINT
850 INPUT "DIAMETER = ";D
860 PRINT"AREA = ";3.14159*D*D/4
870 GOTO 1180
880 PRINT"AREA = CIRCUM SQUARED/4 PI"
890 PRINT
900 INPUT "CIRCUMFERENCE = ";C
910 PRINT"AREA = ";C*C/4/3.14159
920 GOTO 1180
930 PRINT"WHAT IS GIVEN?"
940 PRINT" 1 RADIUS"
950 PRINT" 2 DIAMETER"
960 PRINT" 3 AREA"
970 INPUT A
980 IF A>=1 THEN 1010
990 PRINT"ENTER 1, 2, OR 3"
1000 GOTO 970
1010 IF A>3 THEN 990
1020 PRINT
1030 ON A GOTO 1040,1090,1140
1040 PRINT"C=2 PI*RADIUS"
1050 PRINT
1060 INPUT "RADIUS = ";R
1070 PRINT"CIRCUMFERENCE = ";2*3.14159*R"
1080 GOTO 1180
1090 PRINT"C = PI*DIAMETER"
1100 PRINT
1110 INPUT "DIAMETER = ";D
1120 PRINT"CIRCUMFERENCE = ";3.14159*D
1130 GOTO 1180
1140 PRINT"C = 2*SQR(AREA*PI)"
1150 PRINT
1160 INPUT "AREA = ";A
1170 PRINT"CIRCUMFERENCE = ";2*SQR(A*3.14159)
1180 PRINT
```

Quick Calculations

```
1190 PRINT "CHOOSE:"
1200 PRINT " 1 ANOTHER PROBLEM"
1210 PRINT " 2 END PROGRAM"
1220 INPUT A
1230 IF A=1 THEN 20
1240 IF A<>2 THEN 1220
1250 END
```

Square Root

Computers: Translatable

If you have the TRS-80 4K Color Computer or another computer which does not have the square root function $SQR(n)$, you can use this program to calculate the square root of a number n . You could convert this program to a subroutine to be called whenever you need a square root.

Program 3-6. Square Root

Translatable

```
10 CLS
20 PRINT "*** SQUARE ROOT ***"
30 PRINT
40 PRINT
50 INPUT "NUMBER = ";N
60 PRINT
70 IF N<>0 THEN 100
80 PRINT "SQUARE ROOT = 0"
90 GOTO 250
100 IF N>0 THEN 160
110 PRINT
120 PRINT "PLEASE ENTER NUMBER"
130 PRINT "GREATER THAN ZERO."
140 PRINT
150 GOTO 250
160 PRINT "SQUARE ROOT =";
170 S=N/2
180 T=0
190 U=(N/S-S)/2
200 IF (U=0)+(U=T) THEN 240
210 S=S+U
220 T=U
230 GOTO 190
240 PRINT S
250 PRINT
260 PRINT "CHOOSE:"
270 PRINT " 1 ANOTHER PROBLEM"
280 PRINT " 2 END PROGRAM"
290 A$=INKEY$
```

```

300 IF A$="1" THEN 10
310 IF A$<>"2" THEN 290
320 PRINT
330 END


```

Parallel Resistance

Computers: Translatable

"Parallel Resistance" is an electrical engineering program that illustrates how quickly your computer can calculate a complex equation. In electrical engineering circuit analysis, a number of resistors can be combined in parallel to create a total resistance value. The formula states that the reciprocal of the total is equal to the sum of the reciprocals of each one — in other words, an equation involving fractions.

For example:



where:

$$1/R_T = 1/R_1 + 1/R_2 + 1/R_3$$

For this program, enter the number of resistors, then enter the resistance for each one. The total resistance is calculated and printed.

Program 3-7. Parallel Resistance

Translatable

```

10 CLS
20 PRINT "*** PARALLEL RESISTANCE ***"
30 PRINT
40 PRINT " 1{4 SPACES}1{4 SPACES}1{4 SPACES}1"
50 PRINT "-- = -- + -- + -- + ..."
60 PRINT "RT{3 SPACES}R1{3 SPACES}R2{3 SPACES}R3"
70 PRINT
80 PRINT "YOUR PROBLEM:"
90 INPUT "HOW MANY RESISTORS? ";N
100 N=INT(N)
110 IF N>0 THEN 140

```

Quick Calculations

```
120 PRINT "SORRY, ONE OR MORE"
130 GOTO 90
140 IF N<51 THEN 180
150 PRINT "IF YOU REALLY HAVE MORE"
160 PRINT "THAN 50, SOLVE IN STEPS."
170 GOTO 90
180 PRINT
190 D=0
200 FOR I=1 TO N
210 PRINT "R";I;"=";
220 INPUT R
230 IF R<>0 THEN 260
240 PRINT "SORRY, ZERO NOT ALLOWED"
250 GOTO 210
260 D=D+1/R
270 NEXT I
280 PRINT
290 PRINT "TOTAL R = ";1/D
300 PRINT
310 END
```

Arithmetic Mean

Computers: Translatable

All types of statistical programs can be done with your computer — all you have to do is pick the appropriate formula or procedure and write a short program. This program illustrates the calculation of the arithmetic mean or average of a group of numbers. First, enter the number of items, then enter each value. The total and the average are printed.

Program 3-8. Arithmetic Mean

Translatable

```
10 CLS
20 PRINT "*** ARITHMETIC MEAN ***"
30 PRINT
40 INPUT "HOW MANY ITEMS";N
50 N=INT(N)
60 IF N>0 THEN 100
70 PRINT "PLEASE ENTER A WHOLE NUMBER"
80 PRINT "GREATER THAN ZERO."
90 GOTO 40
100 T=0
110 FOR I=1 TO N
120 PRINT " VALUE";I;" = ";
130 INPUT A
140 T=T+A
150 NEXT I
```

```
160 PRINT
170 PRINT "TOTAL = ";T
180 PRINT "AVERAGE = ";T/N
190 PRINT
200 END
```

Simple Interest

Computers: Translatable

Suppose you have a savings account that pays simple interest. If you deposit a certain amount now, how much will it be worth later? In "Simple Interest," you enter your principal deposit, the interest rate in percent, and the number of years. The future value will be printed.

Another use for this program would be to calculate a simple interest loan. If you borrowed money at simple interest, what would be the total amount due n years from now? Entering the principal loan, the interest rate, and the number of years will produce the total amount due at the end of the loan's term.

Program 3-9. Simple Interest

Translatable

```
10 CLS
20 PRINT "*** SIMPLE INTEREST ***"
30 PRINT
40 PRINT "ENTER PRINCIPAL DEPOSIT."
50 INPUT P
60 IF P>0 THEN 90
70 PRINT "MUST BE MORE THAN ZERO."
80 GOTO 30
90 PRINT
100 PRINT "ENTER INTEREST RATE IN PERCENT"
110 PRINT "SUCH AS 6 FOR 6 PERCENT."
120 INPUT I
130 PRINT
140 INPUT "HOW MANY YEARS";N
150 IF N>0 THEN 180
160 PRINT "MUST BE MORE THAN ZERO."
170 GOTO 130
180 PRINT
190 A=P*(1+N*(I/100))
200 PRINT "VALUE OF";P
210 PRINT "AT INTEREST";I;"PERCENT"
220 PRINT "FOR";N;"YEARS = "
230 PRINT A
240 PRINT
250 END
```


Quick Calculations

Compound Interest

Computers: Translatable

Most banks and savings associations now use compound interest. This program calculates the future value of a present deposit earning compound interest. You will be asked to enter the principal deposit, the interest rate in percent, how many times interest is compounded per year (quarterly would be 4, daily would be 365), and how many years in the analysis. The equation in line 240 involves a power or exponent symbol. Enter your computer's symbol for this function. For example, a VIC-20 would show an arrow (↑) here. Exponentiation is not available on the TRS-80 Color Computer unless Extended Color BASIC is installed. The future value will be printed.

As in Program 3-9, Simple Interest, you could use this program to calculate a compound interest loan's total amount due. Just enter the principal of the loan, the interest rate, times compounded, and the number of years for the loan. The total value will be shown. You could use this to find out how much interest you'll pay on a car loan, for instance.

Program 3-10. Compound Interest

Translatable

```
10 CLS
20 PRINT "*** COMPOUND INTEREST ***"
30 PRINT
40 PRINT "ENTER PRINCIPAL DEPOSIT"
50 INPUT P
60 IF P>0 THEN 90
70 PRINT "MUST BE MORE THAN ZERO."
80 GOTO 30
90 PRINT
100 PRINT "ENTER INTEREST RATE IN PERCENT"
110 INPUT I
120 PRINT
130 PRINT "COMPOUNDED HOW MANY TIMES"
140 INPUT "PER YEAR";Q
150 IF Q>=0 THEN 180
160 PRINT "MUST BE MORE THAN ZERO."
170 GOTO 120
180 PRINT
190 INPUT "HOW MANY YEARS";N
200 IF N>0 THEN 230
210 PRINT "MUST BE MORE THAN ZERO."
220 GOTO 180
230 PRINT
240 A=P*(1+(I/100)/Q)^(N*Q)
250 PRINT "PRINCIPAL";P
```

```
260 PRINT "AT INTEREST";I;"PERCENT"  
270 PRINT "COMPOUNDED";Q;"TIMES PER YEAR"  
280 PRINT "FOR";N;"YEARS ="  
290 PRINT A  
300 PRINT  
310 END
```

Single Payment Present Worth

Computers: Translatable

Have you ever wanted to know how much money you'd have to deposit *today* to have \$10,000 for your retirement? You can find out by using this program. Enter the future sum (10000, for example), the number of years you have until age 65, and a realistic interest rate in percent. The amount you would have to have right now will be calculated.

Line 220 contains a "raise to power" (*exponent*) symbol. Make sure you enter the proper symbol for exponentiation for your computer here. Note that the TRS-80 Color Computer does not have the exponentiation function unless you have Extended Color BASIC installed.

Program 3-11. Single Payment

Translatable

```
10 CLS  
20 PRINT "*** SINGLE PAYMENT ***"  
30 PRINT "*** PRESENT WORTH ***"  
40 PRINT  
50 PRINT "GIVEN A FUTURE SUM,"  
60 PRINT "FIND ITS VALUE TODAY,"  
70 PRINT "N PERIODS EARLIER."  
80 PRINT  
90 PRINT "WHAT IS THE FUTURE SUM?"  
100 INPUT S  
110 IF S>=0 THEN 140  
120 PRINT "MUST BE MORE THAN ZERO."  
130 GOTO 80  
140 PRINT  
150 INPUT "HOW MANY PERIODS";N  
160 IF N>0 THEN 190  
170 PRINT "MUST BE GREATER THAN ZERO"  
180 GOTO 140  
190 PRINT  
200 PRINT "ENTER INTEREST RATE IN PERCENT"  
210 INPUT I  
220 P=S*(1/((1+I/100)^N))  
230 PRINT  
240 PRINT "PRESENT VALUE =";P
```

Quick Calculations

Uniform Series Compound Amount

Computers: Translatable

Suppose you have a savings account earning compound interest. You deposit the same amount each month. After n months, how much will your account be worth?

"Uniform Series" calculates the result of a uniform series of payments. Enter the payment amount, the number of payments, and the interest in percent. The final total, plus interest, will be printed on the screen.

Line 240 contains a power or exponent symbol. Use your computer's symbol here. For the TRS-80 Color Computer, Extended Color BASIC is required.

Program 3-12. Uniform Series

Translatable

```
10 CLS
20 PRINT "*** UNIFORM SERIES ***"
30 PRINT "*** COMPOUND AMOUNT ***"
40 PRINT
50 PRINT "GIVEN A UNIFORM SERIES OF"
60 PRINT "END-OF-PERIOD PAYMENTS,"
70 PRINT "WHAT WILL N PAYMENTS"
80 PRINT "ACCUMULATE TO AT"
90 PRINT "COMPOUND INTEREST I?"
100 PRINT
110 INPUT "ENTER PAYMENT AMOUNT";R
120 IF R>0 THEN 150
130 PRINT "MUST BE MORE THAN ZERO"
140 GOTO 100
150 PRINT
160 INPUT "HOW MANY PAYMENTS";N
170 IF N>0 THEN 200
180 PRINT "MUST BE MORE THAN ZERO"
190 GOTO 150
200 PRINT
210 PRINT "ENTER INTEREST IN PERCENT"
220 INPUT "PER PERIOD";I
230 I=I/100
240 S=R*((1+I)^N-1)/I
250 PRINT
260 PRINT "FINAL VALUE =";S
270 PRINT
280 END
```

Sinking Fund Deposit

Computers: Translatable

You need a certain sum in a future year and want to know how much to deposit each month to reach that amount. Your account earns compound interest. Enter the desired future sum, the number of time periods, and the interest expressed as a percent per period. The program will then calculate the monthly deposit needed.

Make sure you enter your computer's exponent symbol in line 250. Exponentiation is not available on the TRS-80 Color Computer unless Extended Color BASIC is installed.

Program 3-13. Sinking Fund

Translatable

```
10 CLS
20 PRINT "*** SINKING FUND DEPOSIT ***"
30 PRINT
40 PRINT "WHAT UNIFORM SERIES OF"
50 PRINT "END-OF-PERIOD DEPOSITS"
60 PRINT "MUST BE MADE FOR N PERIODS"
70 PRINT "AT COMPOUND INTEREST I TO"
80 PRINT "GET A FUTURE SUM?"
90 PRINT
100 PRINT "ENTER DESIRED FUTURE SUM."
110 INPUT S
120 IF S>=0 THEN 150
130 PRINT "MUST BE MORE THAN ZERO"
140 GOTO 90
150 PRINT
160 INPUT "HOW MANY PERIODS";N
170 IF N>=1 THEN 200
180 PRINT "MUST BE AT LEAST ONE"
190 GOTO 150
200 PRINT
210 PRINT "ENTER INTEREST AS PERCENT"
220 INPUT "PER PERIOD";I
230 I=I/100
240 PRINT
250 R=S*(I/((1+I)^N-1))
260 PRINT "SINKING FUND DEPOSITS ="
270 PRINT R
280 PRINT
290 END
```

Quick Calculations

Loan Payments

Computers: Translatable

You would like to borrow a sum of money from a bank that charges compound interest. What would the monthly payments be? Using this program, you can find out.

Enter the amount you want to borrow, and the number of years for the loan. For example, home mortgages may be 25 or 30 years, but second mortgages are usually only 10 or 15 years. Next, enter the annual interest rate in percent — the interest rate the bank quoted you. The program will compute your monthly payment. By running this program you can quickly see the difference in your monthly payment with different interest rates — or different times — for the same amount of money.

Line 230 contains an exponent symbol. Enter your computer's symbol in this line. Exponentiation is not available on TRS-80 Color Computers unless Extended Color BASIC is installed.

Program 3-14. Loan Payments

Translatable

```
10 CLS
20 PRINT "*** LOAN PAYMENTS **"
30 PRINT
40 PRINT "YOU WANT TO BORROW A"
50 PRINT "CERTAIN SUM OF MONEY."
60 PRINT "IF INTEREST IS COMPOUNDED,"
70 PRINT "WHAT IS THE MONTHLY PAYMENT?"
80 PRINT
90 INPUT "AMOUNT BORROWED";P
100 IF P>0 THEN 130
110 PRINT "AMOUNT MORE THAN ZERO PLEASE"
120 GOTO 80
130 PRINT
140 INPUT "HOW MANY YEARS";Y
150 IF Y>0 THEN 180
160 PRINT "MUST BE MORE THAN ZERO"
170 GOTO 130
180 N=12*Y
190 PRINT
200 PRINT "WHAT IS THE INTEREST RATE"
210 INPUT "IN PERCENT";I
220 I=I/1200
230 F=(1+I)^N
240 M=P*(I*F/(F-1))
250 M=(INT(100*(M+.005)))/100
260 PRINT
270 PRINT "MONTHLY PAYMENT =";M
```

```
280 PRINT
290 END
```

Amortization

Computers: Translatable

"Amortization" is a continuation of Program 3-14, "Loan Payments." After your monthly payment is calculated, you can use this program to see an amortization schedule — a list by month of the amount of principal paid, the amount of interest paid, and the current balance.

Enter the amount borrowed, the number of years for the loan, and the interest rate in percent. The computer will display the monthly payment. The program then gives you the option to see the amortization schedule or end the program. Enter 1 to see the amortization. Pressing 2 will end the program. Since most computers display words faster than we can read, only one line at a time is shown, representing one month's payment. Press any key to see the next line. (For the VIC-20 you'll need to adjust the printing to fit 22 column lines.)

Line 230 contains a power, or exponent, symbol. Use your computer's symbol here. This program cannot be used on TRS-80 Color Computers unless Extended Color BASIC is installed. This is because the exponentiation function is not available in standard Color BASIC.

Program 3-15. Amortization


Translatable

```
10 CLS
20 PRINT "*** AMORTIZATION ***"
30 PRINT
40 PRINT "YOU WANT TO BORROW A"
50 PRINT "CERTAIN SUM OF MONEY."
60 PRINT "IF INTEREST IS COMPOUNDED,"
70 PRINT "WHAT IS THE MONTHLY PAYMENT?"
80 PRINT
90 INPUT "AMOUNT BORROWED";P
100 IF P>0 THEN 130
110 PRINT "AMOUNT MORE THAN ZERO PLEASE"
120 GOTO 80
130 PRINT
140 INPUT "HOW MANY YEARS";Y
150 IF Y>0 THEN 180
160 PRINT "MUST BE MORE THAN ZERO"
170 GOTO 130
```

Quick Calculations

```
180 N=12*Y
190 PRINT
200 PRINT "WHAT IS THE INTEREST RATE"
210 INPUT "IN PERCENT";I
220 I=I/1200
230 F=(1+I)^N
240 M=P*(I*F/(F-1))
250 M=(INT(100*(M+.005)))/100
260 PRINT
270 PRINT "MONTHLY PAYMENT =" ;M
280 PRINT
290 PRINT "CHOOSE:"
300 PRINT " 1 PRINT AMORTIZATION"
310 PRINT " 2 END PROGRAM"
320 INPUT A
330 IF A=2 THEN 510
340 IF A<>1 THEN 320
350 PRINT
360 PRINT "WHILE THE SCHEDULE IS BEING"
370 PRINT "PRINTED, PRESS ANY KEY TO"
380 PRINT "SEE THE NEXT LINE."
390 PRINT
400 PRINT "N PRINCIPAL INTEREST BALANCE"
410 P=(INT(100*(P+.005)))/100
420 FOR C=1 TO N
430 J=(INT(100*(I*P+.005)))/100
440 A=(INT(100*((M-J)+.005)))/100
450 B=(INT(100*((P-A)+.005)))/100
460 P=B
470 PRINT C;TAB(4);A;TAB(14);J;TAB(23);B
480 A$=INKEY$
490 IF A$="" THEN 480
500 NEXT C
510 END
```

Chapter 4
Graphics



Chapter 4

Graphics

The rest of the programs in this book are computer-specific. Each program has been written for a particular machine, and its listing shows what should be entered for that computer. Many of the programs have been translated into other computers' versions of BASIC. Unlike the previous chapters, you won't have to go through the programs and locate commands and symbols to change. It's been done for you.

As before, each program notes which computer or computers the program will RUN on. In many cases, there will be only one explanation of the program's operation, but there will be several program listings. Each listing will be identified with the computer it will operate on. Most of the rest of the programs are listed for one of three computers: the VIC-20, the TRS-80 Color Computer, or the TI-99/4A. Variations of the Color Computer, such as the TRS-80 Model I and the MC-10, or variations of the TI-99/4, such as the TI-99/4A, are included in the list of possible computers where they apply.

This doesn't mean that you can't translate these programs to your computer, even if it is not one of these three machines. It just means that translation may be more difficult. Depending on the program and the computer features it uses, a translation may be simple, or next to impossible. Most of the commands in BASIC are the same for the different microcomputers. Sometimes the punctuation or functions used are slightly different, but in general most of the commands are similar enough to be translated from one computer to another without much problem. The main difference in variations in BASIC is found in the graphic and sound commands. These are very difficult to translate, for each computer uses its own graphics and sound commands. This chapter and Chapter 5 provide programs written for specific computers — graphics and sound commands have made the programs so different that separate listings are given for each computer.

Graphics Demo

Computers: VIC-20; TRS-80 Color Computer; MC-10; TI-99/4A

"Graphics Demo" is a short program that randomly places squares of color on the screen while playing random tones. The program continues until you press any key.

Graphics

Program 4-1. Graphics Demo

VIC-20

```
20 PRINT "{CLR}PRESS A KEY TO STOP."  
30 S=36876:POKE36878,15  
40 P=INT(506*RND(0))  
50 T=INT(254*RND(0)+1)  
60 POKEP+7680,160:POKEP+38400,INT(8*RND(0))  
70 POKES,T  
80 GET E$:IF E$=""THEN40  
90 POKES,0  
100 END
```

Program 4-2. Graphics Demo

TRS-80 Color Computer; MC-10

```
20 CLS  
30 PRINT "PRESS A KEY TO STOP."  
40 I=RND(64)-1  
50 J=RND(32)-1  
60 C=RND(4)  
70 SET(I,J,C)  
80 SOUND RND(255),1  
90 A$=INKEY$  
100 IF A$="" THEN 40  
110 END
```

Program 4-3. Graphics Demo

TI-99/4A

```
110 CALL CLEAR  
120 PRINT "PRESS ANY KEY TO STOP."  
130 FOR C=1 TO 16  
140 CALL COLOR(C,C,C)  
150 NEXT C  
160 RANDOMIZE  
170 CALL SOUND(-50,INT(1600*RND+110),2)  
180 CALL HCHAR(INT(24*RND+1),INT(32*RND+1),8*INT(1  
6*RND+1)+31)  
190 CALL KEY(0,K,S)  
200 IF S<1 THEN 160  
210 END
```

Doodle

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

"Doodle" lets you draw on the screen by pressing the arrow or cursor keys. The cursor starts in the middle of the screen. As you press a cursor or arrow key, the cursor moves in that direction

and leaves a trail. Pressing the space bar changes colors. If you can't see the cursor, it may be the same color as the background screen, or it may be on a color previously drawn.

Here's how the VIC-20 version of Doodle works:

Line	Function
2	Clear screen.
3-5	PRINT instructions.
6-7	Wait for beginning key to be pressed.
8	Clear screen.
10	Initialize variables. P is screen position, S is character screen code number for graphics, and C is color number.
20	Draw square of color.
30	Blink square while waiting for input.
40-160	Depending on cursor key pressed, change screen position.
170-180	If space bar is pressed, change color.
190	RETURN to line 20 to draw new square.
200	END.

To move left or up, remember that you'll have to press SHIFT, then the appropriate cursor key.

Program 4-4. Doodle

VIC-20

```
2 PRINT "{CLR}"
3 PRINT"PRESS THE CRSR KEYS{3 SPACES}TO DRAW."
4 PRINT"{2 DOWN}PRESS THE SPACE BAR{3 SPACES}TO CH
  ANGE COLORS."
5 PRINT"{4 DOWN}PRESS F1 TO START."
6 GETES:IFE$=""THEN6
7 IFASC(E$)<>133THEN6
8 PRINT"{CLR}"
10 P=7932:S=160:C=0
20 R=P+30720:POKEP,S:POKER,C
30 POKER,1:POKER,C:GETES:IFE$=""THEN30
40 E=ASC(E$)
50 IFE<>17THEN80
60 IFP+22>8185THEN30
70 P=P+22:GOTO20
80 IFE<>29THEN110
90 IFP+1>8185THEN30
100 P=P+1:GOTO20
110 IFE<>145THEN140
120 IFP-22<7680THEN30
130 P=P-22:GOTO20
140 IFE<>157THEN170
```

Graphics

```
150 IFP-1<7680THEN30
160 P=P-1:GOTO20
170 IFE<>32THEN30
180 C=C+1:IFC>7THENC=0
190 GOTO20
200 END
```

Doodle works a bit differently on the Color Computer and MC-10:

Line	Function
20	Clear screen.
30-40	PRINT instructions.
50-70	Wait for key to be pressed.
80	Clear screen.
90	Initialize variables. P is screen position, C is color number.
100-110	Calculate character number R.
120-130	Blink square while waiting for input.
140-260	Depending on arrow key pressed, change screen position.
270-280	If space bar is pressed, change color.
290	RETURN to line 100 to draw new square.
300	END.

Program 4-5. Doodle

TRS-80 Color Computer; MC-10

```
20 CLS
30 PRINT "PRESS THE ARROW KEYS TO DRAW."
40 PRINT:PRINT "PRESS THE SPACE BAR":PRINT "TO CHANGE COLORS."
50 PRINT:PRINT"PRESS <ENTER> TO START."
60 E$=INKEY$:IF E$="" THEN 60
70 IF ASC(E$)<>13 THEN 60
80 CLS
90 P=240:C=4
100 R=128+16*(C-1)+15
110 IF R=127 THEN R=128
120 PRINT@P,CHR$(128);:PRINT@P,CHR$(R);
130 E$=INKEY$:IF E$=""THEN 120
140 E=ASC(E$)
150 IF E<>10 THEN 180
160 IF P+32>510 THEN 120
170 P=P+32:GOTO 120
180 IF E<>94 THEN 210
190 IF P-32<0 THEN 120
200 P=P-32:GOTO 120
```

```

210 IF E<>8 THEN 240
220 IF P-1<0 THEN 120
230 P=P-1:GOTO 120
240 IF E<>9 THEN 270
250 IF P+1>510 THEN 120
260 P=P+1:GOTO 120
270 IF E<>32 THEN 120
280 C=C-1:IF C<0 THEN C=4
290 GOTO 100
300 END

```

Doodle works like this on the Model I:

Line	Function
20	Clear screen.
30-40	PRINT instructions.
50-70	Wait for beginning key to be pressed.
80	Clear screen.
90	Initialize starting coordinates X and Y.
100-110	Blink square while waiting for input.
120-240	Depending on key pressed, change screen position.
250	END.

Program 4-6. Doodle

TRS-80 Model I

```

20 CLS
30 PRINT "PRESS THE ARROW KEYS TO DRAW."
40 PRINT:PRINT:PRINT:PRINT
50 PRINT:PRINT "PRESS <ENTER> TO START."
60 E$=INKEY$:IF E$="" THEN 60
70 IF ASC(E$)<>13 THEN 60
80 CLS
90 X=64:Y=23
100 RESET(X,Y):SET(X,Y)
110 E$=INKEY$:IF E$="" THEN 100
120 E=ASC(E$)
130 IF E<>91 THEN 160
140 IF Y-1<0 THEN 100
150 Y=Y-1:GOTO 100
160 IF E<>8 THEN 190
170 IF X-1<0 THEN 100
180 X=X-1:GOTO 100
190 IF E<>9 THEN 220
200 IF X+1>127 THEN 100
210 X=X+1:GOTO 100
220 IF E<>10 THEN 100
230 IF Y+1>47 THEN 100
240 Y=Y+1:GOTO 100
250 END

```

Graphics

Here's how the TI-99/4A version works:

Line	Function
110	Clear screen.
120-130	PRINT instructions.
140-160	Wait for key to be pressed.
170	Clear screen.
180-200	Define 16 colors.
210-230	Define variables. X and Y are screen positions, and C is color number.
240	Calculate character number R.
250-270	Blink square while waiting for input.
280-430	Depending on arrow key pressed, change screen position.
440-480	If space bar is pressed, change colors.
490-530	Check edge condition, then branch to line 250 to draw new square.
540	END.

Program 4-7. Doodle

TI-99/4A

```
110 CALL CLEAR
120 PRINT "PRESS ARROW KEYS TO DRAW."
130 PRINT :: "PRESS THE SPACE BAR" :: "TO CHANGE COLO
RS."
140 PRINT :: "PRESS <ENTER> TO START."
150 CALL KEY(0,K,S)
160 IF K<>13 THEN 150
170 CALL CLEAR
180 FOR I=1 TO 16
190 CALL COLOR(I,I,I)
200 NEXT I
210 X=12
220 Y=15
230 C=16
240 R=31+8*C
250 CALL HCHAR(X,Y,8)
260 CALL HCHAR(X,Y,R)
270 CALL KEY(0,K,S)
280 IF K<>69 THEN 320
290 DX=-1
300 DY=0
310 GOTO 490
320 IF K<>68 THEN 360
330 DX=0
340 DY=1
350 GOTO 490
```

```

360 IF K<>88 THEN 400
370 DX=1
380 DY=0
390 GOTO 490
400 IF K<>83 THEN 440
410 DX=0
420 DY=-1
430 GOTO 490
440 IF K<>32 THEN 250
450 C=C-1
460 IF C>0 THEN 240
470 C=16
480 GOTO 240
490 X=X+DX
500 Y=Y+DY
510 X=INT(24*((X-1)/24-INT((X-1)/24)))+1
520 Y=INT(32*((Y-1)/32-INT((Y-1)/32)))+1
530 GOTO 250
540 END

```

Color Code

Computers: VIC-20; TRS-80 Color Computer; TI-99/4A

"Color Code" is a graphic guessing game, similar to the popular game "Mastermind." The computer chooses four colors and places them in a certain order. Each position may be one of the particular computer's available colors. You then try to guess the computer's color code by pressing the color numbers shown at the top of the screen in the correct order. If you make a mistake before you press the fourth color number, you may erase your guess and start over.

The computer checks your guess. You'll see one kind of marker for each correct color that is in the correct position and a different marker for a color that is in the series but is *not* in the correct position. The markers shown are not in any particular order — you may not know which color goes with which marker. Try to guess the color code in as few tries as possible. If you don't discover the color code before you reach the bottom of the screen, the computer will show you the correct answer.

The VIC version of Color Code works like this:

Line	Function
10-16	PRINT instructions; initialize sound variables.
18	Initialize color numbers.
20-24	Wait for user to press f1 key.
26	PRINT color numbers and color across top of screen.

- 28-30 Draw playing board.
- 32 Choose series of four colors.
- 34 Initialize N for number of tries.
- 36 Label player's guess.
- 38-46 Receive user's four guesses of color numbers.
- 48-54 Check for correct color in correct position.
- 56-60 Check for correct color in any other position.
- 62 Increment number of guesses; return for next try.
- 64-66 Show correct answer.
- 68 Play tune for correct answer.
- 70-76 PRINT option to try again and branch appropriately.
- 78 Subroutine for sound delay loop.
- 80 Clear screen and END.

Program 4-8. Color Code

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```
10 PRINT "{CLR}":PRINTTAB(6); "{BLU}COLOR CODE":PRIN
T "{DOWN}GUESS THE CORRECT"
*12 PRINT "SEQUENCE OF FOUR":PRINT "COLORS. ":PRINT "
{DOWN}GUESS BY PRESSING":PRINT "FOUR COLOR NUMBE
RS. "
14 PRINT "PRESS 0 TO ERASE. ":PRINT "{2 DOWN}{GRN} Z
{SPACE}{BLK}RIGHT COLOR ONLY"
16 PRINT "{DOWN}{RED} S {BLK}RIGHT COLOR IN":PRINT "
{3 SPACES}CORRECT POSITION":S=36876:POKE36878,1
5
18 Z(1)=1:Z(2)=3:Z(3)=6:Z(4)=7:Z(5)=8
20 PRINT "{4 DOWN}{GRN}PRESS {BLU}F1{GRN} TO START.
"
22 GETES:IFES=""THEN22
24 IFASC(ES)<>133THEN22
26 PRINT "{CLR}{BLU}1{BLK}[K]{BLU}3{RED}[K]
{BLU}6{GRN}[K]{BLU}7[K]8{YEL}[K]{BLU}
{3 SPACES}0<"
28 FORP=7725TO8165STEP22:FORA=PTOP+8:POKEA,102:POK
EA+30720,0:NEXTP,P
30 FORP=7748TO8144STEP44:FORA=PTOP+6STEP2:POKEA,16
0:POKEA+30720,1:NEXTP,P
32 FORA=1TO4:J=INT(5*RND(0)+1):X(A)=Z(J):NEXT
34 N=1
36 M=7702+44*N:L=M+30720:POKEM,48+N:POKEL,6
38 FORJ=1TO4:POKES,225:GOSUB78:POKES,0
40 GETES:IFES<>"1"ANDES<>"3"ANDES<>"6"ANDES<>"7"AN
DES<>"8"ANDES<>"0"THEN40
```

```

42 IF E$ <> "0" THEN 46
44 POKE L+2, 1: POKE L+4, 1: POKE L+6, 1: GOTO 38
46 Y(J) = VAL(E$): POKE L+J*2, Y(J)-1: NEXT J
48 J=1
50 FOR A=1 TO 4: W(A)=X(A): IF X(A) <> Y(A) THEN 54
52 POKE L+12+J, 83: POKE L+12+J, 2: Y(A)=0: W(A)=9: J=J+1
54 NEXT A: IF J=5 THEN 68
56 FOR A=1 TO 4: FOR AL=1 TO 4: IF W(A) <> Y(AL) THEN 60
58 POKE L+12+J, 90: POKE L+12+J, 5: Y(AL)=0: W(A)=9: J=J+1
60 NEXT AL, A
62 N=N+1: IF N < 10 THEN 36
64 POKES, 159: GOSUB 78: POKES, 135: GOSUB 78: POKES, 0
66 L=38862: FOR A=1 TO 4: POKE L+2*A, X(A)-1: NEXT A: GOTO 70
*68 FOR A=1 TO 3: POKES, 195: GOSUB 78: POKES, 207: GOSUB 78: POKES, 215: GOSUB 78: POKES, 225: GOSUB 78: GOSUB 78: POKES, 0: NEXT A
70 PRINT "{ 22 DOWN } TRY AGAIN? (Y/N)";
72 GET E$: IF E$ = "Y" THEN 26
74 IF E$ = "N" THEN 80
76 GOTO 72
78 FOR T=1 TO 99: NEXT T: RETURN
80 PRINT "{ CLR }": END

```

Color Code works differently on the Color Computer:

Line	Function
10-60	PRINT instructions.
70-80	Define character numbers for colors C(I).
90-110	Wait for user to press ENTER to start.
120-130	PRINT color numbers and colors across top of screen.
140-160	Draw playing board.
170	Choose series of four colors; initialize N as number of tries.
180-280	Receive user's four guesses of color numbers.
290-320	Check for correct color in correct position.
330-350	Check for correct color in any other position.
360	Increment number of guesses; return for next try.
370-380	Show correct answer.
390	Play tune for correct answer.
400-420	PRINT option to try again and branch appropriately.
430	END.

Program 4-9. Color Code

TRS-80 Color Computer

```

10 CLS:PRINT TAB(6); "*** COLOR CODE ***"
20 PRINT:PRINT "GUESS THE CORRECT SEQUENCE"

```

Graphics

```
30 PRINT "OF FOUR COLORS.":PRINT"GUESS BY PRESSING
FOUR"
40 PRINT "COLOR NUMBERS.":PRINT"PRESS THE BACK ARR
OW TO ERASE."
50 PRINT:PRINT"! RIGHT COLOR ONLY"
60 PRINT"* RIGHT COLOR, CORRECT POSITION"
70 FOR I=2 TO 8:READC(I):NEXT
80 DATA 159,175,191,207,223,239,255
90 PRINT:PRINT"PRESS <ENTER> TO START."
100 E$=INKEY$:IF E$=""THEN100
110 IF ASC(E$)<>13 THEN 100
120 CLS
130 FOR I=2TO8:PRINT STR$(I);CHR$(C(I));:NEXT
140 A$=CHR$(128):B$=CHR$(131)
150 FOR I=65 TO 417 STEP 32:PRINT@I,A$+B$+A$+B$+A$
+B$+A$+B$+A$:NEXT
160 PRINT @449,A$+A$+A$+A$+A$+A$+A$+A$+A$
170 FOR A=1 TO 4:X(A)=RND(6)+2:NEXT:N=1
180 FOR A=1 TO 4
190 SOUND 200,1
200 M=32+32*N+A*2
210 PRINT@M,CHR$(128);:PRINT@M,CHR$(131);
220 E$=INKEY$:IF E$=""THEN 210
230 IF ASC(E$)<>8 THEN 250
240 PRINT @34+32*N,B$+A$+B$+A$+B$;:A=1:GOTO 190GOT
O 190
250 IF ASC(E$)<50 THEN 210
260 IF ASC(E$)>56 THEN 210
270 E=VAL(E$):Y(A)=E:PRINT@M,CHR$(C(E)-12);
280 NEXT A
290 J=1:M=M+5
300 FORA=1TO4:W(A)=X(A):IFX(A)<>Y(A)THEN 320
310 PRINT@M,"*";:Y(A)=0:W(A)=9:J=J+1:M=M+1
320 NEXTA:IFJ=5THEN390
330 FORA=1TO4:FORA1=1TO4:IFW(A)<>Y(A1)THEN350
340 PRINT@M,"!";:Y(A1)=0:W(A)=9:J=J+1:M=M+1
350 NEXTA1,A
360 N=N+1:IFN<12THEN180
370 SOUND125,3:SOUND89,3
380 FORA=1TO4:PRINT@416+2*A,CHR$(C(X(A))-12);:NEXT
:GOTO400
390 FORA=1TO3:SOUND89,1:SOUND125,1:SOUND147,1:SOUN
D176,2:NEXTA
400 PRINT@480,"TRY AGAIN? (Y/N)";
410 E$=INKEY$:IFE$="Y"THEN120
420 IFE$<>"N"THEN410
430 END
```

The same game program, this time on the TI-99/4A, works like this:

Line	Function
100-200	PRINT instructions and define graphics characters.
210-320	Define graphics, colors, and character numbers for colors.
330-350	Wait for user to press any key to start.
360-390	Draw playing board.
400-430	Choose series of four colors.
440-500	PRINT color numbers and colors across top of screen.
510-520	Initialize I for number of tries.
530	If number of guesses is greater than 9, end.
540-550	Label number of try.
560-710	Receive user's four guesses of color numbers.
720-810	Check for correct color in correct position.
820-910	Check for correct color in any other position.
920	Return for next try.
930-980	Show correct answer.
990-1030	Play tune for correct answer.
1040-1070	PRINT option to try again and branch appropriately.
1080-1090	Clear screen and END.

Program 4-10. Color Code

TI-99/4A

```

100 CALL CLEAR
110 PRINT TAB(7);"COLOR CODE"
120 CALL CHAR(93,"3C7EFFFFFFF7E3C")
130 PRINT :::"GUESS THE CORRECT SEQUENCE"
140 PRINT "OF FOUR COLORS."
150 CALL CHAR(120,"3C7EFFFFFFF7E3C")
160 PRINT : "GUESS BY PRESSING THE"
170 PRINT "COLOR NUMBERS."
180 CALL COLOR(12,16,1)
190 PRINT :::CHR$(93);" RIGHT COLOR AND POSITION"
200 PRINT :CHR$(120);" RIGHT COLOR ONLY"
210 FOR I=1 TO 3
220 CALL CHAR((11+I)*8,"0")
230 CALL CHAR((11+I)*8+1,"FFFFFFFFFFFFFF")
240 NEXT I
250 FOR I=1 TO 6
260 READ C(I)
270 NEXT I
280 DATA 96,97,104,105,112,113
290 CALL COLOR(9,3,16)
300 CALL COLOR(10,5,12)
310 CALL COLOR(11,7,14)

```

Graphics

```
320 CALL CHAR(92,"FFFFFFFFFFFFFFFF")
330 PRINT :::"PRESS ANY KEY TO START.";
340 CALL KEY(0,K,S)
350 IF S<1 THEN 340
360 CALL CLEAR
370 FOR I=9 TO 17
380 CALL VCHAR(4,I,92,20)
390 NEXT I
400 FOR I=1 TO 4
410 RANDOMIZE
420 A(I)=INT(RND*6+1)
430 NEXT I
440 FOR I=1 TO 6
450 CALL HCHAR(1,3*I-1,48+I)
460 CALL HCHAR(1,3*I,C(I))
470 NEXT I
480 FOR I=1 TO 7
490 CALL HCHAR(1,21+I,ASC(SEG$( "0-ERASE",I,1)))
500 NEXT I
510 I=0
520 I=I+1
530 IF I>9 THEN 930
540 IA=3+2*I
550 CALL HCHAR(IA,8,48+I)
560 FOR J=1 TO 4
570 JA=8+2*J
580 CALL SOUND(100,1497,2)
590 CALL KEY(0,K,S)
600 CALL HCHAR(IA,JA,92)
610 CALL HCHAR(IA,JA,32)
620 IF S<1 THEN 590
630 IF (K>47)*(K<55)<>1 THEN 590
640 IF K<>48 THEN 690
650 FOR K=J TO 1 STEP -1
660 CALL HCHAR(IA,8+K*2,92)
670 NEXT K
680 GOTO 560
690 B(J)=K-48
700 CALL HCHAR(IA,JA,C(K-48))
710 NEXT J
720 L=1
730 FOR J=1 TO 4
740 D(J)=A(J)
750 IF A(J)<>B(J)THEN 800
760 CALL HCHAR(IA,21+L,93)
770 B(J)=0
780 D(J)=8
790 L=L+1
800 NEXT J
```

```

810 IF L=5 THEN 990
820 FOR J=1 TO 4
830 FOR J1=1 TO 4
840 IF D(J) <> B(J1) THEN 900
850 CALL HCHAR(IA, 21+L, 120)
860 B(J1)=0
870 D(J)=8
880 L=L+1
890 IF L=5 THEN 520
900 NEXT J1
910 NEXT J
920 GOTO 520
930 CALL SOUND(150, 330, 2)
940 CALL SOUND(150, 262, 2)
950 FOR J=1 TO 4
960 CALL HCHAR(23, 8+2*J, C(A(J)))
970 NEXT J
980 GOTO 1040
990 FOR J=1 TO 3
1000 CALL SOUND(100, 262, 2)
1010 CALL SOUND(100, 330, 2)
1020 CALL SOUND(100, 392, 2)
1030 NEXT J
1040 PRINT "TRY AGAIN? (Y/N)";
1050 CALL KEY(0, K, S)
1060 IF K=89 THEN 360
1070 IF K <> 78 THEN 1050
1080 CALL CLEAR
1090 END

```

Matching Shapes

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

"Matching Shapes" uses the color and graphics capabilities of the computer to present a teaching game for preschool children. This program, a drill of matching shapes, begins with a random shape displayed. Four more random shapes are then drawn. One of the four matches the first shape. Press the matching shape's number as your answer. The answer must be correct to continue. Remember to look at the shape, not at the color. After ten problems you have the option to try again.

Each computer's program version is listed below.

Here's how the VIC version works:

Line	Function
2	Initialize variables Y for color and T for sound, turn volume on, and branch past subroutines.

Graphics

- 4-24 Subroutines to draw shapes.
- 26 Subroutine for sound delay.
- 28-30 Read graphics coordinate position W from DATA.
- 32-74 Perform quiz ten times.
- 34 PRINT question.
- 36 Randomly choose shape N and color C, set X equal to the coordinate for drawing, and draw shape to be matched.

- 38 PRINT answer numbers.
- 40 Randomly choose answer number.
- 42-58 Randomly draw four shapes for the four possible answers. Color C is chosen randomly. X is the coordinate for drawing. Lines 44-46 draw the correct answer shape.

- 60-62 Play a tone, then wait for any key to be pressed.
- 64-66 If answer is incorrect, play sound effect and return for another response.
- 68-74 If answer is correct, PRINT three hearts under the correct shape, play a musical sound effect, and return for next problem.

- 76-80 PRINT option to try again and branch appropriately.
- 82 Clear screen and END.

Program 4-11. Matching Shapes

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```
2 Y=30720:T=36876:POKE36878,15:GOTO28
4 POKEX,160:POKEX+21,160:POKEX+22,160:POKEX+23,160
  :POKEX+44,160
6 Z=X+Y:POKEZ,C:POKEZ+21,C:POKEZ+22,C:POKEZ+23,C:P
  OKEZ+44,C:RETURN
*8 FORI=X-1TOX+1:POKEI,160:POKEI+22,160:POKEI+44,16
  0:POKEI+Y,C:POKEI+Y+22,C:POKEI+Y+44,C:NEXT:RETUR
  N
10 POKEX,233:POKEX+1,233:POKEX+21,233:POKEX+22,160
  :POKEX+23,160:POKEX+24,233
12 Z=X+Y:POKEZ,C:POKEZ+1,C:POKEZ+21,C:POKEZ+22,C:P
  OKEZ+23,C:POKEZ+24,C:RETURN
14 POKEX,223:POKEX+22,160:POKEX+23,223:POKEX+44,16
  0:POKEX+45,160:POKEX+46,223
16 Z=X+Y:POKEZ,C:POKEZ+22,C:POKEZ+23,C:POKEZ+44,C:
  POKEZ+45,C:POKEZ+46,C:RETURN
18 POKEX-1,233:POKEX,160:POKEX+1,223:POKEX+21,160:
  POKEX+22,160:POKEX+23,160
```

```

20 POKE X+43,95:POKE X+44,160:POKE X+45,105:Z=X+Y
22 FOR I=Z-1 TO Z+1:POKE I,C:POKE I+22,C:POKE I+44,C:NEXT:RETURN
24 FOR I=X TO X+66 STEP 22:POKE I,160:POKE I+Y,C:POKE I+1,160:POKE I+Y+1,C:NEXT:RETURN
26 FOR D=1 TO 60:NEXT:RETURN
28 FOR I=0 TO 4:READ W(I):NEXT
30 DATA 7754,7947,7952,7957,7962
32 FOR P=1 TO 10
34 PRINT "{CLR}{DOWN}WHICH SHAPE MATCHES?"
36 N=INT(6*RND(0)+1):X=W(0):C=INT(6*RND(0)+2):ONN {SPACE}GOSUB 4,8,10,14,18,24
38 PRINT "{9 DOWN}{3 SPACES}1{4 SPACES}2{4 SPACES}3{4 SPACES}4"
40 A=INT(4*RND(0)+1)
42 FOR J=1 TO 4:X=W(J):C=INT(6*RND(0)+2)
44 IF J<>A THEN 48
46 K(J)=N:ONN GOSUB 4,8,10,14,18,24:GOTO 58
48 B=INT(6*RND(0)+1):IF B=N THEN 48
50 IF J=1 THEN 56
52 FOR L=1 TO J-1:IF B=K(L) THEN 48
54 NEXT L
56 K(J)=B:ONB GOSUB 4,8,10,14,18,24
58 NEXT J
60 POKET,225:GOSUB 26:POKET,0
62 GET E$:IF E$="" THEN 62
64 IF VAL(E$)=A THEN 68
66 POKET,159:GOSUB 26:POKET,135:GOSUB 26:POKET,0:GOTO 62
68 PRINT TAB(A*5-3);"{5 DOWN}{RED}SSS{BLU}"
70 POKET,195:GOSUB 26:POKET,207:GOSUB 26:POKET,215:GOSUB 26
72 POKET,225:GOSUB 26:GOSUB 26:POKET,0
74 NEXT P
76 PRINT "{2 DOWN}TRY AGAIN? (Y/N)"
78 GET E$:IF E$="Y" THEN 32
80 IF E$<>"N" THEN 78
82 PRINT "{CLR}":END

```

The Color Computer version looks a bit different:

Line	Function
20	Branch past subroutines.
30-410	Subroutine to draw shapes.
420-430	Initialize coordinates for positions of shapes.
440	Perform quiz ten times.
450	Clear screen and PRINT question.
460	Randomly choose shape number N and color C and draw shape to be matched.

Graphics

- 470 Randomly choose answer number.
480-570 Randomly draw four shapes for the four possible answers. Color C is chosen randomly. X and Y are coordinates for drawing.
580-590 Play a tone, then wait for any key to be pressed.
600-610 If answer is incorrect, play musical tone and return for another response.
620-640 If answer is correct, PRINT message, play musical effect, and return for next problem.
650-670 PRINT option to try again and branch appropriately.
680 Clear screen and END.

Program 4-12. Matching Shapes

TRS-80 Color Computer; MC-10

```
20 GOTO420
30 L1=X:L2=X+1
40 FOR J=Y TO Y+4 STEP 2
50 FOR I=L1 TO L2
60 SET(I,J,C):SET(I,J+1,C)
70 NEXT I
80 L1=L1-2:L2=L2+2
90 NEXT J
100 RETURN
110 FOR J=Y TO Y+5
120 FOR I=X-4 TO X+5
130 SET(I,J,C)
140 NEXT I,J
150 RETURN
160 L2=X+1
170 FOR J=Y TO Y+4 STEP 2
180 FOR I=X TO L2
190 SET(I,J,C):SET(I,J+1,C)
200 NEXT I
210 L2=L2+2
220 NEXT J
230 RETURN
240 FOR I=X TO X+1
250 FOR J=Y TO Y+5
260 SET(I,J,C)
270 NEXT J,I
280 FOR I=X-4 TO X+5
290 SET(I,Y+2,C):SET(I,Y+3,C)
300 NEXT I
310 RETURN
320 FOR J=Y TO Y+4 STEP 2
330 SET(X,J,C):SET(X,J+1,C)
340 SET(X+1,J,C):SET(X+1,J+1,C)
```

```

350 X=X+2
360 NEXT J:X=X-2
370 SET(X,Y,C):SET(X,Y+1,C)
380 SET(X+1,Y,C):SET(X+1,Y+1,C)
390 SET(X-4,Y+4,C):SET(X-3,Y+4,C)
400 SET(X-4,Y+5,C):SET(X-3,Y+5,C)
410 RETURN
420 FORI=0TO4:READA(I),B(I):NEXT
430 DATA30,2,8,18,24,18,38,18,54,18
440 FORP=1TO10
450 CLS:PRINT"WHICH SHAPE MATCHES?"
460 N=RND(5):X=A(0):Y=B(0):C=RND(3)+1:ONN GOSUB30,
    110,160,240,320
470 D=RND(4)
480 FORM=1TO4:X=A(M):Y=B(M):C=RND(3)+1
490 IFM<>D THEN510
500 K(M)=N:ONN GOSUB30,110,160,240,320:GOTO560
510 E=RND(5):IFE=N THEN510
520 IFM=1THEN550
530 FORL=1TOM-1:IFE=K(L)THEN510
540 NEXTL
550 K(M)=E:ONE GOSUB30,110,160,240,320
560 NEXTM
570 PRINT@388,"1{7 SPACES}2{7 SPACES}3{6 SPACES}4"
580 SOUND200,1
590 E$=INKEY$:IFE$=""THEN590
600 IFVAL(E$)=D THEN 620
610 SOUND125,1:SOUND89,1:GOTO590
620 PRINT@448,"CORRECT!"
630 SOUND89,1:SOUND125,1:SOUND147,1:SOUND176,2
640 NEXTP
650 PRINT:PRINT"TRY AGAIN? (Y/N)";
660 E$=INKEY$:IFE$="Y"THEN440
670 IFE$<>"N"THEN660
680 CLS:END

```

The Model I version of Matching Shapes works like this:

Line	Function
20	Branch past subroutines.
30-390	Draw shapes.
420-540	Draw title screen.
550-570	Wait for ENTER key to be pressed.
580	Perform quiz ten times.
590	Clear screen and PRINT question.
600	Randomly choose shape number N and draw shape to be matched.
610	Randomly choose answer number.

Graphics

- 620-710 Randomly draw four shapes for the four possible answers, making sure none of the shapes are the same.
- 720 Wait for a key to be pressed.
- 730-740 If answer is incorrect, PRINT message and return for another response.
- 750-770 If answer is correct, PRINT message, blink arrows under correct answer, and return for next problem.
- 780-800 PRINT option to try again and branch appropriately.
- 810 Clear screen and END.

Program 4-13. Matching Shapes

TRS-80 Model I

```
20 GOTO 420
30 L1=X+4:L2=X+5
40 FOR J=Y TO Y+6 STEP 3
50 FOR I=L1 TO L2
60 SET(I,J):SET(I,J+1):SET(I,J+2)
70 NEXT I
80 L1=L1-2:L2=L2+2
90 NEXT J
100 RETURN
110 FOR J=Y TO Y+8
120 FOR I=X TO X+15
130 SET(I,J)
140 NEXT I,J
150 RETURN
160 L2=X
170 FOR J=Y TO Y+8
180 FOR I=X TO L2
190 SET(I,J)
200 NEXT I
210 L2=L2+1
220 NEXT J
230 RETURN
240 FOR I=X+4 TO X+7
250 FOR J=Y TO Y+8
260 SET(I,J)
270 NEXT J,I
280 FOR I=X TO X+11
290 SET(I,Y+3):SET(I,Y+4):SET(I,Y+5)
300 NEXT I
310 RETURN
320 FOR J=Y+2 TO Y+5:FOR I=X TO X+13:SET(I,J):NEXT
    I,J
330 FOR I=X+2 TO X+11:SET(I,Y+1):SET(I,Y+6):NEXT I
340 FOR I=X+4 TO X+9:SET(I,Y):SET(I,Y+7):NEXT I
350 RETURN
```

```

360 I=X:II=X+14
370 FOR J=Y TO Y+7:SET(I,J):SET(I+1,J):I=I+2
380 SET(II,J):SET(II+1,J):II=II-2:NEXT J
390 RETURN
420 CLS
430 PRINT@144,"MATCHING SHAPES"
460 X=18:Y=15:GOSUB30
480 X=70:Y=30:GOSUB110
500 X=76:Y=15:GOSUB160
520 X=36:Y=27:GOSUB240
530 X=46:Y=12:GOSUB320
540 X=98:Y=21:GOSUB360
550 PRINT@960,L"PRESS <ENTER> TO START";
560 E$=INKEY$:IFE$=""THEN560
570 IF ASC(E$)<>13 THEN560
580 FOR P=1 TO 10
590 CLS:PRINT"WHICH SHAPE MATCHES?"
600 N=RND(6):X=56:Y=9:ONN GOSUB30,110,160,240,320,
    360
610 D=RND(4)
620 FOR M=1TO4:X=28*M-14:Y=30
630 IF M<>D THEN650
640 K(M)=N:ONN GOSUB30,110,160,240,320,360:GOTO700
650 E=RND(6):IFE=N THEN650
660 IF M=1 THEN690
670 FORL=1TOM-1:IFE=K(L) THEN650
680 NEXTL
690 K(M)=E:ON E GOSUB30,110,160,240,320,360
700 NEXTM
710 PRINT@842,"1{13 SPACES}2{13 SPACES}3{13 SPACES}4"
720 E$=INKEY$:IFE$=""THEN720
730 IFVAL(E$)=D THEN750
740 PRINT@896,E$;" NO, TRY AGAIN.":GOTO720
750 PRINT@896,"CORRECT{9 SPACES}"
760 V=896+14*D-6:FORL=1TO100:PRINT@V,"^^^^^";:PRIN
    T@V,"{5 SPACES}";:NEXTL
770 NEXTP
780 PRINT:PRINT"TRY AGAIN? (Y/N)"
790 E$=INKEY$:IFE$="Y"THEN580
800 IFE$<>"N"THEN790
810 CLS:END

```

The Matching Shapes program works like this on the TI-99/4A:

Line	Function
110	Define R for use in choosing a random color.
120	Branch past subroutines.
130-470	Subroutines to draw shapes.
480-600	Display title while defining graphic shapes.

Graphics

- 610 Perform quiz ten times.
620 Clear screen.
630-680 Randomly choose shape number N and draw shape to be matched.
690 Randomly choose answer number D.
700-870 Randomly draw four shapes for the four possible answers. The correct shape is in position D, and none of the shapes may be the same.
880 PRINT question.
890-910 Play a tone, then wait for a key to be pressed.
920-950 If answer is incorrect, play notes and return for another response.
960-1020 If answer is correct, PRINT three asterisks under correct choice, erase question, play musical effect, and return for next problem.
1030-1060 PRINT option to try again and branch appropriately.
1070-1080 Clear screen and END.

Program 4-14. Matching Shapes

TI-99/4A

```
110 DEF R=INT(12*RND+5)
120 GOTO 480
130 CALL VCHAR(X,Y,96,3)
140 CALL HCHAR(X+1,Y-1,96,3)
150 RETURN
160 FOR I=X TO X+2
170 CALL HCHAR(I,Y-1,104,3)
180 NEXT I
190 RETURN
200 CALL HCHAR(X,Y-1,113)
210 CALL VCHAR(X,Y,112,3)
220 CALL HCHAR(X,Y+1,114)
230 CALL HCHAR(X+1,Y-1,112,3)
240 CALL HCHAR(X+2,Y-1,116)
250 CALL HCHAR(X+2,Y+1,115)
260 RETURN
270 CALL HCHAR(X,Y-1,121)
280 CALL HCHAR(X+1,Y-1,120)
290 CALL HCHAR(X+1,Y,121)
300 CALL HCHAR(X+2,Y-1,120,2)
310 CALL HCHAR(X+2,Y+1,121)
320 RETURN
330 CALL VCHAR(X,Y,128,3)
340 CALL HCHAR(X,Y-1,129)
350 CALL HCHAR(X,Y+1,130)
```

```

360 CALL HCHAR(X+1,Y+1,131)
370 CALL HCHAR(X+2,Y+1,132)
380 CALL HCHAR(X+2,Y-1,133)
390 CALL HCHAR(X+1,Y-1,134)
400 RETURN
410 CALL HCHAR(X,Y,137)
420 CALL HCHAR(X+1,Y-1,138)
430 CALL VCHAR(X+1,Y,136,2)
440 CALL HCHAR(X+1,Y+1,139)
450 CALL HCHAR(X+2,Y-1,140)
460 CALL HCHAR(X+2,Y+1,141)
470 RETURN
480 CALL CLEAR
490 PRINT TAB(7);"MATCHING SHAPES":::::::::::::
500 FOR C=96 TO 136 STEP 8
510 CALL CHAR(C,"FFFFFFFFFFFFFFFF")
520 NEXT C
530 FOR I=1 TO 16
540 READ C,C$
550 CALL CHAR(C,C$)
560 NEXT I
570 DATA 113,00030F1F3F3F7F7F,114,00C0F0F8FCFCFEFE
,115,FEFEFCFCF8F0C,116,7F7F3F3F1F0F03,121,80C0
E0F0F8FCFEFF
580 DATA 129,0101030307070F0F,130,8080C0C0E0E0F0F,
131,F8F8FCFCFCFCF8F8,132,F0F0E0E0C0C0808,133
590 DATA 0F0F070703030101,134,1F1F3F3F3F3F1F1F,137
,18183C3C7E7EFFFF,138,0101030307070F0F,139,808
0C0C0E0E0F0F
600 DATA 140,1F1F3F3F7F7FFFFF,141,F8F8FCFCFEFEFFFF
610 FOR P=1 TO 10
620 CALL CLEAR
630 X=11
640 Y=15
650 RANDOMIZE
660 N=IN=(6*RND+1)
670 CALL COLOR(8+N,R,1)
680 ON N GOSUB 130,160,200,270,330,410
690 D=INT(4*RND+1)
700 FOR M=1 TO 4
710 X=21
720 Y=7*M-1
730 IF M<>D THEN 770
740 J(M)=N
750 ON N GOSUB 130,160,200,270,330,410
760 GOTO 860
770 E=INT(6*RND+1)
780 IF E=N THEN 770
790 IF M=1 THEN 830

```

Graphics

```
800 FOR L=1 TO M-1
810 IF E=J(L)THEN 770
820 NEXT L
830 J(M)=E
840 CALL COLOR(8+E,R,1)
850 ON E GOSUB 130,160,200,270,330,410
860 NEXT M
870 PRINT TAB(4);"1{6 SPACES}2{6 SPACES}3
   {6 SPACES}4"
880 PRINT ::"WHICH SHAPE MATCHES?":::
890 CALL SOUND(150,1497,2)
900 CALL KEY(0,K,S)
910 IF S<1 THEN 900
920 IF K=48+D THEN 960
930 CALL SOUND(100,330,2)
940 CALL SOUND(100,262,2)
950 GOTO 900
960 CALL HCHAR(19,7*D-2,42,3)
970 CALL HCHAR(21,3,32,20)
980 CALL SOUND(100,262,2)
990 CALL SOUND(100,330,2)
1000 CALL SOUND(100,392,2)
1010 CALL SOUND(200,523,2)
1020 NEXT P
1030 PRINT ::"TRY AGAIN? (Y/N)";
1040 CALL KEY(0,K,S)
1050 IF K=89 THEN 600
1060 IF K<>78 THEN 1040
1070 CALL CLEAR
1080 END
```

Counting Shapes

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

"Counting Shapes" is a program designed for preschool children or for anyone learning the concept of counting. One to nine shapes are drawn on the screen. You simply press the correct *number* key as an answer. To continue, the answer must be correct. After ten problems, you have the option to try again.

This program, although simpler than "Matching Shapes," does use some of the same shapes. All three versions of Counting Shapes work in the same general way.

The instructions to draw the shapes are in subroutines at the beginning of each program. The coordinates for the nine possible positions are then defined. A random color C is chosen for the objects, a random number N for the number of objects, and a

random number S for the shape. A FOR-NEXT loop draws the correct number of objects.

Program 4-15. Counting Shapes

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```

2 Y=30720:T=36876:POKE36878,15:GOTO28
4 POKE X,160:POKE X+21,160:POKE X+22,160:POKE X+23,160
  :POKE X+44,160
6 Z=X+Y:POKE Z,C:POKE Z+21,C:POKE Z+22,C:POKE Z+23,C:P
  OKE Z+44,C:RETURN
*8 FOR I=X-1 TO X+1:POKE I,160:POKE I+22,160:POKE I+44,16
  0:POKE I+Y,C:POKE I+Y+22,C:POKE I+Y+44,C:NEXT:RETUR
  N
10 POKE X,233:POKE X+1,223:POKE X+21,233:POKE X+22,160
  :POKE X+23,160:POKE X+24,223
12 Z=X+Y:POKE Z,C:POKE Z+1,C:POKE Z+21,C:POKE Z+22,C:P
  OKE Z+23,C:POKE Z+24,C:RETURN
14 POKE X,223:POKE X+22,160:POKE X+23,223:POKE X+44,16
  0:POKE X+45,160:POKE X+46,223
16 Z=X+Y:POKE Z,C:POKE Z+22,C:POKE Z+23,C:POKE Z+44,C:
  POKE Z+45,C:POKE Z+46,C:RETURN
18 POKE X-1,233:POKE X,160:POKE X+1,223:POKE X+21,160:
  POKE X+22,160:POKE X+23,160
20 POKE X+43,95:POKE X+44,160:POKE X+45,105:Z=X+Y
22 FOR I=Z-1 TO Z+1:POKE I,C:POKE I+22,C:POKE I+44,C:NEX
  T:RETURN
24 FOR I=X TO X+66 STEP 22:POKE I,160:POKE I+Y,C:POKE I+1,
  160:POKE I+Y+1,C:NEXT:RETURN
26 FOR D=1 TO 60:NEXT D:RETURN
28 PRINT "{CLR}"
30 PRINT "{3 RIGHT}{BLU}COUNTING SHAPES"
32 C=INT(6*RND(0)+2):X=7794:GOSUB 4
34 C=INT(6*RND(0)+2):X=7843:GOSUB 8
36 C=INT(6*RND(0)+2):X=7850:GOSUB 10
38 C=INT(6*RND(0)+2):X=7925:GOSUB 14
40 C=INT(6*RND(0)+2):X=7954:GOSUB 18
42 C=INT(6*RND(0)+2):X=7981:GOSUB 24
44 FOR I=1 TO 9:READ W(I):NEXT
46 DATA 7771,7842,7803,7875,7924,7973,7956,8005,808
  0
48 PRINT "{17 DOWN}PRESS RETURN TO START.";
50 GET E$:IF E$="" THEN 50
52 IF ASC(E$) <> 13 THEN 50
54 FOR K=1 TO 10
56 PRINT "{CLR}"
58 C=INT(6*RND(0)+2)

```


Graphics

```
60 N=INT(9*RND(0)+1)
62 S=INT(6*RND(0)+1)
64 FORL=1TON:X=W(L)
66 ON S GOSUB4,8,10,14,18,24
68 NEXTL
70 PRINT"{HOME}HOW MANY OBJECTS?"
72 POKET,225:FORD=1TO100:NEXTD:POKET,0
74 GETES$:IFES$=""THEN74
76 PRINTE$
78 IFASC(E$)=48+N THEN86
80 POKET,159:GOSUB26:POKET,135:GOSUB26:POKET,0
82 PRINT"{HOME}{DOWN} {LEFT}";
84 GOTO72
*86 POKET,195:GOSUB26:POKET,207:GOSUB26:POKET,215:G
OSUB26:POKET,225:GOSUB26:GOSUB26:POKET,0
88 NEXT K
90 PRINT"{CLR}"
92 PRINT"TRY AGAIN? (Y/N)"
94 GETES$:IFES$="Y"THEN54
96 IFES$<>"N" THEN 94
98 PRINT"{CLR}"
100 END
```

Program 4-16. Counting Shapes

TRS-80 Color Computer; MC-10

```
20 GOTO 420
30 L1=X:L2=X+1
40 FOR J=Y TO Y+4 STEP 2
50 FOR I=L1 TO L2
60 SET(I,J,C):SET(I,J+1,C)
70 NEXT I
80 L1=L1-2:L2=L2+2
90 NEXT J
100 RETURN
110 FOR J=Y TO Y+5
120 FOR I=X TO X+9
130 SET(I,J,C)
140 NEXT I,J
150 RETURN
160 L2=X+1
170 FOR J=Y TO Y+6 STEP 2
180 FOR I=X TO L2
190 SET(I,J,C):SET(I,J+1,C)
200 NEXT I
210 L2=L2+2
220 NEXT J
230 RETURN
240 FOR I=X TO X+1
250 FOR J=Y TO Y+5
```

```

260 SET(I,J,C)
270 NEXT J,I
280 FOR I=X-4 TO X+5
290 SET(I,Y+2,C):SET(I,Y+3,C)
300 NEXT I
310 RETURN
320 FOR J=Y TO Y+4 STEP 2
330 SET(X,J,C):SET(X,J+1,C)
340 SET(X+1,J,C):SET(X+1,J+1,C)
350 X=X+2
360 NEXT J:X=X-2
370 SET(X,Y,C):SET(X,Y+1,C)
380 SET(X+1,Y,C):SET(X+1,Y+1,C)
390 SET(X-4,Y+4,C):SET(X-3,Y+4,C)
400 SET(X-4,Y+5,C):SET(X-3,Y+5,C)
410 RETURN
420 CLS
430 PRINT @72,"COUNTING SHAPES"
440 C=RND(3)+1
450 X=10:Y=8:GOSUB 30
460 C=RND(3)+1
470 X=20:Y=20:GOSUB 110
480 C=RND(3)+1
490 X=44:Y=16:GOSUB 160
500 C=RND(3)+1
510 X=30:Y=10:GOSUB 240
520 C=RND(3)+1
530 X=52:Y=8:GOSUB 320
540 FOR I=1 TO 9:READ X1(I),Y1(I):NEXT I
550 DATA 8,4,16,14,28,6,38,14,44,4,52,16,4,20,28,2
    2,48,24
560 PRINT @480,"PRESS ENTER TO START.";
570 E$=INKEY$:IF E$="" THEN 570
580 IF ASC(E$)<>13 THEN 570
590 FOR K=1 TO 10
600 CLS
610 C=RND(3)+1
620 N=RND(9)
630 S=RND(5)
640 FOR L=1 TO N
650 X=X1(L):Y=Y1(L)
660 ON S GOSUB 30,110,160,240,320
670 NEXT L
680 PRINT "HOW MANY OBJECTS?"
690 SOUND 227,2
700 E$=INKEY$:IF E$="" THEN 700
710 PRINT @20,E$
720 IF ASC(E$)=48+N THEN 760
730 SOUND 125,1:SOUND 89,1

```

Graphics

```
740 PRINT @20, " "  
750 GOTO 700  
760 SOUND 89,1:SOUND 125,1:SOUND 147,1:SOUND 176,2  
770 NEXT K  
780 CLS  
790 PRINT @258, "TRY AGAIN? (Y/N)"  
800 E$=INKEY$:IF E$="Y" THEN 590  
810 IF E$<>"N" THEN 800  
820 CLS  
830 END
```

Program 4-17. Counting Shapes

TRS-80 Model I

```
20 GOTO 420  
30 L1=X+4:L2=X+5  
40 FOR J=Y TO Y+6 STEP 3  
50 FOR I=L1 TO L2  
60 SET(I,J):SET(I,J+1):SET(I,J+2)  
70 NEXT I  
80 L1=L1-2:L2=L2+2  
90 NEXT J  
100 RETURN  
110 FOR J=Y TO Y+8  
120 FOR I=X TO X+15  
130 SET(I,J)  
140 NEXT I,J  
150 RETURN  
160 L2=X  
170 FOR J=Y TO Y+8  
180 FOR I=X TO L2  
190 SET(I,J)  
200 NEXT I  
210 L2=L2+1  
220 NEXT J  
230 RETURN  
240 FOR I=X+4 TO X+7  
250 FOR J=Y TO Y+8  
260 SET(I,J)  
270 NEXT J,I  
280 FOR I=X TO X+11  
290 SET(I,Y+3):SET(I,Y+4):SET(I,Y+5)  
300 NEXT I  
310 RETURN  
320 FOR J=Y+2 TO Y+5:FOR I=X TO X+13:SET(I,J  
):NEXT I,J  
330 FOR I=X+2 TO X+11:SET(I,Y+1):SET(I,Y+6):  
NEXT I  
340 FOR I=X+4 TO X+9:SET(I,Y):SET(I,Y+7):NEX  
T I
```

```

350 RETURN
360 I=X:II=X+14
370 FOR J=Y TO Y+7:SET(I,J):SET(I+1,J):I=I+2
380 SET(II,J):SET(II+1,J):II=II-2:NEXT J
390 RETURN
420 CLS
430 PRINT@144,"COUNTING SHAPES"
460 X=18:Y=15:GOSUB30
480 X=70:Y=30:GOSUB110
500 X=76:Y=15:GOSUB160
520 X=36:Y=27:GOSUB240
530 X=46:Y=12:GOSUB320
540 X=98:Y=21:GOSUB360
550 FORI=1TO9:READX1(I),Y1(I):NEXTI
560 DATA46,12,70,30,98,13,36,27,20,17,98,27,
    76,15,2,9,8,33
570 PRINT@960,"PRESS <ENTER> TO START.";
580 E$=INKEY$:IFE$=""THEN580
590 IF ASC(E$)<>13THEN580
600 FOR K=1 TO 10
610 CLS
630 N=RND(9)
640 S=RND(6)
650 FOR L=1TON
660 X=X1(L):Y=Y1(L)
670 ON S GOSUB30,110,160,240,320,360
680 NEXTL
690 PRINT"HOW MANY OBJECTS?"
710 E$=INKEY$:IFE$=""THEN710
720 PRINT@20,E$
730 IF ASC(E$)=48+N THEN770
740 PRINT@84,"SORRY, TRY AGAIN.":FOR II=1TO5
    00:NEXTII
750 PRINT@20," ":PRINT@84,"{17 SPACES}"
760 GOTO710
770 FOR II=1TO10:PRINT@84,"{7 SPACES}":PRINT
    @84,"CORRECT":NEXTII
775 FOR II=1TO200:E$=INKEY$:IFE$<>""THEN780
776 NEXTII
780 NEXTK
790 CLS
795 PRINT@212,"GOOD WORK!"
800 PRINT@516,"TRY AGAIN? (Y/N)"
810 E$=INKEY$:IFE$="Y" THEN600
820 IFE$<>"N"THEN810
830 CLS
840 END

```

Graphics

Program 4-18. Counting Shapes

TI-99/4A

```
110 GOTO 470
120 CALL VCHAR(X,Y,96,3)
130 CALL HCHAR(X+1,Y-1,96,3)
140 RETURN
150 FOR I=X TO X+2
160 CALL HCHAR(I,Y-1,104,3)
170 NEXT I
180 RETURN
190 CALL HCHAR(X,Y-1,113)
200 CALL VCHAR(X,Y,112,3)
210 CALL HCHAR(X,Y+1,114)
220 CALL HCHAR(X+1,Y-1,112,3)
230 CALL HCHAR(X+2,Y-1,116)
240 CALL HCHAR(X+2,Y+1,115)
250 RETURN
260 CALL HCHAR(X,Y-1,121)
270 CALL HCHAR(X+1,Y-1,120)
280 CALL HCHAR(X+1,Y,121)
290 CALL HCHAR(X+2,Y-1,120,2)
300 CALL HCHAR(X+2,Y+1,121)
310 RETURN
320 CALL VCHAR(X,Y,128,3)
330 CALL HCHAR(X,Y-1,129)
340 CALL HCHAR(X,Y+1,130)
350 CALL HCHAR(X+1,Y+1,131)
360 CALL HCHAR(X+2,Y+1,132)
370 CALL HCHAR(X+2,Y-1,133)
380 CALL HCHAR(X+1,Y-1,134)
390 RETURN
400 CALL HCHAR(X,Y,137)
410 CALL HCHAR(X+1,Y-1,138)
420 CALL VCHAR(X+1,Y,136,2)
430 CALL HCHAR(X+1,Y+1,139)
440 CALL HCHAR(X+2,Y-1,140)
450 CALL HCHAR(X+2,Y+1,141)
460 RETURN
470 CALL CLEAR
480 PRINT TAB(7);"COUNTING SHAPES"::::::::::::::
490 FOR C=96 TO 136 STEP 8
500 CALL CHAR(C,"FFFFFFFFFFFFFFFF")
510 NEXT C
520 FOR I=1 TO 16
530 READ C,C$
540 CALL CHAR(C,C$)
550 NEXT I
```

```

560 DATA 113,00030F1F3F3F7F7F,114,00C0F0F8FCFCFEFE
    ,115,FEFEFCFCF8F0C,116,7F7F3F3F1F0F03,121,80C0
    E0F0F8FCFEFF
570 DATA 129,0101030307070F0F,130,8080C0C0E0E0F0F,
    131,F8F8FCFCFCFCF8F8,132,F0F0E0E0C0C0808,133
580 DATA 0F0F070703030101,134,1F1F3F3F3F3F1F1F,137
    ,18183C3C7E7EFFFF,138,0101030307070F0F,139,808
    0C0C0E0E0F0F
590 DATA 140,1F1F3F3F7F7FFFFF,141,F8F8FCFCFEFEFFFF
600 FOR I=1 TO 9
610 READ X1(I),Y1(I)
620 NEXT I
630 DATA 8,5,9,11,7,17,10,23,14,7,13,16,14,28,19,1
    1,18,22
640 FOR K=1 TO 10
650 CALL CLEAR
660 C=INT(12*RND+5)
670 N=INT(9*RND+1)
680 S=INT(6*RND+1)
690 CALL COLOR(S+8,C,1)
700 FOR L=1 TO N
710 X=X1(L)
720 Y=Y1(L)
730 ON S GOSUB 120,150,190,260,320,400
740 NEXT L
750 PRINT "HOW MANY OBJECTS?"
760 CALL SOUND(100,1497,2)
770 CALL KEY(0,KE,ST)
780 IF ST<1 THEN 770
790 CALL HCHAR(23,22,KE)
800 IF KE=48+N THEN 850
810 CALL SOUND(100,330,2)
820 CALL SOUND(100,262,2)
830 CALL HCHAR(23,22,32)
840 GOTO 760
850 CALL SOUND(100,262,2)
860 CALL SOUND(100,330,2)
870 CALL SOUND(100,392,2)
880 CALL SOUND(200,523,2)
890 NEXT K
900 CALL CLEAR
910 PRINT "TRY AGAIN? (Y/N)"
920 CALL KEY(0,KE,ST)
930 IF KE=89 THEN 640
940 IF KE<>78 THEN 920
950 CALL CLEAR
960 END

```

Locating Points

Computers: VIC-20; TRS-80 Color Computer; MC-10; TI-99/4A

A computer can easily use a tutorial program because it can respond differently to each different answer. Color and graphics make learning even more fun. "Locating Points" teaches you how to locate points on a rectangular coordinate grid. This concept is often taught at the third-grade level, and reviewed in algebra classes.

A random example point is first displayed, with the coordinates labeled. You can view as many examples as you want. Next, a randomly chosen point is shown, and you enter your estimate of its coordinates. You can then move to the last part of the program, where you're given a point's coordinates. Using the proper movement keys (each program tells you what they are), you move the point until it is at the desired coordinates. The computer will tell you if you've placed the point correctly.

This is how the VIC version works:

Line	Function
2	PRINT title screen.
4	Define volume and sound.
6-7	Define string variables for grid. Delay loop.
8-26	Draw grid and show example point.
30-51	Present problem to give coordinates for given point.
52-54	PRINT instructions.
55-74	Present problem to locate point with given coordinates.
75-79	PRINT option to have another problem, start over, or end program. The program then shifts appropriately.
80-83	Label point and draw yellow lines from point to axis.
84	Calculate graphics memory location.
86	Play music for incorrect answer.
88	Subroutine for correct answer.
89	Delay loop for music.
90-92	PRINT grid.
94	Clear keyboard buffer.
96-99	PRINT "PRESS RETURN" and wait for response.
100	Clear screen and END.

Program 4-19. Locating Points

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```
2 PRINT "{CLR}{5 DOWN}{3 SPACES}{BLK}LOCATING .POINT  
S"
```

```

4 POKE36878,15:S=36876
6 A$="{GRN}{2 SPACES}-{2 SPACES}-{2 SPACES}-
  {2 SPACES}-{2 SPACES}-{2 SPACES}-":B$="{GRN}+
  **+***+***+**+***+****"
7 FORI=1TO3000:NEXT
8 GOSUB90
10 PRINT"A POINT IS LOCATED BY AN X-COORDINATE AND
  A Y-COORDINATE (X,Y). "
12 X=INT(4*RND(0))+1:Y=INT(3*RND(0))+1
14 GOSUB84:POKEC,81:POKEC1,2
18 GOSUB80
22 PRINT:PRINT"{GRN}ANOTHER EXAMPLE? (Y/N){BLU}";:
  GOSUB94
24 GETR$:IFR$="Y"THEN8
26 IFR$<>"N"THEN24
30 GOSUB90:PRINT"WHERE IS THE POINT?":PRINT:PRINT"
  {BLK}({RED}?{BLK},{RED}?{BLK})";
31 X=INT(6*RND(0)):Y=INT(5*RND(0)):GOSUB84:POKEC,8
  1:POKEC1,2
32 GOSUB94
33 GETR$:IFR$=""THEN33
34 IFASC(R$)<48ORASC(R$)>57THEN33
35 POKE8099,ASC(R$):POKE38819,2:X$=R$
36 GETR$:IFR$=""THEN36
37 IFASC(R$)<48ORASC(R$)>57THEN36
38 POKE8101,ASC(R$):POKE38821,2:Y$=R$
39 IFVAL(X$)<>XTHEN42
40 IFVAL(Y$)=YTHEN48
*42 GOSUB86:PRINT"{2 SPACES}{BLK}(";RIGHT$(STR$(X),
  1);",";RIGHT$(STR$(Y),1);)":GOSUB80:GOSUB96:GO
  TO30
48 GOSUB88:PRINT"{GRN}PRESS":PRINT"1 SAME TYPE PRO
  BLEM{3 SPACES}2 CONTINUE PROGRAM";:GOSUB94
50 GETR$:IFR$="1"THEN30
51 IFR$<>"2"THEN50
52 PRINT"{CLR}{BLU}YOU WILL BE GIVEN THE COORDINAT
  ES (X,Y).":PRINT:PRINT"PRESS F1 TO MOVE UP"
*53 PRINT:PRINTTAB(6);"F3 TO MOVE LEFT":PRINT:PRINT
  TAB(6);"F5 TO MOVE RIGHT":PRINTTAB(6);"F7 TO MO
  VE DOWN"
*54 PRINT:PRINT"MOVE THE POINT TO THE CORRECT LOCAT
  ION,{5 SPACES}THEN PRESS <RETURN>.{3 DOWN}":GOS
  UB96
55 GOSUB90:X=INT(6*RND(0)):Y=INT(5*RND(0)):A=0:B=0
56 PRINT"LOCATE (";RIGHT$(STR$(X),1);",";RIGHT$(ST
  R$(Y),1);)":GOSUB94
57 POKES,231:GOSUB89:P=7990-B*66+3*A:POKEP,81:POKE
  P+30720,3
58 GETR$:IFR$=""THEN58

```


Graphics

```
59 IFASC(R$)=13THEN73
61 IFASC(R$)<>133THEN64
62 B=B+1:IFB>4THENB=4
63 GOTO72
64 IFASC(R$)<>134THEN67
65 A=A-1:IFA<0THENA=0
66 GOTO72
67 IFASC(R$)<>135THEN70
68 A=A+1:IFA>5THENA=5
69 GOTO72
70 IFASC(R$)<>136THEN58
71 B=B-1:IFB<0THENB=0
72 POKEP,91:POKEP+30720,5:GOTO57
73 GOSUB84:IFP=CTHEN75
74 GOSUB86:POKEC,81:POKEC1,2:GOSUB96:GOTO52
*75 GOSUB88:PRINT"{GRN}PRESS":PRINT" 1 SAME TYPE PR
    OBLEM{3 SPACES}2 START PROGRAM OVER{2 SPACES}3
    {SPACE}END PROGRAM";:GOSUB94
76 GETR$:IFR$="1"THEN52
77 IFR$="2"THEN8
78 IFR$="3"THEN100
79 GOTO76
80 FORI=C1+1TOC1+5:POKEI,2:NEXT
81 POKEC+1,40:POKEC+2,X+48:POKEC+3,44:POKEC+4,Y+48
    :POKEC+5,41
82 FORI=C1+22TO38710+X*3STEP22:POKEI,7:NEXT
83 FORI=C1-1TOC1-X*3STEP-1:POKEI,7:NEXT:RETURN
84 C=7990-Y*66+3*X:C1=C+30720:RETURN
86 POKES,159:GOSUB89:POKES,135:GOSUB89:RETURN
*88 PRINT"{RED} CORRECT!":POKES,195:GOSUB89:POKES,2
    07:GOSUB89:POKES,215:GOSUB89:POKES,225:GOSUB89
89 FORI=1TO150:NEXT:POKES,0:RETURN
*90 PRINT"{CLR}{BLK}{2 SPACES}Y":PRINTA$:PRINT"
    {BLK}4";B$:PRINTA$:PRINTA$:PRINT" {BLK}3";B$:PR
    INTA$:PRINTA$:PRINT" {BLK}2";B$:PRINTA$:PRINTA$
    :PRINT" {BLK}1";B$:PRINTA$:PRINTA$
92 PRINT" {BLK}0";B$;"{BLK}X{2 SPACES}0{2 SPACES}1
    {2 SPACES}2{2 SPACES}3{2 SPACES}4{2 SPACES}5
    {BLU}":PRINT:RETURN
94 FORI=1TO10:GETR$:NEXT:RETURN
96 PRINT:PRINT:PRINT"{GRN}PRESS RETURN";:GOSUB94
97 GETR$:IFR$=""THEN97
98 IFASC(R$)<>13THEN97
99 RETURN
100 PRINT"{CLR}{BLU}":END
```

Here's how the program works on the Color Computer:

Line	Function
1	Branch past subroutines.
2-3	PRINT grid.
4-7	PRINT "PRESS ENTER" and wait for response.
8-9	Choose point and calculate graphics print position.
10	Calculate coordinates to SET point.
11	Play music for incorrect response.
12	Play music for correct response.
13-14	PRINT title screen and define string variables for grid graphics.
15-18	Draw grid with example point.
19-30	Present problem to find coordinates for given point.
31-32	PRINT instructions.
33-52	Present problem to locate point with given coordinates.
53-57	PRINT option to have another problem, start over, or end program; program branches appropriately. Clear screen and END.

Program 4-20. Locating Points

TRS-80 Color Computer; MC-10

```

1 GOTO13
2 CLS:PRINT " Y";A$:FORX=3TO1STEP-1:PRINT " ";RIGHT$(STR$(X),1);B$:PRINT " ";A$:PRINT " ";A$:NEXT
3 PRINT@321,"0";B$:PRINT@354,"0{4 SPACES}1
  {4 SPACES}2{4 SPACES}3{4 SPACES}4{4 SPACES}5 X"
  :PRINT:RETURN
4 PRINT@496,"PRESS <ENTER>";
5 R$=INKEY$:IFR$=""THEN5
6 IFASC(R$)<>13THEN5
7 RETURN
8 X=RND(5)
9 Y=RND(3):A=322-96*Y+X*5:PRINT@A,CHR$(159);CHR$(1
  59):RETURN
10 B=4+X*10:C=20-6*Y:FORI=C+2 TO20:SET(B,I,4):SET(
  B+2,I,4):NEXT:FORI=B-2 TO4STEP-1:SET(I,C,4):NEX
  T:RETURN
11 SOUND125,2:SOUND89,2:RETURN
12 SOUND89,2:SOUND125,2:SOUND147,2:SOUND176,4:RETU
  RN
13 CLS:PRINT@197,"** LOCATING POINTS **":C$=CHR$(1
  75):D$=C$+C$+C$+C$+C$:E$="{3 SPACES}"+C$+C$:A$=
  E$:B$=D$
14 FORI=1TO4:A$=A$+E$:B$=B$+D$:NEXT:A$=C$+C$+A$:B$
  =C$+C$+B$+C$+C$:FORD=1TO2000:NEXT
  
```

Graphics

```
15 GOSUB2:PRINT"A POINT HAS AN X-COORDINATE
   {5 SPACES}AND A Y-COORDINATE (X,Y).":X=RND(4):G
   OSUB9:X$=RIGHT$(STR$(X),1)
16 Y$=RIGHT$(STR$(Y),1):PRINT@A+2,(" +X$+", "+Y$+"
   ");:GOSUB10:PRINT@480,"ANOTHER EXAMPLE? (Y/N)";
17 R$=INKEY$:IFR$="Y"THEN15
18 IFR$<>"N"THEN17
19 GOSUB2:GOSUB8:PRINT@416,"WHAT ARE THE COORDINAT
   ES?":PRINT"{5 SPACES}{?,?}"
20 U$=INKEY$:IFU$=""THEN20
21 PRINT@454,U$;
22 V$=INKEY$:IFV$=""THEN22
23 PRINT@456,V$;
24 IFX<>VAL(U$) THEN26
25 IFY=VAL(V$) THEN27
26 GOSUB11:GOSUB10:PRINT@460,"LOCATION IS (;RIGHT
   $(STR$(X),1);",",";RIGHT$(STR$(Y),1);")";:GOSUB4:
   GOT019
27 GOSUB12:PRINT@460,"CORRECT!":GOSUB4:PRINT@496,L
   "{13 SPACES}";
28 PRINT@416,"PRESS 1 FOR SAME TYPE PROBLEM
   {9 SPACES}2 TO CONTINUE PROGRAM"
29 R$=INKEY$:IFR$="1"THEN19
30 IFR$<>"2"THEN29
31 CLS:PRINT@66,"YOU WILL BE GIVEN THE":PRINT" CO
   ORDINATES.":PRINT@162,"USE THE ARROW KEYS TO MO
   VE"
32 PRINT" THE POINT TO THE CORRECT":PRINT" LOCAT
   ION, THEN PRESS <ENTER>.:GOSUB4
33 CLS:GOSUB2:X=RND(5):Y=RND(3):PRINT@416,"LOCATE
   (;RIGHT$(STR$(X),1);",",";RIGHT$(STR$(Y),1);")":
   D=4:E=20:B=4:C=20
34 SET(B,C,2):SET(B+2,C,2):SOUND227,2
35 R$=INKEY$:IFR$=""THEN35
36 IFASC(R$)=13THEN49
37 IFASC(R$)<>9THEN40
38 D=B+10:IFD>54THEND=54
39 GOTO48
40 IFASC(R$)<>8THEN43
41 D=B-10:IFD<4THEND=4
42 GOTO48
43 IFASC(R$)<>94THEN46
44 E=C-6:IFE<2THENE=2
45 GOTO48
46 IFASC(R$)<>10THEN35
47 E=C+6:IFE>20THENE=20
48 SET(B,C,3):SET(B+2,C,3):B=D:C=E:GOTO34
49 D=4+X*10:E=20-6*Y:IFB<>D THEN51
50 IFC=E THEN52
```

```

51 GOSUB11:SET(D,E,4):SET(D+2,E,4):GOSUB10:GOSUB4:
   GOTO33
52 GOSUB12:PRINT@448,"CORRECT!":GOSUB4
53 PRINT@496,"{13 SPACES}";:PRINT@416,"PRESS 1 FOR
   SAME TYPE PROBLEM":PRINT"{6 SPACES}2 TO START P
   ROGRAM OVER":PRINT@486,"3 TO END PROGRAM";
54 R$=INKEY$:IFR$="1"THEN33
55 IFR$="2"THEN15
56 IFR$<>"3"THEN54
57 CLS:END

```

The TI-99/4A version is slightly different, although the display is much the same as the other computers' programs.

Line	Function
100	Define random function.
110-400	PRINT title screen and define graphics characters.
410-430	Play music for incorrect answer.
440-480	Play music for correct answer.
490-550	PRINT grid.
560-600	PRINT "PRESS ENTER" and wait for response.
610-770	Present random example of a point with coordinates.
790-1150	Present problem to give coordinates for given point.
1160-1200	PRINT instructions.
1210-1710	Present problem to locate point with given coordinates.
1720-1770	Give option to repeat problem or end, and then branch appropriately.
1780-1830	Choose random point.
1840-1870	Draw vertical red line from point to X-axis.
1880-1910	Draw horizontal red line from point to Y-axis.
1920	END.

Program 4-21. Locating Points

TI-99/4A

```

100 DEF R(N)=INT(N*RND+1)
110 CALL CLEAR
120 PRINT "{4 SPACES}*****":
   {4 SPACES}*";TAB(23);"*": "{4 SPACES}* LOCATING
   POINTS *"
130 PRINT "{4 SPACES}*";TAB(23);"*": "{4 SPACES}***
   *****"::~::~:
140 A$="1818181818181818"
150 B$="181818FFFF181818"
160 C$="000000FFFF"

```

Graphics


```
170 FOR C=96 TO 112 STEP 8
180 CALL CHAR(C,A$)
190 CALL CHAR(C+1,B$)
200 CALL CHAR(C+2,C$)
210 NEXT C
220 CALL CHAR(120,"183C7EFFFF7E3C18")
230 CALL CHAR(128,"183C7EFFFF7E3C18")
240 CALL CHAR(129,"000000000030C30C")
250 CALL CHAR(130,"030C30C")
260 CALL CHAR(94,"00102828444482FE")
270 CALL COLOR(10,5,1)
280 CALL COLOR(11,10,1)
290 CALL COLOR(12,11,1)
300 CALL COLOR(13,7,1)
310 CALL CHAR(140,"1010101010101")
320 CALL CHAR(141,"000000FF")
330 CALL CHAR(142,"101010F")
340 CALL COLOR(14,13,1)
350 A$="` h h h h h h h"
360 B$="ajjijjijjijjijjijjijjijji"
370 C$="abbabbabbabbabbabbabb"
380 CALL CLEAR
390 CALL COLOR(2,2,1)
400 GOTO 610
410 CALL SOUND(100,330,2)
420 CALL SOUND(100,262,2)
430 RETURN
440 CALL SOUND(100,262,2)
450 CALL SOUND(100,330,2)
460 CALL SOUND(100,392,2)
470 CALL SOUND(200,523,2)
480 RETURN
490 CALL CLEAR
500 PRINT "{4 SPACES}Y": "{4 SPACES}";A$: "
    {4 SPACES}";A$: "{3 SPACES}4";B$: "{4 SPACES}";A
    $: "{4 SPACES}";A$: "{3 SPACES}3";B$
510 PRINT "{4 SPACES}";A$: "{4 SPACES}";A$: "
    {3 SPACES}2";B$
520 PRINT "{4 SPACES}";A$: "{4 SPACES}";A$: "
    {3 SPACES}1";B$: "{4 SPACES}";A$: "{4 SPACES}";A
    $: "{3 SPACES}0";C$
530 PRINT "{4 SPACES}0 1 2 3 4 5 6 7": "::
540 CALL HCHAR(20,31,88)
550 RETURN
560 PRINT TAB(16);"PRESS <ENTER>";
570 CALL KEY(0,K,S)
580 IF K<>13 THEN 570
590 CALL HCHAR(24,18,32,13)
600 RETURN
```

```
610 GOSUB 490
620 PRINT "THE LOCATION OF A POINT IS": "GIVEN BY I
    TS X-COORDINATE"
630 PRINT "AND Y-COORDINATE (X,Y)"
640 RANDOMIZE
650 X=R(5)
660 GOSUB 1790
670 GOSUB 1840
680 CALL HCHAR(Y1,X1+2,40)
690 CALL HCHAR(Y1,X1+3,48+X)
700 CALL HCHAR(Y1,X1+4,44)
710 GOSUB 1880
720 CALL HCHAR(Y1,X1+5,48+Y)
730 CALL HCHAR(Y1,X1+6,41)
740 PRINT : "WANT ANOTHER EXAMPLE? (Y/N)";
750 CALL KEY(0,K,S)
760 IF K=89 THEN 610
770 IF K<>78 THEN 750
780 CALL CLEAR
790 PRINT "YOU WILL BE SHOWN A POINT.": "PRESS THE
    NUMBER OF THE"
800 PRINT : "X-COORDINATE THEN THE"
810 PRINT : "NUMBER OF THE Y-COORDINATE.": ":::::"
820 GOSUB 560
830 CALL CLEAR
840 GOSUB 490
850 PRINT :::
860 RANDOMIZE
870 GOSUB 1780
880 CALL HCHAR(21,7,40)
890 CALL HCHAR(21,9,44)
900 CALL HCHAR(21,11,41)
910 CALL KEY(0,K,S)
920 CALL HCHAR(21,8,63)
930 CALL HCHAR(21,8,32)
940 IF S<1 THEN 910
950 CALL HCHAR(21,8,K)
960 X2=K
970 CALL KEY(0,K,S)
980 CALL HCHAR(21,10,63)
990 CALL HCHAR(21,10,32)
1000 IF S<1 THEN 970
1010 CALL HCHAR(21,10,K)
1020 Y2=K
1030 IF X2<>X+48 THEN 1100
1040 IF Y2<>Y+48 THEN 1100
1050 GOSUB 440
1060 PRINT "PRESS": "1 FOR SAME TYPE PROBLEM": "2 TO
    CONTINUE PROGRAM";
```


Graphics

```
1070 CALL KEY(0,K,S)
1080 IF K=49 THEN 830
1090 IF K=50 THEN 1160 ELSE 1070
1100 GOSUB 410
1110 GOSUB 1840
1120 GOSUB 1880
1130 PRINT "THE CORRECT ANSWER IS (";STR$(X);",";S
TR$(Y);")"
1140 GOSUB 560
1150 GOTO 830
1160 CALL CLEAR
1170 PRINT "NOW YOU WILL BE GIVEN THE": "COORDINAT
ES."
1180 PRINT : "USE THE ARROW KEYS TO MOVE": "THE POI
NT TO THE CORRECT"
1190 PRINT : "PLACE, THEN PRESS <ENTER>." : : : :
1200 GOSUB 560
1210 CALL CLEAR
1220 GOSUB 490
1230 RANDOMIZE
1240 X=R(7)
1250 Y=R(4)
1260 X1=7+3*X
1270 Y1=17-3*Y
1280 PRINT : "PLOT (";STR$(X);",";STR$(Y);")": :
1290 C1=97
1300 A=17
1310 A1=A
1320 B=7
1330 B1=B
1340 CALL HCHAR(A,B,120)
1350 CALL KEY(0,K,S)
1360 IF S<1 THEN 1350
1370 IF K=13 THEN 1620
1380 IF K<>69 THEN 1430
1390 IF A=5 THEN 1350
1400 CALL GCHAR(A-3,B,C)
1410 A=A-3
1420 GOTO 1570
1430 IF K<>88 THEN 1480
1440 IF A=17 THEN 1350
1450 CALL GCHAR(A+3,B,C)
1460 A=A+3
1470 GOTO 1570
1480 IF K<>83 THEN 1530
1490 IF B=7 THEN 1350
1500 CALL GCHAR(A,B-3,C)
1510 B=B-3
1520 GOTO 1570
```

```
1530 IF K<>68 THEN 1350
1540 IF B=28 THEN 1350
1550 CALL GCHAR(A,B+3,C)
1560 B=B+3
1570 CALL HCHAR(A1,B1,C1)
1580 A1=A
1590 B1=B
1600 C1=C
1610 GOTO 1340
1620 CALL SOUND(150,1397,2)
1630 CALL GCHAR(Y1,X1,C)
1640 IF C=120 THEN 1710
1650 GOSUB 410
1660 CALL HCHAR(Y1,X1,128)
1670 GOSUB 1840
1680 GOSUB 1880
1690 GOSUB 560
1700 GOTO 1210
1710 GOSUB 440
1720 PRINT "PRESS":"1 FOR SAME TYPE PROBLEM":"2 TO
      END PROGRAM";
1730 CALL KEY(0,K,S)
1740 IF K=49 THEN 1210
1750 IF K<>50 THEN 1730
1760 CALL CLEAR
1770 STOP
1780 X=R(7)
1790 Y=R(4)
1800 X1=7+3*X
1810 Y1=17-3*Y
1820 CALL HCHAR(Y1,X1,128)
1830 RETURN
1840 FOR I=Y1+1 TO 17
1850 CALL HCHAR(I,X1,112)
1860 NEXT I
1870 RETURN
1880 FOR I=X1-1 TO 7 STEP -1
1890 CALL HCHAR(Y1,I,114)
1900 NEXT I
1910 RETURN
1920 END
```

Chapter 5
Sound



Chapter 5

Sound

Few, if any, computers produce sound effects the same way. The VIC-20, for example, uses different commands to create sound than does the Commodore 64, even though both machines are Commodore computers. Translating BASIC to duplicate sound effects is difficult because of this. You have to know how both computers produce the sound in order to translate a program from one computer's version of BASIC to another.

The programs in this chapter, "Musical Keyboard," "Treble Clef," and "Bass Clef," are all examples of computer-generated sound. Since it is so hard to translate sound commands, all three programs have been listed for each of three computers: the VIC-20, the TRS-80 Color Computer, and the TI-99/4A. By comparing the programs, you can see how the creation of sound was translated from one machine's version of BASIC to another.

Musical Keyboard

Computers: VIC-20; TRS-80 Color Computer; MC-10; TI-99/4A

This program won't teach you how to play the piano, but it will help you remember the keys. The musical keyboard displayed on the screen consists of white and black keys. The program shows how the white keys are named — with the first seven letters of the alphabet. After the letter names are shown, there is a short quiz. A key is randomly chosen, and you must press the key's letter name. If the letter pressed is correct, the correct note is played. You must answer correctly to continue. The quiz consists of ten keys.

"Musical Keyboard" on the VIC works like this:

Line	Function
2	Branch past subroutines.
3-16	Define letter name of note AA\$, tone S to be played, and coordinate CC of key.
20-25	Draw the keyboards.
26-29	Wait for RETURN to be pressed.
30-31	PRINT title screen, define SS to play music, and turn on volume.
32-33	Play scale for title screen.
34-42	PRINT instructions.
50-60	Present quiz for ten notes.
51	Clear screen and randomly choose and draw one of two keyboards.
52-55	Randomly choose key and blink an asterisk on the key.

Sound

- 56-57 PRINT message, then wait for key to be pressed.
58-59 If name is incorrect, sound a noise and return for another response.
60 If name is correct, PRINT the letter name on the key, play the tone, and then go to next key.
61-63 PRINT option to try again and branch appropriately.
64 Clear screen and END.

Program 5-1. Musical Keyboard

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```
2 GOTO30
3 AA$="C":S=131:CC=0:RETURN
4 AA$="D":S=145:CC=3:RETURN
5 AA$="E":S=158:CC=6:RETURN
6 AA$="F":S=161:IFK=1THENCC=8:RETURN
7 CC=0:RETURN
8 AA$="G":S=173:IFK=1THENCC=11:RETURN
9 CC=3:RETURN
10 AA$="A":S=181:IFK=1THENCC=14:RETURN
11 CC=6:RETURN
12 AA$="B":S=189:IFK=1THENCC=17:RETURN
13 CC=9:RETURN
14 AA$="C":S=192:CC=11:RETURN
15 AA$="D":S=200:CC=14:RETURN
16 AA$="E":S=206:CC=17:RETURN
20 FORKI=1TO10:PRINT" {BLK}{RVS}{2 SPACES}{OFF}
  {RVS}{2 SPACES}{OFF} B {RVS}{2 SPACES}{OFF}
  {RVS}{2 SPACES}{OFF} {RVS}{2 SPACES}{OFF} B":NE
XT
21 FORKI=1TO3:PRINT"{2 SPACES}{G}{2 SPACES}{G}
  B{2 SPACES}{G} B{2 SPACES}{G}
  B":NEXT:RETURN
22 FORKI=1TO10:PRINT" {BLK}{RVS}{2 SPACES}{OFF}
  {RVS}{2 SPACES}{OFF} {RVS}{2 SPACES}{OFF} B
  {RVS}{2 SPACES}{OFF} {RVS}{2 SPACES}{OFF} B":NE
XT
23 FORKI=1TO3:PRINT"{2 SPACES}{G}{2 SPACES}{G}
  {2 SPACES}{G} B{2 SPACES}{G}
  B":NEXT:RETURN
24 FORI=1TO7:PRINT"{2 SPACES}{RVS} {OFF} {RVS}
  {OFF} {RVS} {OFF} B {RVS} {OFF} {RVS} {OFF} B
  {RVS} {OFF} {RVS} {OFF} {RVS} ":NEXT:RETURN
25 FORI=1TO3:PRINT"{2 SPACES}B B B B B B B B B B":
  NEXT:PRINT"[22 Y] ":RETURN
26 PRINT"{2 DOWN}{GRN}PRESS RETURN";
```

```

27 GETA$: IFA$="" THEN 27
28 IF ASC(A$) <> 13 THEN 27
29 PRINT "{CLR}{BLU}": RETURN
30 PRINT "{CLR}{BLU}{3 DOWN}{2 SPACES}LET'S LEARN N
OTES": PRINT "{2 DOWN}{4 SPACES}THE KEYBOARD
{7 DOWN}"
31 SS=36876: POKE 36878, 15
32 DATA 131, 145, 158, 161, 173, 181, 189, 192
33 FOR I=1 TO 8: READS: POKE SS, S: FORD=1 TO 200: NEXT D, I: PO
KE SS, 0
34 PRINT "{CLR}{DOWN}{BLU}A PIANO OR ORGAN": PRINT "K
EYBOARD HAS GROUPS{3 SPACES}OF TWO BLACK KEYS"
35 PRINT "AND THREE BLACK KEYS. {2 DOWN}{BLK}": GOSU
B 24: GOSUB 25: GOSUB 26
*36 PRINT "{DOWN}LOOK AT A SET OF": PRINT "TWO {BLK}BL
ACK{BLU} KEYS:": PRINT "{DOWN}THE NAMES OF THE KE
YS ARE C, D, AND E."
37 PRINT "{BLK}{DOWN}": FOR I=1 TO 7: PRINT "{3 SPACES}
[H]{3 SPACES}{RVS}{2 SPACES}{OFF}{3 SPACES}
{RVS}{2 SPACES}{OFF}{4 SPACES}{H}": NEXT
38 PRINT "{3 SPACES}{H}{4 SPACES}{H}{4 SPACES}
{H}{4 SPACES}{H}{6 SPACES}{H} {RED}C{BLK}
{2 SPACES}{H} {RED}D{BLK}{2 SPACES}{H}
{RED}E{BLK}{2 SPACES}{H}": PRINT "{3 SPACES}
{H}{4 SPACES}{H}{4 SPACES}{H}{4 SPACES}
{H}"
39 PRINT "[22 Y]": GOSUB 26
*40 PRINT "THE LETTER NAMES OF{3 SPACES}THE KEYS ARE
THE{6 SPACES}ALPHABET LETTERS{6 SPACES}UP TO G
.{3 DOWN}{BLK}"
41 GOSUB 24: PRINT "{2 SPACES}B{SHIFT-SPACE}B
{SHIFT-SPACE}B{SHIFT-SPACE}B{SHIFT-SPACE}B
{SHIFT-SPACE}B{SHIFT-SPACE}B{SHIFT-SPACE}B
{SHIFT-SPACE}B{SHIFT-SPACE}B{2 SPACES}{RED}F
{BLK}B{RED}G{BLK}B{RED}A{BLK}B{RED}B{BLK}B{PUR}
C{BLK}B{PUR}D{BLK}B{PUR}E{BLK}B{RED}F{BLK}B
{RED}G{BLK}B{RED}A{BLK}B{RED}B{BLK}";
42 PRINT "{2 SPACES}B{SHIFT-SPACE}B{SHIFT-SPACE}B
{SHIFT-SPACE}B{SHIFT-SPACE}B{SHIFT-SPACE}B
{SHIFT-SPACE}B{SHIFT-SPACE}B{SHIFT-SPACE}B
{SHIFT-SPACE}B{SHIFT-SPACE}[22 Y]": GOSUB 26
50 FOR I=1 TO 10
51 PRINT "{CLR}{BLK}": K=INT(RND(1)*2)+1: ON K GOSUB 20,
22: PRINT "[22 T]"
52 N=INT(RND(1)*7)+1: ON K GOTO 53, 54
53 ON N GOSUB 3, 4, 5, 6, 8, 10, 12: GOTO 55
54 ON N GOSUB 6, 8, 10, 12, 14, 15, 16
55 POKE 7944+CC, 42: CS=38664+CC: FORD=1 TO 20: POKE CS, 7:
POKE CS, 2: NEXT

```

Sound

```
56 PRINT "{BLU}{3 DOWN}NAME THE NOTE"  
57 GETA$: IFA$="" THEN 57  
58 IFA$=AA$ THEN 60  
59 POKE 36877, 128: FORD=1 TO 400: NEXT: POKE 36877, 0: GOTO  
57  
60 POKE 7944+CC, ASC(A$)-64: POKESS, S: FORD=1 TO 400: NEX  
TD: POKESS, 0: NEXT I  
61 PRINT "{3 DOWN}{GRN}TRY AGAIN (Y/N)"  
62 GETA$: IFA$="Y" THEN 50  
63 IFA$ <> "N" THEN 62  
64 PRINT "{CLR}{BLU}": END
```

The Color Computer version of the program works somewhat differently:

Line	Function
0	Clear string space and branch past subroutines.
1-4	Wait for user to press ENTER.
5-15	Define sound S and letter name H\$.
16	PRINT title.
17	Define string variables to PRINT keyboard.
18-33	Present instructions to learn the letter names of the keys.
34	Title for quiz.
35-36	Draw keyboard.
37-45	Present quiz. Choose a key randomly and blink a character under the key. If the answer is correct, the correct note plays. The key must be named correctly to continue.
46-48	PRINT option to try again and branch appropriately.
49	Clear screen and END.

Program 5-2. Musical Keyboard

TRS-80 Color Computer; MC-10

```
0 CLEAR 300: GOTO 16  
1 PRINT @496, "PRESS <ENTER>";  
2 G$=INKEY$: IF G$="" THEN 2  
3 IF ASC(G$) <> 13 THEN 2  
4 CLS: RETURN  
5 S=5: H$="F": RETURN  
6 S=32: H$="G": RETURN  
7 S=58: H$="A": RETURN  
8 S=78: H$="B": RETURN  
9 S=89: H$="C": RETURN  
10 S=108: H$="D": RETURN  
11 S=125: H$="E": RETURN  
12 S=133: H$="F": RETURN
```

```

13 S=147:H$="G":RETURN
14 S=159:H$="A":RETURN
15 S=170:H$="B":RETURN
16 CLS:PRINT " ** THE MUSICAL KEYBOARD **"
17 A$=CHR$(159)+CHR$(154)+CHR$(144):B$=CHR$(159)+C
HR$(159)+CHR$(149):L$=CHR$(159)+A$+A$+A$+B$+A$+
A$+B$+A$+A$+A$+CHR$(159)
18 PRINT@129,"EACH KEY ON THE KEYBOARD":PRINT" HAS
  A SPECIFIC LETTER NAME.":PRINT@225,"THE LETTER
  S WE USE ARE":PRINT" A B C D E F G"19 PRINT " T
  HEN THE LETTERS START OVER--":PRINT " ABCDEFGAB
  CDEFG ...":GOSUB1
20 FORI=0TO192STEP32:PRINT@I,L$:NEXT:C$=CHR$(159)+
CHR$(159)+CHR$(149):M$=""
21 FORI=1TO10:M$=M$+C$:NEXT:M$=CHR$(159)+M$+CHR$(1
59)
22 PRINT@224,M$:PRINT@256,M$:PRINT@289,"F G A B
  C D E F G A B":PRINT@353,"THE KEYBOARD
  HAS GROUPS OF":PRINT" 3 BLACK KEYS ALTERNATING"
23 PRINT" WITH GROUPS OF 2 BLACK KEYS.":GOSUB 1:PR
INT" ONE OF THE EASIEST KEYS":PRINT" TO FIND IS
  'C'."
24 PRINT@97,"FIND A GROUP OF TWO BLACK KEYS."
25 FORI=170TO266STEP32:PRINT@I,A$+A$+CHR$(159)+CHR
$(159):NEXT
26 FORI=298TO330STEP32:PRINT@I,B$+B$+CHR$(159)+CHR
$(159):NEXT
27 PRINT@385,"JUST TO THE LEFT OF THE":PRINT@417,"
  TWO BLACK KEYS IS 'C'.".":PRINT@330,"C";:GOSUB1
28 PRINT @33,"'D' IS BETWEEN THE":PRINT " TWO BLAC
K KEYS.":PRINT " 'E' IS ON THE RIGHT."
29 FORI=202TO298STEP32:PRINT@I,A$+A$+CHR$(159)+CHR
$(159):NEXT
30 FORI=330TO362STEP32:PRINT@I,B$+B$+CHR$(159)+CHR
$(159):NEXT
31 PRINT@394,"C D E":GOSUB1:PRINT@33,"THE KEYS B
Y THE GROUP OF":PRINT" 3 BLACK KEYS ARE:"
32 FORI=202TO298STEP32:PRINT@I,A$+A$+A$+CHR$(159)+
CHR$(159):NEXT
33 FORI=330TO362STEP32:PRINT@I,B$+B$+B$+CHR$(159)+
CHR$(159):NEXT:PRINT@394,"F G A B":GOSUB1
34 PRINT @228,"NOW FOR A QUIZ . . .":GOSUB1
35 CLS:FORI=0TO192STEP32:PRINT@I,L$:NEXT
36 PRINT@224,M$:PRINT@256,M$:PRINT@394,"NAME THE K
EY"
37 FOR I=1 TO 10
38 N=RND(11):IF N=N1 THEN 38
39 P=286+3*N:N1=N:PRINT @P,CHR$(198)
40 G$=INKEY$:IF G$="" THEN 40

```


Sound

```
41 ON N GOSUB 5,6,7,8,9,10,11,12,13,14,15
42 IF G$=H$ THEN 44
43 SOUND 125,1:SOUND 89,1:GOTO 40
44 PRINT@P,G$:SOUNDS,15:PRINT@P," "
45 NEXT
46 PRINT@394," GOOD !!{4 SPACES}":PRINT@448,"TRY A
  GAIN? (Y/N)"
47 E$=INKEY$:IF E$="Y" THEN 35
48 IF E$<>"N" THEN 47
49 CLS:END
```

Here's how the TI version of Musical Keyboard works:

Line	Function
110-140	PRINT title screen.
150-230	Define characters, strings, and colors for graphics.
240-270	READ note name ASCII code and frequency for each key.
280-420	Present instructions.
430-700	Present quiz. Choose a key randomly and blink a question mark on the key. If the answer key pressed is correct, the correct note plays. R chooses one of two keyboards displayed.
710-740	PRINT option to try again and branch appropriately.
750-860	Draw musical keyboard.
870-900	Wait for ENTER key to be pressed.
910-920	Clear screen and END.

Program 5-3. Musical Keyboard

TI-99/4A

```
110 DIM N(13,2),L$(2),M$(2)
120 CALL CLEAR
130 CALL SCREEN(8)
140 PRINT " ** MUSICAL KEYBOARD **":.....:
150 CALL CHAR(97,"Ø")
160 CALL CHAR(98,"FFFFFFFFFFFFFF")
170 CALL CHAR(99,"Ø1Ø1Ø1Ø1Ø1Ø1Ø1")
180 CALL CHAR(100,"8Ø8Ø8Ø8Ø8Ø8Ø8Ø8")
190 L$(1)="dabbabbacdabbabbabbacdabbabb"
200 L$(2)="bbabbabbacdabbabbacdabbabbab"
210 M$(1)="dacdacdacdacdacdacdacdacdac"
220 M$(2)="cdacdacdacdacdacdacdacdacd"
230 CALL COLOR(9,2,16)
240 FOR C=1 TO 13
250 READ N(C,1),N(C,2)
260 NEXT C
```

```

270 DATA 67,262,68,294,69,330,70,349,71,392,65,440
    ,66,494,67,523,68,587,69,659,70,698,71,784,65,
    880
280 R=1
290 GOSUB 750
300 PRINT ::"THE LETTER NAMES OF THE KEYS"::"ARE T
    HE FIRST 7 LETTERS"::"OF THE ALPHABET."
310 DATA 67,68,69,70,71,65,66,67,68
320 FOR C=4 TO 28 STEP 3
330 READ G
340 CALL HCHAR(15,C,G)
350 NEXT C
360 PRINT : "THE KEYBOARD HAS GROUPS OF": "TWO AND
    THREE BLACK KEYS."
370 GOSUB 870
380 GOSUB 750
390 PRINT : "YOU MAY REMEMBER THAT JUST": "LEFT OF
    THE TWO BLACK KEYS": "IS THE KEY CALLED 'C'." :
    ::
400 CALL HCHAR(14,4,67)
410 CALL HCHAR(14,25,67)
420 GOSUB 870
430 SC=0
440 FOR T=1 TO 10
450 F=0
460 RANDOMIZE
470 R=INT(2*RND+1)
480 X=INT(9*RND+1)
490 J=3*X+R
500 GOSUB 750
510 PRINT ::::TAB(8);"NAME THE KEY"::::
520 IF R=1 THEN 540
530 X=X+4
540 CALL HCHAR(14,J,63)
550 CALL KEY(0,K,S)
560 CALL COLOR(4,16,16)
570 CALL COLOR(4,7,16)
580 IF S<1 THEN 550
590 IF K=N(X,1)THEN 630
600 CALL SOUND(400,-8,2)
610 F=1
620 GOTO 550
630 CALL HCHAR(14,J,K)
640 CALL SOUND(600,N(X,2),2)
650 CALL SOUND(1,N(X,2),30)
660 IF F=1 THEN 680
670 SC=SC+1
680 NEXT T
690 CALL COLOR(4,2,1)

```

Sound

```
700 PRINT "SCORE";SC;"OUT OF 10"  
710 PRINT ::"TRY AGAIN? (Y/N)"  
720 CALL KEY(0,K,S)  
730 IF K=89 THEN 430  
740 IF K=78 THEN 910 ELSE 720  
750 CALL CLEAR  
760 FOR C=1 TO 12  
770 PRINT L$(R)  
780 IF R=2 THEN 800  
790 CALL HCHAR(23,31,97)  
800 NEXT C  
810 FOR C=1 TO 3  
820 PRINT M$(R)  
830 IF R=2 THEN 850  
840 CALL HCHAR(23,31,97)  
850 NEXT C  
860 RETURN  
870 PRINT : "PRESS <ENTER>." ;  
880 CALL KEY(0,K,S)  
890 IF K<>13 THEN 880  
900 RETURN  
910 CALL CLEAR  
920 END
```

Treble Clef Notes

Computers: VIC-20; TRS-80 Color Computer; MC-10; TI-99/4A

"Treble Clef Notes" can help you learn to read the notes of the treble clef. You can more easily remember the notes by thinking of the word F-A-C-E for the letter names of the notes, starting with the bottom space. You can remember the phrase Every Good Boy Does Fine, which has been a standard phrase in teaching the treble clef line notes. The first letter of each word in the phrase represents the letter name of the note, starting with the bottom line of the treble staff.

After the names of the letters are introduced, there is a quiz of ten notes. A note is chosen randomly, and you press the key with the letter name of the note. If the answer is correct, the right note will play. You must be correct to continue the game. After ten problems, you have the option to try again or end the program.

Treble Clef works like this on the VIC:

Line	Function
2	Dimension N\$ for the name of the note and the tone, turn volume on, and define S for sound.
3	READ values for N\$ from data.

- 8-12 PRINT title screen and play scale.
- 14-24 PRINT menu screen of options and then branch accordingly.
- 30-35 PRINT instructions to learn names of the space notes.
- 37 Quiz for naming ten notes chosen randomly.
- 40-46 Instructions to learn names of the line notes.
- 48 Quiz for naming ten notes chosen randomly.
- 50-63 Draw treble staff with note A chosen randomly.
- 64-66 Clear the note.
- 70-86 Perform the quiz.
- 90 DATA for note names and tones.
- 95-98 Wait for RETURN to be pressed.
- 100 Perform quiz including all notes.
- 200 Clear screen and END.

Program 5-4. Treble Clef

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```

2 DIMN$(8,1):POKE36878,15:S=36876
3 FORI=0TO8:READN$(I,0),N$(I,1):NEXT
8 PRINT"{CLR}{BLU}{2 DOWN}{2 SPACES}LET'S LEARN NOTES{2 DOWN}":PRINTTAB(5);"TREBLE CLEF"
12 FORI=0TO8:POKES,VAL(N$(I,1)):FORJ=1TO200:NEXTJ,I:POKES,0
*14 PRINT"{CLR}{BLU}{2 DOWN}{5 SPACES}TREBLE CLEF{2 DOWN}":PRINT"CHOOSE:":PRINT"{DOWN} 1 NOTES ON N SPACES":PRINT"{DOWN} 2 NOTES ON LINES"
16 PRINT"{DOWN} 3 ALL TREBLE NOTES":PRINT"{DOWN} 4 END PROGRAM":GETA$
18 GETA$:IFA$=""THEN18
20 IFASC(A$)<49ORASC(A$)>52THEN18
22 ONVAL(A$)GOSUB30,40,100,200
24 GOTO14
30 PRINT"{CLR}{DOWN}THERE ARE FOUR SPACES ON THE STAFF."
31 PRINT"THE NAMES OF THE NOTES ON THE SPACES ARE:":PRINT"{2 DOWN}{BLK}{21 Y}"
*32 PRINTTAB(11);"{RED}E{BLK}{DOWN}":PRINT"{21 Y}":PRINTTAB(9);"{RED}C{BLK}{DOWN}":PRINT"{21 Y}"
*33 PRINTTAB(7);"{RED}A{BLK}{DOWN}":PRINT"{21 Y}":PRINTTAB(5);"{RED}F{BLK}{DOWN}":PRINT"{21 Y}"
35 PRINT"{BLU}REMEMBER THE WORD {RED}FACE":GOSUB95:Z=1

```

Sound

```
37 FORI=1TO10:A=(INT(RND(1)*4)+1)*2-1:GOSUB70:NEXT
:Z=0:RETURN
40 PRINT"{CLR}THERE ARE FIVE LINES{2 SPACES}ON THE
TREBLE STAFF.{DOWN}"
41 PRINT"THE NAMES OF THE NOTESON THE LINES ARE:
{BLK}":PRINT"{2 DOWN}CCCCCCCCCCCCCCCCCCCC{RED}F
{BLK}CC"
42 PRINT"{DOWN}CCCCCCCCCCCCCCCC{RED}D{BLK}CCCCCC
{2 DOWN}CCCCCCCCCCCC{RED}B{BLK}CCCCCCCCCC
{2 DOWN}CCCCCCC{RED}G{BLK}CCCCCCCCCCCCCCC"
*43 PRINT"{DOWN}CCC{RED}E{BLK}CCCCCCCCCCCCCCCCCCC":G
OSUB95:PRINT"TO HELP YOU REMEMBER{2 SPACES}THE
{SPACE}NOTES, THINK OF"
44 PRINT"THIS PHRASE:";PRINT"{3 DOWN}{BLK}CCCCCCCC
CCCCCCCC{RED}F{BLU}INE{BLK}C{2 DOWN}CCCCCCCCC
CCC{RED}D{BLU}OES{BLK}CCCCC"
*45 PRINT"{DOWN}CCCCCCCCCCC{RED}B{BLU}OY{BLK}CCCCCCC
CC{2 DOWN}CCCCCCC{RED}G{BLU}OOD{BLK}CCCCCCCCCCCCC
{2 DOWN}C{RED}E{BLU}VERY{BLK}CCCCCCCCCCCCCCCCC"
46 GOSUB95
48 FORI=1TO10:A=(INT(RND(1)*5))*2:GOSUB70:NEXT:RET
URN
50 PRINT"{CLR}{BLK}{2 DOWN}{3 SPACES}UI{DOWN}
{2 LEFT}GG":PRINT"3 YTT17 Y}{2 SPACES}
M}N"
52 PRINT"2 YNO18 Y} M} G}"
55 PRINT"Y}PUOM17 Y} M} G{2 SPACES}":PRI
NT"Y}P}Y}GP17 Y}{2 SPACES}MBN"
60 PRINT"3 Y}B18 Y}{3 SPACES}B{DOWN}{2 LEFT}
JK":PRINT"{4 DOWN}{BLU}NAME THE NOTE";
61 IFZ=1THENPRINT"{YEL}{2 SPACES}B{UP}{LEFT}B{UP}
{LEFT}UCCI{DOWN}{3 LEFT}WWH{DOWN}{LEFT}H{DOWN}
{4 LEFT}BJKH{DOWN}{4 LEFT}JFFK"
*62 B=22*A:POKE7935-B,85:POKE7936-B,67:POKE7937-B,8
3:POKE7957-B,74:POKE7958-B,70:POKE7959-B,75
63 FORII=38655-BTO38657-B:POKEII,2:POKEII+22,2:NEX
T:RETURN
*64 POKE7935-B,32:POKE7936-B,32:POKE7937-B,32:POKE7
957-B,119:POKE7958-B,119:POKE7959-B,119:RETURN
*66 POKE7935-B,119:POKE7936-B,119:POKE7937-B,119:PO
KE7957-B,32:POKE7958-B,32:POKE7959-B,32:RETURN
70 GOSUB50
72 GETA$:IFA$=""THEN72
74 IFA$=N$(A,0)THEN78
76 POKE36877,128:FORD=1TO60:NEXT:POKE36877,0:GOTO7
2
78 POKES,VAL(N$(A,1)):POKE7962-B,ASC(N$(A,0))-64:P
OKE36882-B,0
80 FORD=1TO300:NEXT:ONA+1GOSUB64,66,64,66,64,66,64
,66,64
```

```

82 IFA/2=INT(A/2)THENPOKE7962-B,119:GOTO86
84 POKE7962-B,32
86 POKES,0:RETURN
90 DATAE,158,F,161,G,173,A,181,B,189,C,192,D,200,E
  ,206,F,208
95 PRINT"{GRN}PRESS RETURN";
96 GETA$:IFA$=""THEN96
97 IFASC(A$)<>13THEN96
98 PRINT"{CLR}{BLU}":RETURN
100 FORI=1TO10:A=INT(RND(1)*9):GOSUB70:NEXT:RETURN
200 PRINT"{CLR}{BLU}":END

```

On the Color Computer, Treble Clef works in this way:

Line Function

- 1 GOTO line 22, where the PRINT statements for the title screen and instructions begin.
- 2-5 Routine for pressing the ENTER key to continue reading the instruction screens.
- 6-15 Set tones and assign letters to H\$ for the ten possible notes.
- 16-21 Clear the screen and draw the treble staff on the screen.
- 22-28 PRINT the title screen and instructions. At the end of each screen of instructions, the player must press the ENTER key to continue.
- 29-30 After drawing the treble staff, place notes in the correct positions, as well as PRINT the mnemonics such as F-A-C-E and Every Good Boy Does Fine.
- 31-35 Perform the quiz, randomly selecting one of the ten possible notes. As a key is pressed, the note plays and its letter is displayed. If the answer is correct, the reward message shows and the player is asked if another note is wanted.
- 36 Wrong answer sound and return to line 33 for another key to be pressed.
- 37-38 Reward sound and message. Player is asked if another round is desired.
- 39-40 Begin the game again if Y was pressed or end the program if N was pressed.

Program 5-5. Treble Clef

TRS-80 Color Computer, MC-10

```

1 GOTO22
2 PRINT@464,"PRESS <ENTER>";
3 E$=INKEY$:IFE$=""THEN3

```

Sound

```
4 IFASC(E$) <> 13 THEN 3
5 CLS: RETURN
6 S=108: H$="D": RETURN
7 S=125: H$="E": RETURN
8 S=133: H$="F": RETURN
9 S=147: H$="G": RETURN
10 S=159: H$="A": RETURN
11 S=170: H$="B": RETURN
12 S=176: H$="C": RETURN
13 S=185: H$="D": RETURN
14 S=193: H$="E": RETURN
15 S=197: H$="F": RETURN
16 CLS: PRINT I$+D$+CHR$(156)+CHR$(156)+F$+B$: PRINT@
32, I$+CHR$(154)+F$+CHR$(149)+D$+B$: PRINT@64, K$+
K$+CHR$(146)+J$+CHR$(145)+C$+A$
17 PRINT@96, I$+CHR$(154)+CHR$(158)+CHR$(153)+F$+B$
: PRINT@128, K$+J$+CHR$(144)+C$+CHR$(145)+K$+A$: P
RINT@160, G$+CHR$(158)+CHR$(153)+F$+CHR$(149)+G$
+B$
18 PRINT@192, K$+CHR$(145)+C$+CHR$(145)+C$+CHR$(145
)+C$+CHR$(146)+C$+A$: PRINT@224, F$+CHR$(154)+F$+
CHR$(151)+D$+CHR$(149)+F$+CHR$(149)+B$
19 PRINT@256, K$+CHR$(145)+K$+CHR$(146)+J$+CHR$(145
)+A$: PRINT@288, G$+CHR$(155)+CHR$(150)+CHR$(157)
+D$+CHR$(154)+CHR$(158)+CHR$(156)+CHR$(151)+B$
20 PRINT@320, K$+J$+J$+CHR$(146)+K$+A$: PRINT@352, F$
+F$+CHR$(158)+F$+CHR$(154)+G$+B$
21 PRINT@384, G$+F$+CHR$(150)+CHR$(156)+CHR$(149)+G
$+B$: PRINT@416, I$+G$+F$+B$: RETURN
22 CLS: PRINT "{4 SPACES} ** TREBLE CLEF NOTES **": A$
="": B$="": C$=CHR$(147): D$=CHR$(159): FOR I=1 TO 21:
A$=A$+C$: B$=B$+D$: NEXT
23 F$=D$+D$: G$=F$+D$: I$=G$+G$: J$=C$+C$: K$=J$+C$
24 PRINT@97, "EACH NOTE DRAWN ON THE": PRINT " TREBLE
STAFF REPRESENTS A": PRINT " CERTAIN NOTE ON THE
KEYBOARD. "
25 PRINT@225, "IF YOU LEARN THE LETTER NAMES
{3 SPACES} OF THE NOTES YOU CAN FIND THE
{3 SPACES} CORRECT NOTE ON THE KEYBOARD.": GOSUB 2
26 PRINT@33, "THERE ARE TWO PHRASES TO": PRINT " HELP
LEARN THE TREBLE NOTES.": PRINT@129, "THE WORD '
F A C E' REPRESENTS"
27 PRINT " THE NOTES ON THE SPACES.": PRINT@257, "THE
LINE NOTES ARE THE FIRST": PRINT " LETTERS OF EA
CH WORD IN"
28 PRINT " THIS PHRASE.": PRINT@353, "EVERY GOOD BOY
DOES FINE.": PRINT " -{5 SPACES}-{4 SPACES}-
{3 SPACES}-{4 SPACES}-": GOSUB 2
```

```

29 GOSUB16:PRINT@400,"F A C E";:PRINT@304,"F";:
  PRINT@243,"A";:PRINT@182,"C";:PRINT@121,"E";:GO
  SUB2
30 GOSUB16:PRINT@331,"EVERY";:PRINT@272,"GOOD";:PR
  INT@212,"BOY";:PRINT@151,"DOES";:PRINT@90,"FINE
  ";:GOSUB2
31 FORI=1TO10:GOSUB16:PRINT@458,"NAME THE NOTE":N=
  RND(10):Z=N*2
32 FORX=41TO46:FORY=Z TOZ+3:SET(X,Y,4):NEXTY,X
33 E$=INKEY$:IFE$=" "THEN33
34 ONN GOSUB15,14,13,12,11,10,9,8,7,6
35 IFE$=H$THEN37
36 SOUND125,1:SOUND89,1:GOTO33
37 PRINT@N*32+54,E$;:SOUNDS,15:NEXT
38 CLS:PRINT@234,"GOOD WORK!":PRINT@320,"TRY AGAIN
  ? (Y/N)"
39 E$=INKEY$:IFE$="Y"THEN31
40 IFE$<>"N"THEN39
41 CLS:END

```

The TI version of Treble Clef operates like this:

Line	Function
110-130	PRINT title screen.
140-170	READ note name and frequency for each of the notes on the clef.
180-280	Define characters for graphics. L\$ will print a line.
290-440	Present instructions to learn the names of the notes.
450-750	Perform the quiz of ten notes chosen randomly. Note is played when correct key is pressed. The answer must be correct to continue.
760-810	PRINT option to try again and branch appropriately.
820-860	Draw the treble staff.
870-910	Wait for ENTER key to be pressed.
920-930	Clear screen and END.

Program 5-6. Treble Clef

TI-99/4A

```

110 DIM N(9,2)
120 CALL CLEAR
130 PRINT TAB(6);"** TREBLE CLEF **"::~::~~::~~::~~::~~::~~::~~:::
140 FOR C=1 TO 9
150 READ N(C,1),N(C,2)
160 NEXT C
170 DATA 70,698,69,659,68,587,67,523,66,494,65,440
  ,71,392,70,349,69,330
180 FOR C=96 TO 131

```


Sound

```
190 READ C$
200 CALL CHAR(C,C$)
210 NEXT C
220 DATA 000000FF,0000384482828282,828282828282828
2,848488FF889090A,0000000000010102,A0A0C0C0C04
0404
230 DATA 040810FF2040808,404040FF2020202,010204040
810101,202020202020202,202040FF4040404,000000FF
F0304081
240 DATA 101010FF1010101,000000FF18040201,80808080
8080808,102020202040404,0808080808080808,C0202
01010080808
250 DATA 808080FF4040402,404040FF2020100C,040404FF
04040404,2020101008040403,0202020202010101,080
8101020408
260 DATA 804038FF,010101FF01010101,030C30FF,010101
0101010101,0101010111110E,00000000000F30C,0000
000000F00C03
270 DATA 0102020404040201,708808304040004,80404020
2020408,8060100F,010608F
280 L$="``````````"
290 GOSUB 820
300 PRINT : "THINK OF THE WORD 'FACE' FOR": : "THE NO
TE NAMES ON SPACES.": :
310 CALL HCHAR(14,14,70)
320 CALL HCHAR(12,17,65)
330 CALL HCHAR(10,20,67)
340 CALL HCHAR(8,23,69)
350 GOSUB 870
360 GOSUB 820
370 PRINT : "MEMORIZE THIS PHRASE TO HELP": : "LEARN
LINE NOTES E G B D F.": :
380 DATA 15,69,15,86,15,69,15,82,15,89,13,71,13,79
,13,79,13,68,11,66,11,79,11,89,9,68,9,79
390 DATA 9,69,9,83,7,70,7,73,7,78,7,69
400 FOR I=9 TO 28
410 READ C,G
420 CALL HCHAR(C,I,G)
430 NEXT I
440 GOSUB 870
450 GOSUB 820
460 PRINT TAB(8); "NAME THE NOTE": : : :
470 FOR T=1 TO 10
480 RANDOMIZE
490 X=INT(9*RND+1)
500 J=5+X
510 CALL HCHAR(J,20,125)
520 CALL HCHAR(J,21,126)
530 CALL HCHAR(J+1,19,127)
```

```

540 CALL HCHAR(J+1,22,129)
550 CALL HCHAR(J+2,20,130)
560 CALL HCHAR(J+2,21,131)
570 CALL HCHAR(J+1,21,128)
580 CALL SOUND(150,1397,4)
590 CALL KEY(0,K,S)
600 IF S<1 THEN 590
610 IF K=N(X,1)THEN 640
620 CALL SOUND(200,-5,4)
630 GOTO 590
640 CALL HCHAR(J+1,21,N(X,1))
650 CALL SOUND(500,N(X,2),2)
660 CALL SOUND(1,N(X,2),30)
670 IF X/2=INT(X/2)THEN 720
680 CALL HCHAR(J,20,32,2)
690 CALL HCHAR(J+1,19,96,4)
700 CALL HCHAR(J+2,20,32,2)
710 GOTO 750
720 CALL HCHAR(J,20,96,2)
730 CALL HCHAR(J+1,19,32,4)
740 CALL HCHAR(J+2,20,96,2)
750 NEXT T
760 PRINT "GOOD! TRY AGAIN? (Y/N)";
770 CALL KEY(0,K,S)
780 IF K=78 THEN 920
790 IF K<>89 THEN 770
800 CALL HCHAR(24,13,32,15)
810 GOTO 450
820 CALL CLEAR
830 PRINT TAB(9);"TREBLE CLEF"::::
840 PRINT " a": " b": "``c``"&L$: " de": "``fg``
    "&L$; " h i": "`jklm`"&L$
850 PRINT " nop q": "`rst`t"&L$: " u v w": "``xyz`"&L
    $: " {": " {": " !:"
860 RETURN
870 PRINT :TAB(16);"PRESS <ENTER>";
880 CALL KEY(0,K,S)
890 IF K<>13 THEN 880
900 CALL HCHAR(24,18,32,13)
910 RETURN
920 CALL CLEAR
930 END

```

In this program, you'll see spaces between quotation marks, such as " ". Type in the quote marks and press the space bar the correct number of times. If it's hard to tell how many spaces to enter, you can use the line above as a reference. For example, in line 850, you can tell that three spaces are needed near the line's end by counting the characters in line 840 directly above the gap.

Bass Clef Notes

Computers: VIC-20; TRS-80 Color Computer; MC-10; TI-99/4A

Similar to Treble Clef, this program introduces you to bass clef musical notes. The letter names can be remembered by two animal phrases — All Cows Eat Grass for the space notes, and Great Big Dogs Fight Animals for the line notes. The first letter of each word in the phrase is the letter name of the musical note. Both letter strings start with the notes at the bottom and move upward.

A note is randomly drawn on the bass staff, and you must press the key with the letter name of that note. If the answer is correct, the correct note or a short musical tune will play. After a quiz of ten notes, you can try again or end the program.

The program explanation for each computer's version is almost identical to those for Treble Clef Notes, except that this program uses bass clef notes. Refer to the program descriptions for Treble Clef if you want to see how each program works.

Program 5-7. Bass Clef

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```
2 DIMN$(8,1):POKE36878,15:S=36874:FORI=0TO8:READN$(I,0),N$(I,1):NEXT
3 Y$="{20 Y}"
6 PRINT"{CLR}{BLU}{2 DOWN}{2 SPACES}LET'S LEARN NOTES{2 DOWN}":PRINTTAB(6);"BASS CLEF"
8 FORI=0TO8:POKES,VAL(N$(I,1)):FORJ=1TO200:NEXTJ,I:POKES,0
*10 PRINT"{CLR}{BLU}{2 DOWN}{6 SPACES}BASS CLEF{2 DOWN}":PRINT" CHOOSE:";PRINT"{DOWN} 1 NOTES{SPACE}ON SPACES":PRINT"{DOWN}{SHIFT-SPACE}2 NOTES ON LINES"
12 PRINT"{DOWN} 3 ALL BASS NOTES":PRINT"{DOWN} 4 END PROGRAM":GETA$
14 GETA$:IFA$=""THEN14
15 IFASC(A$)<49ORASC(A$)>52THEN14
16 ONVAL(A$)GOSUB30,70,80,99
18 GOTO10
*30 PRINT"{CLR}{DOWN}THERE ARE FOUR SPACES ON THE STAFF.{DOWN}":PRINT"THE NAMES OF THE NOTES ON THE SPACES ARE:"
*32 PRINT"{2 DOWN}{BLK}",Y$:PRINTTAB(11);"{RED}G{BLK}{DOWN}":PRINTY$:PRINTTAB(9);"{RED}E{BLK}{DOWN}":PRINTY$:PRINTTAB(7);"{RED}C{BLK}{DOWN}":PRINTY$:PRINTTAB(5);"{RED}A{BLK}{DOWN}":PRINTY$
$
```

```

34 PRINT"{BLU}REMEMBER THE PHRASE{3 SPACES}{RED}A
{BLK}LL {RED}C{BLK}OWS {RED}E{BLK}AT {RED}G
{BLK}RASS":GOSUB95
* 35 PRINT"{CLR}{BLK}{4 DOWN}":PRINTY$:PRINTTAB(14);
"{RED}G{BLU}RASS{BLK}":PRINTY$:PRINTTAB(10);"
{RED}E{BLU}AT{BLK}":PRINTY$:PRINTTAB(5);"{RED}C
{BLU}OWS{BLK}":PRINTY$:PRINT" {RED}A{BLU}LL
{BLK}"
36 PRINTY$:PRINT"{5 DOWN}":GOSUB95
38 FORI=1TO10:A=(INT(RND(1)*4)+1)*2-1:GOSUB40:NEXT
:RETURN
40 PRINT"{CLR}{BLK}{4 DOWN}[2 Y]N[Y]M[17 Y]
{SPACE}[N]{3 SPACES}M Q":PRINT"[2 Y]K
[2 Y]P[Y]Q[14 Y]{5 SPACES}[M]"
42 PRINT"[5 Y]N[16 Y]{4 SPACES}N":PRINT"
[3 Y]N[18 Y]":PRINTY$;"[2 Y]"
44 PRINT"{4 DOWN}{BLU}NAME THE NOTE"
* 46 B=22*A:POKE7935-B,85:POKE7936-B,67:POKE7937-B,7
3:POKE7957-B,74:POKE7958-B,70:POKE7959-B,75
48 FORII=38655-BTO38657-B:POKEII,2:POKEII+22,2 :NE
XT
50 GETA$:IFA$=""THEN50
51 IFA$=N$(A,0)THEN54
52 POKE36877,128:FORD=1TO60:NEXT:POKE36877,0:GOTO5
0
54 POKES,VAL(N$(A,1)):POKE7962-B,ASC(N$(A,0))-64:P
OKE38682-B,0
56 FORD=1TO300:NEXT:ONA+1GOSUB66,68,66,68,66,68,66
,68,66
58 IFA/2=INT(A/2)THENPOKE7962-B,119:GOTO60
59 POKE7962-B,32
60 POKES,0:RETURN
* 66 POKE7935-B,32:POKE7936-B,32:POKE7937-B,32:POKE7
957-B,119:POKE7958-B,119:POKE7959-B,119:RETURN
* 68 POKE7935-B,119:POKE7936-B,119:POKE7937-B,119:PO
KE7957-B,32:POKE7958-B,32:POKE7959-B,32:RETURN
70 PRINT"{CLR}{BLU}THERE ARE FIVE LINES{2 SPACES}O
N THE BASS STAFF.{DOWN}"
71 PRINT"THE NAMES OF THE NOTESON THE LINES ARE:"
PRINT"{2 DOWN}{BLK}*****{RED}A
{BLK}**{DOWN}"
72 PRINT"*****{RED}F{BLK}*****{2 DOWN}
*****{RED}D{BLK}*****{2 DOWN}
*****{RED}B{BLK}*****"
* 73 PRINT"{DOWN}***{RED}G{BLK}*****{DO
WN}":GOSUB95:PRINT"TO HELP YOU REMEMBER{2 SPACE
S}THE NOTES, THINK OF"
74 PRINT"THIS PHRASE:"PRINT"{3 DOWN}{BLK}*****
*****{RED}A{BLU}NIMALS {BLK}*{2 DOWN}*****
{RED}F{BLU}I GHT{BLK}*****"

```

Sound

```
*75 PRINT "{DOWN}*****{RED}D{BLU}OGS{BLK}*****  
**{2 DOWN}*****{RED}B{BLU}IG{BLK}*****  
{2 DOWN}*{RED}G{BLU}REAT{BLK}*****"  
76 PRINT "{DOWN}":GOSUB95  
77 FOR I=1 TO 10:A=(INT(RND(1)*5))*2:GOSUB40:NEXT:RET  
URN  
80 FOR I=1 TO 10:A=INT(RND(1)*9):GOSUB40:NEXT:RETURN  
90 DATAG,175,A,183,B,191,C,195,D,201,E,207,F,209,G  
,215,A,219  
95 PRINT "{GRN}PRESS RETURN";  
96 GETA$:IFA$=""THEN96  
97 IFASC(A$)<>13THEN96  
98 PRINT "{CLR}{BLU}":RETURN  
99 PRINT "{CLR}{BLU}":END
```

Program 5-8. Bass Clef

TRS-80 Color Computer; MC-10

```
1 GOTO18  
2 PRINT@464,"PRESS <ENTER>";  
3 G$=INKEY$:IF G$=""THEN 3  
4 IFASC(G$)<>13THEN3  
5 CLS:RETURN  
6 H$="G":RETURN  
7 H$="A":RETURN  
8 H$="B":RETURN  
9 H$="C":RETURN  
10 H$="D":RETURN  
11 H$="E":RETURN  
12 H$="F":RETURN  
13 CLS:PRINTB11$+B$:PRINT@32,B11$+B$:PRINT@64,B11$  
+B$:PRINT@96,A3$+A1$+CHR$(145)+A3$+CHR$(146)+CH  
R$(145)+A1$+A$  
14 PRINT@128,B3$+CHR$(144)+B3$+B2$+CHR$(155)+CHR$(  
149)+BL$:PRINT@160,A3$+A3$+A3$+A1$+CHR$(145)+A1  
$+A$  
15 PRINT@192,B6$+B2$+B2$+CHR$(149)+BL$:PRINT@224,A  
3$+A3$+A3$+CHR$(146)+A3$+A$:PRINT@256,B6$+B2$+C  
HR$(158)+CHR$(151)+B1$+B$  
16 PRINT@288,A3$+A3$+A1$+CHR$(146)+A3$+A$:PRINT@32  
0,B6$+CHR$(155)+CHR$(151)+B3$+B$  
17 PRINT@352,A3$+A3$+A3$+A2$+A$:PRINT@384,B11$+B$:  
PRINT@416,B11$+B$:RETURN  
18 CLS:PRINT"{5 SPACES}** BASS CLEF NOTES **":A$=""  
":B$="":A1$=CHR$(147):B1$=CHR$(159)  
19 FOR I=1 TO 21:A$=A$+A1$:B$=B$+B1$:NEXT I  
20 B2$=B1$+B1$:B3$=B2$+B1$:B6$=B3$+B3$:A2$=A1$+A1$  
:A3$=A2$+A1$  
21 B11$=B6$+B3$+B2$:BL$=B2$+CHR$(151)+B6$+B6$+B6$  
22 PRINT@97,"THE PHRASES TO HELP LEARN THE {3 SPACE  
S}BASS CLEF NOTES BOTH":PRINT" CONCERN ANIMALS."
```

```

23 PRINT@225,"REMEMBER, THE FIRST LETTER OF
   {3 SPACES}EACH WORD IS THE NAME OF NOTE."
24 PRINT@321,"FOR THE SPACES, THINK--":PRINT" ALL
   COWS EAT GRASS.":PRINT" -{3 SPACES}-{4 SPACES}-
   {3 SPACES}--":GOSUB 2
25 GOSUB 13
26 PRINT@392,"ALL COWS EAT GRASS";:PRINT@335,"ALL"
   ;:PRINT@274,"COWS";:PRINT@214,"EAT";:PRINT@153,
   "GRASS";:GOSUB2
27 PRINT @33,"THE PHRASE TO LEARN FOR THE
   {5 SPACES}NOTES ON THE LINES OF THE{7 SPACES}BA
   SS CLEF IS:"
28 PRINT @193,"GREAT BIG DOGS FIGHT ANIMALS.
   {3 SPACES}-{5 SPACES}-{3 SPACES}-{4 SPACES}-
   {5 SPACES}--"
29 PRINT @353,"REMEMBER TO START THE PHRASE
   {4 SPACES}ON THE BOTTOM LINE OF THE CLEF.":GOSU
   B 2
30 GOSUB13:PRINT@418,"GREAT BIG DOGS FIGHT ANIMALS
   ";
31 PRINT @366,"GREAT";:PRINT @305,"BIG";:PRINT @24
   3,"DOGS";:PRINT @182,"FIGHT";:PRINT @121,"ANIMA
   LS";:GOSUB 2
32 FOR I=1 TO 10
33 GOSUB13:PRINT@458,"NAME THE NOTE"
34 N=RND(9):IFN=N1 THEN34
35 Y1=N*2+2:N1=N:FORX=41TO46:FORY=Y1 TOY1+3:SET(X,
   Y,4):NEXT Y,X
36 G$=INKEY$:IF G$="" THEN 36
37 ON N GOSUB 7,6,12,11,10,9,8,7,6
38 IFG$=H$THEN40
39 SOUND125,1:SOUND89,1:GOTO36
40 PRINT@(N+1)*32+54,G$;:SOUND88,2:SOUND125,2:SOUN
   D147,2:SOUND176,4
41 NEXT I
42 CLS:PRINT@170,"GOOD WORK!":PRINT@320,"TRY AGAIN
   ? (Y/N)"
43 E$=INKEY$:IF E$="Y" THEN 32
44 IF E$<>"N" THEN 43
45 CLS:END

```

Program 5-9. Bass Clef

TI-99/4A


```

110 DIM N(9,2)
120 CALL CLEAR
130 PRINT TAB(7);"** BASS CLEF **"::~::~~::~~::~~::~~::~~::~~:::
140 FOR I=1 TO 9
150 READ N(I,1),N(I,2)
160 NEXT I


```

Sound

```
170 DATA 65,220,71,196,70,175,69,165,68,147,67,131
,66,123,65,110,71,110
180 FOR I=96 TO 114
190 READ C$
200 CALL CHAR(I,C$)
210 NEXT I
220 DATA 000000FF,000000FF0F10608,000000FFFF,00000
0FF80700C03,010204181020204,8040202010080804,4
0583CFF3C18
230 DATA 040202FF01010101,0101020204040408,080810F
F2040808,0101020408,000E1F1F1F0E
240 DATA 000000000000F30C,0000000000F00C03,01020204
04040201,708808304040004,804040202020408,80601
00F,010608F
250 L$="``````````"
260 GOSUB 760
270 PRINT : "LEARN THIS PHRASE FOR THE": "NOTES ON
SPACES, A C E G.":
280 GOSUB 800
290 DATA 13,30,15,65,15,76,15,76,15,32,13,67,13,79
,13,87,13,83,13,32,11,69,11,65,11,84,11,32
300 DATA 9,71,9,82,9,65,9,83,9,83
310 GOSUB 860
320 GOSUB 760
330 PRINT : "THIS PHRASE HELPS YOU KNOW": "THE LINE
NOTES, G B D F A.":
340 GOSUB 800
350 DATA 8,32,16,71,16,82,16,69,16,65,16,84,14,66,
14,73,14,71,12,68,12,79,12,71,12,83,10,70,10,73
360 DATA 10,71,10,72,10,84,8,65,8,78,8,73,8,77,8,6
5,8,76,8,83,8,32
370 GOSUB 860
380 GOSUB 760
390 PRINT TAB(8); "NAME THE NOTE": : : : :
400 FOR T=1 TO 10
410 RANDOMIZE
420 X=INT(9*RND+1)
430 J=6+X
440 CALL HCHAR(J,20,108)
450 CALL HCHAR(J,21,109)
460 CALL HCHAR(J+1,19,110)
470 CALL HCHAR(J+1,22,112)
480 CALL HCHAR(J+2,20,113)
490 CALL HCHAR(J+2,21,114)
500 CALL HCHAR(J+1,21,111)
510 CALL SOUND(150,1397,4)
520 CALL KEY(0,K,S)
530 IF S<1 THEN 520
540 IF K=N(X,1) THEN 570
```

Chapter 6
Math
Competency



Chapter 6

Math Competency

The six programs in this chapter generate story or word problems for mathematics, much like those found on standardized math tests given in high school. Each program concentrates on one type of word problem, presenting a problem and waiting for a response from you. "Buying Items," for example, displays a list of five items, along with their prices. Questions are then presented, such as asking you to total the items' costs.

Illustrating how word or story problems are formed, these programs could be used by anyone interested in mathematics, from those in the third grade to high school students.

If you enter an incorrect answer while using these programs, you'll see a reminder of how to get to the right answer, and the problem will repeat. If you answer correctly, you'll always have a choice of continuing with another problem or quitting the problem.

Buying Items

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I; TI-99/4

"Buying Items" gives a list of five items along with their prices. The first question asks you to total all five items. The second question asks which two items could be purchased, given a certain amount of money. This question is in multiple-choice form.

There are three different categories for price lists. The number A is randomly chosen as 1, 2, or 3. School supplies is number 1, a toy store is number 2, and a grocery store is number 3. The items I\$ are read in as data in an array I\$(A,C), where I\$(2,4) would mean the name of an item in category 2 (toys) and the fourth item listed.

The data for each item includes a minimum price $I(A,C,1)$ and a maximum price $I(A,C,2)$. For the actual price list for the problem, the price P is a random number from the minimum to the maximum. A subroutine is used to convert the price calculated as a number of cents to a dollar value for printing in the problem. The price P is a whole number of cents.

A random number F is chosen to be 1 or 2 so that the wording of the problem will vary. If F is 1 the question is, "How much will it cost to buy all the items on the list?" If F is 2, the question has a person's name. Six names are read in as data, chosen at random. If the name number N is greater than 3, the person is a boy; otherwise, a girl; and the printed question uses the appropriate pronouns.

Math Competency

M is a random amount of money based on a range from the *lowest* possible price of the two least expensive items in the list to a *maximum* price that would still be less than any other two items. The correct answer for the multiple-choice question is the phrase H\$(A), and it is placed in a random position R, one of the four choices. The incorrect choices use one item from the first three in the list, and one item from the last two in the list. The computer makes sure the four phrases are different.

Here's how the program works on the VIC:

Line	Function
1-2	PRINT title screen, DIMension variables, READ six names from data, and turn on volume for sound.
3	READ from DATA items and minimum and maximum prices, initialize S for sound statements, and branch past subroutines.
4-7	Wait for user to press RETURN.
9	Delay loop for sound.
10-13	Convert cost in cents to dollars.
16	Play music for correct answer.
25	DIMension H and S for multiple-choice answers and define H\$ answers.
28	Wait for user to press RETURN.
30	Randomly choose category and initialize total TP=0.
32-36	PRINT problem, randomly choosing price for each item.
38-41	PRINT one of two forms of the question.
44	Receive answer.
46-48	If answer is incorrect, give the right answer with an explanation.
60	If answer is correct, play sound.
62	Clear lower part of screen.
64-81	PRINT second question.
82-87	PRINT multiple-choice answers.
88	Receive answer.
89-90	If answer is correct, play sound.
92-96	PRINT option to try again and branch appropriately.
100-102	If answer is incorrect, PRINT explanation.
120-122	DATA statements.
200	Clear screen and END.

Program 6-1. Buying Items

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```

1 PRINT "{CLR}{2 DOWN}{BLU}{3 SPACES}MATH COMPETENC
Y":PRINT "{2 DOWN}{4 SPACES}BUYING ITEMS{8 DOWN}"
2 DIMI$(3,5),I(3,5,2),N$(6),J(5):FORC=1TO6:READN$(
C):NEXT:POKE36878,15
3 FORA=1TO3:FORC=1TO5:READI$(A,C),I(A,C,1),I(A,C,2
):NEXTC,A:S=36876:GOTO25
4 PRINT "{GRN}PRESS RETURN";
5 GETA$:IFA$=""THEN5
6 IFASC(A$)<>13THEN5
7 RETURN
9 FORDE=1TO100:NEXT:RETURN
10 P$=STR$(P):IFLEN(P$)=2THENP$="0"+RIGHT$(P$,1)
11 PR$=RIGHT$(P$,2):PL$=LEFT$(P$,LEN(P$)-2):IFLEN(
PL$)<2THENPL$="{SHIFT-SPACE}"
13 P$="$"+PL$+"."+PR$:RETURN
16 POKES,195:GOSUB9:POKES,207:GOSUB9:POKES,215:GOS
UB9
18 POKES,225:GOSUB9:GOSUB9:POKES,0:RETURN
20 DIMH$(3),S$(4):H$(1)="PENCIL AND ERASER"
25 H$(2)="BALL AND TRUCK":H$(3)="CANDY AND FRUIT"
28 GOSUB4
30 A=INT(RND(0)*3+1):TP=0
32 PRINT "{CLR}{BLU}GIVEN THIS PRICE LIST:{BLK}"
34 FORC=1TO5:D=I(A,C,2)-I(A,C,1):P=I(A,C,1)+INT(RN
D(0)*D+1):GOSUB10:TP=TP+P
36 PRINTTAB(4);I$(A,C);TAB(13);P$:NEXT
38 F=INT(RND(0)*2+1)
39 IFF=1THENPRINT "{DOWN}{BLU}HOW MUCH WILL IT COST
TO BUY ALL THE ITEMS{2 SPACES}ON THE LIST?":GO
TO44
40 N=INT(RND(0)*6+1):PRINT "{DOWN}{BLU}";N$(N);" WA
NTS TO BUY":PRINT"EVERYTHING ON THE"
41 PRINT"LIST.{2 SPACES}WHAT WOULD THE TOTAL COST
{SPACE}BE?"
44 PRINT "{DOWN}{RED}$";:INPUT X
46 IFABS(X-TP/100)<.001THEN60
47 POKES,159:GOSUB9:POKES,135:GOSUB9:POKES,0
* 48 PRINT "{PUR}{DOWN}ADD ALL FIVE NUMBERS. THE TOTA
L IS ";:P=TP:GOSUB10:PRINTP$:PRINT "{DOWN}":GOSU
B4:GOTO30
60 GOSUB16
62 FORD=7834TO8185:POKED,32:NEXT
64 PRINT "{6 UP}{BLU}":IFF=1THENPRINT"IF YOU COULD
{SPACE}ONLY":GOTO70

```

Math Competency

```
65 PRINT "{UP} IF ";N$(N); " COULD ONLY"
70 IFA=1THENM=INT(RND(1)*5)+25:GOTO76
72 IFA=2THENM=INT(RND(1)*36)+239:GOTO76
74 M=INT(RND(1)*18)+100
76 P=M:GOSUB10:IFF=1THENPRINT "{2 UP} SPEND ";P$:GOT
O78
77 PRINT "SPEND ";P$
78 PRINT "WHICH OF THESE PAIRS{2 SPACES}OF ITEMS ON
THE LIST"
79 IFF=1THENPRINT "COULD YOU BUY?":GOTO82
80 IFN<4THENPRINT "COULD SHE BUY?":GOTO82
81 PRINT "COULD HE BUY?"
82 R=INT(RND(1)*4)+1:PRINT "{DOWN}":FORV=1TO4:IFV=R
THENS$(V)=H$(A):GOTO87
83 X=INT(RND(1)*2)+4:S$(V)=I$(A,X):X=INT(RND(1)*3)
+1:S$(V)=S$(V)+" AND "+I$(A,X)
84 IFV=1THEN87
85 FORV1=1TOV-1:IFS$(V1)=S$(V)THEN83
86 NEXTV1
87 PRINTCHR$(64+V); " "+S$(V):NEXTV:PRINT "{RED}";
88 GETA$:IFA$=""THEN88
89 PRINTA$:IFASC(A$)<>64+RTHEN100
90 GOSUB16
92 PRINT "{PUR}{DOWN}TRY AGAIN?(Y/N)";
93 GETA$:IFA$=""THEN93
94 IFA$="Y"THEN30
95 IFA$="N"THEN200
96 GOTO93
100 POKES,159:GOSUB9:POKES,135:GOSUB9:POKES,0
102 PRINT "{PUR}THE TOTAL OF THE TWO{2 SPACES}ITEMS
MUST BE LESS{4 SPACES}THAN ";P$:GOTO92
120 DATANANCY,CINDY,CHERY,DAVID,RANDY,LANCE,PENCIL
,8,15,ERASER,2,10
121 DATA NOTEBOOK,35,99,RULER,29
122 DATA49,PAPER,59,90,DOLL,249,599,BALL,49,89,TRU
CK,100,150,GAME,270,500,MODEL,300,700
123 DATACANDY,20,50,MEAT,123,425,FRUIT,24,50,CHIPS
,100,257,BREAD,100,179
200 PRINT "{CLR}{BLU}":END
```

The program looks and works differently on the Color Computer and MC-10:

Line	Function
1	DIMension variables, READ from DATA the items and minimum and maximum prices.
2	Define H\$ for multiple-choice answers and branch past subroutines.

- 3-6 Wait for user to press ENTER.
- 7-10 Convert cost in cents to dollars.
- 11-12 Clear screen, initialize total price TP, randomly choose category, and PRINT list of items with random prices.
- 13 Ask question, receive answer and compare input to correct answer.
- 14 If input answer was incorrect, show how to get correct answer and return for another problem.
- 15 If answer is correct, PRINT message and clear lower part of screen.
- 16-18 Randomly choose cost.
- 19-24 PRINT multiple-choice question.
- 25 Wait for user to press a key for answer.
- 26-27 If answer is incorrect, give correct answer.
- 28 If answer is correct, PRINT message.
- 29-31 PRINT option to try again and branch appropriately.
- 32-33 DATA statements.
- 34 Clear screen and END.

Program 6-2. Buying Items

TRS-80 Color Computer; MC-10; TRS-80 Model I (with changes)

```

1 DIMI$(3,5),I(3,5,2),J(5),H$(3),S$(4):FORA=1TO3:FORC=1TO5:READI$(A,C),I(A,C,1),I(A,C,2):NEXTC,A
2 H$(1)="PENCIL AND ERASER":H$(2)="BALL AND TRUCK":H$(3)="CANDY AND FRUIT":GOTO11
3 PRINT@495,"PRESS <ENTER>";
4 A$=INKEY$:IFA$=""THEN4
5 IF ASC(A$)<>13 THEN 4
6 RETURN
7 P$=STR$(P)
8 IFLEN(P$)=2THENP$=" 0"+RIGHT$(P$,1)
9 PR$=RIGHT$(P$,2):PL$=LEFT$(P$,LEN(P$)-2):IFLEN(PL$)<2THENPL$=" "+PL$
10 P$="$"+PL$+"." "+PR$:RETURN
11 CLS:TP=0:A=RND(3):PRINT"GIVEN THIS PRICE LIST:"
12 FORC=1TO5:D=I(A,C,2)-I(A,C,1):P=I(A,C,1)+RND(D):GOSUB7:TP=TP+P:PRINTTAB(4);I$(A,C);TAB(13);P$:NEXT:PRINT
13 PRINT"HOW MUCH WILL IT COST TO BUY{4 SPACES}ALL THE ITEMS ON THE LIST?":PRINT"$";:INPUTX:IFABS(X-TP/100)<.001THEN15
14 PRINT:PRINT"ADD ALL FIVE NUMBERS.":PRINT"THE TOTAL IS ";:P=TP:GOSUB7:PRINTP$:GOSUB3:GOTO11

```


Math Competency

```
15 PRINT:PRINT"CORRECT!":FORC=1TO100:NEXT:FORC=1TO
8:PRINT@192+32*C," ":NEXT
16 IFA=1THENM=RND(5)+25:GOTO19
17 IFA=2THENM=RND(36)+239:GOTO19
18 M=RND(18)+100
19 P=M:GOSUB7:PRINT@224,"IF YOU COULD ONLY SPEND "
;P$:PRINT"WHICH OF THESE PAIRS OF ITEMS
{3 SPACES}ON THE LIST COULD YOU BUY?"
20 R=RND(4):FORV=1TO4:IFV=R THENM$(V)=H$(A):GOTO24
21 X=RND(2)+3:S$(V)=I$(A,X):X=RND(3):S$(V)=S$(V)+
AND "+I$(A,X):IFV=1THEN24
22 FORV1=1TOV-1:IFS$(V1)=S$(V)THEN21
23 NEXTV1
24 PRINTTAB(3);CHR$(64+V);" "+S$(V):NEXTV
25 A$=INKEY$:IFA$=""THEN25
26 PRINTA$:IFASC(A$)=64+R THEN28
27 PRINT"THE TOTAL OF THE TWO ITEMS MUST BE LESS T
HAN ";P$;"--";CHR$(64+R):GOTO29
28 PRINT"CORRECT!"
29 PRINT@480,"TRY AGAIN? (Y/N)";
30 A$=INKEY$:IFA$="Y"THEN11
31 IFA$<>"N"THEN30
32 DATAPENCIL,8,15,ERASER,2,10,NOTEBOOK,35,99,RULE
R,29,49,PAPER,59,90,DOLL,249,599
33 DATA BALL,49,89,TRUCK,100,150,GAME,270,500,MODE
L,300,700,CANDY,20,50,MEAT,123,425,FRUIT,24,50,
CHIPS,100,257,BREAD,100,179
34 CLS:END
```

For the TRS-80 Model I computer, make the following changes in the above listing:

Insert CLEAR 100 at the beginning of line 1.

Change line 3 to PRINT: PRINT "PRESS <ENTER>";

In line 15 change @192+32*C to @320+64*C

In line 19 change @224 to @448

In line 29 delete @480.

Buying Items on the TI-99/4A works this way:

Line	Function
100-150	Clear screen; PRINT title screen; define characters and colors.
160	DIMension variables.
170-290	READ from DATA six names N\$ and items I\$ with minimum and maximum prices I.
300-320	Define H\$ for multiple-choice answers.
330	Branch past subroutines.

- 340-370 Wait for ENTER key to be pressed.
 380-400 Play music for incorrect answer.
 410-450 Play music for correct answer.
 460-540 Convert cost in cents to dollars.
 550-570 Randomly choose one of three categories and initialize total cost TP = 0.
 580-660 PRINT problem, randomly choosing price for each item.
 670-720 Draw a colored box around list.
 730-820 PRINT one of two forms of the question.
 830 Receive answer.
 840-910 If answer is incorrect, give the right answer with an explanation and return for another problem.
 920-930 If answer is correct, play sound then clear lower part of screen.
 940-1150 PRINT second question.
 1160-1300 PRINT multiple-choice answers.
 1310-1340 Receive answer.
 1350-1360 If answer is correct, play sound.
 1370-1400 PRINT option to try again and branch appropriately.
 1410-1440 If answer is incorrect, play different sound effect and show correct answer.
 1450-1460 Clear screen and END.

Program 6-3. Buying Items

TI-99/4A

```

100 CALL CLEAR
110 PRINT TAB(6); "MATH COMPETENCY"
120 CALL CHAR(136, "080402FF020408")
130 PRINT ::TAB(7); "BUYING ITEMS"
140 CALL COLOR(14,9,16)
150 PRINT :::::::::::
160 DIM I$(3,5),I(3,5,2),N$(6),J(5),H$(3),S$(4)
170 FOR C=1 TO 6
180 READ N$(C)
190 NEXT C
200 FOR A=1 TO 3
210 FOR C=1 TO 5
220 READ I$(A,C),I(A,C,1),I(A,C,2)
230 NEXT C
240 NEXT A
250 DATA ANGIE,CINDY,CHERY,RICKY,BOBBY,RANDY,PENCIL,8,15
260 DATA ERASER,2,10,NOTEBOOK,35,99,RULER,29,49
270 DATA PAPER,59,90,DOLL,249,599,BALL,49,89,TRUCK,100,150
    
```

Math Competency

```
280 DATA GAME,270,500,MODEL,300,700,CANDY,20,50
290 DATA MEAT,123,425,FRUIT,24,50,CHIPS,100,257,BR
    EAD,100,179
300 H$(1)="PENCIL AND ERASER"
310 H$(2)="BALL AND TRUCK"
320 H$(3)="CANDY AND FRUIT"
330 GOTO 550
340 PRINT TAB(15);"PRESS <ENTER>";
350 CALL KEY(0,K,S)
360 IF K<>13 THEN 350
370 RETURN
380 CALL SOUND(100,330,2)
390 CALL SOUND(150,262,2)
400 RETURN
410 CALL SOUND(100,262,2)
420 CALL SOUND(100,330,2)
430 CALL SOUND(100,392,2)
440 CALL SOUND(200,523,2)
450 RETURN
460 P$=STR$(P)
470 IF LEN(P$)>1 THEN 490
480 P$="0"&P$
490 IF LEN(P$)>2 THEN 510
500 P$=" "&P$
510 PR$=SEG$(P$,LEN(P$)-1,2)
520 PL$=SEG$(P$,1,LEN(P$)-2)
530 P$="$"&PL$&". "&PR$
540 RETURN
550 RANDOMIZE
560 A=INT(RND*3+1)
570 TP=0
580 CALL CLEAR
590 PRINT "GIVEN THIS PRICE LIST:":
600 FOR C=1 TO 5
610 D=I(A,C,2)-I(A,C,1)
620 P=I(A,C,1)+INT(RND*D+1)
630 GOSUB 460
640 TP=TP+P
650 PRINT TAB(6);I$(A,C);TAB(15);P$
660 NEXT C
670 R=INT(RND*13+4)
680 CALL COLOR(13,R,R)
690 CALL HCHAR(18,6,128,18)
700 CALL VCHAR(19,6,128,5)
710 CALL VCHAR(19,23,128,5)
720 CALL HCHAR(24,6,128,18)
730 F=INT(RND*2+1)
740 IF F=2 THEN 790
750 PRINT ::"HOW MUCH WILL IT COST"
```

Math Competency

```
760 PRINT "TO BUY ALL THE ITEMS"  
770 PRINT "ON THE LIST?"  
780 GOTO 830  
790 N=INT(RND*6+1)  
800 PRINT ":N$(N);" WANTS TO BUY"  
810 PRINT "EVERYTHING ON THE LIST."  
820 PRINT "WHAT WILL THE TOTAL COST BE?"  
830 INPUT "$":X  
840 IF ABS(X-TP/100)<.001 THEN 920  
850 GOSUB 380  
860 PRINT ":ADD ALL FIVE NUMBERS."  
870 P=TP  
880 GOSUB 460  
890 PRINT "THE TOTAL IS ";P$:::  
900 GOSUB 340  
910 GOTO 550  
920 GOSUB 410  
930 CALL HCHAR(20,1,32,128)  
940 IF F=2 THEN 970  
950 PRINT "IF YOU COULD ONLY SPEND"  
960 GOTO 980  
970 PRINT "IF ";N$(N);" COULD ONLY SPEND"  
980 IF A<>1 THEN 1010  
990 M=INT(RND*5+25)  
1000 GOTO 1050  
1010 IF A<>2 THEN 1040  
1020 M=INT(RND*36+239)  
1030 GOTO 1050  
1040 M=INT(RND*18+100)  
1050 P=M  
1060 GOSUB 460  
1070 PRINT P$;" , WHICH OF THESE PAIRS"  
1080 PRINT "OF ITEMS COULD ";  
1090 IF F<>1 THEN 1120  
1100 PRINT "YOU BUY?"::  
1110 GOTO 1160  
1120 IF N>3 THEN 1150  
1130 PRINT "SHE BUY?"::  
1140 GOTO 1160  
1150 PRINT "HE BUY?"::  
1160 R=INT(RND*4+1)  
1170 FOR V=1 TO 4  
1180 IF V=R THEN 1280  
1190 X=INT(RND*2+4)  
1200 S$(V)=I$(A,X)  
1210 X=INT(RND*3+1)  
1220 S$(V)=S$(V)&" AND "&I$(A,X)  
1230 IF V=1 THEN 1290  
1240 FOR V1=1 TO V-1
```

Math Competency

```
1250 IF S$(V1)=S$(V)THEN 1190
1260 NEXT V1
1270 GOTO 1290
1280 S$(V)=H$(A)
1290 PRINT TAB(3);CHR$(64+V);" "&S$(V)
1300 NEXT V
1310 CALL SOUND(150,1397,2)
1320 CALL KEY(0,K,S)
1330 IF (K<65)+(K>68)THEN 1320
1340 CALL HCHAR(K-45,4,42)
1350 IF K<>64+R THEN 1410
1360 GOSUB 410
1370 PRINT : "TRY AGAIN? (Y/N)";
1380 CALL KEY(0,K,S)
1390 IF K=89 THEN 550
1400 IF K=78 THEN 1450 ELSE 1380
1410 GOSUB 380
1420 CALL HCHAR(19+R,3,136)
1430 PRINT : "THE TOTAL OF THE TWO ITEMS MUST BE L
    ESS THAN ";P$
1440 GOTO 1370
1450 CALL CLEAR
1460 END
```

Sales Tax

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

This word problem program shows you how to calculate sales tax on purchased items. One of three categories is chosen randomly, and then four items are listed with random prices. Sales tax is selected randomly, from 2 to 5 percent. You enter the total cost of the items, including sales tax as your answer.

Here's how the three versions of the program work:

Line	Function
110-200	READ from DATA the title of store T\$, item P\$(I,J), and base price P(I,J).
210	Clear screen.
220	Initialize total cost A.
230-280	PRINT problem with random tax T.
290-310	PRINT random title or kind of store.
320-370	PRINT four items with random prices.
380-410	Draw separator line and ask for total cost.
420-430	Calculate total, plus tax. If input value is correct, shift to appropriate line.

- 440-570 If input value is incorrect, show how to get right answer and return for another problem.
- 580-650 If answer is correct, present option to try again and branch appropriately.
- 660-690 Convert number to dollars and cents for printing.
- 700-710 Clear screen and END.

Program 6-4. Sales Tax

VIC-20

```
110 FORI=1TO4
120 READT$(I)
130 FORJ=1TO4
140 READP$(I,J),P(I,J)
150 NEXTJ
160 NEXTI
170 DATAHARDWARE,HAMMER,1500,PLIERS,300,SAW,600,NA
    ILS,100
180 DATACLOTHES,BELT,400,TIE,500,SHIRT,600,PANTS,2
    000
190 DATATOYS,BALL,100,CAR,200,GAME,500,DOLL,600
200 DATASUPPLIES,PAPER,400,CLIPS,100,PENCILS,100,E
    NVELOPES,200
210 PRINT"{CLR}"
220 A=0
230 PRINT"SALES TAX ON THE"
240 T=INT(4*RND(0)+2)
250 PRINT"FOLLOWING ITEMS IS":PRINTT;"PER CENT, OR
    ";"$."0"+RIGHT$(STR$(T),1)
260 PRINT"FOR EACH DOLLAR SPENT."
270 PRINT"WHAT IS THE TOTAL":PRINT"COST?"
280 PRINT
290 I=INT(4*RND(0)+1)
300 PRINTTAB(5);T$(I)
310 PRINT
320 FORJ=1TO4
330 C=P(I,J)+25*(INT(4*RND(0)))
340 GOSUB660
350 PRINTP$(I,J),C$
360 A=A+C
370 NEXTJ
380 PRINT"#####"
390 PRINT
400 INPUT "TOTAL COST = $";B
410 PRINT
420 TX=1+T/100
430 TA=INT(((A/100)*TX+.005)*100)/100
440 IF ABS(B-TA)<.01 THEN 580
450 C=A
```

Math Competency

```
460 GOSUB660
470 PRINT"ADD COSTS FOR TOTAL."
480 PRINT"PRICE OF ITEMS = ";PRINTC$
490 PRINT"MULTIPLY BY ";T/100;" FOR"
500 PRINT"TAX, THEN ADD."
510 PRINT
520 C=TA*100
530 GOSUB660
540 PRINT"TOTAL COST ={SHIFT-SPACE}";C$
550 PRINT
560 INPUT "PRESS <RETURN>";E$
570 GOTO210
580 PRINT"CORRECT!"
590 PRINT
600 PRINT"CHOOSE: 1 TRY AGAIN"
610 PRINTTAB(8);"2 END"
620 GET E$
630 IF E$="1" THEN 210
640 IF E$="2" THEN 700
650 GOTO620
660 C$=STR$(C)
670 L=LEN(C$)
680 C$="$"+LEFT$(C$,L-2)+". "+RIGHT$(C$,2)
690 RETURN
700 PRINT"{CLR}"
710 END
```

Program 6-5. Sales Tax

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
105 DIMT$(4),P$(4,4),P(4,4)
110 FOR I=1 TO 4
120 READ T$(I)
130 FOR J=1 TO 4
140 READ P$(I,J),P(I,J)
150 NEXT J
160 NEXT I
170 DATA HARDWARE,HAMMER,1500,PLIERS,300,SAW
,600,SCREWDRIVER,100
180 DATA CLOTHES,BELT,400,TIE,500,SHIRT,600,
PANTS,2000
190 DATA TOYS,BALL,100,CAR,200,GAME,500,DOLL
,600
200 DATA SUPPLIES,PAPER,400,CLIPS,100,PENCIL
S,100,ENVELOPES,200
210 CLS
220 A=0
230 PRINT "SALES TAX ON THE FOLLOWING"
240 T=RND(4)+2
250 PRINT "ITEMS IS";T;"PER CENT, OR"
```

Math Competency

```
260 PRINT "$." + RIGHT$(STR$(T), 1) + " FOR EACH
    DOLLAR SPENT."
270 PRINT "WHAT IS THE TOTAL COST?"
280 PRINT
290 I=RND(4)
300 PRINT TAB(8);T$(I)
310 PRINT
320 FOR J=1 TO 4
330 C=P(I,J)+25*(RND(4)-1)
340 GOSUB 660
350 PRINT P$(I,J),C$
360 A=A+C
370 NEXT J
380 PRINT "#####"
390 PRINT
400 INPUT "TOTAL COST = $ ";B
410 PRINT
420 TX=1+T/100
430 TA=INT(((A/100)*TX+.005)*100)/100
440 IF ABS(B-TA)<.01 THEN 580
450 C=A
460 GOSUB 660
470 PRINT "ADD COSTS FOR TOTAL."
480 PRINT "PRICE OF ITEMS = ";C$
490 PRINT "MULTIPLY BY ";T/100;" FOR TAX,"
500 PRINT "THEN ADD."
510 PRINT
520 C=TA*100
530 GOSUB 660
540 PRINT "TOTAL COST = ";C$
550 PRINT
560 INPUT "PRESS <ENTER>";E$
570 GOTO 210
580 PRINT "CORRECT!"
590 PRINT
600 PRINT "CHOOSE: 1 TRY AGAIN"
610 PRINT TAB(8);"2 END"
620 E$=INKEY$
630 IF E$="1" THEN 210
640 IF E$="2" THEN 700
650 GOTO 620
660 C$=STR$(C)
670 L=LEN(C$)
680 C$="$"+LEFT$(C$,L-2)+". "+RIGHT$(C$,2)
690 RETURN
700 CLS
710 END
```


Math Competency

Program 6-6. Sales Tax

TI-99/4A

```
110 FOR I=1 TO 4
120 READ T$(I)
130 FOR J=1 TO 4
140 READ P$(I,J),P(I,J)
150 NEXT J
160 NEXT I
170 DATA HARDWARE,HAMMER,1500,PLIERS,300,SAW,600,N
    AILS,100
180 DATA CLOTHES,BELT,400,TIE,500,SHIRT,600,PANTS,
    2000
190 DATA TOYS,BALL,100,CAR,200,GAME,500,DOLL,600
200 DATA SUPPLIES,PAPER,400,CLIPS,100,PENCILS,100,
    ENVELOPES,200
210 CALL CLEAR
220 A=0
230 PRINT "SALES TAX ON THE FOLLOWING"
235 RANDOMIZE
240 T=INT(4*RND+2)
250 PRINT "ITEMS IS";T;"PER CENT, OR"
260 PRINT "$.0";STR$(T);" FOR EACH DOLLAR SPENT."
270 PRINT "WHAT IS THE TOTAL COST?"
280 PRINT
290 I=INT(4*RND+1)
300 PRINT TAB(8);T$(I)
310 PRINT
320 FOR J=1 TO 4
330 C=P(I,J)+25*(INT(4*RND))
340 GOSUB 660
350 PRINT P$(I,J),C$
360 A=A+C
370 NEXT J
380 PRINT "#####"
390 PRINT
400 INPUT "TOTAL COST = $ ":B
410 PRINT
420 TX=1+T/100
430 TA=INT(((A/100)*TX+.005)*100)/100
440 IF ABS(B-TA)<.01 THEN 580
450 C=A
460 GOSUB 660
470 PRINT "ADD COSTS FOR TOTAL."
480 PRINT "PRICE OF ITEMS = ";C$
490 PRINT "MULTIPLY BY ";T/100;" FOR TAX,"
500 PRINT "THEN ADD."
510 PRINT
520 C=TA*100
530 GOSUB 660
```

```
540 PRINT "TOTAL COST = ";C$
550 PRINT
560 INPUT "PRESS <ENTER> ":E$
570 GOTO 210
580 PRINT "CORRECT!"
590 PRINT
600 PRINT "CHOOSE: 1 TRY AGAIN"
610 PRINT TAB(9);"2 END"
620 CALL KEY(0,K,S)
630 IF K=49 THEN 210
640 IF K=50 THEN 700
650 GOTO 620
660 C$=STR$(C)
670 L=LEN(C$)
680 C$="$"&SEG$(C$,1,L-2)&". "&SEG$(C$,L-1,2)
690 RETURN
700 CALL CLEAR
710 END
```

Earning Money

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

"Earning Money," another math competency program, provides you with word problems that use addition and multiplication. The program presents two kinds of problems for you to look at. The first are relatively simple, asking you to calculate how much money a person makes in a week, given an hourly wage and the number of hours worked, for example. Once you answer correctly, you'll have the option of continuing with the simpler problems, or moving on to more difficult ones.

This second group of word problems is more complicated, asking you to calculate the money earned in several weeks, for instance, giving you the amount earned each hour, day, or week. As with the first group of problems, you'll use multiplication and addition to arrive at your answer. If you enter an incorrect answer, the program will show you how the total should have been computed, and then ask if you want to continue. You'll have the choice of going on or quitting the program entirely.

The names of the people in the problems are read in as N\$(I) and T\$(I), where I is a subscript from 0 to 5. The ways of earning money are read in as phrases J\$(I).

The wage earned is a random number from \$1 to \$3.25 in amounts divisible by 25¢.

A name is chosen with the random number N, and the number of hours in the first problem is a random number H. For the

Math Competency

second type of problem, the number of weeks is a random number W from 2 to 20 weeks. The third type of problem chooses a random name, a random job, and a random number of weeks W from 2 to 9 weeks.

The VIC version works like this:

Line	Function
2-4	PRINT title screen and DIMension variables name N\$, job J\$, and name T\$.
6	READ from DATA the names and jobs, turn on volume for sound, initialize S for sound, and branch.
10-13	Wait for RETURN key to be pressed.
14-15	Convert price to dollar and cent value.
18	Delay loop for sounds.
20	Play sounds for incorrect answer.
22	Play sounds for correct answer.
28	Wait for RETURN key to be pressed.
30	Randomly choose name, hours, and wage.
32-40	PRINT problem and check input answer.
42-47	Procedure for correct answer. PRINT option to try again or continue and then branch appropriately.
49-52	If answer is incorrect, show how to get correct answer and return for another problem.
54-64	Randomly choose variables and PRINT next problem. Compare input answer.
66-69	Procedure for correct answer. PRINT option to try again or continue and then branch appropriately.
70-72	If answer is incorrect, show how to get correct answer and return for another problem of same type.
80-88	Randomly choose variables and PRINT next problem.
90	Input answer and compare with correct answer.
92-95	Procedure for correct answer. PRINT option to try again and branch appropriately.
97-98	If answer is incorrect, show how to get correct answer, and then return for another problem of same type.
100-102	DATA statement.
110	Clear screen and END.

Program 6-7. Earning Money

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```

2 PRINT"{CLR}{BLU}{2 DOWN}{3 SPACES}MATH COMPETENC
Y":PRINT"{3 DOWN}{4 SPACES}EARNING MONEY{9 DOWN}
"
4 DIMN$(5),J$(5),T$(5)
6 FORI=0TO5:READN$(I),J$(I),T$(I):NEXT:POKE36878,1
5:S=36876:GOTO28

10 PRINT"{DOWN}{GRN}PRESS RETURN";
11 GETA$:IFA$=""THEN11
12 IFASC(A$)<>13THEN11
13 PRINT"{CLR}{BLU}":RETURN
14 P=100+25*INT(RND(0)*10)
15 P$=STR$(P):PL$=LEFT$(P$,LEN(P$)-2):PR$=RIGHT$(P
$,2):P$="$"+PL$+"."+PR$:RETURN
18 FORV=1TO100:NEXT:RETURN
20 POKES,159:GOSUB18:POKES,135:GOSUB18:POKES,0:RET
URN
*22 POKES,195:GOSUB18:POKES,207:GOSUB18:POKES,215:G
OSUB18:POKES,225:GOSUB18:GOSUB18:POKES,0:RETURN
28 GOSUB10
30 N=INT(RND(0)*6):H=8+INT(RND(0)*11):GOSUB14
32 PRINT"{CLR}{BLU}";N$(N);" WORKS";H;"HOURS":IFN<
3THENPRINT"PER WEEK.{2 SPACES}HE EARNS":GOTO36
34 PRINT"PER WEEK.{2 SPACES}SHE EARNS"
36 PRINTP$;" PER HOUR.":IFN<3THENPRINT"HOW MUCH DO
ES HE EARN":GOTO40
38 PRINT"HOW MUCH DOES SHE EARN";
40 PRINT"IN A WEEK?{DOWN}{BLK}":INPUT"$";D:D1=P*H/
100:IFABS(D-D1)>.001THEN49
42 GOSUB22:PRINT"{4 DOWN}"
44 PRINT"{GRN}PRESS 1 TO TRY AGAIN{8 SPACES}2 TO C
ONTINUE"
45 GETA$:IFA$="1"THEN30
46 IFA$="2"THEN54
47 GOTO45
49 GOSUB20
*50 PRINT"{DOWN}{RED}MULTIPLY";H;"HOURS BY":PRINTP$
;" PER HOUR.":P=H*P:GOSUB15:PRINT"{DOWN}THE ANS
WER IS ";P$
52 PRINT"{3 DOWN}":GOTO28
54 PRINT"{CLR}{BLU}";N=INT(RND(0)*5):H=INT(RND(0)
*11)+8:GOSUB14:PRINTN$(N);" EARNS ";P$;" PER"
56 IFN<3THENPRINT"HOURL.{2 SPACES}HE WORKS";:GOTO60
58 PRINT"HOURL.{2 SPACES}SHE WORKS";

```

Math Competency

```
60 PRINTH:PRINT"HOURS PER WEEK.{2 SPACES}HOW":IFN<
3THENPRINT"MUCH WILL HE EARN IN":GOTO64
62 PRINT"MUCH WILL SHE EARN IN"
64 W=INT(RND(0)*19+2):PRINTW;"WEEKS? {DOWN}{BLK}":I
NPUT"$";D:D1=P*H*W/100:IFABS(D-D1)>.001THEN70
66 GOSUB22:PRINT"{4 DOWN}{GRN}PRESS 1 TO TRY AGAIN
{8 SPACES}2 TO CONTINUE"
67 GETA$:IFA$="1"THEN54
68 IFA$="2"THEN80
69 GOTO67
*70 GOSUB20:PRINT"{DOWN}{RED}MULTIPLY";H;"HOURS BY"
:PRINTP$;" PER HOUR. THEN":PRINT"MULTIPLY BY";W
;"WEEKS."
72 PRINT"{DOWN}THE ANSWER IS ";P=H*P*W:GOSUB15:PR
INTP$:PRINT"{3 DOWN}":GOSUB10:GOTO54
80 J=INT(RND(0)*5):T=INT(RND(0)*5):GOSUB14:W=INT(R
ND(0)*8)+2
82 PRINT"{CLR}{BLU}";T$(T);" EARNED ";P$:PRINT"LAS
T WEEK":PRINTJ$(J)
84 IFT<3THENPRINT"{DOWN}IF HE EARNED THIS":GOTO88
86 PRINT"{DOWN}IF SHE EARNED THIS"
88 PRINT"AMOUNT EVERY WEEK,":PRINT"WHAT WOULD THE
{SPACE}TOTAL{2 SPACES}INCOME BE FOR";W;"WEEKS?
{DOWN}{BLK}"
90 INPUT"$";D:D1=P*W/100:IFABS(D-D1)>.001THEN97
92 GOSUB22:PRINT"{4 DOWN}{GRN}TRY AGAIN? (Y/N)"
93 GETA$:IFA$="Y"THEN80
94 IFA$="N"THEN110
95 GOTO93
97 GOSUB20:PRINT"{DOWN}{RED}MULTIPLY ";P$;" PER":P
RINT"WEEK BY";W;"WEEKS.":P=P*W:GOSUB15
98 PRINT"{DOWN}THE ANSWER IS ";P$:PRINT"{DOWN}":GO
SUB10:GOTO80
100 DATASAM,DOING ODD JOBS.,PAUL,JOE,MOWING LAWN.
,JACK,BOB,TENDING CHILDREN,MARK,ANN
102 DATARUNNING ERRANDS,JANE,SUE,DOING HOUSEWORK,J
UDY,KAY,DELIVERING ADS,DAWN
110 PRINT"{CLR}{BLU}":END
```

Earning Money on the Color Computer works this way:

Line	Function
10	Clear screen, PRINT title screen, DIMension variables, READ from DATA the name N\$, job J\$, and name T\$, and branch past subroutines.
20-50	Wait for ENTER key to be pressed.
60-100	Convert price to dollar value.
110-160	Clear screen and PRINT problem, randomly choosing name, hours, and wage.

- 170 Receive answer, calculate correct answer, and compare answers.
- 180 Procedure for correct answer.
- 190-220 PRINT option to try again and branch appropriately.
- 230 If answer is incorrect, show how to get correct answer and return for another problem.
- 240-300 PRINT problem, receive student's answer, calculate correct answer, and compare answers.
- 310 PRINT message for correct answer.
- 320-340 PRINT option to try again and branch appropriately.
- 350-360 If answer is incorrect, show how to get correct answer and return for another problem of same type.
- 370-410 PRINT problem.
- 420 Receive student's answer, calculate correct answer, and compare answers.
- 430 PRINT message for correct answer.
- 440-460 PRINT option to try again and branch appropriately.
- 470 If answer is incorrect, show how to get correct answer and return for another problem of same type.
- 480-490 DATA statements.
- 500 Clear screen and END.

Program 6-8. Earning Money

TRS-80 Color Computer; MC-10; TRS-80 Model I (with changes)

```

10 CLS:PRINT "MATH COMPETENCY":PRINT "EARNING MONEY"
   :DIMN$(5),J$(5),T$(5):FORI=0TO5:READN$(I),J$(I)
   ,T$(I):NEXT:FORI=1TO500:NEXT:GOTO110
20 PRINT @495,"PRESS <ENTER>";
30 A$=INKEY$:IF A$="" THEN 30
40 IF ASC(A$)<>13 THEN 30
50 RETURN
60 P=100+25*RND(10)
70 P$=STR$(P)
80 IF LEN(P$)=2 THEN P$=" 0"+RIGHT$(P$,1)
90 PR$=RIGHT$(P$,2):PL$=LEFT$(P$,LEN(P$)-2):IFLEN(
   PL$)<2THENPL$=" "+PL$
100 P$="$"+PL$+"."+"PR$:RETURN
110 CLS:N=RND(6)-1:H=8+RND(10):GOSUB60:PRINTN$(N);
   " WORKS";H;"HOURS PER WEEK.":IFN<3THENPRINT"HE
   EARNS ";:GOTO130
120 PRINT "SHE EARNS ";
130 PRINT P$;" PER HOUR.":PRINT
140 IF N<3 THEN PRINT "HOW MUCH DOES HE EARN":GOTO
   160
150 PRINT "HOW MUCH DOES SHE EARN"

```

Math Competency

```
160 PRINT "IN A WEEK? $";
170 INPUTD:D1=P*H/100:IF ABS(D-D1)>.001 THEN 230
180 PRINT"CORRECT!":FOR C=1 TO 100:NEXT
190 PRINT @480,"TRY AGAIN? (Y/N)";
200 A$=INKEY$:IF A$="Y" THEN 110
210 IF A$="N" THEN 240
220 GOTO200
230 PRINT:PRINT"MULTIPLY";H;"HOURS BY ";P$:PRINT"P
ER HOUR.":P=H*P:GOSUB70:PRINT"THE ANSWER IS ";
P$:GOSUB20:GOTO110
240 CLS:N=RND(6)-1:H=8+RND(10):GOSUB60:PRINTN$(N);
" EARNS ";P$;" PER HOUR.":PRINT
250 IF N<3 THEN PRINT "HE WORKS";:GOTO 270
260 PRINT "SHE WORKS";
270 PRINT H;"HOURS PER WEEK.":PRINT
280 IF N<3 THEN PRINT "HOW MUCH WILL HE EARN IN":G
OTO 300
290 PRINT "HOW MUCH WILL SHE EARN IN"
300 W=RND(19)+1:PRINT W;"WEEKS? $";:INPUTD:D1=P*H
*W/100:IFABS(D-D1)>.001THEN350
310 PRINT"CORRECT!"
320 PRINT @480,"TRY AGAIN? (Y/N)";
330 A$=INKEY$:IF A$="Y" THEN 240
340 IF A$="N" THEN 370 ELSE 330
350 PRINT:PRINT"MULTIPLY";H;"HOURS BY ";P$:PRINT"P
ER HOUR.":PRINT:PRINT"THEN MULTIPLY BY";W;"WEE
KS.":P=H*P*W:GOSUB70
360 PRINT:PRINT"THE ANSWER IS ";P$:GOSUB20:GOTO240
370 CLS:J=RND(6)-1:T=RND(6)-1:GOSUB60:W=RND(8)+1
380 PRINT T$(T);" EARNED ";P$;" LAST WEEK":PRINT J
$(J);".":PRINT
390 IF T<3 THEN PRINT "IF HE EARNED THIS AMOUNT":G
OTO 410
400 PRINT "IF SHE EARNED THIS AMOUNT"
410 PRINT "EVERY WEEK, WHAT WOULD THE TOTALINCOME
BE FOR";W;"WEEKS?"
420 INPUT "$";D:D1=P*W/100:IF ABS(D-D1)>.001 THEN
470
430 PRINT "CORRECT!"
440 PRINT @480,"TRY AGAIN? (Y/N)";
450 A$=INKEY$:IF A$="Y" THEN 370
460 IF A$="N" THEN 500 ELSE 450
470 PRINT:PRINT"MULTIPLY ";P$;" PER WEEK":PRINT"BY
";W;"WEEKS.":P=P*W:GOSUB70:PRINT:PRINT"THE ANS
WER IS ";P$:GOSUB20:GOTO370
480 DATA SAM,DOING ODD JOBS,JOHN,JOE,MOWING LAWNS,
ANDY,BOB,TENDING CHILDREN,MARK,ANN
490 DATA RUNNING ERRANDS,LENA,SUE,DOING HOUSEWORK,
AURA,KAY,DELIVERING ADS,DAWN
500 CLS:END
```

For the TRS-80 Model I version, make the following changes in the above listing:

Change line 20 to PRINT:PRINT "PRESS <ENTER>";
Delete @480, in lines 190, 320, and 440.

Here's the TI-99/4A version:

Line	Function
100-130	Clear screen and PRINT title screen.
140	DIMension variables for name N\$, job J\$, and name T\$.
150-190	READ from DATA for names and jobs.
200	Branch past subroutines.
210-240	Wait for ENTER to be pressed.
250-270	Play sound for incorrect answer.
280-320	Play music for correct answer.
330-360	Convert number to dollars.
370-520	Clear screen, PRINT problem with random name, hours, and wage.
530	Receive answer.
540	Calculate correct answer.
550-560	If answer is correct, play sound effect.
570-600	PRINT option to try again and branch appropriately.
610-670	If answer is incorrect, play sound and show how to get correct answer; return for another problem.
680-840	Clear screen and PRINT problem with random name, wage, and weeks.
850	Receive answer.
860	Calculate correct answer.
870-880	If answer is correct, play appropriate sound.
890-920	PRINT option to try again and branch appropriately.
930-1020	If answer is incorrect, play sound, show how to get correct answer, and then return for another problem.
1030-1160	Clear screen and PRINT problem with random name, wage, and weeks.
1170	Receive answer.
1180	Calculate correct answer.
1190-1200	If answer is correct, play musical effect.
1210-1240	PRINT option to try again and branch appropriately.
1250-1320	If answer is incorrect, play sound and show how to get correct answer, and return for another problem of the same type.
1330-1340	Clear screen and END.

Math Competency

Program 6-9. Earning Money

TI-99/4A

```
100 CALL CLEAR
110 PRINT TAB(6);"MATH COMPETENCY"
120 PRINT ::TAB(7);"EARNING MONEY"
130 PRINT :::::::::::
140 DIM N$(5),J$(5),T$(5)
150 FOR I=0 TO 5
160 READ N$(I),J$(I),T$(I)
170 NEXT I
180 DATA SAM,DOING ODD JOBS,JOHN,JOE,MOWING LAWN,
ANDY,BOB,TENDING CHILDREN,MARK,ANN
190 DATA RUNNING ERRANDS,LENA,SUE,DOING HOUSEWORK,
AURA,KAY,DELIVERING ADS,DAWN
200 GOTO 370
210 PRINT :TAB(15);"PRESS <ENTER>";
220 CALL KEY(0,K,S)
230 IF K<>13 THEN 220
240 RETURN
250 CALL SOUND(100,330,2)
260 CALL SOUND(150,262,2)
270 RETURN
280 CALL SOUND(100,262,2)
290 CALL SOUND(100,330,2)
300 CALL SOUND(100,392,2)
310 CALL SOUND(200,523,2)
320 RETURN
330 P=100+25*INT(RND*10)
340 P$=STR$(P)
350 P$="$"&SEG$(P$,1,LEN(P$)-2)&". "&SEG$(P$,LEN(P$)
)-1,2)
360 RETURN
370 CALL CLEAR
380 RANDOMIZE
390 N=INT(RND*6)
400 H=8+INT(RND*11)
410 GOSUB 330
420 PRINT N$(N);" WORKS";H;"HOURS PER WEEK."
430 IF N<3 THEN 460
440 PRINT : "SHE EARNS ";
450 GOTO 470
460 PRINT : "HE EARNS ";
470 PRINT P$;" PER HOUR."
480 IF N<3 THEN 510
490 PRINT : "HOW MUCH DOES SHE EARN"
500 GOTO 520
510 PRINT : "HOW MUCH DOES HE EARN"
520 PRINT : "IN A WEEK?":
530 INPUT "$":D
540 D1=P*H/100
```

```
550 IF ABS(D-D1)>.001 THEN 610
560 GOSUB 280
570 PRINT ::"TRY AGAIN? (Y/N)"
580 CALL KEY(0,K,S)
590 IF K=89 THEN 370
600 IF K=78 THEN 680 ELSE 580
610 GOSUB 250
620 PRINT : "MULTIPLY";H;"HOURS BY ";P$::"PER HOUR.
    "
630 P=H*P
640 GOSUB 340
650 PRINT : "THE ANSWER IS ";P$
660 GOSUB 210
670 GOTO 370
680 CALL CLEAR
690 RANDOMIZE
700 N=INT(RND*6)
710 H=INT(RND*11)+8
720 GOSUB 330
730 PRINT N$(N); " EARNS ";P$;" PER HOUR."
740 IF N<3 THEN 770
750 PRINT : "SHE WORKS";
760 GOTO 780
770 PRINT : "HE WORKS";
780 PRINT H;"HOURS PER WEEK."
790 IF N<3 THEN 820
800 PRINT : "HOW MUCH WILL SHE EARN IN"
810 GOTO 830
820 PRINT : "HOW MUCH WILL HE EARN IN"
830 W=INT(RND*19)+2
840 PRINT :W;"WEEKS?": :
850 INPUT "$":D
860 D1=P*H*W/100
870 IF ABS(D-D1)>.001 THEN 930
880 GOSUB 280
890 PRINT ::"TRY AGAIN? (Y/N)"
900 CALL KEY(0,K,S)
910 IF K=89 THEN 680
920 IF K=78 THEN 1030 ELSE 900
930 GOSUB 250
940 PRINT : "MULTIPLY";H;"HOURS BY"
950 PRINT :P$;" PER HOUR. "
960 PRINT : "THEN MULTIPLY BY";W;" WEEKS."
970 PRINT : "THE ANSWER IS ";
980 P=H*P*W
990 GOSUB 340
1000 PRINT P$:::
1010 GOSUB 210
1020 GOTO 680
1030 CALL CLEAR
```

Math Competency

```
1040 J=INT(RND*6)
1050 T=INT(RND*6)
1060 GOSUB 330
1070 W=INT(RND*8)+2
1080 PRINT T$(T);" EARNED ";P$;" LAST WEEK"
1090 PRINT :J$(J);"."
1100 IF T<3 THEN 1130
1110 PRINT : "IF SHE EARNED THIS AMOUNT"
1120 GOTO 1140
1130 PRINT : "IF HE EARNED THIS AMOUNT"
1140 PRINT : "EVERY WEEK, WHAT WOULD THE"
1150 PRINT : "TOTAL INCOME BE FOR"
1160 PRINT :W;"WEEKS?": :
1170 INPUT "$":D
1180 D1=P*W/100
1190 IF ABS(D-D1)>.001 THEN 1250
1200 GOSUB 280
1210 PRINT : "TRY AGAIN? (Y/N)";
1220 CALL KEY(0,K,S)
1230 IF K=89 THEN 1030
1240 IF K=78 THEN 1330 ELSE 1220
1250 GOSUB 250
1260 PRINT : "MULTIPLY ";P$;" PER WEEK"
1270 PRINT : "BY";W;" WEEKS."
1280 P=P*W
1290 GOSUB 340
1300 PRINT : "THE ANSWER IS ";P$: :
1310 GOSUB 210
1320 GOTO 1030
1330 CALL CLEAR
1340 END
```

Weekly Expenses

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

Multiplication and addition are used in this word problem drill program. One of three summer activities is listed on the screen, along with an itemized list of costs for a week. Summer camp, a sports clinic, or a canyon resort are the week-long activities, and you have to add the total weekly cost of your stay. As soon as you enter the correct answer for the addition section of the problem, the program will ask you to calculate a total, given those weekly expenses, for a random number of weeks. When you respond to this part of the problem, you can either continue with another question or quit the program.

All three computer versions work much the same way:

Line	Function
20	DIMENSION variables for activity A\$, name N\$, and phrase P\$.
30-140	READ from DATA the names, phrases, and activity title and itemized fees.
150	Clear screen.
160-190	Randomly choose one of three places and PRINT introduction.
200-350	Randomly calculate fees for each item and total cost T.
360-390	Get answer.
400-430	If answer is incorrect, give correct total.
440-470	Randomly choose number of weeks and ask for cost.
480-490	Receive answer.
500-600	If answer is incorrect, show how to get right answer and return for another problem.
610	If answer is correct, PRINT message.
620-670	PRINT option to try again and branch appropriately.
680-720	Convert cost to dollars for printing.
730-740	Clear screen and END.

Program 6-10. Weekly Expenses

VIC-20

```

20 DIMA$(3,5),N$(6),P$(3)
30 FORA=1TO3
40 READN$(A),N$(A+3),P$(A)
50 FORB=1TO5
60 READA$(A,B)
70 NEXTB
80 NEXTA
90 DATALENA,ANDY,IS GOING TO
100 DATACAMP BEAVER,CAMP FEE,HORSE RIDING,TENNIS L
    ESSONS,CRAFT SUPPLIES
110 DATAAURA,BILL,WILL ATTEND
120 DATASPORTS CLINIC,TUITION,UNIFORM FEE,EQUIPMEN
    T FEE,SPECIAL EVENTS
130 DATAMARY,JOHN,WILL STAY AT
140 DATALOGAN CANYON,CAMP FEE,T-SHIRTS,ACTIVITY FE
    E,SUPPLIES
150 PRINT"{CLR}"
160 I=INT(3*RND(0))+1
170 PRINT"HERE ARE THE EXPENSES"
180 PRINT"FOR ONE WEEK AT":PRINTA$(I,1);"."
190 PRINT
200 C=1000*INT(5*RND(0))+1)+4000
210 GOSUB680
220 PRINTA$(I,2);TAB(15);C$
    
```

Math Competency

```
230 T=C
240 C=25*INT(12*RND(0)+1)+275
250 GOSUB680
260 PRINTA$(I,3);TAB(15);C$
270 T=T+C
280 C=50*INT(8*RND(0)+1)+150
290 GOSUB680
300 PRINTA$(I,4);TAB(15);C$
310 T=T+C
320 C=50*INT(5*RND(0)+1)+50
330 GOSUB680
340 PRINTA$(I,5);TAB(15);C$
350 T=T+C
360 PRINT
370 PRINT"TOTAL EXPENSES FOR":PRINT"ONE WEEK = ";
380 INPUT"$";A
390 PRINT
400 IFA=T/100THEN440
410 PRINT"ADD THE NUMBERS TO"
420 PRINT"GET TOTAL{SHIFT-SPACE}";T/100
430 PRINT
440 W=INT(7*RND(0)+2)
450 PRINTN$(INT(6*RND(0)+1));" ";P$(INT(3*RND(0)+1
))
460 PRINTA$(I,1):PRINT"FOR";W;"WEEKS."
470 PRINT"WHAT WILL IT COST?"
480 INPUT "$";A
490 PRINT
500 IF A=W*T/100 THEN 610
510 PRINT"MULTIPLY TOTAL EXPENSEPER WEEK TIMES";W;
"WEEKS "
530 C=T*W
540 GOSUB680
550 PRINT"$";T/100;"*";W;"= ";C$
560 PRINT
570 PRINT"PRESS RETURN."
580 GET E$:IFE$=""THEN580
590 IFASC(E$)=13THEN150
600 GOTO580
610 PRINT"CORRECT!"
620 PRINT
630 PRINT"PRESS 1 TRY AGAIN"
640 PRINTTAB(6);"2 END PROGRAM"
650 GETE$:IFE$="1"THEN150
660 IFE$<>"2"THEN650
670 GOTO730
680 M$=STR$(C)
690 IFLEN(M$)=4 THEN M$="{SHIFT-SPACE}"+M$
700 L=LEN(M$)
710 C$="$"+LEFT$(M$,L-2)+". "+RIGHT$(M$,2)
```

```
720 RETURN
730 PRINT "{CLR}"
740 END
```

Program 6-11. Weekly Expenses

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
20 DIM A$(3,5),N$(6),P$(3)
30 FOR A=1 TO 3
40 READ N$(A),N$(A+3),P$(A)
50 FOR B=1 TO 5
60 READ A$(A,B)
70 NEXT B
80 NEXT A
90 DATA LENA,ANDY,IS GOING TO
100 DATA CAMP BEAVER,CAMP FEE,HORSE RIDING,TENNIS
    LESSONS,CRAFT SUPPLIES
110 DATA AURA,BILL,WILL ATTEND
120 DATA SPORTS CLINIC,TUITION,UNIFORM FEE,EQUIPME
    NT FEE,SPECIAL EVENTS
130 DATA MARY,JOHN,WILL STAY AT
140 DATA LOGAN CANYON,CAMP FEE, T-SHIRTS,ACTIVITY
    FEE,SUPPLIES
150 CLS
160 I=RND(3)
170 PRINT "HERE ARE THE EXPENSES FOR"
180 PRINT "ONE WEEK AT ";A$(I,1);"."
190 PRINT
200 C=RND(5)*1000+4000
210 GOSUB 680
220 PRINT A$(I,2),C$
230 T=C
240 C=RND(12)*25+275
250 GOSUB 680
260 PRINT A$(I,3),C$
270 T=T+C
280 C=RND(8)*50+150
290 GOSUB 680
300 PRINT A$(I,4),C$
310 T=T+C
320 C=RND(5)*50+50
330 GOSUB 680
340 PRINT A$(I,5),C$
350 T=T+C
360 PRINT
370 PRINT "TOTAL EXPENSES FOR ONE WEEK ="
380 INPUT "$";A
390 PRINT
400 IF A=T/100 THEN 440
410 PRINT "ADD THE NUMBERS TO GET"
```

Math Competency

```
420 PRINT "TOTAL ";T/100
430 PRINT
440 W=RND(7)+1
450 PRINT N$(RND(6));" ";P$(RND(3))
460 PRINT A$(I,1);" FOR";W;"WEEKS."
470 PRINT "WHAT WILL IT COST?"
480 INPUT "$";A
490 PRINT
500 IF A=W*T/100 THEN 610
510 PRINT "MULTIPLY TOTAL EXPENSES"
520 PRINT "PER WEEK TIMES";W;"WEEKS:"
530 C=T*W
540 GOSUB 680
550 PRINT "$";T/100;"*";W;"= ";C$
560 PRINT
570 PRINT "PRESS ENTER."
580 E$=INKEY$:IF E$="" THEN 580
590 IF ASC(E$)=13 THEN 150
600 GOTO 580
610 PRINT "CORRECT!"
620 PRINT
630 PRINT "PRESS 1 ANOTHER PROBLEM"
640 PRINT TAB(6);"2 END PROGRAM"
650 E$=INKEY$:IF E$="1" THEN 150
660 IF E$<>"2" THEN 650
670 GOTO 730
680 M$=STR$(C)
690 IF LEN(M$)=4 THEN M$=" "+M$
700 L=LEN(M$)
710 C$="$"+LEFT$(M$,L-2)+". "+RIGHT$(M$,2)
720 RETURN
730 CLS
740 END
```

Program 6-12. Weekly Expenses

TI-99/4A

```
20 DIM A$(3,5),N$(6),P$(3)
30 FOR A=1 TO 3
40 READ N$(A),N$(A+3),P$(A)
50 FOR B=1 TO 5
60 READ A$(A,B)
70 NEXT B
80 NEXT A
90 DATA LENA,ANDY,IS GOING TO
100 DATA CAMP BEAVER,CAMP FEE,HORSE RIDING,TENNIS,
    CRAFTS
110 DATA AURA,BILL,WILL ATTEND
120 DATA SPORTS CLINIC,TUITION,UNIFORM FEE,EQUIPME
    NT FEE,SPECIAL EVENT
```

Math Competency

```
130 DATA MARY,JOHN,WILL STAY AT
140 DATA LOGAN CANYON,CAMP FEE,T-SHIRTS,ACTIVITY F
    EE,SUPPLIES
150 CALL CLEAR
155 RANDOMIZE
160 I=INT(3*RND+1)
170 PRINT "HERE ARE THE EXPENSES FOR"
180 PRINT "ONE WEEK AT ";A$(I,1);"."
190 PRINT
200 C=INT(5*RND+1)*1000+4000
210 GOSUB 680
220 PRINT A$(I,2),C$
230 T=C
240 C=INT(12*RND+1)*25+275
250 GOSUB 680
260 PRINT A$(I,3),C$
270 T=T+C
280 C=INT(8*RND+1)*50+150
290 GOSUB 680
300 PRINT A$(I,4),C$
310 T=T+C
320 C=INT(5*RND+1)*50+50
330 GOSUB 680
340 PRINT A$(I,5),C$
350 T=T+C
360 PRINT
370 PRINT "TOTAL EXPENSES FOR ONE WEEK"
380 INPUT "$":A
390 PRINT
400 IF A=T/100 THEN 440
410 PRINT "ADD THE NUMBERS TO GET"
420 PRINT "TOTAL ";T/100
430 PRINT
440 W=INT(7*RND+2)
450 PRINT N$(INT(6*RND+1));" ";P$(INT(3*RND+1))
460 PRINT A$(I,1);" FOR";W;"WEEKS."
470 PRINT "WHAT WILL IT COST?"
480 INPUT "$":A
490 PRINT
500 IF A=W*T/100 THEN 610
510 PRINT "MULTIPLY TOTAL EXPENSES"
520 PRINT "PER WFEK TIMES";W;"WEEKS:"
530 C=T*W
540 GOSUB 680
550 PRINT "$";T/100;"*";W;"=" ";C$
560 PRINT
570 PRINT "PRESS ENTER."
580 CALL KFY(0,K,S)
590 IF K=13 THEN 150
```


Math Competency

```
600 GOTO 580
610 PRINT "CORRECT!"
620 PRINT
630 PRINT "PRESS 1 ANOTHER PROBLEM"
640 PRINT TAB(7);"2 END PROGRAM"
650 CALL KEY(0,K,S)
660 IF K=49 THEN 150
670 IF K=50 THEN 730 ELSE 650
680 M$=STR$(C)
690 IF LEN(M$)>3 THEN 700
695 M$=" "&M$
700 L=LEN(M$)
710 C$="$"&SEG$(M$,1,L-2)&". "&SEG$(M$,L-1,2)
720 RETURN
730 CALL CLEAR
740 END
```

Saving Money

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

"Saving Money" gives you a chance to brush up on your division skills. The program first chooses one of six names to use in the problem. One of three major items, with its price, is selected. The computer also randomly selects a number to represent the number of weeks you'll have to save for this item. You have to figure out how much money you'd need to save each week to purchase the item.

The price of the item is determined by a base value multiplied by a random number in a range from 1 to F. The base value, multiplier, and value F are all READ from DATA statements in the program.

Here's how the program works:

Line	Line	Function
VIC-20	Color Computer and TI-99/4A	
30	120	Clear screen.
40-110	130-200	READ from DATA six names N\$, the item A\$ with a base price B, a multiplier M, and random factor F.
120	210	Clear screen.
130-140	220-230	Randomly PRINT name.
150-160	240-250	Randomly choose one of three items.
170-190	260-280	Calculate and PRINT cost of item.
200-270	290-360	Randomly choose number of weeks and PRINT problem.

280	370	Calculate correct answer.
290	380	Receive answer.
300-370	390-460	If answer is incorrect, show how to get right answer, and return for another problem.
380-400	470-490	PRINT message for correct answer.
410-460	500-550	PRINT option to try again and branch appropriately.
470-510	560-600	Convert cost into dollars.
520-530	610	END.

Program 6-13. Saving Money

VIC-20

For Commodore 64 version, add line 25: 25 POKE 53281,1

```
30 PRINT "{CLR}"
40 FOR C=1 TO 6
50 READ N$(C)
60 NEXT C
70 FOR C=1 TO 3
80 READ A$(C),B(C),M(C),F(C)
90 NEXT C
100 DATA SALLY,HEIDI,NANCY,BRENT,DAVID,CHUCK
110 DATA BIKE,80,5,7,STEREO,90,5,14,COMPUTER,100,1
    0,10
120 PRINT "{CLR}{BLU}"
130 R6=INT(6*RND(0)+1)
140 PRINT N$(R6); " WANTS TO BUY A"
150 R3=INT(3*RND(0)+1)
160 PRINT A$(R3); "."
170 T=B(3)+M(R3)*INT(F(R3)*RND(0)+1)
180 GOSUB 480
190 PRINT "IT WILL COST ";C$; "."
200 P$="HE"
210 IF R6>=4 THEN 230
220 P$="SHE"
230 W=10*INT(4*RND(0)+1)
240 PRINT "IF ";P$; " SAVES FOR"
245 PRINTW;"WEEKS,"
250 PRINT "HOW MUCH WILL ";N$(R6)"
260 PRINT "NEED TO SAVE EACH"
265 PRINT "WEEK?"
270 PRINT "{BLK}"
280 S=(T/W)/100
290 INPUT ANS
300 IF ABS(ANS-S)<.01 THEN 380
310 PRINT "{RED}TOTAL COST ";C$
320 PRINT "DIVIDED BY";W;"WKS ="
```

Math Competency

```
330 GOSUB 470
340 PRINTC$
350 PRINT"{GRN}"
360 INPUT "PRESS <RETURN>";E$
370 GOTO 120
380 PRINT
390 PRINT"{RED}CORRECT!"
400 PRINT
410 PRINT"{BLK}ANOTHER PROBLEM?"
420 PRINT"(Y/N)"
430 GET E$
440 IF E$="Y" THEN 120
450 IF{2 SPACES}E$="N" THEN 520
460 GOTO 430
470 T=S+.005
480 T=INT(100*T)
490 T$=STR$(T)
500 C$="$"+LEFT$(T$,LEN(T$)-2)+". "+RIGHT$(T$,2)
510 RETURN
520 PRINT"{CLR}{BLU}"
530 END
```

Program 6-14. Saving Money

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
120 CLS
130 FOR C=1 TO 6
140 READ N$(C)
150 NEXT C
160 FOR C=1 TO 3
170 READ A$(C),B(C),M(C),F(C)
180 NEXT C
190 DATA SALLY,HEIDI,NANCY,BRENT,DAVID,CHUCK
200 DATA BIKE,80,5,7,STEREO,90,5,14,COMPUTER,100,1
0,10
210 CLS
220 R6=RND(6)
230 PRINT N$(R6);" WANTS TO BUY A"
240 R3=RND(3)
250 PRINT A$(R3);"."
260 T=B(R3)+M(R3)*RND(F(R3))
270 GOSUB 570
280 PRINT "IT WILL COST ";C$;"."
290 P$="HE"
300 IF R6>=4 THEN 320
310 P$="SHE"
320 W=10*RND(4)
330 PRINT "IF ";P$;" SAVES FOR";W;"WEEKS,"
340 PRINT "HOW MUCH WILL ";N$(R6);" NEED"
350 PRINT "TO SAVE EACH WEEK?"
```

```
360 PRINT
370 S=(T/W)/100
380 INPUT ANS
390 IF ABS(ANS-S)<.01 THEN 470
400 PRINT "TOTAL COST ";C$
410 PRINT "DIVIDED BY";W;"WEEKS ="
420 GOSUB 560
430 PRINT C$
440 PRINT
450 INPUT "PRESS <ENTER>";E$
460 GOTO 210
470 PRINT
480 PRINT "CORRECT!"
490 PRINT
500 PRINT "ANOTHER PROBLEM?"
510 PRINT "(Y/N)"
520 E$=INKEY$
530 IF E$="Y" THEN 210
540 IF E$="N" THEN 610
550 GOTO 520
560 T=S
570 T=INT(100*T)
580 T$=STR$(T)
590 C$="$"+LEFT$(T$,LEN(T$)-2)+"."+RIGHT$(T$,2)
600 RETURN
610 END
```

Program 6-15. Saving Money

TI-99/4A

```
120 CALL CLEAR
130 FOR C=1 TO 6
140 READ N$(C)
150 NEXT C
160 FOR C=1 TO 3
170 READ A$(C),B(C),M(C),F(C)
180 NEXT C
190 DATA SALLY,HEIDI,NANCY,BRENT,DAVID,CHUCK
200 DATA BIKE,80,5,7,STEREO,90,5,14,COMPUTER,100,1
   0,10
210 CALL CLEAR
215 RANDOMIZE
220 R6=INT(6*RND+1)
230 PRINT N$(R6);" WANTS TO BUY A"
240 R3=INT(3*RND+1)
250 PRINT A$(R3);"."
260 T=B(R3)+M(R3)*INT(F(R3)*RND+1)
270 GOSUB 570
280 PRINT "IT WILL COST ";C$;"."
```

Math Competency

```
290 P$="HE"
300 IF R6>=4 THEN 320
310 P$="SHE"
320 W=10*INT(4*RND+1)
330 PRINT "IF ";P$;" SAVES FOR";W;" WEEKS, "
340 PRINT "HOW MUCH WILL ";N$(R6);" NEED"
350 PRINT "TO SAVE EACH WEEK?"
360 PRINT
370 S=(T/W)/100
380 INPUT ANS
390 IF ABS(ANS-S)<.01 THEN 470
400 PRINT "TOTAL COST ";C$
410 PRINT "DIVIDED BY";W;" WEEKS ="
420 GOSUB 560
430 PRINT C$
440 PRINT
450 INPUT "PRESS <ENTER>":E$
460 GOTO 210
470 PRINT
480 PRINT "CORRECT!"
490 PRINT
500 PRINT "ANOTHER PROBLEM?"
510 PRINT "(Y/N)"
520 CALL KEY(0,K,S)
530 IF K=89 THEN 210
540 IF K=78 THEN 610
550 GOTO 520
560 T=S
570 T=INT(100*T)
580 T$=STR$(T)
590 C$="$"&SEG$(T$,1,LEN(T$)-2)&". "&SEG$(T$,LEN(T$)
)-1,2)
600 RETURN
610 END
```

Chapter 7
Homework
Helpers

Handwritten text, possibly a name or title, appearing as a faint watermark or bleed-through from the reverse side of the page.

Chapter 7

Homework Helpers

Computers work best when they do repetitive tasks. Checking your homework is just that — it's repetitive, time-consuming, and often boring. The programs in this chapter, which have all been adapted for use on one of three computers, are designed to quickly check your homework.

You should do your problems on paper, as usual, showing all the steps in getting an answer, and then use these programs to check your results. These programs, because they are quite simple, show only an answer, not the step-by-step process of solving the problem.

After each problem, you can press 1 for another problem, or 2 to end the program.

Division

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

Most division problems can be checked and corrected with a calculator. However, if there is a remainder, a calculator converts it to a decimal equivalent. When you first learn division, you learn to divide and specify a remainder if the number does not divide evenly. "Division" keeps the answer in quotient plus remainder form. When you enter the divisor and dividend, the quotient and remainder are printed.

The INTeger function is used in these programs to calculate the whole number part of the quotient. The remainder is then the difference between the original number and the integer portion, multiplied by the divisor.

Program 7-1. Division

VIC-20

```
110 PRINT "{CLR}"
120 PRINT "DIVISION--REMAINDER"
130 PRINT
140 PRINT
150 PRINTTAB(11); "QUOTIENT"
160 PRINTTAB(9); "-----"
170 PRINT " DIVISOR ) DIVIDEND"
180 PRINT:PRINT
190 INPUT "DIVISOR:{2 SPACES}";D
200 IF D<>0 THEN 230
210 PRINT "DIVISOR CANNOT = 0"
220 GOTO 190
230 INPUT "DIVIDEND: ";N
240 Q=INT(N/D)
```



```
170 PRINT " DIVISOR %DIVIDEND"::::
180 INPUT "DIVISOR: ":I
190 IF I<>0 THEN 220
200 PRINT : "SORRY, DIVISOR CANNOT = 0"::
210 GOTO 180
220 INPUT "DIVIDEND: ":D
230 Q=INT(D/I)
240 R=D-Q*I
250 PRINT : "QUOTIENT =";Q;" R";R
260 PRINT ::: "CHOOSE: "
270 PRINT " 1 ANOTHER PROBLEM"
280 PRINT " 2 END PROGRAM"
290 CALL KEY(0,K,S)
300 IF K=49 THEN 130
310 IF K<>50 THEN 290
320 CALL CLEAR
330 END
```

All Factors

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

This program will find all the factors of a number that you enter. (Factors are all the possible divisors that will yield an integer answer.) For example, the number 12 has the factors 12, 6, 4, 3, 2, and 1. You must enter a number greater than 1. Large numbers will take longer for the computer to factor.

Program 7-4. Find All Factors

VIC-20

```
110 PRINT "{CLR}"
120 PRINT "FINDING ALL FACTORS"
130 PRINT
140 PRINT "ENTER NUMBER TO BE"
150 INPUT "FACTORED: ";N
160 IF N>1 THEN 190
170 PRINT "ENTER NUMBER > 1"
180 GOTO 130
190 PRINT "FACTORS OF";N;" ARE"
200 PRINT N;
210 B=INT(N/2+1)
220 FOR C=2 TO B
230 IF N/C<>INT(N/C) THEN 280
240 B=N/C
250 PRINT B;
260 IF B=1 THEN 300
270 IF B=2 THEN 290
280 NEXT C
```

Homework Helpers

```
290 PRINT " 1"
300 PRINT
310 PRINT"CHOOSE:"
320 PRINT" 1 ANOTHER PROBLEM"
330 PRINT" 2 END PROGRAM"
340 GETA$
350 IF A$="1" THEN 110
360 IF A$<>"2" THEN 340
370 PRINT"{CLR}":END
```

Program 7-5. Find All Factors

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
110 CLS
120 PRINT"FINDING ALL FACTORS"
130 PRINT
140 PRINT"ENTER NUMBER TO BE"
150 INPUT "FACTORED: ";N
160 IF N>1 THEN 190
170 PRINT"ENTER NUMBER > 1"
180 GOTO130
190 PRINT"FACTORS OF";N;"ARE"
200 PRINT N;
210 B=INT(N/2+1)
220 FOR C=2 TO B
230 IF N/C<>INT(N/C) THEN 280
240 B=N/C
250 PRINT B;
260 IF B=1 THEN 300
270 IF B=2 THEN 290
280 NEXT C
290 PRINT " 1"
300 PRINT
310 PRINT "CHOOSE:"
320 PRINT " 1 ANOTHER PROBLEM"
330 PRINT " 2 END PROGRAM"
340 A$=INKEY$
350 IF A$="1" THEN 110
360 IF A$<>"2" THEN 340
370 CLS:END
```

Program 7-6. Find All Factors

TI-99/4A

```
110 CALL CLEAR
120 PRINT "FINDING ALL FACTORS"
130 PRINT
140 PRINT "ENTER NUMBER TO BE"
150 INPUT "FACTORED: ":N
160 IF N>1 THEN 190
```

```
170 PRINT "ENTER NUMBER > 1"
180 GOTO 130
190 PRINT : "FACTORS OF";N;"ARE"
200 PRINT N;
210 B=INT(N/2+1)
220 FOR C=2 TO B
230 IF N/C<>INT(N/C)THEN 280
240 B=N/C
250 PRINT B;
260 IF B=1 THEN 300
270 IF B=2 THEN 290
280 NEXT C
290 PRINT " 1"
300 PRINT
310 PRINT "CHOOSE:"
320 PRINT " 1 ANOTHER PROBLEM"
330 PRINT " 2 END PROGRAM"
340 CALL KEY(0,K,S)
350 IF K=49 THEN 110
360 IF K<>50 THEN 340
370 CALL CLEAR
380 END
```

Prime Factors

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

"Prime Factors" returns a list of the prime factors for a number you enter. Another term for this process is *complete factorization*. All prime factors multiplied together will yield the original number. For example, the prime factors of the number 12 are 2, 2, and 3. You must enter a number greater than 1.

Program 7-7. Prime Factors

VIC-20

```
110 PRINT"{CLR}"
120 PRINT"PRIME FACTORS OR"
130 PRINT"COMPLETE FACTORIZATION"
140 PRINT
150 PRINT"ENTER NUMBER TO BE"
160 INPUT "FACTORED: ";N
170 IF N>1 THEN 200
180 PRINT"NUMBER MUST BE > 1"
190 GOTO 140
200 PRINT
210 PRINT"THE PRIME FACTORS ARE:"
220 G=INT(N/2)
230 FOR I=2 TO G
240 IF N/I<>INT(N/I) THEN 290
```

Homework Helpers

```
250 N=N/I
260 G=N
270 PRINT I;
280 GOTO 230
290 NEXT I
300 IF N=1 THEN 320
310 PRINT N
320 PRINT
330 PRINT
340 PRINT"CHOOSE:"
350 PRINT" 1 ANOTHER PROBLEM"
360 PRINT" 2 END PROGRAM"
370 GET A$
380 IF A$="1" THEN 110
390 IF A$<>"2" THEN 370
400 PRINT"{CLR}":END
```

Program 7-8. Prime Factors

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
110 CLS
120 PRINT"PRIME FACTORS OR"
130 PRINT"COMPLETE FACTORIZATION"
140 PRINT
150 PRINT"ENTER NUMBER TO BE"
160 INPUT "FACTORED: ";N
170 IF N>1 THEN 200
180 PRINT"NUMBER MUST BE > 1"
190 GOTO 140
200 PRINT
210 PRINT"THE PRIME FACTORS ARE:"
220 G=INT(N/2)
230 FOR I=2 TO G
240 IF N/I<>INT(N/I) THEN 290
250 N=N/I
260 G=N
270 PRINT I;
280 GOTO 230
290 NEXT I
300 IF N=1 THEN 320
310 PRINT N
320 PRINT
330 PRINT
340 PRINT"CHOOSE:"
350 PRINT" 1 ANOTHER PROBLEM"
360 PRINT" 2 END PROGRAM"
370 A$=INKEY$
380 IF A$="1" THEN 110
390 IF A$<>"2" THEN 370
400 CLS:END
```

Program 7-9. Prime Factors

TI-99/4A

```
110 CALL CLEAR
120 PRINT "PRIME FACTORS OR"
130 PRINT "COMPLETE FACTORIZATION"
140 PRINT
150 PRINT "ENTER NUMBER TO BE"
160 INPUT "FACTORED: ":N
170 IF N>1 THEN 200
180 PRINT "NUMBER MUST BE > 1"
190 GOTO 140
200 PRINT
210 PRINT "THE PRIME FACTORS ARE:"
220 G=INT(N/2)
230 FOR I=2 TO G
240 IF N/I<>INT(N/I)THEN 290
250 N=N/I
260 G=N
270 PRINT I;
280 GOTO 230
290 NEXT I
300 IF N=1 THEN 320
310 PRINT N
320 PRINT
330 PRINT
340 PRINT "CHOOSE:"
350 PRINT " 1 ANOTHER PROBLEM"
360 PRINT " 2 END PROGRAM"
370 CALL KEY(0,K,S)
380 IF K=49 THEN 110
390 IF K<>50 THEN 370
400 CALL CLEAR
410 END
```

Greatest Common Factor

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

This program will find the greatest common factor (GCF) of two numbers. This mathematical concept is usually introduced before the concept of simplifying fractions. For example, if the two numbers entered are 12 and 18, the greatest common factor is 6. Both numbers can be evenly divided by 6. Although they may also both be divided by 3 or 2, 6 is the largest factor. You enter two numbers, and the computer will show the greatest common factor.

Homework Helpers

Program 7-10. Greatest Common Factor

VIC-20

```
110 PRINT"{CLR}"
120 PRINT"GREATEST COMMON FACTOR"
130 PRINT"OF TWO NUMBERS"
140 PRINT
150 INPUT "FIRST NUMBER ";M
160 IF M>1 THEN 190
170 PRINT"ENTER NUMBER > 1"
180 GOTO 140
190 IF{2 SPACES}M<10000 THEN 220
200 PRINT"MUST BE < 10000"
210 GOTO140
220 PRINT
230 INPUT "SECOND NUMBER ";N
240 IF N>1 THEN 270
250 PRINT"ENTER NUMBER > 1"
260 GOTO 220
270 IF N<10000 THEN 300
280 PRINT"MUST BE < 10000"
290 GOTO 220
300 PRINT
310 PRINT"GREATEST COMMON FACTOR ={SHIFT-SPACE}";
320 IF M=N THEN G=M:GOTO 450
330 IF{2 SPACES}M<N THEN 370
340 MM=M
350 M=N
360 N=MM
370 FOR I=1 TO M
380 IF (M/I)<>INT(M/I)THEN 430
390 J=M/I
400 IF (N/J)<>INT(N/J)THEN 430
410 G=J
420 GOTO 450
430 NEXT I
440 G=1
450 PRINT G
460 PRINT
470 PRINT"CHOOSE:"
480 PRINT" 1 ANOTHER PROBLEM"
490 PRINT" 2 END PROGRAM"
500 GET A$
510 IF A$="1" THEN 110
520 IF A$<>"2" THEN 500
530 PRINT"{CLR}":END
```

Program 7-11. Greatest Common Factor

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
110 CLS
120 PRINT"GREATEST COMMON FACTOR"
130 PRINT"OF TWO NUMBERS"
140 PRINT
150 INPUT "FIRST NUMBER ";M
160 IF M>1 THEN 190
170 PRINT"ENTER NUMBER > 1"
180 GOTO 140
190 IF M<10000 THEN 220
200 PRINT"MUST BE < 10000"
210 GOTO 140
220 PRINT
230 INPUT "SECOND NUMBER ";N
240 IF N>1 THEN 270
250 PRINT"ENTER NUMBER > 1"
260 GOTO 220
270 IF N<10000 THEN 300
280 PRINT"MUST BE < 10000"
290 GOTO 220
300 PRINT
310 PRINT"GREATEST COMMON FACTOR ="
320 IF M=N THEN G=M:GOTO 450
330 IF M<N THEN 370
340 MM=M
350 M=N
360 N=MM
370 FOR I=1 TO M
380 IF (M/I)<>INT(M/I)THEN 430
390 J=M/I
400 IF (N/J)<>INT(N/J)THEN 430
410 G=J
420 GOTO 450
430 NEXT I
440 G=1
450 PRINT G
460 PRINT
470 PRINT"CHOOSE:"
480 PRINT" 1 ANOTHER PROBLEM"
490 PRINT" 2 END PROGRAM"
500 A$=INKEY$
510 IF A$="1" THEN 110
520 IF A$<>"2" THEN 500
530 CLS:END
```


Homework Helpers

Program 7-12. Greatest Common Factor

TI-99/4A

```
110 CALL CLEAR
120 PRINT "GREATEST COMMON FACTOR"
130 PRINT "OF TWO NUMBERS"
140 PRINT
150 INPUT "FIRST NUMBER ":M
160 IF M>1 THEN 190
170 PRINT "ENTER NUMBER > 1"
180 GOTO 140
190 IF M<10000 THEN 220
200 PRINT "MUST BE < 10000"
210 GOTO 140
220 PRINT
230 INPUT "SECOND NUMBER ":N
240 IF N>1 THEN 270
250 PRINT "ENTER NUMBER > 1"
260 GOTO 220
270 IF N<10000 THEN 300
280 PRINT "MUST BE < 10000"
290 GOTO 220
300 PRINT
310 PRINT "GREATEST COMMON FACTOR ="
320 IF M<>N THEN 350
330 G=M
340 GOTO 470
350 IF M<N THEN 390
360 MM=M
370 M=N
380 N=MM
390 FOR I=1 TO M
400 IF (M/I)<>INT(M/I)THEN 450
410 J=M/I
420 IF (N/J)<>INT(N/J)THEN 450
430 G=J
440 GOTO 470
450 NEXT I
460 G=1
470 PRINT G
480 PRINT
490 PRINT "CHOOSE:"
500 PRINT " 1 ANOTHER PROBLEM"
510 PRINT " 2 END PROGRAM"
520 CALL KEY(0,K,S)
530 IF K=49 THEN 110
540 IF K<>50 THEN 520
550 CALL CLEAR
560 END
```

Least Common Multiple

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

Use this program to find the least common multiple of two or three numbers. First, enter 2 or 3 for the number of given numbers. Next enter the numbers. The program will return the least common multiple.

For example, enter three numbers, such as 12, 18, and 6. The least common multiple is 36. 36 may be divided evenly by 12, 18, and 6, and is the smallest number that can be divided by all three.

This concept is usually introduced to students before they learn about adding and subtracting fractions with unlike denominators. Another name for this concept is *smallest common denominator*.

Program 7-13. Least Common Multiple

VIC-20

```
110 PRINT "{CLR}"
120 PRINT "LEAST COMMON MULTIPLE"
130 PRINT "OF 2 OR 3 NUMBERS"
140 PRINT
150 PRINT "HOW MANY NUMBERS--2, 3?"
160 GET A$:IF A$="" THEN 160
170 IF ASC(A$)<50 THEN 160
180 IF ASC(A$)>51 THEN 160
190 PRINT A$;" NUMBERS"
200 FOR C=1 TO VAL(A$)
210 PRINT
220 PRINT "NUMBER";C;
230 INPUT N(C)
240 IF N(C)>1 THEN 270
250 PRINT "NUMBER MUST BE > 1"
260 GOTO 210
270 IF N(C)<1000 THEN 300
280 PRINT "NUMBER MUST BE < 1000"
290 GOTO 210
300 NEXT C
310 IF A$="3" THEN 440
320 IF N(1)<>N(2) THEN 350
330 L=N(1)
340 GOTO 630
350 IF N(1)<N(2) THEN 390
360 NN=N(1)
370 N(1)=N(2)
380 N(2)=NN
390 FOR C=1 TO N(1)
```

Homework Helpers

```
400 IF C*N(2)/N(1)=INT(C*N(2)/N(1)) THEN L=C*N(2):
    GOTO 630
410 NEXT C
420 L=N(1)*N(2)
430 GOTO 630
440 S=0
450 FOR C=1 TO 2
460 IF N(C)<=N(C+1) THEN 510
470 NN=N(C)
480 N(C)=N(C+1)
490 N(C+1)=NN
500 S=1
510 NEXT C
520 IF S=1 THEN 440
530 FOR C=1 TO N(2)
540 F=C*N(3)
550 IF (F/N(1)=INT(F/N(1))) AND (F/N(2)=INT(F/N(2)
)) THEN L=F:GOTO 630
560 NEXT C
570 M=N(2)*N(3)
580 FOR C=1 TO N(1)
590 F=C*M
600 IF F/N(1)=INT(F/N(1)) THEN L=F:GOTO 630
610 NEXT C
620 L=M*N(1)
630 PRINT
640 PRINT"LEAST COMMON MULTIPLE IS ";
650 PRINTL
660 PRINT
670 PRINT"CHOOSE:"
680 PRINT" 1 ANOTHER PROBLEM"
690 PRINT" 2 END PROGRAM"
700 GET A$
710 IF A$="1" THEN 110
720 IF A$<>"2" THEN 700
730 PRINT"{CLR}":END
```

Program 7-14. Least Common Multiple

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
110 CLS
120 PRINT"LEAST COMMON MULTIPLE"
130 PRINT"OF 2 OR 3 GIVEN NUMBERS"
140 PRINT
150 PRINT"HOW MANY NUMBERS--2 OR 3?"
160 A$=INKEY$:IF A$="" THEN 160
170 IF ASC(A$)<50 THEN 160
180 IF ASC(A$)>51 THEN 160
190 PRINT A$;" NUMBERS"
```

Homework Helpers

```
200 FOR C=1 TO VAL(A$)
210 PRINT
220 PRINT"NUMBER";C;
230 INPUT N(C)
240 IF N(C)>1 THEN 270
250 PRINT"NUMBER MUST BE > 1"
260 GOTO 210
270 IF N(C)<1000 THEN 300
280 PRINT"NUMBER MUST BE < 1000"
290 GOTO 210
300 NEXT C
310 IF A$="3" THEN 440
320 IF N(1)<>N(2) THEN 350
330 L=N(1)
340 GOTO 630
350 IF N(1)<N(2) THEN 390
360 NN=N(1)
370 N(1)=N(2)
380 N(2)=NN
390 FOR C=1 TO N(1)
400 IF C*N(2)/N(1)=INT(C*N(2)/N(1)) THEN L=C*N(2):
    GOTO 630
410 NEXT C
420 L=N(1)*N(2)
430 GOTO 630
440 S=0
450 FOR C=1 TO 2
460 IF N(C)<=N(C+1) THEN 510
470 NN=N(C)
480 N(C)=N(C+1)
490 N(C+1)=NN
500 S=1
510 NEXT C
520 IF S=1 THEN 440
530 FOR C=1 TO N(2)
540 F=C*N(3)
550 IF (F/N(1)=INT(F/N(1))) AND (F/N(2)=INT(F/N(2)
)) THEN L=F:GOTO 630
560 NEXT C
570 M=N(2)*N(3)
580 FOR C=1 TO N(1)
590 F=C*M
600 IF F/N(1)=INT(F/N(1)) THEN L=F:GOTO 630
610 NEXT C
620 L=M*N(1)
630 PRINT
640 PRINT"LEAST COMMON MULTIPLE IS"
650 PRINTL
660 PRINT
```

Homework Helpers

```
670 PRINT"CHOOSE:"
680 PRINT" 1 ANOTHER PROBLEM"
690 PRINT" 2 END PROGRAM"
700 A$=INKEY$
710 IF A$="1" THEN 110
720 IF A$<>"2" THEN 700
730 CLS:END
```

Program 7-15. Least Common Multiple

TI-99/4A

```
110 CALL CLEAR
120 PRINT "LEAST COMMON MULTIPLE"
130 PRINT "OF 2 OR 3 GIVEN NUMBERS"
140 PRINT
150 PRINT "HOW MANY NUMBERS --2 OR 3?"
160 CALL KEY(0,K,S)
170 IF K<50 THEN 160
180 IF K>51 THEN 160
190 PRINT K-48;" NUMBERS"
200 FOR C=1 TO K-48
210 PRINT
220 PRINT "NUMBER";C;
230 INPUT N(C)
240 IF N(C)>1 THEN 270
250 PRINT "NUMBER MUST BE > 1"
260 GOTO 210
270 IF N(C)<1000 THEN 300
280 PRINT "NUMBER MUST BE < 1000"
290 GOTO 210
300 NEXT C
310 IF K=51 THEN 460
320 IF N(1)<>N(2)THEN 350
330 L=N(1)
340 GOTO 690
350 IF N(1)<N(2)THEN 390
360 NN=N(1)
370 N(1)=N(2)
380 N(2)=NN
390 FOR C=1 TO N(1)
400 IF C*N(2)/N(1)<>INT(C*N(2)/N(1))THEN 430
410 L=C*N(2)
420 GOTO 690
430 NEXT C
440 L=N(1)*N(2)
450 GOTO 690
460 SW=0
470 FOR C=1 TO 2
480 IF N(C)<=N(C+1)THEN 530
490 NN=N(C)
```

```
500 N(C)=N(C+1)
510 N(C+1)=NN
520 SW=1
530 NEXT C
540 IF SW=1 THEN 460
550 FOR C=1 TO N(2)
560 F=C*N(3)
570 IF (F/N(1)=INT(F/N(1)))+(F/N(2)=INT(F/N(2)))<>
-2 THEN 600
580 L=F
590 GOTO 690
600 NEXT C
610 M=N(2)*N(3)
620 FOR C=1 TO N(1)
630 F=C*M
640 IF F/N(1)<>INT(F/N(1))THEN 670
650 L=F
660 GOTO 690
670 NEXT C
680 L=M*N(1)
690 PRINT : "LEAST COMMON MULTIPLE IS"
700 PRINT L
710 PRINT : "CHOOSE:"
720 PRINT " 1 ANOTHER PROBLEM"
730 PRINT " 2 END PROGRAM"
740 CALL KEY(0,K,S)
750 IF K=49 THEN 110
760 IF K<>50 THEN 740
770 CALL CLEAR
780 END
```

Equivalent Fractions

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

"Equivalent Fractions" can quickly find the unknown in problems such as $1/2=?/8$. The fractions are of the form: $A/B=C/D$. You first press the letter for the unknown, then enter values for the other three numbers as the computer prompts you. The equivalent fractions will be printed.

Program 7-16. Equivalent Fractions

VIC-20

```
110 PRINT "{CLR}"
120 PRINT TAB(7); "A{5 SPACES}C"
130 PRINT TAB(7); "-{2 SPACES}={2 SPACES}-"
140 PRINT TAB(7); "B{5 SPACES}D"
150 PRINT:PRINT
```

Homework Helpers

```
160 PRINT"WHICH IS THE UNKNOWN?"
170 PRINT"CHOOSE A, B, C, OR D"
180 GETA$:IF A$=""THEN 180
190 K=ASC(A$)
200 IF K<65 THEN 180
210 IF K>68 THEN 180
220 PRINT
230 ON K-64 GOTO 240,350,490,600
240 INPUT "ENTER B =";B
250 IF B<>0 THEN 280
260 PRINT"B CANNOT = 0{DOWN}"
270 GOTO 240
280 INPUT "ENTER C =";C
290 INPUT "ENTER D =";D
300 IF D<>0 THEN 330
310 PRINT"D CANNOT = 0{DOWN}"
320 GOTO 290
330 A=INT(100*(B*C/D+.005))/100
340 GOTO 730
350 INPUT "ENTER A =";A
360 IF A<>0 THEN 390
370 PRINT"SORRY, A<>0{DOWN}"
380 GOTO 350
390 INPUT "ENTER C =";C
400 IF C<>0 THEN 430
410 PRINT"SORRY, C<>0{DOWN}"
420 GOTO 390
430 INPUT "ENTER D =";D
440 IF D<>0 THEN 470
450 PRINT"D CANNOT = 0{DOWN}"
460 GOTO 430
470 B=INT(100*(A*D/C+.005))/100
480 GOTO 730
490 INPUT "ENTER A =";A
500 INPUT "ENTER B =";B
510 IF B<>0 THEN 540
520 PRINT"B CANNOT = 0{DOWN}"
530 GOTO 500
540 INPUT "ENTER D =";D
550 IF D<>0 THEN 580
560 PRINT"D CANNOT = 0{DOWN}"
570 GOTO 540
580 C=INT(100*(A*D/B+.005))/100
590 GOTO 730
600 INPUT "ENTER A =";A
610 IF A<>0 THEN 640
620 PRINT"SORRY, A<>0{DOWN}"
630 GOTO 600
640 INPUT "ENTER B =";B
```

```

650 IF B<>0 THEN 680
660 PRINT"B CANNOT = 0{DOWN}"
670 GOTO 640
680 INPUT "ENTER C =";C
690 IF C<>0 THEN 720
700 PRINT"SORRY, C<>0{DOWN}"
710 GOTO 680
720 D=INT(100*(B*C/A+.005))/100
730 PRINT"{CLR}"
740 PRINT TAB(4);A,C
750 PRINT TAB(4);"----- = -----"
760 PRINT TAB(4);B,D
770 PRINT
780 PRINT"CHOOSE:"
790 PRINT" 1 ANOTHER PROBLEM"
800 PRINT" 2 END PROGRAM"
810 GET A$
820 IF A$="1" THEN 110
830 IF A$<>"2" THEN 810
840 PRINT"{CLR}":END

```

Program 7-17. Equivalent Fractions

TRS-80 Color Computer; MC-10; TRS-80 Model I

```

110 CLS
120 PRINT TAB(10);"A{5 SPACES}C"
130 PRINT TAB(10);"- = -"
140 PRINT TAB(10);"B{5 SPACES}D"
150 PRINT:PRINT
160 PRINT"WHICH IS THE UNKNOWN?"
170 PRINT"CHOOSE A, B, C, OR D"
180 A$=INKEY$:IF A$=""THEN 180
190 K=ASC(A$)
200 IF K<65 THEN 180
210 IF K>68 THEN 180
220 PRINT
230 ON K-64 GOTO 240,350,490,600
240 INPUT "ENTER B =";B
250 IF B<>0 THEN 280
260 PRINT"B CANNOT = 0":PRINT
270 GOTO 240
280 INPUT "ENTER C =";C
290 INPUT "ENTER D =";D
300 IF D<>0 THEN 330
310 PRINT"D CANNOT = 0":PRINT
320 GOTO 290
330 A=INT(100*(B*C/D+.005))/100
340 GOTO 730
350 INPUT "ENTER A =";A
360 IF A<>0 THEN 390

```


Homework Helpers

```
370 PRINT"SORRY, A<>0":PRINT
380 GOTO 350
390 INPUT "ENTER C =";C
400 IF C<>0 THEN 430
410 PRINT"SORRY, C<>0":PRINT
420 GOTO 390
430 INPUT "ENTER D =";D
440 IF D<>0 THEN 470
450 PRINT"D CANNOT = 0":PRINT
460 GOTO430
470 B=INT(100*(A*D/C+.005))/100
480 GOTO 730
490 INPUT "ENTER A =";A
500 INPUT "ENTER B =";B
510 IF B<>0 THEN 540
520 PRINT"B CANNOT = 0":PRINT
530 GOTO 500
540 INPUT "ENTER D =";D
550 IF D<>0 THEN 580
560 PRINT"D CANNOT = 0":PRINT
570 GOTO 540
580 C=INT(100*(A*D/B+.005))/100
590 GOTO 730
600 INPUT "ENTER A =";A
610 IF A<>0 THEN 640
620 PRINT"SORRY, A<>0":PRINT
630 GOTO 600
640 INPUT "ENTER B =";B
650 IF B<>0 THEN 680
660 PRINT"B CANNOT = 0":PRINT
670 GOTO 640
680 INPUT "ENTER C =";C
690 IF C<>0 THEN 720
700 PRINT"SORRY, C<>0":PRINT
710 GOTO 680
720 D=INT(100*(B*C/A+.005))/100
730 CLS
740 PRINT TAB(7);A,C
750 PRINT TAB(7);"----- = -----"
760 PRINTTAB(7);B,D
770 PRINT
780 PRINT"CHOOSE:"
790 PRINT" 1 ANOTHER PROBLEM"
800 PRINT" 2 END PROGRAM"
810 A$=INKEY$
820 IF A$="1" THEN 110
830 IF A$<>"2" THEN 810
840 CLS:END
```

Program 7-18. Equivalent Fractions

TI-99/4A

```
110 CALL CLEAR
120 PRINT TAB(10);"A{5 SPACES}C"
130 PRINT TAB(10);"- = -"
140 PRINT TAB(10);"B{5 SPACES}D":::
150 PRINT "WHICH IS THE UNKNOWN?"
160 PRINT "CHOOSE A, B, C, OR D.":::
170 CALL KEY(0,K,S)
180 IF (K<65)+(K>68)THEN 170
190 ON K-64 GOTO 200,310,450,560
200 INPUT "ENTER B = ":B
210 IF B<>0 THEN 240
220 PRINT "B CANNOT = 0"::
230 GOTO 200
240 INPUT "ENTER C = ":C
250 INPUT "ENTER D = ":D
260 IF D<>0 THEN 290
270 PRINT "D CANNOT = 0"::
280 GOTO 250
290 A=INT(100*(B*C/D+.005))/100
300 GOTO 690
310 INPUT "ENTER A = ":A
320 IF A<>0 THEN 350
330 PRINT "SORRY, A<>0"::
340 GOTO 310
350 INPUT "ENTER C = ":C
360 IF C<>0 THEN 390
370 PRINT "SORRY, C<>0"::
380 GOTO 350
390 INPUT "ENTER D = ":D
400 IF D<>0 THEN 430
410 PRINT "D CANNOT = 0"::
420 GOTO 390
430 B=INT(100*(A*D/C+.005))/100
440 GOTO 690
450 INPUT "ENTER A = ":A
460 INPUT "ENTER B = ":B
470 IF B<>0 THEN 500
480 PRINT "B CANNOT = 0"::
490 GOTO 460
500 INPUT "ENTER D = ":D
510 IF D<>0 THEN 540
520 PRINT "D CANNOT = 0"::
530 GOTO 500
540 C=INT(100*(A*D/B+.005))/100
550 GOTO 690
560 INPUT "ENTER A = ":A
570 IF A<>0 THEN 600
```

Homework Helpers

```
580 PRINT "SORRY, A<>0":  
590 GOTO 560  
600 INPUT "ENTER B = ":B  
610 IF B<>0 THEN 640  
620 PRINT "B CANNOT = 0":  
630 GOTO 600  
640 INPUT "ENTER C = ":C  
650 IF C<>0 THEN 680  
660 PRINT "SORRY, C<>0":  
670 GOTO 640  
680 D=INT(100*(B*C/A+.005))/100  
690 CALL CLEAR  
700 PRINT TAB(7);A,C  
710 PRINT TAB(7);"----- = -----"  
720 PRINT TAB(7);B,D  
730 PRINT :::"CHOOSE:"  
740 PRINT " 1 ANOTHER PROBLEM"  
750 PRINT " 2 END PROGRAM"  
760 CALL KEY(0,K,S)  
770 IF K=49 THEN 110  
780 IF K<>50 THEN 760  
790 CALL CLEAR  
800 END
```

Simplifying Fractions

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

Enter a numerator, then a denominator. The computer simplifies or reduces the fraction to its lowest terms, or tells you if it cannot be simplified. This algorithm first checks which is larger, the numerator or the denominator. The first factor to be checked is the smaller number. If either the numerator or the denominator is an odd number, even factors will be eliminated by choosing a step size of -2 in the checking loop.

Although you usually reduce fractions starting with the smallest factors, the computer starts with the largest possible factor and decreases for each check.

Program 7-19. Simplifying Fractions

VIC-20

```
110 PRINT"{CLR}"  
120 PRINT"SIMPLIFYING FRACTIONS"  
130 PRINT  
140 INPUT "NUMERATOR ={2 SPACES}";N  
150 INPUT "DENOMINATOR =" ;D  
160 IF D<>0 THEN 190  
170 PRINT"DENOMINATOR CANNOT = 0"
```

```
180 GOTO 150
190 IF N<>0 THEN 220
200 PRINT:PRINTN;"/";D;" = 0"
210 GOTO 390
220 IF D>N THEN 250
230 L=D
240 GOTO 260
250 L=N
260 S=-2
270 IF D/2<>INT(D/2)THEN 290
280 IF N/2=INT(N/2)THEN 300
290 S=-1
300 FOR C=L TO 2 STEP S
310 A=N/C
320 IF A<>INT(A) THEN 350
330 B=D/C
340 IF B=INT(B)THEN 380
350 NEXT C
360 PRINT:PRINTN;"/";D:PRINT" CANNOT BE SIMPLIFIED
"
370 GOTO 390
380 PRINT:PRINTN;"/";D;" = ";A;"/";B
390 PRINT"{DOWN}CHOOSE:"
400 PRINT" 1 ANOTHER PROBLEM"
410 PRINT" 2 END PROGRAM"
420 GET A$
430 IF A$="1" THEN 110
440 IF A$<>"2" THEN 420
450 PRINT"{CLR}":END
```

Program 7-20. Simplifying Fractions

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
110 CLS
120 PRINT"SIMPLIFYING FRACTIONS"
130 PRINT
140 INPUT "NUMERATOR = ";N
150 INPUT "DENOMINATOR =";D
160 IF D<>0 THEN 190
170 PRINT"DENOMINATOR CANNOT = 0"
180 GOTO 150
190 IF N<>0 THEN 220
200 PRINT:PRINTN;"/";D;" = 0"
210 GOTO 390
220 IF D>N THEN 250
230 L=D
240 GOTO 260
250 L=N
260 S=-2
270 IF D/2<>INT(D/2)THEN 290
```

Homework Helpers

```
280 IF N/2=INT(N/2)THEN 300
290 S=-1
300 FOR C=L TO 2 STEP S
310 A=N/C
320 IF A<>INT(A) THEN 350
330 B=D/C
340 IF B=INT(B)THEN 380
350 NEXT C
360 PRINT:PRINTN;"/";D;" CANNOT BE SIMPLIFIED"
370 GOTO390
380 PRINT:PRINTN;"/";D;" = ";A;"/";B
390 PRINT:PRINT"CHOOSE:"
400 PRINT" 1 ANOTHER PROBLEM"
410 PRINT" 2 END PROGRAM"
420 A$=INKEY$
430 IF A$="1" THEN 110
440 IF A$<>"2" THEN 420
450 CLS:END
```

Program 7-21. Simplifying Fractions

TI-99/4A

```
110 CALL CLEAR
120 PRINT "SIMPLIFYING FRACTIONS"::::
130 INPUT "NUMERATOR ={3 SPACES}":N
140 INPUT "DENOMINATOR = ":D
150 IF D<>0 THEN 180
160 PRINT "DENOMINATOR CANNOT = 0"::
170 GOTO 140
180 IF D>N THEN 210
190 LIM=D
200 GOTO 220
210 LIM=N
220 S=-2
230 IF D/2<>INT(D/2)THEN 250
240 IF N/2=INT(N/2)THEN 260
250 S=-1
260 FOR C=LIM TO 2 STEP S
270 A=N/C
280 IF A<>INT(A)THEN 310
290 B=D/C
300 IF B=INT(B)THEN 340
310 NEXT C
320 PRINT :N;"/";D;" CANNOT BE SIMPLIFIED"
330 GOTO 350
340 PRINT :N;"/";D;" = ";A;"/";B
350 PRINT :::"CHOOSE:"
360 PRINT " 1 ANOTHER PROBLEM"
370 PRINT " 2 END PROGRAM"
```

```
380 CALL KEY(0,K,S)
390 IF K=49 THEN 110
400 IF K<>50 THEN 380
410 CALL CLEAR
420 END
```

Multiplying Fractions

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

You can use this program to multiply from two to nine fractions. First, press a number key to represent the total number of fractions, then enter each numerator and denominator as you're prompted. The program multiplies the fractions and simplifies the final answer. To simplify fractions, the denominators are arranged in order, and the largest denominator is the first factor tried.

Program 7-22. Multiplying Fractions

VIC-20

```
110 PRINT"{CLR}"
120 PRINT"MULTIPLYING FRACTIONS"
130 PRINT"{DOWN}HOW MANY FRACTIONS? ";
140 GET A$:IF A$=""THEN 140
150 IF ASC(A$)<50 THEN 140
160 IF ASC(A$)>57 THEN 140
170 PRINTA$
180 C=VAL(A$)
190 NT=1
200 DT=1
210 FOR I=1 TO C
220 PRINT"{DOWN}FRACTION";I
230 INPUT" NUMERATOR{3 SPACES}";N(I)
240 NT=NT*N(I)
250 INPUT" DENOMINATOR ";D(I)
260 IF D(I)<>0 THEN 290
270 PRINT"DENOMINATOR CANNOT = 0"
280 GOTO 250
290 DT=DT*D(I)
300 NEXT I
310 PRINT"{2 DOWN}** MULTIPLY **{DOWN}"
320 FOR I=1 TO C
330 PRINTN(I);"/";D(I)
340 NEXT I
350 PRINT"*****"
360 FOR I=1 TO C
370 A=NT/D(I)
380 IF A<>INT(A)THEN 430
390 B=DT/D(I)
```

Homework Helpers

```
400 IF B<>INT(B)THEN 430
410 NT=A
420 DT=B
430 NEXT I
440 SW=0
450 FOR I=1 TO C-1
460 IF D(I)<=D(I+1) THEN 510
470 DD=D(I)
480 D(I)=D(I+1)
490 D(I+1)=DD
500 SW=1
510 NEXT I
520 IF SW=1 THEN 440
530 L=D(C)
540 FOR I=L TO 2 STEP -1
550 A=NT/I
560 IF A<>INT(A)THEN 590
570 B=DT/I
580 IF B=INT(B)THEN 620
590 NEXT I
600 A=NT
610 B=DT
620 IF A>=B THEN 650
630 PRINT:PRINTA;"/";B
640 GOTO 710
650 W=INT(A/B)
660 R=A-W*B
670 IF R<>0 THEN 700
680 PRINT W
690 GOTO 710
700 PRINTW:"{3 SPACES}";R;"/";B
710 PRINT"{DOWN}CHOOSE:"
720 PRINT" 1 ANOTHER PROBLEM"
730 PRINT" 2 END PROGRAM"
740 GET A$
750 IF A$="1" THEN 110
760 IF A$<>"2" THEN 740
770 PRINT"{CLR}":END
```

Program 7-23. Multiplying Fractions

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
110 CLS
120 PRINT"MULTIPLYING FRACTIONS"
130 PRINT:PRINT"HOW MANY FRACTIONS?";
140 A$=INKEY$:IF A$=""THEN 140
150 IF ASC(A$)<50 THEN 140
160 IF ASC(A$)>57 THEN 140
170 PRINTA$
180 C=VAL(A$)
```

Homework Helpers

```
190 NT=1
200 DT=1
210 FOR I=1 TO C
220 PRINT"FRACTION";I
230 INPUT"  NUMERATOR = ";N(I)
240 NT=NT*N(I)
250 INPUT"  DENOMINATOR =";D(I)
260 IF D(I)<>0 THEN 290
270 PRINT"DENOMINATOR CANNOT = 0":PRINT
280 GOTO 250
290 DT=DT*D(I)
300 NEXT I
310 PRINT:PRINT:PRINT"*** MULTIPLY ***":PRINT
320 FOR I=1 TO C
330 PRINTN(I);"/";D(I)
340 NEXT I
350 PRINT"-----"
360 FOR I=1 TO C
370 A=NT/D(I)
380 IF A<>INT(A)THEN 430
390 B=DT/D(I)
400 IF B<>INT(B)THEN 430
410 NT=A
420 DT=B
430 NEXT I
440 SW=0
450 FOR I=1 TO C-1
460 IF D(I)<=D(I+1) THEN 510
470 DD=D(I)
480 D(I)=D(I+1)
490 D(I+1)=DD
500 SW=1
510 NEXT I
520 IF SW=1 THEN 440
530 L=D(C)
540 FOR I=L TO 2 STEP -1
550 A=NT/I
560 IF A<>INT(A)THEN 590
570 B=DT/I
580 IF B<=INT(B)THEN 620
590 NEXT I
600 A=NT
610 B=DT
620 IF A>=B THEN 650
630 PRINT:PRINTA;"/";B
640 GOTO 710

650 W=INT(A/B)
660 R=A-W*B
670 IF R<>0 THEN 700
```


Homework Helpers

```
680 PRINT W
690 GOTO 710
700 PRINTW;"{3 SPACES}";R;"/";B
710 PRINT:PRINT"CHOOSE:"
720 PRINT" 1 ANOTHER PROBLEM"
730 PRINT" 2 END PROGRAM"
740 A$=INKEY$
750 IF A$="1" THEN 110
760 IF A$<>"2" THEN 740
770 CLS:END
```

Program 7-24. Multiplying Fractions

TI-99/4A

```
110 CALL CLEAR
120 PRINT "MULTIPLYING FRACTIONS"::::
130 PRINT "HOW MANY FRACTIONS?"::::
140 CALL KEY(0,K,S)
150 IF (K<50)+(K>57)THEN 140
160 CALL HCHAR(21,23,K)
170 C=K-48
180 NT=1
190 DT=1
200 FOR I=1 TO C
210 PRINT "FRACTION";I
220 INPUT "{4 SPACES}NUMERATOR ={3 SPACES}":N(I)
230 NT=NT*N(I)
240 INPUT "{4 SPACES}DENOMINATOR = ":D(I)
250 IF D(I)<>0 THEN 280
260 PRINT "DENOMINATOR CANNOT BE ZERO.":
270 GOTO 240
280 DT=DT*D(I)
290 NEXT I
300 PRINT :::"** MULTIPLY **":
310 FOR I=1 TO C
320 PRINT STR$(N(I));"/";STR$(D(I))
330 NEXT I
340 PRINT "-----"
350 FOR I=1 TO C
360 A=NT/D(I)
370 IF A<>INT(A)THEN 420
380 B=DT/D(I)
390 IF B<>INT(B)THEN 420
400 NT=A
410 DT=B
420 NEXT I
430 SW=0
440 FOR I=1 TO C-1
450 IF D(I)<=D(I+1)THEN 500
460 DD=D(I)
```

```
470 D(I)=D(I+1)
480 D(I+1)=DD
490 SW=1
500 NEXT I
510 IF SW=1 THEN 430
520 L=D(C)
530 FOR I=L TO 2 STEP -1
540 A=NT/I
550 IF A<>INT(A)THEN 580
560 B=DT/I
570 IF B=INT(B)THEN 610
580 NEXT I
590 A=NT
600 B=DT
610 IF A>=B THEN 640
620 PRINT ::STR$(A);"/";STR$(B)
630 GOTO 700
640 W=INT(A/B)
650 R=A-W*B
660 IF R<>0 THEN 690
670 PRINT W
680 GOTO 700
690 PRINT W;" {3 SPACES}";STR$(R);"/";STR$(B)
700 PRINT :::"CHOOSE:";" 1 ANOTHER PROBLEM"
710 PRINT " 2 END PROGRAM";
720 CALL KEY(0,K,S)
730 IF K=49 THEN 110
740 IF K<>50 THEN 720
750 CALL CLEAR
760 END
```

Dividing Fractions

Computers: VIC-20; TRS-80 Color Computer; MC-10; TRS-80 Model I;
TI-99/4A

You enter the numerator and denominator of each fraction, and the first fraction is divided by the second. The final answer is printed in simplified form.

Program 7-25. Dividing Fractions

VIC-20

```
110 PRINT"{CLR}"
120 PRINT"THE FIRST FRACTION IS DIVIDED BY THE"
130 PRINT"SECOND FRACTION:"
140 PRINT"(N1/D1) / (N2/D2){DOWN}"
150 INPUT "NUMERATOR 1{SHIFT-SPACE}";N1
160 INPUT "DENOMINATOR 1 ";D1
170 IF D1<>0 THEN 200
180 PRINT"SORRY CANNOT = 0"
```

Homework Helpers

```
190 GOTO 160
200 PRINT
210 INPUT "NUMERATOR 2 ";N2
220 IF N2<>0 THEN 250
230 PRINT"SORRY CANNOT = 0"
240 GOTO 210
250 INPUT "DENOMINATOR 2 ";D2
260 IF D2<>0 THEN 290
270 PRINT"SORRY CANNOT = 0"
280 GOTO 250
290 PRINT
300 TN=N1*D2
310 TD=D1*N2
320 PRINTN1;"/";D1
330 PRINT"*****"
340 PRINTN2;"/";D2
350 PRINT"{DOWN}EQUALS"
360 IF N1<>0 THEN 390
370 PRINT"0"
380 GOTO 580
390 IF TN<TD THEN PL=TN:GOTO 410
400 PL=TD
410 FOR P=PL TO 2 STEP -1
420 A=TN/P
430 IF A<>INT(A) THEN 460
440 B=TD/P
450 IF B=INT(B) THEN 490
460 NEXT P
470 A=TN
480 B=TD
490 IF A>=B THEN 520
500 PRINTA;"/";B
510 GOTO 580
520 IF B=1 THEN 570
530 C=INT(A/B)
540 R=A-C*B
550 PRINT C;" {2 SPACES}";R;"/";B
560 GOTO 580
570 PRINT A
580 PRINT"{DOWN}CHOOSE:"
590 PRINT" 1 ANOTHER PROBLEM"
600 PRINT" 2 END PROGRAM"
610 GET A$
620 IF A$="1" THEN 110
630 IF A$<>"2" THEN 610
640 PRINT"{CLR}":END
```

Program 7-26. Dividing Fractions

TRS-80 Color Computer; MC-10; TRS-80 Model I

```
110 CLS
120 PRINT"THE FIRST FRACTION IS DIVIDED"
130 PRINT"BY THE SECOND FRACTION:"
140 PRINT"(N1/D1) / (N2/D2)":PRINT
150 INPUT "NUMERATOR 1 = ";N1
160 INPUT "DENOMINATOR 1 =";D1
170 IF D1<>0 THEN 200
180 PRINT"SORRY CANNOT = 0"
190 GOTO 160
200 PRINT
210 INPUT "NUMERATOR 2 = ";N2
220 IF N2<>0 THEN 250
230 PRINT"SORRY CANNOT = 0"
240 GOTO 210
250 INPUT "DENOMINATOR 2 =";D2
260 IF D2<>0 THEN 290
270 PRINT"SORRY CANNOT = 0"
280 GOTO 250
290 PRINT
300 TN=N1*D2
310 TD=D1*N2
320 PRINTN1;"/";D1
330 PRINT"-----"
340 PRINTN2;"/";D2
350 PRINT:PRINT"EQUALS"
360 IF N1<>0 THEN 390
370 PRINT"0"
380 GOTO 580
390 IF TN<TD THEN PL=TN:GOTO 410
400 PL=TD
410 FOR P=PL TO 2 STEP -1
420 A=TN/P
430 IF A<>INT(A) THEN 460
440 B=TD/P
450 IF B=INT(B) THEN 490
460 NEXT P
470 A=TN
480 B=TD
490 IF A>=B THEN 520
500 PRINTA;"/";B
510 GOTO 580
520 IF B=1 THEN 570
530 C=INT(A/B)
540 R=A-C*B
550 PRINT C;" " " ;R;"/";B
560 GOTO 580
570 PRINT A
```

Homework Helpers

```
580 PRINT:PRINT"CHOOSE:"
590 PRINT" 1 ANOTHER PROBLEM"
600 PRINT" 2 END PROGRAM";
610 A$=INKEY$
620 IF A$="1" THEN 110
630 IF A$<>"2" THEN 610
640 CLS:END
```

Program 7-27. Dividing Fractions

TI-99/4A

```
110 CALL CLEAR
120 PRINT "THE FIRST FRACTION IS"
130 PRINT "DIVIDED BY THE"
140 PRINT "SECOND FRACTION."
150 PRINT ::TAB(10);"N1/D1"
160 PRINT TAB(9);"-----"
170 PRINT TAB(10);"N2/D2":::
180 INPUT "ENTER N1 = ":N1
190 INPUT "ENTER D1 = ":D1
200 IF D1<>0 THEN 230
210 PRINT "DENOMINATOR CANNOT BE ZERO."::
220 GOTO 190
230 PRINT
240 INPUT "ENTER N2 = ":N2
250 IF N2<>0 THEN 280
260 PRINT "SORRY, CANNOT = 0"
270 GOTO 230
280 INPUT "ENTER D2 = ":D2
290 IF D2<>0 THEN 320
300 PRINT "DENOMINATOR CANNOT BE ZERO."::
310 GOTO 280
320 NT=N1*D2
330 DT=D1*N2
340 PRINT :::STR$(N1);"/";STR$(D1)
350 PRINT "-----"
360 PRINT STR$(N2);"/";STR$(D2)
370 PRINT :::"EQUALS"::
380 IF NT<DT THEN 410
390 L=DT
400 GOTO 420
410 L=NT
420 FOR I=L TO 2 STEP -1
430 A=NT/I
440 IF A<>INT(A)THEN 470
450 B=DT/I
460 IF B=INT(B)THEN 500
470 NEXT I
480 A=NT
```

```
490 B=DT
500 IF A>=B THEN 530
510 PRINT :;STR$(A);"/";STR$(B)
520 GOTO 620
530 IF B<>1 THEN 560
540 PRINT :;A
550 GOTO 620
560 C=INT(A/B)
570 R=A-C*B
580 IF R=0 THEN 610
590 PRINT C;" ";STR$(R);"/";STR$(B)
600 GOTO 620
610 PRINT C
620 PRINT :;:"CHOOSE: ":" 1 ANOTHER PROBLEM"
630 PRINT " 2 END PROGRAM";
640 CALL KEY(0,K,S)
650 IF K=49 THEN 110
660 IF K<>50 THEN 640
670 CALL CLEAR
680 END
```

Adding Fractions

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A
"Adding Fractions" has two main options, adding fractions with like denominators, such as $\frac{1}{12} + \frac{5}{12} + \frac{7}{12}$, or adding fractions with unlike denominators, such as $\frac{1}{2} + \frac{1}{3} + \frac{1}{4}$. The program will add up to nine fractions with *like* denominators or up to five fractions with *unlike* denominators.

If you choose the option of like denominators, first press the total number of fractions to be added. Then enter the denominator, followed by the numerators.

If you choose the unlike denominators option, press a number key from 2 to 5 for the number of fractions. The numerator and then the denominator are entered for each fraction.

The fractions are added, the problem is rewritten, and then the answer is printed in simplified terms.

Program 7-28. Adding Fractions

VIC-20

```
110 PRINT"{CLR}"
120 PRINT"ADDING FRACTIONS"
130 PRINT"{DOWN}CHOOSE:"
140 PRINT"1 LIKE DENOMINATORS"
150 PRINT"2 UNLIKE DENOMINATORS"
160 GET A$
```

Homework Helpers

```
170 IF A$="2" THEN 410
180 IF A$<>"1" THEN 160
190 PRINT"{CLR}"
200 CH=1
210 PRINT"ADDING FRACTIONS WITH"
220 PRINT"LIKE DENOMINATORS"
230 PRINT"{DOWN}HOW MANY FRACTIONS? ";
240 GET A$:IF A$="" THEN 240
250 IF ASC(A$)<49 THEN 240
260 IF ASC(A$)>57 THEN 240
270 PRINTA$
280 C=VAL(A$)
290 INPUT "DENOMINATOR{2 SPACES}=" ;DT
300 IF DT<>0 THEN 330
310 PRINT"DENOMINATOR CANNOT = 0"
320 GOTO 290
330 PRINT"{DOWN}ENTER NUMERATORS"
340 NT=0
350 FOR I=1 TO C
360 INPUT N(I)
370 NT=NT+N(I)
380 D(I)=DT
390 NEXT I
400 GOTO 730
410 PRINT"{CLR}"
420 PRINT"ADDING UP TO FIVE"
430 PRINT"FRACTIONS WHICH HAVE"
440 PRINT"UNLIKE DENOMINATORS"
450 PRINT"{DOWN}HOW MANY FRACTIONS? ";
460 GET A$:IF A$="" THEN 460
470 IF ASC(A$)<49 THEN 460
480 IF ASC(A$)>53 THEN 460
490 PRINTA$
500 C=VAL(A$)
510 NT=0
520 DT=1
530 FOR I=1 TO C
540 PRINT "FRACTION";I
550 INPUT " NUMERATOR{3 SPACES}";N(I)
560 INPUT " DENOMINATOR ";D(I)
570 IF D(I)<>0 THEN 600
580 PRINT"DENOMINATOR CANNOT = 0"
590 GOTO 560
600 IF I=1 THEN 640
610 FOR J=1 TO I-1
620 IF D(I)=D(J) THEN 660
630 NEXT J
640 F=D(I)
650 GOTO 670
660 F=1
```

```
670 DT=DT*F
680 NEXT I
690 FOR I=1 TO C
700 F=DT/D(I)
710 NT=NT+N(I)*F
720 NEXT I
730 PRINT"{CLR}"
740 PRINT"ADDING FRACTIONS{DOWN}"
750 FOR I=1 TO C
760 PRINTN(I);"/";D(I)
770 NEXT I
780 PRINT"*****"
790 IF DT>NT THEN 820
800 L=DT
810 GOTO 830
820 L=NT
830 S=-2
840 IF DT/2<>INT(DT/2)THEN 860
850 IF NT/2=INT(NT/2)THEN 870
860 S=-1
870 FOR I=L TO 2 STEP S
880 A=NT/I
890 IF A<>INT(A) THEN 920
900 B=DT/I
910 IF B=INT(B) THEN 950
920 NEXT I
930 A=NT
940 B=DT
950 PRINTA;"/";B
960 IF A<B THEN 1030
970 W=INT(A/B)
980 R=A-W*B
990 IF R<>0 THEN 1020
1000 PRINT"OR";W
1010 GOTO 1030
1020 PRINT"OR";W;"{2 SPACES}";R;"/";B
1030 PRINT"{DOWN}CHOOSE:"
1040 PRINT" 1 ANOTHER PROBLEM"
1050 PRINT" 2 END PROGRAM"
1060 GET A$
1070 IF A$="1" THEN 110
1080 IF A$<>"2" THEN 1060
1090 PRINT"{CLR}":END
```

Program 7-29. Adding Fractions

TRS-80 Color Computer; TRS-80 Model I

```
110 CLS
120 PRINT"ADDING FRACTIONS"
130 PRINT:PRINT"CHOOSE:"
```


Homework Helpers

```
140 PRINT " 1 LIKE DENOMINATORS"
150 PRINT " 2 UNLIKE DENOMINATORS"
160 A$=INKEY$
170 IF A$="2" THEN 410
180 IF A$<>"1" THEN 160
190 CLS
200 CH=1
210 PRINT"ADDING FRACTIONS WITH"
220 PRINT"LIKE DENOMINATORS"
230 PRINT:PRINT"HOW MANY FRACTIONS? ";
240 A$=INKEY$:IF A$=""THEN 240
250 IF ASC(A$)<49 THEN 240
260 IF ASC(A$)>57 THEN 240
270 PRINTA$
280 C=VAL(A$)
290 INPUT "DENOMINATOR = ";DT
300 IF DT<>0 THEN 330
310 PRINT"DENOMINATOR CANNOT = 0"
320 GOTO 290
330 PRINT:PRINT"ENTER NUMERATORS"
340 NT=0
350 FOR I=1 TO C
360 INPUT N(I)
370 NT=NT+N(I)
380 D(I)=DT
390 NEXT I
400 GOTO 730
410 CLS
420 PRINT"ADDING UP TO FIVE"
430 PRINT"FRACTIONS WHICH MAY HAVE"
440 PRINT"UNLIKE DENOMINATORS"
450 PRINT:PRINT"HOW MANY FRACTIONS? ";
460 A$=INKEY$:IF A$=""THEN 460
470 IF ASC(A$)<49 THEN 460
480 IF ASC(A$)>53 THEN 460
490 PRINTA$
500 C=VAL(A$)
510 NT=0
520 DT=1
530 FOR I=1 TO C
540 PRINT "FRACTION";I
550 INPUT " NUMERATOR = ";N(I)
560 INPUT " DENOMINATOR =";D(I)
570 IF D(I)<>0 THEN 600
580 PRINT"DENOMINATOR CANNOT = 0"
590 GOTO 560
600 IF I=1 THEN 640
610 FOR J=1 TO I-1
620 IF D(I)=D(J) THEN 660
```

```
630 NEXT J
640 F=D(I)
650 GOTO 670
660 F=1
670 DT=DT*F
680 NEXT I
690 FOR I=1 TO C
700 F=DT/D(I)
710 NT=NT+N(I)*F
720 NEXT I
730 CLS
740 PRINT"ADDING FRACTIONS":PRINT
750 FOR I=1 TO C
760 PRINTN(I);"/";D(I)
770 NEXT I
780 PRINT"-----"
790 IF DT>NT THEN 820
800 L=DT
810 GOTO 830
820 L=NT
830 ST=-2
840 IF DT/2<>INT(DT/2)THEN 860
850 IF NT/2=INT(NT/2)THEN 870
860 ST=-1
870 FOR I=L TO2 STEP ST
880 A=NT/I
890 IF A<>INT(A) THEN 920
900 B=DT/I
910 IF B=INT(B) THEN 950
920 NEXT I
930 A=NT
940 B=DT
950 PRINTA;"/";B
960 IF A<B THEN 1030
970 W=INT(A/B)
980 R=A-W*B
990 IF R<>0 THEN 1020
1000 PRINT"OR";W
1010 GOTO 1030
1020 PRINT"OR";W;" ";R;"/";B
1030 PRINT:PRINT"CHOOSE:"
1040 PRINT" 1 ANOTHER PROBLEM"
1050 PRINT" 2 END PROGRAM";
1060 A$=INKEY$
1070 IF A$="1" THEN 110
1080 IF A$<>"2" THEN 1060
1090 CLS:END
```

Homework Helpers

Program 7-30. Adding Fractions

TI-99/4A

```
110 CALL CLEAR
120 PRINT "*** ADDING FRACTIONS ***"
130 PRINT :::"CHOOSE:"
140 PRINT : "1 LIKE DENOMINATORS"
150 PRINT : "2 UNLIKE DENOMINATORS":::
160 CALL KEY(0,K,S)
170 IF K=50 THEN 410
180 IF K<>49 THEN 160
190 CALL CLEAR
200 CH=1
210 PRINT "ADDING FRACTIONS WITH"
220 PRINT "LIKE DENOMINATORS"
230 PRINT :::"HOW MANY FRACTIONS?"
240 CALL KEY(0,K,S)
250 IF (K<50)+(K>57)THEN 240
260 CALL HCHAR(23,23,K)
270 C=K-48
280 PRINT :::"WHAT IS THE DENOMINATOR?"
290 INPUT DT
300 IF DT<>0 THEN 330
310 PRINT "DENOMINATOR CANNOT = 0"
320 GOTO 280
330 PRINT :::"ENTER THE NUMERATORS":::
340 NT=0
350 FOR I=1 TO C
360 INPUT N(I)
370 NT=NT+N(I)
380 D(I)=DT
390 NEXT I
400 GOTO 720
410 CALL CLEAR
420 PRINT "ADDING UP TO FIVE"
430 PRINT "FRACTIONS WHICH MAY HAVE"
440 PRINT "UNLIKE DENOMINATORS"
450 PRINT :::"HOW MANY FRACTIONS?":::
460 CALL KEY(0,K,S)
470 IF (K<50)+(K>53)THEN 460
480 CALL HCHAR(22,23,K)
490 C=K-48
500 NT=0
510 DT=1
520 FOR I=1 TO C
530 PRINT "FRACTION";I
540 INPUT "{3 SPACES}NUMERATOR ={3 SPACES}":N(I)
550 INPUT "{3 SPACES}DENOMINATOR = ":D(I)
560 IF D(I)<>0 THEN 590
570 PRINT : "DENOMINATOR CANOT BE ZERO":::
580 GOTO 550
```

Homework Helpers

```
590 IF I=1 THEN 630
600 FOR J=1 TO I-1
610 IF D(I)=D(J)THEN 650
620 NEXT J
630 F=D(I)
640 GOTO 660
650 F=1
660 DT=DT*F
670 NEXT I
680 FOR I=1 TO C
690 F=DT/D(I)
700 NT=NT+N(I)*F
710 NEXT I
720 CALL CLEAR
730 PRINT "*** ADDING FRACTIONS ***":
740 FOR I=1 TO C
750 PRINT STR$(N(I));"/";STR$(D(I))
760 NEXT I
770 PRINT "-----":
780 IF DT>NT THEN 810
790 L=DT
800 GOTO 820
810 L=NT
820 ST=-2
830 IF DT/2<>INT(DT/2)THEN 850
840 IF NT/2=INT(NT/2)THEN 860
850 ST=-1
860 FOR I=L TO 2 STEP ST
870 A=NT/I
880 IF A<>INT(A)THEN 910
890 B=DT/I
900 IF B=INT(B)THEN 940
910 NEXT I
920 A=NT
930 B=DT
940 PRINT STR$(A);"/";STR$(B)
950 IF A<B THEN 1020
960 W=INT(A/B)
970 R=A-W*B
980 IF R<>0 THEN 1010
990 PRINT "OR";W
1000 GOTO 1020
1010 PRINT : "OR ";W;" ";STR$(R);"/";STR$(B)
1020 PRINT : "CHOOSE:" 1 ANOTHER PROBLEM"
1030 PRINT " 2 END PROGRAM"
1040 CALL KEY(0,K,S)
1050 IF K=49 THEN 110
1060 IF K<>50 THEN 1040
1070 CALL CLEAR
1080 END
```

Homework Helpers

Simultaneous Equations

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A

You can solve simultaneous equations using the matrix inversion technique with this program, which solves up to six equations with six unknowns. If you have a computer with more memory, you can increase the number of equations by changing the DIMENSION statement and the statement limiting your choice of the number of equations. If you have a 40-column or 80-column screen, you may wish to change the printing.

First, you enter the number of equations and unknowns (the degree of the matrix). Next, enter the coefficients row by row with the corresponding B elements in the matrix of equations $[A] \cdot [X] = [B]$. Here's an example problem.

Given three equations with three unknowns:

$$x_1 + x_2 + x_3 = 12$$

$$2x_1 + x_2 + 3x_3 = 25$$

$$x_1 + 3x_2 + 2x_3 = 25$$

Rewritten in matrix form, this is:

$$\begin{bmatrix} 1 & 1 & 1 \\ 2 & 1 & 3 \\ 1 & 3 & 2 \end{bmatrix} * \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix} = \begin{bmatrix} 12 \\ 25 \\ 25 \end{bmatrix}$$

The group of numbers on the left-hand side of the matrix represents A , in the equation $[A] * [X] = [B]$. A in the equation $x_1 + x_2 + x_3 = 12$ is always 1, while A in the equation $2x_1 + x_2 + 3x_3 = 25$ equals 2, 1, and 3.

To solve this system using the program, you enter the degree of the matrix as 3 for the *three* equations with three unknowns. In order, the following numbers are entered:

$$A(1,1) = 1$$

$$A(1,2) = 1$$

$$A(1,3) = 1$$

$$B(1) = 12$$

$$A(2,1) = 2$$

$$A(2,2) = 1$$

$$A(2,3) = 3$$

$$B(2) = 25$$

$$\begin{aligned} A(3,1) &= 1 \\ A(3,2) &= 3 \\ A(3,3) &= 2 \\ B(3) &= 25 \end{aligned}$$

Notice that the numbers you enter correspond to the values of A you saw in the matrix form of the equation.

The solution is then printed:

$$\begin{aligned} X(1) &= 3 \\ X(2) &= 4 \\ X(3) &= 5 \end{aligned}$$

Program 7-31. Simultaneous Equations

VIC-20

```

110 GOTO 240
120 REM SUB TO SWITCH ROWS
130 FOR C=I+1 TO N
140 IF W(C,I)=0 THEN 200
150 FOR L=1 TO N
160 DW=W(I,L):W(I,L)=W(C,L):W(C,L)=DW
170 NEXT L
180 DB=B(I):B(I)=B(C):B(C)=DB
190 RETURN
200 NEXT C
210 PRINT "{DOWN} DETERMINANT = 0."
220 PRINT "NO UNIQUE SOLUTION. {DOWN}"
230 GOTO 740
240 PRINT "{CLR}"
250 DIM A(6,6),W(6,6),X(6),B(6)
260 PRINT "SOLUTION OF"
270 PRINT "SIMULTANEOUS EQUATIONS"
280 PRINT "MATRIX INVERSION":PRINT "TECHNIQUE TO SOL
VE"
290 PRINT " [A][X] = [B]"
350 PRINT "{DOWN} NUMBER OF EQUATIONS OR DEGREE OF MA
TRIX, "
360 PRINT "N = ";
370 GET I$:IF I$="" THEN 370
380 IF ASC(I$)<50 OR ASC(I$)>54 THEN 370
390 N=VAL(I$):PRINT N
400 PRINT "{DOWN} ENTER [A] VALUES WITH"
410 PRINT "CORRESPONDING [B]"
420 PRINT "FOR EACH EQUATION. {DOWN}"
430 FOR I=1 TO N:FOR J=1 TO N
440 PRINT "A("+RIGHT$(STR$(I),1)+" , "+RIGHT$(STR$(J
),1)+" ) = " ; INPUT A(I,J)
450 W(I,J)=A(I,J):NEXT J
460 PRINT "B("+RIGHT$(STR$(I),1)+" ) = " ; INPUT B(I
)

```

Homework Helpers

```
470 PRINT:NEXT I
480 REM INVERT MATRIX A
490 FOR I=1 TO N
500 IF W(I,I)=0 THEN GOSUB 130
510 W(I,I)=1/W(I,I)
520 FOR J=1 TO N
530 IF J-I=0 THEN 590
540 W(J,I)=W(J,I)*W(I,I)
550 FOR K=1 TO N
560 IF K-I=0 THEN 580
570 W(J,K)=W(J,K)-W(J,I)*W(I,K)
580 NEXT K
590 NEXT J
600 FOR K=1 TO N
610 IF K-I=0 THEN 630
620 W(I,K)=-W(I,I)*W(I,K)
630 NEXT K
640 NEXT I
650 PRINT"{2 DOWN}SOLUTION VECTOR X:{DOWN}"
660 FOR I=1 TO N
670 X(I)=0
680 FOR J=1 TO N
690 X(I)=X(I)+W(I,J)*B(J)
700 NEXT J
710 PRINT"X("+RIGHT$(STR$(I),1)+") = ";X(I)
720 NEXT I
730 PRINT
740 END
```

Program 7-32. Simultaneous Equations

TRS-80 Color Computer; TRS-80 Model I

```
110 GOTO 240
120 REM SUB TO SWITCH ROWS
130 FOR C=I+1 TO N
140 IF W(C,I)=0 THEN 200
150 FOR L=1 TO N
160 DW=W(I,L):W(I,L)=W(C,L):W(C,L)=DW
170 NEXT L
180 DB=B(I):B(I)=B(C):B(C)=DB
190 RETURN
200 NEXT C
210 PRINT:PRINT "SORRY, DETERMINANT = 0."
220 PRINT "NO UNIQUE SOLUTION.":PRINT
230 GOTO 740
240 CLS
250 DIM A(6,6),W(6,6),X(6),B(6)
260 PRINT"SOLUTION OF SIMULTANEOUS"
270 PRINT"EQUATIONS"
280 PRINT:PRINT"MATRIX INVERSION TECHNIQUE"
```

Homework Helpers

```
290 PRINT"TO SOLVE [A][X] = [B]"
300 PRINT:PRINT"PRESS <ENTER>"
310 I$=INKEY$
320 IF I$<>CHR$(13) THEN 310
330 CLS
340 PRINT"[A]*[X] = [B]"
350 PRINT"PRESS NUMBER OF EQUATIONS OR"
360 PRINT"DEGREE OF THE MATRIX, N = ";
370 I$=INKEY$:IF I$="" THEN 370
380 IF ASC(I$)<50 OR ASC(I$)>54 THEN 370
390 N=VAL(I$):PRINTN
400 PRINT:PRINT "ENTER [A] VALUES WITH"
410 PRINT "CORRESPONDING [B] VALUE"
420 PRINT "FOR EACH EQUATION.":PRINT
430 FOR I=1 TO N:FOR J=1 TO N
440 PRINT "A("+RIGHT$(STR$(I),1)+", "+RIGHT$(STR$(J
),1)+") = ";:INPUT A(I,J)
450 W(I,J)=A(I,J):NEXT J
460 PRINT "B("+RIGHT$(STR$(I),1)+") = ";:INPUT B(I
)
470 PRINT:NEXT I
480 REM INVERT MATRIX A
490 FOR I=1 TO N
500 IF W(I,I)=0 THEN GOSUB 130
510 W(I,I)=1/W(I,I)
520 FOR J=1 TO N
530 IF J-I=0 THEN 590
540 W(J,I)=W(J,I)*W(I,I)
550 FOR K=1 TO N
560 IF K-I=0 THEN 580
570 W(J,K)=W(J,K)-W(J,I)*W(I,K)
580 NEXT K
590 NEXT J
600 FOR K=1 TO N
610 IF K-I=0 THEN 630
620 W(I,K)=-W(I,I)*W(I,K)
630 NEXT K
640 NEXT I
650 PRINT:PRINT:PRINT "SOLUTION VECTOR X:":PRINT
660 FOR I=1 TO N
670 X(I)=0
680 FOR J=1 TO N
690 X(I)=X(I)+W(I,J)*B(J)
700 NEXT J
710 PRINT "X("+RIGHT$(STR$(I),1)+") = ";X(I)
720 NEXT I
730 PRINT
740 END
```


Homework Helpers

Program 7-33. Simultaneous Equations


TI-99/4A

```
110 CALL CLEAR
120 PRINT "SOLVING SIMULTANEOUS"
130 PRINT "EQUATIONS BY THE"
140 PRINT "MATRIX INVERSION TECHNIQUE"
150 PRINT ::"SOLVE [A][X] = [B]"
160 PRINT ::"ENTER DEGREE OF THE MATRIX"
170 PRINT "OR NUMBER OF EQUATIONS"::
180 INPUT "N = ":N
190 IF N<10 THEN 220
200 PRINT ::"N MUST BE < 10"::
210 GOTO 180
220 IF N>1 THEN 250
230 PRINT ::"1<N<10{3 SPACES}TRY AGAIN"::
240 GOTO 180
250 PRINT ::"THE COEFFICIENTS OF X"
260 PRINT "ARE THE 'A' MATRIX."
270 PRINT ::"INPUT THE VALUES ROW BY ROW:"
280 PRINT ::"A(1,1),A(1,2),A(1,3),..."
290 PRINT "A(2,1),A(2,2),A(2,3),..."
300 PRINT "."::"."::"."::
310 PRINT "A(N,1),A(N,2),...,A(N,N)"::::
320 FOR I=1 TO N
330 FOR J=1 TO N
340 INPUT "A("&STR$(I)&","&STR$(J)&") = ":A(I,J)
350 W(I,J)=A(I,J)
360 NEXT J
370 PRINT
380 INPUT "B("&STR$(I)&") = ":B(I)
390 PRINT ::
400 NEXT I
410 PRINT ::"--SOLVING--"::
420 REM INVERT MATRIX A
430 FOR C=1 TO N
440 IF W(C,C)<>0 THEN 460
450 GOSUB 710
460 W(C,C)=1/W(C,C)
470 FOR D=1 TO N
480 IF (D-C)=0 THEN 540
490 W(D,C)=W(D,C)*W(C,C)
500 FOR E=1 TO N
510 IF (E-C)=0 THEN 530
520 W(D,E)=W(D,E)-W(D,C)*W(C,E)
530 NEXT E
540 NEXT D
550 FOR E=1 TO N
560 IF (E-C)=0 THEN 580
570 W(C,E)=-W(C,C)*W(C,E)
```

```

580 NEXT E
590 NEXT C
600 PRINT :: "SOLUTION VECTOR X:":
610 FOR I=1 TO N
620 X(I)=0
630 FOR J=1 TO N
640 X(I)=X(I)+W(I,J)*B(J)
650 NEXT J
660 PRINT : "X("&STR$(I)&") = ";X(I)
670 NEXT I
680 PRINT ::
690 GOTO 870
700 REM SUB TO SWITCH ROWS
710 FOR F=C+1 TO N
720 IF W(F,C)=0 THEN 820
730 FOR E=1 TO N
740 DW=W(C,E)
750 W(C,E)=W(F,E)
760 W(F,E)=DW
770 NEXT E
780 DB=B(C)
790 B(C)=B(F)
800 B(F)=DB
810 GOTO 860
820 NEXT F
830 PRINT "SORRY, DETERMINANT=0."
840 PRINT "NO UNIQUE SOLUTION."
850 GOTO 870
860 RETURN
870 PRINT : "PRESS 1 FOR ANOTHER PROBLEM"
880 PRINT TAB(7); "2 TO END PROGRAM"
890 CALL KEY(0,K,S)
900 IF K=49 THEN 110
910 IF K<>50 THEN 890
920 CALL CLEAR
930 END

```



Chapter 8
Computer
Sorting

REPRODUCED FROM
THE NATIONAL ARCHIVES
SERIALS SECTION

Chapter 8

Computer Sorting

One of the functions of a computer is to organize data. You can use a sort routine to arrange numbers in ascending or descending order or to alphabetize lists. There are many kinds of sort routines. Four sorts are presented here.

To illustrate the procedure, 50 random numbers are generated and printed, then sorted. In a regular program you would list the actual data to sort, then perhaps call a sort as a subroutine.

The numbers used in these programs are contained in the A array. The DIMENSION statement at the beginning of the programs allow for the 50 numbers of the example. When you place these sorts in your own programs, make sure the value n in DIM A(n) reflects the total number of items to be sorted.

These sort programs arrange numbers in ascending order (smaller to larger). To change to descending order, change the less than signs to greater than signs. To use strings, when you're using names as data, for example, put dollar signs after the variable names to indicate string variables.

Bubble Sort

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A

The bubble sort is a simple interchange sort and can be used for lists that do not contain many numbers, or for lists in which the numbers are not much out of order — it is slow for other uses. The program compares each number to the next number and exchanges numbers where necessary. If an exchange has been made during a pass through all the numbers, the loop of comparisons starts over.

Program 8-1. Bubble Sort

VIC-20

```
110 DIM A(50)
120 FOR I=1 TO 50: A(I)=INT(100*RND(0))+1): PRINTA(I); : N
    EXT: PRINT: PRINT
130 L=49
140 S=0: FOR I=1 TO L: IFA(I) <= A(I+1) THEN I60
150 D=A(I): A(I)=A(I+1): A(I+1)=D: S=1: L=I
160 NEXT
170 IFS=1 THEN I40
180 FOR I=1 TO 50: PRINTA(I); : NEXT
190 END
```

Sorting

Program 8-2. Bubble Sort

TRS-80 Color Computer; TRS-80 Model I

```
110 DIM A(50)
120 FOR I=1 TO 50:A(I)=RND(100):PRINTA(I);:NEXT:PRINT
    :PRINT
130 L=49
140 S=0:FOR I=1 TO L:IFA(I)<=A(I+1)THEN160
150 D=A(I):A(I)=A(I+1):A(I+1)=D:S=1:L=I
160 NEXT
170 IFS=1THEN140
180 FOR I=1 TO 50:PRINTA(I);:NEXT
190 END
```

Program 8-3. Bubble Sort

TI-99/4A

```
110 DIM A(50)
120 FOR I=1 TO 50
130 RANDOMIZE
140 A(I)=INT(RND*100+1)
150 PRINT A(I);
160 NEXT I
170 PRINT ::
200 LIM=49
210 SW=0
220 FOR I=1 TO LIM
230 IF A(I)<=A(I+1)THEN 290
240 AA=A(I)
250 A(I)=A(I+1)
260 A(I+1)=AA
270 SW=1
280 LIM=I
290 NEXT I
300 IF SW=1 THEN 210
500 FOR I=1 TO 50
510 PRINT A(I);
520 NEXT I
530 END
```

Shell Sort

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A

The shell sort is considerably faster than the bubble sort because the number of comparisons to be made is reduced.

In an array of N numbers, the program first determines a number B so that $2^B < N < 2^{B+1}$, then initializes B to $2^B - 1$. The main loop varies a counter I from 1 to $N - B$. Within the loop, the program checks if $A(I) < A(I+B)$. If so, it increments I and continues the comparisons. If not, it exchanges $A(I)$ and $A(I+B)$.

and changes the subscript. When I reaches the value of N, B is reduced by a factor of 2 and the loop starts over. When B=0, the sort is complete.

As with the bubble sort, this program generates 50 random numbers. To use this sort in your own programs, you'd have to enter DATA and have the program READ it into the array. Make sure you also change the values throughout the sort routine to show the number of items to be sorted.

Program 8-4. Shell Sort

VIC-20

```
110 DIM A(50)
120 FOR I=1 TO 50:A(I)=INT(100*RND(0)+1):PRINTA(I);:N
    EXT:PRINT:PRINT
130 B=1
140 B=2*B:IFB<=50 THEN 140
150 B=INT(B/2):IFB=0 THEN 200
160 FOR I=1 TO 50-B:C=I
170 D=C+B:IFA(C)<=A(D) THEN 190
180 X=A(C):A(C)=A(D):A(D)=X:C=C-B:IFC>0 THEN 170
190 NEXT:GOTO 150
200 FOR I=1 TO 50:PRINTA(I);:NEXT
210 END
```

Program 8-5. Shell Sort

TRS-80 Color Computer; TRS-80 Model I

```
110 DIM A(50)
120 FOR I=1 TO 50:A(I)=RND(100):PRINTA(I);:NEXT:PRINT
    :PRINT
130 B=1
140 B=2*B:IFB<=50 THEN 140
150 B=INT(B/2):IFB=0 THEN 200
160 FOR I=1 TO 50-B:C=I
170 D=C+B:IFA(C)<=A(D) THEN 190
180 X=A(C):A(C)=A(D):A(D)=X:C=C-B:IFC>0 THEN 170
190 NEXT:GOTO 150
200 FOR I=1 TO 50:PRINTA(I);:NEXT
210 END
```

Program 8-6. Shell Sort

TI-99/4A

```
110 DIM A(50)
120 FOR I=1 TO 50
130 RANDOMIZE
140 A(I)=INT(RND*100+1)
150 PRINT A(I);
160 NEXT I
```


Sorting

```
170 PRINT ::
200 B=1
210 B=2*B
220 IF B<=50 THEN 210
230 B=INT(B/2)
240 IF B=0 THEN 500
250 FOR I=1 TO 50-B
260 C=I
270 D=C+B
280 IF A(C) <=A(D) THEN 340
290 AA=A(C)
300 A(C)=A(D)
310 A(D)=AA
320 C=C-B
330 IF C>0 THEN 270
340 NEXT I
350 GOTO 230
500 FOR I=1 TO 50
510 PRINT A(I);
520 NEXT I
530 END
```

Maximum and Minimum Sort

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A

This program is also faster than the bubble sort if the numbers are mixed up quite a bit. The program goes through all the numbers and places the lowest value in the first spot of the array, and the highest value in the last spot of the array. Each pass through the numbers finds the minimum and maximum of the numbers left and places them at the endpoints.

Program 8-7. Maximum and Minimum Sort

VIC-20

```
110 DIMA(50):N=50:S=1
120 FORI=1TO50:A(I)=INT(100*RND(0)+1):PRINTA(I);:N
    EXT:PRINT:PRINT
130 L=A(S):J=S:U=L:K=S
140 FORI=S TO N
150 IFA(I)>U THENU=A(I):K=I
160 IFA(I)<L THENL=A(I):J=I
170 NEXT
180 IFJ=N THENJ=K
190 D=A(N):A(N)=A(K):A(K)=D:N=N-1
200 D=A(S):A(S)=A(J):A(J)=D:S=S+1
210 IFN>S THEN130
220 FORI=1TO50:PRINTA(I);:NEXT
230 END
```

Program 8-8. Maximum and Minimum Sort

TRS-80 Color Computer; TRS-80 Model I

```

110 DIM A(50):N=50:S=1
120 FORI=1TO50:A(I)=RND(100):PRINTA(I);:NEXT:PRINT
    :PRINT
130 L=A(S):J=S:U=L:K=S
140 FORI=S TON
150 IFA(I)>U THENU=A(I):K=I
160 IFA(I)<L THENL=A(I):J=I
170 NEXT
180 IFJ=N THENJ=K
190 D=A(N):A(N)=A(K):A(K)=D:N=N-1
200 D=A(S):A(S)=A(J):A(J)=D:S=S+1
210 IFN>S THEN130
220 FORI=1TO50:PRINTA(I);:NEXT
230 END

```

Program 8-9. Maximum and Minimum Sort

TI-99/4A

```

110 DIM A(50)
120 N=50
130 FOR I=1 TO 50
140 RANDOMIZE
150 A(I)=INT(RND*100+1)
160 PRINT A(I);
170 NEXT I
180 PRINT ::
200 S=1
210 MN=A(S)
220 IMIN=S
230 MX=MN
240 IMAX=S
250 FOR I=S TO N
260 IF A(I)<=MX THEN 290
270 MX=A(I)
280 IMAX=I
290 IF A(I)>=MN THEN 320
300 MN=A(I)
310 IMIN=I
320 NEXT I
330 IF IMIN<>N THEN 350
340 IMIN=IMAX
350 AA=A(N)
360 A(N)=A(IMAX)
370 A(IMAX)=AA
380 N=N-1
390 AA=A(S)
400 A(S)=A(IMIN)

```

Sorting

```
410 A(IMIN)=AA
420 S=S+1
430 IF N>S THEN 210
500 FOR I=1 TO 50
510 PRINT A(I);
520 NEXT I
530 END
```

Quick Sort

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A
"Quick Sort" can be thought of as a combination of the maximum and minimum procedure and the shell sort. It is usually faster than either.

Program 8-10. Quick Sort

VIC-20

```
110 DIMA(50)
120 FORI=1TO50:A(I)=INT(100*RND(0))+1):PRINTA(I);:N
    EXT:PRINT:PRINT
130 N=50:S(1)=1:S(2)=N:T=1
140 IFT=0THEN260
150 T=T-1:I=2*T:L=S(I+1):M=S(I+2):X=A(L):J=L:K=M+1
160 K=K-1:IFK=J THEN220
170 IFX<=A(K)THEN160
180 A(J)=A(K)
190 J=J+1:IFK=J THEN220
200 IFX>=A(J)THEN190
210 A(K)=A(J):GOTO160
220 A(J)=X:IFM-J<2THEN240
230 I=2*T:S(I+1)=J+1:S(I+2)=M:T=T+1
240 IFK-L<2THEN140
250 I=2*T:S(I+1)=L:S(I+2)=K-1:T=T+1:GOTO140
260 FORI=1TON:PRINTA(I);:NEXT
270 END
```

Program 8-11. Quick Sort

TRS-80 Color Computer; TRS-80 Model I

```
110 DIM A(50)
120 FORI=1TO50:A(I)=RND(100):PRINTA(I);:NEXT:PRINT
    :PRINT
130 N=50:S(1)=1:S(2)=N:T=1
140 IFT=0THEN260
150 T=T-1:I=2*T:L=S(I+1):M=S(I+2):X=A(L):J=L:K=M+1
160 K=K-1:IFK=J THEN220
170 IFX<=A(K)THEN160
180 A(J)=A(K)
190 J=J+1:IFK=J THEN220
200 IFX>=A(J)THEN190
```

```

210 A(K)=A(J):GOTO160
220 A(J)=X:IFM-J<2THEN240
230 I=2*T:S(I+1)=J+1:S(I+2)=M:T=T+1
240 IFK-L<2THEN140
250 I=2*T:S(I+1)=L:S(I+2)=K-1:T=T+1:GOTO140
260 FORI=1TON:PRINTA(I);:NEXT
270 END

```

Program 8-12. Quick Sort

TI-99/4A

```

110 CALL CLEAR
120 DIM A(50)
130 N=50
140 FOR I=1 TO N
150 RANDOMIZE
160 A(I)=INT(RND*100)+1
170 PRINT A(I);
180 NEXT I
190 PRINT
200 S(1)=1
210 S(2)=N
220 T=1
230 IF T=0 THEN 520
240 T=T-1
250 I=2*T
260 L=S(I+1)
270 M=S(I+2)
280 X=A(L)
290 J=L
300 K=M+1
310 K=K-1
320 IF K=J THEN 400
330 IF X<=A(K)THEN 310
340 A(J)=A(K)
350 J=J+1
360 IF K=J THEN 400
370 IF X>=A(J)THEN 350
380 A(K)=A(J)
390 GOTO 310
400 A(J)=X
410 IF M-J<2 THEN 460
420 I=2*T
430 S(I+1)=J+1
440 S(I+2)=M
450 T=T+1
460 IF K-L<2 THEN 230
470 I=2*T
480 S(I+1)=L
490 S(I+2)=K-1

```

Sorting

```
500 T=T+1
510 GOTO 230
520 PRINT
530 FOR I=1 TO N
540 PRINT A(I);
550 NEXT I
560 END
```

Test Scores

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A
"Test Scores" illustrates the use of a sort routine in an actual program. Look this over to see how you could use a sort routine in a program of your own.

Students' scores are shown for three tests, along with a total and an average. The total scores are then sorted in descending order, and the total, average, and student's name are PRINTed.

The DATA statements contain these items separated by commas: last name, first name, and the scores for three tests. The computer READs the last name and first name and prints them on the screen. To save memory, the first name, a space, and the last name are combined for N\$(). Next, the three scores are READ and PRINTed. The total is calculated and displayed, and the average of the three scores is shown. As the test scores are being READ, the class totals S() are also calculated. After the individual scores are shown, the average for each test and the overall average are displayed.

To prevent names from scrolling too fast on the screen, a line count L keeps track of how much is on the screen. You must press ENTER or RETURN to continue.

The bubble sort is used to arrange the names in descending order by total score. In the sort, the >= sign is used instead of the <= sign to sort from the largest to the smallest number.

Program 8-13. Test Scores

VIC-20

```
110 PRINT "{CLR}"
120 DIM N$(17), S(2), P(17)
130 L=0: S(0)=0: S(1)=0: S(2)=0: V=0
140 FOR I=0 TO 17: READ N$(I), F$: PRINT N$(I); ", "; F$: N$(I)=F$+" "+N$(I): P(I)=0
150 FOR J=0 TO 2: READ T(J): P(I)=P(I)+T(J): S(J)=S(J)+T(J): PRINT T(J); : NEXT J
160 A=INT(P(I)/3+.5): PRINT "{2 SPACES}"; P(I); A: V=V+A
170 L=L+3: IFL=18 THEN GOSUB 480
```

```

180 NEXTI
190 FORJ=0TO2:S(J)=INT(S(J)/18+.5):NEXT
200 PRINT"{DOWN}AVERAGES":PRINTS(0);S(1);S(2);"
    {3 SPACES}";INT(V/18+.5):M=16
210 W=0:FORI=0TOM:IFP(I)>=P(I+1)THEN240
220 D=P(I):P(I)=P(I+1):P(I+1)=D:W=1:M=I
230 D$=N$(I):N$(I)=N$(I+1):N$(I+1)=D$
240 NEXTI
250 IFW=1THEN210
260 GOSUB480
270 FORI=0TO17:PRINTP(I);INT(P(I)/3+.5);N$(I)
290 NEXTI:GOTO510
300 DATA ADAMS,BOB,91,82,71
310 DATA BROWN,ANDY,96,84,72
320 DATA CARSON,GRANT,94,79,67
330 DATA ELLIS,CINDY,82,88,65
340 DATA HANSEN,ANGIE,96,67,72
350 DATA HANSEN,KELLY,72,97,65
360 DATA JENSEN,ED,88,71,71
370 DATA JENSEN,LENA,71,97,78
380 DATA KENT,RICHARD,77,73,78
390 DATA LARSEN,JOHN,92,82,83
400 DATA MARTIN,CHERY,88,71,97
410 DATA NELSON,RANDY,80,72,99
420 DATA ROGERS,BRIAN,78,77,98
430 DATA SMITH,AURA,97,80,64
440 DATA SMITH,BILL,62,67,87
450 DATA SMITH,JIM,64,81,87
460 DATA TAYLOR,JENNY,77,61,64
470 DATA WHITE,LEWIS,94,65,94
480 L=00:PRINT"{DOWN}PRESS <RETURN>";
490 GETES:IFE$<>CHR$(13)THEN490
500 PRINT"{CLR}":RETURN
510 END

```

Program 8-14. Test Scores

TRS-80 Color Computer; TRS-80 Model I

```

110 CLS:CLEAR250
120 DIMN$(17),S(2),P(17)
130 L=0:S(0)=0:S(1)=0:S(2)=0:V=0
140 FORI=0TO17:READN$(I),F$:PRINTN$(I);", ";F$,:N$(I)=F$+" "+N$(I):P(I)=0
150 FORJ=0TO2:READT(J):P(I)=P(I)+T(J):S(J)=S(J)+T(J):PRINTT(J);:NEXTJ
160 A=INT(P(I)/3+.5):PRINT"{5 SPACES}";P(I);A:V=V+A
170 L=L+2:IFL=14GOSUB480
180 NEXTI

```

Sorting

```
190 FORJ=0TO2:S(J)=INT(S(J)/18+.5):NEXT
200 PRINT:PRINT"AVERAGES",S(0);S(1);S(2);"
    {10 SPACES}";INT(V/18+.5):M=16
210 W=0:FORI=0TOM:IFP(I)>=P(I+1)THEN240
220 D=P(I):P(I)=P(I+1):P(I+1)=D:W=1:M=I
230 D$=N$(I):N$(I)=N$(I+1):N$(I+1)=D$
240 NEXTI
250 IFW=1THEN210
260 GOSUB480
270 FORI=0TO17:PRINTP(I);INT(P(I)/3+.5);N$(I)
280 L=L+1:IFL=12THENGOSUB480
290 NEXTI:GOTO510
300 DATA ADAMS,BOB,91,82,71
310 DATA BROWN,ANDY,96,84,72
320 DATA CARSON,GRANT,94,79,67
330 DATA ELLIS,CINDY,82,88,65
340 DATA HANSEN,ANGIE,96,67,72
350 DATA HANSEN,KELLY,72,97,65
360 DATA JENSEN,ED,88,71,71
370 DATA JENSEN,LENA,71,97,78
380 DATA KENT,RICHARD,77,73,78
390 DATA LARSEN,JOHN,92,82,83
400 DATA MARTIN,CHERY,88,71,97
410 DATA NELSON,RANDY,80,72,99
420 DATA ROGERS,BRIAN,78,77,98
430 DATA SMITH,AURA,97,80,64
440 DATA SMITH,BILL,62,67,87
450 DATA SMITH,JIM,64,81,87
460 DATA TAYLOR,JENNY,77,61,64
470 DATA WHITE,LEWIS,94,65,94
480 L=0:PRINT:PRINT"PRESS <ENTER>";
490 E$=INKEY$:IFE$<>CHR$(13)THEN490
500 CLS:RETURN
510 PRINT:END
```

Program 8-15. Test Scores

TI-99/4A

```
110 CALL CLEAR
120 DIM N$(17),S(2),P(17)
130 FOR I=0 TO 17
140 READ N$(I),F$
150 PRINT :N$(I);", ";F$
160 N$(I)=F$&" "&N$(I)
170 P(I)=0
180 FOR J=0 TO 2
190 READ T(J)
200 P(I)=P(I)+T(J)
210 S(J)=S(J)+T(J)
220 PRINT T(J);
```

```


230 NEXT J
240 A=INT(P(I)/3+.5)
250 PRINT " ";P(I);A
260 V=V+A
270 L=L+3
280 IF L<21 THEN 300
290 GOSUB 730
300 NEXT I
310 FOR J=0 TO 2
320 S(J)=INT(S(J)/18+.5)
330 NEXT J
340 PRINT : "AVERAGES"
350 PRINT S(0);S(1);S(2); "{7 SPACES}"; INT(V/18+.5)
360 M=16
370 W=0
380 FOR I=0 TO M
390 IF P(I)>=P(I+1) THEN 480
400 D=P(I)
410 P(I)=P(I+1)
420 P(I+1)=D
430 W=1
440 M=I
450 D$=N$(I)
460 N$(I)=N$(I+1)
470 N$(I+1)=D$
480 NEXT I
490 IF W=1 THEN 370
500 GOSUB 730
510 FOR I=0 TO 17
520 PRINT P(I); INT(P(I)/3+.5); N$(I)
530 NEXT I
540 GOTO 790
550 DATA ADAMS,BOB,91,82,71
560 DATA BROWN,ANDY,96,84,72
570 DATA CARSON,GRANT,94,79,67
580 DATA ELLIS,CINDY,82,88,65
590 DATA HANSEN,ANGIE,96,67,72
600 DATA HANSEN,KELLY,72,97,65
610 DATA JENSEN,ED,88,71,71
620 DATA JENSEN,LENA,71,97,78
630 DATA KENT,RICHARD,77,73,78
640 DATA LARSEN,JOHN,92,82,83
650 DATA MARTIN,CHERY,88,71,97
660 DATA NELSON,RANDY,80,72,99
670 DATA ROGERS,BRIAN,78,77,98
680 DATA SMITH,AURA,97,80,64
690 DATA SMITH,BILL,62,67,87
700 DATA SMITH,JIM,64,81,87
710 DATA TAYLOR,JENNY,77,61,64

```


Sorting

```
720 DATA WHITE,LEWIS,94,65,94
730 L=0
740 PRINT : "PRESS <ENTER>";
750 CALL KEY(0,K,ST)
760 IF K<>13 THEN 750
770 CALL CLEAR
780 RETURN
790 END
```

Chapter 9
Fun in
Less Than 4K



Chapter 9

Fun in Less Than 4K

It is possible to design and write games for computers with limited memory. You just have to make compromises from time to time. Nonetheless, games for such computers as the VIC, the Color Computer, and the TI-99 can be graphically entertaining and educational. Even with only 4K of memory, the game can have plenty of variety, enough to hold even the most avid player's attention.

Almost all the games in this chapter have been translated into versions for several computers. Each computer, then, has its own listing for you to look at and type in.

Some of the games, such as "Typing" or "States and Capitals," are teaching games that test your skill and knowledge. Others, such as "Defend," are simply fun to play. There are even games that are a cross between the two, such as "Multiplication Invasion," where the learning takes place in a game setting. Not only are the games fun to play, but you'll also learn something as you play.

There are even two programs, "Bake a Cake" and "Cookie File," which are really not games at all. They're still fun to type in and use, however, so I included them in this final chapter. You can use them as recipe holders, or you can alter them to include your own unusual recipes. Both have been listed only for the VIC-20, so if you have a different computer, you'll have to study the program to see if it can be translated to your machine's version of BASIC.

Typing

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I

The letters on most microcomputer keyboards are in the same position as on a standard typewriter. With the capabilities of the computer, a learn-to-type program works very well. The graphic abilities of a computer can actually draw the hands on the screen, with the letters to be typed above the appropriate fingers.

This program teaches touch typing, using the computer, by showing the home position of the fingers — fingers on A, S, D, F, and J, K, L, and ; (: on Commodore computers). An introduction is presented, then a random letter drill. For more proficiency, there is a drill of actual words using the letters on the home keys.

Here's how the program works on the VIC-20:

Line	Function
2	PRINT title screen.
3	Turn volume on and define M for sound.

Fun in Less Than 4K

- 6 Define string variables to create graphics.
8-12 READ from DATA the variables position P, letter P\$, graphic letter L, and sound S.
14-16 PRINT instructions.
18-22 Draw hands.
24 PRINT letters above appropriate fingers.
26-30 Wait for any key to be pressed.
32-48 Perform drill to type a letter as it appears above the finger.
50-56 Clear screen and READ phrases A\$ from DATA. RESTORE is included to use the DATA again, ignoring the first 32 items.
58-80 Perform drill for phrases. A phrase is chosen randomly. If it is typed correctly, A\$(J) is set equal to "" so the phrase cannot be used again. If it is typed incorrectly, the phrase may be used again. Five phrases must be typed correctly to complete the drill.
82-90 PRINT option to practice letters, practice words, or end, and branch appropriately.
92-94 Check letter typed.
96 Delay loop for music.
98-104 Wait for RETURN key to be pressed.
106 Clear screen and END.

Program 9-1. Typing

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```
* 2 PRINT "{CLR}{5 DOWN}":PRINTTAB(8);"TYPING":PRINTT
  AB(8);"{2 DOWN}UNIT 1":PRINTTAB(4);"{DOWN}HOME P
  OSITION"
  4 POKE36878,15:M=36876
  6 F$="{RVS}[K] {OFF}":G$="{RVS} [L]{OFF}":H$=F
    $+" "+G$+F$:I$=G$+F$+" "+G$
  8 FORI=1TO8:READP(I),P$(I),L(I),S(I):NEXT
  10 DATA8010,A,1,195,7969,S,19,201,7949,D,4,207,797
    4,F,6,209
  12 DATA7978,J,10,215,7959,K,11,219,7983,L,12,223,8
    030,"":",58,225
  14 PRINT"{3 DOWN}YOU WILL SEE A DIAGRAMOF THE HAND
    S."
  16 PRINT"PLACE YOUR FINGERS ON THE KEYS AS SHOWN."
    :GOSUB98
```

```

*18 PRINT"{CLR}{RED}{13 DOWN}":PRINTTAB(5);G$;TAB(1
  4);F$:PRINT"{2 SPACES}";H$;SPC(3);I$:PRINT"
  {2 SPACES}";H$;SPC(3);I$
  20 FORC=1TO4:PRINTG$+H$;SPC(3);I$+F$:NEXT
  22 PRINT"{RVS}{9 SPACES}{OFF}{3 SPACES}{RVS}
  {9 SPACES}{OFF}{RVS}{9 SPACES}{OFF}{3 SPACES}
  {RVS}{9 SPACES}{OFF}";
  24 FORC=1TO8:POKEM,S(C):POKEP(C),L(C):POKEP(C)+307
  20,0:GOSUB96:POKEM,0:NEXTC
  26 PRINT"{HOME}{BLU}PLACE YOUR FINGERS IN POSITION
  ."
  28 PRINT"{GRN}PRESS ANY KEY TO":PRINT"CONTINUE."
  30 GETES:IFES$=""THEN30
*32 PRINT"{HOME}{BLU}TYPE EACH LETTER AS{3 SPACES}I
  T APPEARS.{4 SPACES}":PRINT"{31 SPACES}"
  34 FORI=1TO8:POKEP(I),32:NEXT
  36 FORI=1TO3:FORJ=1TO8:POKEM,S(J):POKEP(J),L(J):GO
  SUB92:NEXTJ,I
  38 FORI=1TO30
  40 J=INT(8*RND(0))+1:IFJ=K THEN40
  42 K=J:POKEM,S(J):POKEP(J),L(J):GOSUB92:NEXTI
  44 PRINT"{HOME}CHOOSE:{17 SPACES}1 TRY AGAIN":PRIN
  T"{2 SPACES}2 CONTINUE PROGRAM"
  46 GETES:IFES$="1"THEN32
  48 IFES$<>"2"THEN46
  50 PRINT"{CLR}":RESTORE:FORI=1TO32:READES:NEXT
  52 DATA"A SAD LAD:","A FAD:","ASK A LAD:","A SAD FA
  D,A LAD ASKS DAD
  54 DATA"ALFALFA:","ALAS A SAD DAD,"DAD ASKS A LAD:"
  ,"ASK DAD:"
  56 FORI=1TO9:READA$(I):NEXT
  58 FORI=1TO5
  60 J=INT(9*RND(0))+1:IFA$(J)=""THEN60
*62 PRINT"{CLR}TYPE THE PHRASE":PRINT"THEN PRESS <R
  ETURN>{5 DOWN}":PRINTTAB(3);A$(J):PRINTTAB(3);
  64 B$="":FORK=1TO17
  66 GETES:IFES$=""THEN66
  68 IFASC(E$)=13THEN72
  70 PRINTE$;:B$=B$+E$:NEXTK
  72 IFB$=A$(J)THEN76
  74 POKEM,159:GOSUB96:POKEM,135:GOSUB96:POKEM,0:PRI
  NT:PRINT"{DOWN}{PUR}{3 SPACES}WRONG":GOSUB98:GO
  TO60
  76 FORD=1TOI:POKE7930+D,83:POKE7930+D+30720,2:NEXT
  D
*78 POKEM,195:GOSUB96:POKEM,207:GOSUB96:POKEM,215:G
  OSUB96:POKEM,225:GOSUB96:GOSUB96:POKEM,0:A$(J)=
  ""
  80 NEXTI

```

Fun in Less Than 4K

```
*82 PRINT:PRINT"{5 DOWN}{BLU}CHOOSE:":PRINT"  
  {2 SPACES}1 PRACTICE LETTERS{4 SPACES}2 PRACTIC  
  E WORDS":PRINT"{2 SPACES}3 END PROGRAM"  
84 GETES:IFES="1"THEN18  
86 IFES="2"THEN50  
88 IFES="3"THEN106  
90 GOTO84  
92 GETES:IFES<>P$(J)THEN92  
94 POKEM,0:POKEP(J),32:RETURN  
96 FORD=1TO99:NEXT:RETURN  
98 PRINT"{DOWN}{GRN}PRESS <RETURN>{BLU}"  
100 GETES:IFES=" "THEN100  
102 IFASC(E$)<>13THEN100  
104 RETURN  
106 PRINT"{CLR}{BLU}":END
```

The program's organization is slightly different on the Color Computer:

Line	Function
1-2	READ from DATA position P, letter P\$, and sound S.
3-4	Detect key pressed and compare it with correct letter.
5-7	PRINT title screen and instructions.
8-14	Draw hands with letters above appropriate fingers.
15	Clear letters.
16-24	Perform drill of typing letters — three times in order from left to right, then 30 random letters. A letter appears, and you must type it correctly to continue.
25-27	PRINT instructions, RESTORE DATA and ignore first 24 items; then READ words and phrases A\$ in from DATA for next drill.
28-41	Perform drill of typing words and phrases, which are chosen randomly from A\$ array. If a phrase is typed correctly, A\$(L) is set equal to "" (null) so it won't be chosen again. You must type five phrases correctly to complete the drill. B\$ is the phrase you type.
42-46	PRINT option for more practice or end program, branch appropriately, clear screen and END.

Program 9-2. Typing

TRS-80 Color Computer; MC-10

```
1 FORI=1TO8:READP(I),P$(I),S(I):NEXT:GOTO5  
2 DATA322,A,89,261,S,108,232,D,125,267,F,133,276,J  
  ,147,247,K,159,282,L,170,349,;,176  
3 E$=INKEY$:IFES<>P$(J) THEN3  
4 PRINT@P(J)," ";:RETURN
```

```

5 CLS:PRINT@109,"TYPING":PRINT@205,"UNIT 1":PRINT@
  265,"HOME POSITION":PRINT@352,"YOU WILL SEE A DI
  AGRAM OF THE"
6 PRINT"HANDS. PLACE YOUR FINGERS ON{3 SPACES}THE
  KEYS AS SHOWN.":PRINT:PRINT"PRESS ANY KEY TO ST
  ART."
7 E$=INKEY$:IF E$="" THEN 7
8 CLS:PRINT"PLACE YOUR FINGERS IN POSITION.":E$=CH
  R$(191)+CHR$(191):P$=E$+" "+E$+" "+E$+"
  {8 SPACES}"+"E$+" "+E$+" "+E$
9 PRINT@295,E$:PRINT@311,E$:PRINT@324,P$:PRINT@356
  ,P$:P$=E$+" "+P$+" "+E$:PRINT@385,P$:PRINT@417,P
  $:PRINT@449,P$:PRINT@461,E$::PRINT@465,E$:
10 P$=E$+E$+E$+E$+E$+CHR$(191):PRINT@481,P$+" "+E$
  +" "+E$+" "+P$:
11 FORI=1TO8:SOUNDS(I),2:PRINT@P(I),P$(I)::NEXT
12 PRINT@429,"SPACE":PRINT@64,"PRESS ENTER TO CON
  TINUE."
13 E$=INKEY$:IF E$="" THEN 13
14 IF ASC(E$)<>13 THEN 13
15 FORI=1TO8:PRINT@P(I)," ";:NEXT
16 PRINT @64,"TYPE THE LETTER AS IT APPEARS."
17 FORI=1TO3:FORJ=1TO8:SOUNDS(J),2:PRINT@P(J),P$(J
  );:GOSUB3:NEXTJ,I
18 FORI=1TO30
19 J=RND(8):IFJ=L THEN19
20 SOUNDS(J),2:PRINT@P(J),P$(J)::GOSUB3:L=J:NEXT
21 PRINT@0,"PRESS 1 TO TRY AGAIN{10 SPACES}":PRINT@
  38,"2 TO CONTINUE":PRINT"{30 SPACES}"
22 E$=INKEY$:IFE$="2"THEN25
23 IFE$<>"1"THEN22
24 PRINT@0,"{20 SPACES}":PRINT@38,"{13 SPACES}":GOTO
  16
25 CLS:PRINT"NOW TRY TYPING WORDS.":PRINT"TYPE THE
  PHRASE SHOWN":PRINT"THEN PRESS ENTER.":RESTORE
  :FORI=1TO24:READE$:NEXT
26 DATA A SAD LAD;,A FAD;,ASK A LAD;,A SAD FAD,A L
  AD ASKS DAD;,ASK DAD;,ALFALFA;,ALAS A SAD DAD;,
  DAD ASKS A LAD;
27 FORI=1TO9:READA$(I):NEXT
28 FORI=1TO5
29 L=RND(9):IFA$(L)=""THEN29
30 B$="":PRINT@298,A$(L):SOUND227,1
31 FORJ=1TO20
32 E$=INKEY$:IFE$=""THEN32
33 IFASC(E$)=13THEN35
34 PRINT@329+J,E$:B$=B$+E$:NEXTJ
35 IFB$=A$(L) THEN40
36 SOUND125,2:SOUND89,2:PRINT@416,"WRONG":PRINT@44
  8,"PRESS ENTER"

```


Fun in Less Than 4K

```
37 E$=INKEY$:IFE$=""THEN37
38 IFASC(E$)<>13THEN37
39 PRINT@416,"{5 SPACES}":PRINT@448,"{11 SPACES}":I
  =I-1:GOTO41
40 SOUND89,1:SOUND125,1:SOUND147,1:SOUND176,2:A$(L
  )=""
41 PRINT@298,"{15 SPACES}":PRINT@330,"{20 SPACES}":N
  EXTI
42 CLS:PRINT"PRESS 1 MORE PRACTICE ON LETTERS
  {6 SPACES}2 MORE PRACTICE ON PHRASES{6 SPACES}3
  END PROGRAM"
43 E$=INKEY$:IFE$="1"THEN8
44 IFE$="2"THEN25
45 IFE$<>"3"THEN43
46 CLS:END
```

"Typing" works this way on the TRS-80 Model I:

Line	Function
10	Branch past subroutines.
20	SET points for graphics.
100-170	Define print position P and letter P\$ to print the letter above the finger.
200-210	Check letter typed. If it's correct, it's erased.
300-360	PRINT title screen and instructions.
490-730	Draw hands on screen.
750-770	Label fingers with letters.
780-840	When ENTER is pressed, the labels are cleared.
845-970	Perform drill to type letters. Letters are presented in order from left to right three times, then randomly for 30 letters. The letter must be typed correctly to continue.
980-1000	PRINT instructions for phrase drill.
1010-1030	READ words and phrases from DATA for A\$.
1040-1220	Perform drill typing phrases. If a phrase has been typed correctly, A\$(L) is set equal to "" (null) so it won't be chosen again. Five phrases must be typed correctly to complete the drill. B\$ is what you enter.
1230-1300	PRINT option to practice more or end program and branch appropriately.
1310	Clear screen and END.

Program 9-3. Typing

TRS-80 Model I

```
10 GOTO300
20 FORJ=BTO41:FORI=A TOA+6:SET(I,J):NEXTI,J:RETURN
```

```

100 P=642:P$="A":PRINT@P,P$;:RETURN
110 P=519:P$="S":PRINT@P,P$;:RETURN
120 P=460:P$="D":PRINT@P,P$;:RETURN
130 P=529:P$="F":PRINT@P,P$;:RETURN
140 P=558:P$="J":PRINT@P,P$;:RETURN
150 P=499:P$="K":PRINT@P,P$;:RETURN
160 P=568:P$="L":PRINT@P,P$;:RETURN
170 P=701:P$="";:PRINT@P,P$;:RETURN
200 E$=INKEY$:IFE$<>P$ THEN200
210 PRINT@P," ";:RETURN
300 CLS
310 PRINT@218,"T Y P I N G"
320 PRINT@469,"UNIT 1: HOME POSITION"
330 PRINT@768,"YOU WILL SEE A DIAGRAM OF THE HANDS
."
340 PRINT"PLACE YOUR FINGERS ON THE 'HOME' KEYS."
350 PRINT:PRINT"PRESS ANY KEY TO START."
360 E$=INKEY$:IFE$=" "THEN360
490 CLS
500 PRINT"PLACE YOUR HANDS IN THIS POSITION ON THE
KEYS."
510 A=2:B=33:GOSUB20
520 A=12:B=27:GOSUB20
530 A=22:B=24:GOSUB20
540 A=32:B=27:GOSUB20
550 FORJ=42TO47
560 FORI=2TO38:SET(I,J):NEXTI,J
590 FORJ=45TO47:FORI=42TO50
600 SET(I,J):NEXTI,J
610 FORJ=42TO44:FORI=44TO50
620 SET(I,J):NEXTI,J
630 A=120:B=33:GOSUB20
640 A=110:B=27:GOSUB20
650 A=100:B=24:GOSUB20
660 A=90:B=27:GOSUB20
670 FORJ=42TO47:FORI=90TO126:SET(I,J):NEXTI,J
710 FORJ=45TO47:FORI=78TO86:SET(I,J):NEXTI,J
730 FORJ=42TO44:FORI=78TO84:SET(I,J):NEXTI,J
750 GOSUB100:GOSUB110:GOSUB120:GOSUB130
760 GOSUB140:GOSUB150:GOSUB160:GOSUB170
770 PRINT@860,"SPACE BAR";
780 PRINT@64,"PRESS ENTER TO CONTINUE."
790 E$=INKEY$:IFE$=" "THEN790
800 IFASC(E$)<>13THEN790
810 PRINT@0,"{46 SPACES}"
830 P=642:GOSUB210:P=519:GOSUB210:P=460:GOSUB210:P
=529:GOSUB210
840 P=701:GOSUB210:P=568:GOSUB210:P=499:GOSUB210:P
=558:GOSUB210
845 PRINT@0,"TYPE EACH LETTER AS IT APPEARS."

```

Fun in Less Than 4K

```
846 PRINT"{18 SPACES}"
850 FORI=1TO3
860 GOSUB100:GOSUB200:GOSUB110:GOSUB200:GOSUB120:G
  OSUB200:GOSUB130:GOSUB200
870 GOSUB140:GOSUB200:GOSUB150:GOSUB200:GOSUB160:G
  OSUB200:GOSUB170:GOSUB200
880 NEXTI
890 FORI=1TO30
900 L=RND(8):IFL=L1 THEN900
910 ONL GOSUB100,110,120,130,140,150,160,170
920 GOSUB200:L1=L
930 NEXTI
940 PRINT@0,"PRESS 1 TO TRY AGAIN{14 SPACES}"
950 PRINT"{6 SPACES}2 TO CONTINUE"
960 E$=INKEY$:IFE$="1"THEN845
970 IFE$<>"2"THEN960
980 CLS
990 PRINT"NOW TRY TYPING WORDS."
1000 PRINT"TYPE THE PHRASE SHOWN THEN PRESS ENTER.
  "
1010 RESTORE
1020 DATA A SAD LAD; ,A FAD; ,ASK A LAD; ,A SAD FAD,A
  LAD ASKS DAD; ,ASK DAD; ,ALFALFA; ,ALAS A SAD D
  AD; ,DAD ASKS A LAD;
1030 FORI=1TO9:READA$(I):NEXTI
1040 FORI=1TO5
1050 L=RND(9):IFA$(L)=" "THEN1050
1055 B$=""
1060 PRINT@470,A$(L)
1070 FORJ=1TO20
1080 E$=INKEY$:IFE$=""THEN1080
1085 IFASC(E$)=13THEN1120
1090 PRINT@533+J,E$
1100 B$=B$+E$
1110 NEXTJ
1120 IFB$=A$(L) THEN1195
1130 PRINT@832,"WRONG"
1140 PRINT@896,"PRESS ENTER."
1150 E$=INKEY$:IFE$=""THEN1150
1160 IFASC(E$)<>13THEN1150
1170 PRINT@832,"{5 SPACES}"
1180 PRINT@896,"{12 SPACES}"
1185 I=I-1
1190 GOTO1200
1195 A$(L)=" "
1200 PRINT@470,"{15 SPACES}"
1210 PRINT@534,"{20 SPACES}"
1220 NEXTI
1230 CLS
```

```
1240 PRINT"PRESS 1 MORE PRACTICE ON LETTERS"  
1250 PRINTTAB(6);"2 MORE PRACTICE ON PHRASES"  
1260 PRINTTAB(6);"3 END PROGRAM"  
1270 E$=INKEY$  
1280 IFE$="1"THEN490  
1290 IFE$="2"THEN980  
1300 IFE$<>"3"THEN1270  
1310 CLS:END
```

Teeth Wisdom

Computers: VIC-20; TI-99/4A

"Teeth Wisdom" teaches the names of the teeth — central incisors, lateral incisors, cuspids, bicuspid, and molars. The computer draws the teeth on the screen with the labels. After you've studied the names, press ENTER or RETURN to clear the labels. A quiz is given by blinking certain teeth. To answer, you press the number corresponding to the name.

Teeth Wisdom works like this on the VIC:

Line	Function
5-6	PRINT title screen.
7	READ names of teeth in N\$ array and create delay.
8	Define B\$ and C\$ strings for use in clearing labels.
9	Turn on volume and set S equal to a speaker value for sounds.
10-12	PRINT instruction screen.
15	Change screen color to pink.
16-20	Draw and label teeth.
22	Wait for RETURN to be pressed before continuing.
30-32	Clear labels.
35	PRINT "NAME THE TEETH".
40	Set elements of W\$ array equal to element of N\$ array.
42-48	Randomly list names of teeth.
50-66	Perform quiz.
52-54	Randomly choose teeth.
56-58	Blink appropriate teeth while waiting for response.
60-62	If answer is incorrect, play sound and return for another response.
65-66	If answer is correct, play appropriate music, set A(I)=0 so tooth will not be chosen again, and go to next problem.
70-72	PRINT option to try again and branch appropriately.

Fun in Less Than 4K

- 74-76 If answer is Y for *yes*, clear previous list of answers and branch to beginning of quiz.
- 80-89 Blink teeth by POKEing memory locations with color numbers.
- 92 Delay loop for sounds.
- 94-97 Wait for RETURN key to be pressed before continuing program.
- 99 DATA containing names of teeth.
- 100 Clear screen, return to normal screen and border colors, and END.

Program 9-4. Teeth Wisdom

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```
5 PRINT"{CLR}{BLU}{4 DOWN}{2 SPACES}*****
**":PRINT"{SHIFT-SPACE} *";SPC(14);"":PRINT"
{2 SPACES}* TEETH WISDOM *"
6 PRINT"{2 SPACES}*";SPC(14);"":PRINT"{2 SPACES}*
*****"
7 FORC=1TO5:READN$(C):NEXT:FORC=1TO900:NEXT
8 B$="{16 SPACES}":C$="{10 SPACES}"
9 POKE36878,15:S=36876
10 PRINT"{CLR}{2 DOWN}YOU WILL SEE A DIAGRAMOF THE
TEETH WITH THE NAMES OF THE TEETH."
11 PRINT"{DOWN}WHEN YOU KNOW THE{5 SPACES}NAMES, P
RESS <RETURN>."
*12 PRINT"{DOWN}THE LABELS WILL CLEAR AND YOU WILL
{SPACE}BE GIVEN A QUIZ.":PRINT"{3 DOWN}{GRN}PRE
SS RETURN{BLU}":GOSUB94
15 POKE36879,169
*16 PRINT"{CLR}{DOWN}":PRINTTAB(4);"CENTRAL INCISOR
S":PRINTTAB(4);"T":PRINTTAB(4);"T{A}LATERAL I
NCISORS{2 SPACES}{WHT}ZQQZ"
18 PRINT" Z{4 SPACES}Z{BLU}**CUSPIDS":PRINT"
{WHT}A";SPC(6);"A {BLU}I":PRINT"{WHT}A";SPC(6);
"A {BLU}KBICUSPIDS"
20 PRINT"{WHT}X";SPC(6);"X {BLU}I":PRINT"{WHT}X";S
PC(6);"X {BLU}{Q}*MOLARS":PRINT"{WHT}X";SPC(
6);"X {BLU}K"
22 PRINT"{3 DOWN}{GRN}PRESS RETURN":GOSUB94
*30 PRINT"{UP}";B$:PRINT"{HOME}{2 DOWN}{4 SPACES}";
B$:PRINT"{5 SPACES}":PRINT"{6 SPACES}";B$:PRINT
TAB(7);C$:PRINTTAB(9);" ":PRINTTAB(9);C$
32 PRINTTAB(9);" ":PRINTTAB(9);C$:PRINTTAB(9);" "
35 PRINT"{DOWN}{BLU}NAME THE TEETH{DOWN}"
```

```

40 FORC=1TO5:W$(C)=N$(C):NEXT
42 FORC=1TO5
44 I=INT(5*RND(0)+1)
46 IFW$(I)=""THEN44
48 PRINTC;W$(I):A(I)=C:W$(I)="" :NEXTC
50 FORC=1TO5
52 I=INT(5*RND(0)+1)
54 IFA(I)=0THEN52
56 ON I GOSUB80,82,84,86,88
58 GETA$:IFA$=""THEN56
60 IF ASC(A$)=A(I)+48THEN 65
62 POKES,159:GOSUB92:POKES,135:GOSUB92:POKES,0:GOT
O56
*65 POKES,195:GOSUB92:POKES,207:GOSUB92:POKES,215:G
OSUB92:POKES,225:GOSUB92:GOSUB92:POKES,0:A(I)=0
66 NEXTC
70 PRINT"{DOWN}{GRN}TRY AGAIN? (Y/N){BLU}";
72 GETA$:IFA$="N"THEN100
74 IFA$<>"Y"THEN72
76 PRINT"{7 UP}{BLU}":FORC=1TO7:PRINTB$;"
{3 SPACES}":NEXT:PRINT"{8 UP}":GOTO40
80 POKE38513,2:POKE38514,2:POKE38513,1:POKE38514,1
:RETURN
82 POKE38512,2:POKE38515,2:POKE38512,1:POKE38515,1
:RETURN
84 POKE38533,2:POKE38538,2:POKE38533,1:POKE38538,1
:RETURN
*86 POKE38554,2:POKE38561,2:POKE38576,2:POKE38583,2
:POKE38554,1:POKE38561,1:POKE38576,1:POKE38583,
1:RETURN
88 POKE38598,2:POKE38605,2:POKE38620,2:POKE38627,2
:POKE38642,2:POKE38649,2
89 POKE38598,1:POKE38605,1:POKE38620,1:POKE38627,1
:POKE38642,1:POKE38649,1:RETURN
92 FORD=1TO150:NEXT:RETURN
94 FORD=1TO10:GETA$:NEXT
95 GETA$:IFA$=""THEN95
96 IF ASC(A$)<>13THEN95
97 RETURN
99 DATACENTRAL INCISORS,LATERAL INCISORS,CUSPIDS,B
ICUSPIDS,MOLARS
100 PRINT"{CLR}":POKE36879,27:END

```

Here's the explanation for the TI version:

Line	Function
110-170	Clear screen and PRINT title screen.
180-230	Define graphics characters and blink asterisks on screen.
240-330	DATA containing graphics definitions.

Fun in Less Than 4K

- 340-360 Define color sets 9 through 13 as white on light red for teeth.
- 370-390 Define color sets for light red on transparent for graphics surrounding teeth.
- 400-510 Clear screen, PRINT instructions, and define strings as groups of characters for later printing.
- 520-560 READ in names of five groups of teeth as N\$ array and set the W\$ array elements equal to the N\$ array elements.
- 570 PRINT message to press ENTER and wait for response.
- 580-690 Clear screen and PRINT teeth with labels.
- 700 PRINT message to press ENTER and wait for response.
- 710-760 Clear message and labels.
- 770 PRINT quiz title.
- 780-850 Randomly PRINT names of teeth on screen from the W\$ array. A(I) will be the correct corresponding answer.
- 860-1060 Perform quiz.
- 870-880 Randomly choose teeth.
- 890-920 Blink teeth blue and white while waiting for response.
- 930-940 If number 1-5 is pressed, show which number was pressed; otherwise, return to line 890.
- 950-990 If answer is incorrect, play sound effect and return for another response.
- 1000-1030 If answer is correct, play music.
- 1040 Clear answer chosen.
- 1050-1060 Set element A to zero so tooth will not be chosen again and return for next problem.
- 1070-1100 PRINT option to try again, wait for response, and branch appropriately.
- 1110-1140 If you want to try again, set W\$ array elements equal to names of teeth and branch to beginning of exercise.
- 1150 Stop.
- 1160-1190 PRINT "PRESS ENTER" and wait for response.
- 1200-1210 Clear screen and END.

Program 9-5. Teeth Wisdom

TI-99/4A

```

110 CALL CLEAR
120 PRINT TAB(4); "*****"
130 PRINT TAB(4); "*" ; TAB(21); "*"
140 PRINT TAB(4); "* TEETH WISDOM  *"
150 PRINT TAB(4); "*" ; TAB(21); "*"
160 PRINT TAB(4); "*****": :::::
170 PRINT :::
180 FOR C=94 TO 157
190 CALL COLOR(2,13,1)
200 READ C$
210 CALL CHAR(C,C$)
220 CALL COLOR(2,16,1)
230 NEXT C
240 DATA 0000001F1010101,000000FF
250 DATA 00000F1F3F7F7F7F,000083C7C7E7E7EF,0000E0F
0F8FCFEFE,FFFFFFFFFFFFFFFF,EFEFEFEFEFEFEF,0
''
260 DATA 70F8FCFCFEFEFEFF,FEF8C,0E1F3F3F7F7F7FFF,7
F1F03,,,,,00000000000000F8,010101010101,FCFEFF
FFFECF83
270 DATA 000000000000001F,3F7FFFFFF7F3F1F0C,8080808
0808,,,0F1F1F1F1F1F0E,E0F8F8F8F8F8F,071F1F1F1F
1F0F
280 DATA F0F8F8F8F8F87,1F3F3F3F3F3F1F,F0F8F8F8F8F8
F,0F1F1F1F1F1F0F,F8FCFCFCFCFCF8,7FFFFFFFFFFF7F
290 DATA E0F0F0F0F0F0E,070F0F0F0F0F07,FFFFFFFFFFFF
FE,0303030303030301,FFFFFFFFFFFFFFFF,C0C0C0C0C
0C0C08,""
300 DATA FFFFCF0E0C08,FFFF7E181,FFFF3F0F070301,FF
FEFECFCF8F8F8,FF7F7F3F3F1F1F1F,F0F0F0F0F0F0F0
F
310 DATA 0F0F0F0F0F0F0F0F,E0E0E0E0C0C0808,07070707
03030101,0000000000071FFF,0000031FFFFFFFFF,00F
FFFFFFFFFFFFFFFF
320 DATA 0000C0F8FFFFFFFF,000000000070F8FF,01070F1
F3F3F7FFF,80E0F0F8FCFCFEFF,010103030307070F,80
80C0C0C0E0E0F
330 DATA 0F0F1F1F1F3F3F3F,F0F0F8F8F8FCFCFC,3F7F7F7
F7F7F7F7F,FCFEFEFEFEFEFEFE
340 FOR C=9 TO 13
350 CALL COLOR(C,16,10)
360 NEXT C
370 CALL COLOR(14,10,1)
380 CALL COLOR(15,10,1)
390 CALL COLOR(16,10,1)
400 CALL CLEAR
410 CALL COLOR(2,2,1)
420 PRINT "YOU WILL SEE A DIAGRAM OF"

```


Fun in Less Than 4K

```
430 PRINT : "THE TEETH WITH THE NAMES"
440 PRINT : "OF THE TEETH."
450 A$=CHR$(128)&CHR$(129)
460 B$=CHR$(130)&CHR$(131)
470 PRINT :: "WHEN YOU KNOW THE NAMES,"
480 PRINT : "PRESS <ENTER>."
490 D$=CHR$(132)&CHR$(133)&CHR$(134)
500 PRINT :: "THE LABELS WILL CLEAR AND"
510 PRINT : "YOU WILL BE GIVEN A QUIZ.":::
520 FOR C=1 TO 5
530 READ N$(C)
540 W$(C)=N$(C)
550 NEXT C
560 DATA CENTRAL INCISORS,LATERAL INCISORS,CUSPIDS
    ,BICUSPIDS,MOLARS
570 GOSUB 1160
580 CALL CLEAR
590 PRINT TAB(8);"^_CENTRAL INCISORS"
600 PRINT TAB(5);CHR$(145)&CHR$(146)&CHR$(147)&CHR
$(148)&CHR$(149)
610 PRINT TAB(4);CHR$(150);"e`abe";CHR$(151);"_LA
TERAL INCISORS"
620 PRINT " ";CHR$(150);"phcdcjs";CHR$(151)
630 PRINT " ";CHR$(152);"qrieeektu";CHR$(153);"CUS
PIDS"
640 PRINT " ";CHR$(154);"xyeeeeez{";CHR$(155);"BIC
USPIDS"
650 PRINT " ";CHR$(156);" |{, }e";CHR$(136)&CHR$(137
)&CHR$(138);"e{, }";CHR$(127)&CHR$(157)
660 PRINT " e";A$;CHR$(139);" ";CHR$(140);B$;"e"
;"MOLARS"
670 PRINT " e";A$;CHR$(141);" ";CHR$(142);B$;"e"
680 PRINT " ";D$;CHR$(143);" ";CHR$(144);D$
690 PRINT " eee eee":::
700 GOSUB 1160
710 CALL HCHAR(23,16,32,13)
720 CALL HCHAR(10,10,32,18)
730 CALL HCHAR(12,13,32,18)
740 CALL HCHAR(14,15,32,7)
750 CALL HCHAR(15,15,32,9)
760 CALL HCHAR(17,15,32,6)
770 PRINT TAB(8);"NAME THE TEETH"::
780 FOR C=1 TO 5
790 RANDOMIZE
800 I=INT(5*RND+1)
810 IF W$(I)="" THEN 800
820 PRINT TAB(9);C;W$(I)
830 A(I)=C
840 W$(I)=""
```

```
850 NEXT C
860 FOR C=1 TO 5
870 I=INT(5*RND+1)
880 IF A(I)=0 THEN 870
890 CALL KEY(0,K,S)
900 CALL COLOR(I+8,6,10)
910 CALL COLOR(I+8,16,10)
920 IF S<1 THEN 890
930 IF (K<49)+(K>53)THEN 890
940 CALL HCHAR(18+K-48,11,62)
950 IF K-48=A(I)THEN 1000
960 CALL SOUND(150,330,0)
970 CALL SOUND(150,262,0)
980 CALL VCHAR(19,11,32,5)
990 GOTO 890
1000 CALL SOUND(150,262,0)
1010 CALL SOUND(150,330,0)
1020 CALL SOUND(150,392,0)
1030 CALL SOUND(200,523,0)
1040 CALL VCHAR(19,11,32,5)
1050 A(I)=0
1060 NEXT C
1070 PRINT :: "TRY AGAIN? (Y/N)"
1080 CALL KEY(0,K,S)
1090 IF K=78 THEN 1200
1100 IF K<>89 THEN 1080
1110 FOR C=1 TO 5
1120 W$(C)=N$(C)
1130 NEXT C
1140 GOTO 580
1150 STOP
1160 PRINT TAB(14); "PRESS <ENTER>"
1170 CALL KEY(0,K,S)
1180 IF K<>13 THEN 1170
1190 RETURN
1200 CALL CLEAR
1210 END
```

In this program, you'll see spaces between quotation marks, such as " ". Type in the quote marks and press the space bar the correct number of times. If it's hard to tell how many spaces to enter, you can use the line above as a reference. For example, in line 690, you can tell that five spaces are needed between the e's by counting the characters in line 680 directly above the gap.

Bake a Cake

Computer: VIC-20

Baking a cake, but you don't have a recipe? Here's a program for

Fun in Less Than 4K

the VIC-20 that will help. The program contains recipes for several cakes. You can select the recipe, see it on the screen, and even enlarge or reduce it. If you would like to enlarge the recipe enter a number such as 3 to triple the recipe. To reduce it, enter a decimal fraction, such as .5 to halve the recipe. The converted recipe will then appear on the screen.

You don't know what to make? Select the ingredient list. As the inventory is listed, press Y for yes if you have the ingredient or N if you don't. After the inventory list is complete, the computer will tell you which recipes you can make, given that inventory.

For convenience in programming, the amounts in the recipes are given in decimals. For example, $\frac{2}{3}$ cup sugar is written as .67 c. sugar.

The DATA statements in lines 101-107 contain the measure, then the ingredient for 25 ingredients. Line 3 READs A\$, the measure, and B\$(N,0), the name of the ingredient, then assigns I\$(N) equal to the measure plus a space plus the ingredient name. B\$ is used in the inventory list, and I\$ is used in printing the recipe. Z is the number of ingredients, minus one (because the subscripts start with the number zero).

As you are typing the DATA statements, you may notice two or more commas together with nothing between them (,,,). Be sure to get the right number of commas as you are typing. The commas indicate a null string, or a string variable equal to "".

The recipes are in the DATA statements in lines 109-123. The first item is the name of the cake. The next items are the amounts of the ingredients in the following order:

cups shortening	number of eggs
cups flour	number of egg whites
cups sugar	tsp. red food coloring
cups brown sugar	ounces chocolate
tsp. baking powder	tbbsp. cocoa
tsp. salt	tsp. vanilla
tsp. soda	tsp. cinnamon
cups cherry juice	tsp. nutmeg
number of cherries	tsp. vinegar
cups bananas (mashed)	cups salad oil
cups sauerkraut	cups water
cups milk	cups oatmeal
cups buttermilk	

The DATA will contain a null string if the cake does not contain that ingredient.

An example is the first part of line 109, the data for Banana Cake:
DATA BANANA, .67,2.5,1.67,,1,1,1,,1,,, .67,2,,,,,,,,,,,,,

The name of the cake is BANANA. The recipe is .67 cup shortening (two-thirds), 2.5 cups flour, 1.67 cup sugar, 1 tsp. baking powder, 1 tsp. salt, 1 tsp. soda, 1 cup mashed bananas, .67 cup buttermilk, and 2 eggs.

To print a recipe, the data is RESTORED first, then the first 50 ingredients and measures are read and ignored (line 73). The key you press to choose a recipe is E\$; the ASCII code will be 65 for A, 66 for B, and so on. Lines 75-77 figure out which recipe was chosen and read through the title and ingredients to get to the appropriate recipe. Line 79 PRINTs the title of the cake. Line 81 READs the amount from the DATA statement. If the amount is a null or zero, that ingredient is not PRINTed. However, if there is a value, the value is shown. The corresponding measure and name of the ingredient are READ from the I\$ array.

Line 83 defines variables M(I) for measure and C\$(I) for ingredient for only those ingredients in the recipe. These values are used in printing the converted recipe, lines 97-99.

For the inventory list, the computer keeps track of your Y or N answers in the BR(N,1) array, where N varies from 0 to 24 for the ingredients. Line 33 checks to see if an N is stored as a no answer for flour, sugar, or salt. If any one of these three items has a no, no cakes can be baked. Y is a variable for the number of Y answers; if there are not enough ingredients with a Y answer, you cannot make a cake (checked in line 35).

Line 47 RESTOREs the DATA, then ignores the first 50 items. Lines 49-51 check through the recipe for each cake. If there is an amount listed for an ingredient, the corresponding B\$(N,1) value is checked. If it is N for no, you are lacking one of the ingredients required for the cake. The rest of the ingredients are skipped over, and the computer goes to the next recipe. If each of the required ingredients also has a B\$ value of Y for yes, the cake can be made and the name of the cake is printed.

These are real recipes that really work. No baking instructions are given because there isn't room in the VIC version, but usually the person baking knows how to bake the cake and just needs to be reminded of the amount of each ingredient. In case you want to try these recipes, all of these recipes are for cakes to be baked in two 9-inch layers. Mix the shortening with the sugars, add the dry ingredients alternately with the liquid, and then add the eggs and vanilla. Bake at 350° to 375°.

In the cherry cake, cut up the cherries before adding to the mixture. For the oatmeal cake, first boil the water, and then add the oatmeal. Let that mixture cool while you mix everything else — add the oatmeal mixture last. The “wacky” cake you can mix all at once — just dump everything into one big bowl and mix it up. For the red velvet cake, combine the ingredients as usual, except for the vinegar and soda. Mix the vinegar and soda together and fold into the rest of the batter. The sauerkraut is a moist chocolate cake — just don't tell anyone what it is until after they've eaten it. Rinse the sauerkraut well, then chop it into small pieces before combining it with the rest of the cake batter.

This program comes very close to filling all the available memory. As with other VIC-20 programs in this book, be sure to use the abbreviations for all the BASIC words, such as ? for PRINT and D SHIFT A for DATA in the lines marked with the asterisk.

Be sure you copy the DATA statements exactly for the correct recipes. There are no spaces except in the names of the cakes. If there are commas together, do not put spaces between the commas.

Program 9-6. Bake a Cake

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, “BASIC Keyword Abbreviations,” for the proper abbreviations.)

```
1 DIMI$(24),B$(24,1),M(11),C$(11):Z=24
3 FORN=0TOZ:READA$,B$(N,0):I$(N)=A$+" "+B$(N,0):NEXT
5 PRINT"{CLR}{BLU}{5 DOWN}{5 RIGHT}BAKE A CAKE"
7 PRINT"{2 DOWN}CHOOSE:":PRINT"{DOWN} 1 NEED TO KNOW":PRINT"{3 SPACES}WHAT CAN BE MADE."
9 PRINT"{DOWN} 2 WANT TO SEE":PRINT"{3 SPACES}A CERTAIN RECIPE":PRINT"{DOWN} 3 END PROGRAM"
11 GETES$:IFES$="3"THEN200
13 IFES$="2"THEN61
15 IFES$<>"1"THEN11
17 PRINT"{CLR}{BLU}IN THE FOLLOWING LIST PRESS 'Y' IF YOU HAVE THE INGREDIENT,"
19 PRINT"'N' IF YOU DO NOT.{4 SPACES}PRESS 'S' TO {SPACE}START{4 SPACES}OVER.{2 DOWN}"
21 Y=0:FORN=0TOZ:PRINTB$(N,0);"{2 SPACES}{RED}-- {BLK}";:GOSUB150
23 GETES$:IFES$="S"THEN17
25 IFES$="N"THENPRINT"N{BLU}":GOTO31
27 IFES$<>"Y"THEN23
29 PRINT"Y{BLU}":Y=Y+1
31 B$(N,1)=E$:NEXTN:C=0:PRINT"{2 DOWN}YOU CAN MAKE :"
```

```

33 IFB$(1,1)="N"ORB$(2,1)="N"ORB$(5,1)="N"THEN37
35 IFY>7THEN47
37 PRINT"NOTHING TODAY.":PRINT"YOU NEED MORE SUPPL
IES"
39 PRINT"{2 DOWN}{GRN}PRESS RETURN{BLU}";:GOSUB150
41 GETES$:IFE$=""THEN41
43 IF ASC(E$)<>13THEN41
45 GOTO5
47 RESTORE:FORN=0TOZ*2+1:READE$:NEXT:READA$
49 FORN=0TOZ:READE$:IFE$=""ORE$=""0"THEN53
51 IFB$(N,1)="N"THENFORI=N+1TOZ:READE$:NEXTI:GOTO5
5
53 NEXTN:PRINTA$;" CAKE":C=C+1
55 READA$:IFA$<>"Z"THEN49
57 IFC=0THEN37
59 PRINT"{DOWN}GO AHEAD AND BAKE!":GOTO39
*61 PRINT"{CLR}{BLU}{DOWN}CHOOSE:{2 DOWN}":PRINT"A
{SPACE}BANANA CAKE":PRINT"B CHERRY CAKE":PRINT"
C CHOCOLATE CAKE"
63 PRINT"D DEVIL'S FOOD CAKE{3 SPACES}E GOLD LAYER
CAKE":PRINT"F OATMEAL CAKE"
65 PRINT"G RED VELVET CAKE":PRINT"H SAUERKRAUT CAK
E":PRINT"I SPICE CAKE"
67 PRINT"J TWO-EGG CAKE":PRINT"K WACKY CAKE":PRINT
"L WHITE CAKE":GOSUB150
69 GETES$:IFE$=""THEN69
71 A=ASC(E$):IFA<65ORA>76THEN69
73 RESTORE:PRINT"{CLR}{BLU}";:FORN=0TO2*Z+1:READE$
:NEXT
75 IFA=65THEN79
77 FORN=1TOA-65:READA$:FORI=0TOZ:READE$:NEXTI,N
79 READA$:PRINTA$;" CAKE{DOWN}":I=0
81 FORN=0TOZ:READE$:IFE$=""ORVAL(E$)=0THEN85
83 M(I)=VAL(E$):C$(I)=I$(N):PRINTM(I);TAB(6);C$(I)
:I=I+1
85 NEXTN
87 PRINT"{2 DOWN}{RED}CONVERT RECIPE? (Y/N){BLU}"
89 GETES$:IFE$="N"THEN39
91 IFE$<>"Y"THEN89
93 PRINT"{DOWN}MULTIPLY BY WHAT":PRINT"NUMBER OR D
ECIMAL?":PRINT"{RED} --{BLU}";
95 INPUTF:IFF<=0THENPRINT"SORRY, F>0":GOTO93
97 F=INT(F*100)/100:PRINT"{CLR}{BLU}";F;"TIMES ORI
GINAL{DOWN}":PRINTA$;" CAKE{DOWN}"
99 FORN=0TOI-1:PRINTF*M(N);TAB(6);C$(N):NEXT:GOTO8
7
101 DATAC.,SHORTENING,C.,FLOUR,C.,SUGAR,C.,BROWN S
UGAR,TSP.,BAKING PDR,TSP.
103 DATASALT,TSP.,SODA,C.,CHERRY JUICE,,CHERRIES,C
.,BANANAS,C.,SAUERKRAUT,C.,MILK

```

Fun in Less Than 4K

```

105  DATAC.,BUTTERMILK,,EGGS,,EGG WHITES,TSP.,RED C
    OLOR,ØZ.,CHOCOLATE,TBSP.,COCOA,TSP.
107  DATAVANILLA,TSP.,CINNAMON,TSP.,NUTMEG,TSP.,VIN
    EGAR,C.,SALAD OIL,C.,WATER,C.,OATMEAL
*109  DATABANANA,.67,2.5,1.67,,1,1,1,,1,,.67,2,,,
    ,,CHERRY,.5,2.25,1.33,,3,.5,,.25
111  DATA16,,,5,,,4,,,,,CHOCOLATE,.67,2.5,1.
    75,,,,.5,1,,2,,2,,1,,1.25,Ø
*113  DATADEVIL'S FOOD,.67,2.25,2,,1,1,1,,,,1.25,,3
    ,,1,3,,,,,GOLD LAYER,.5,2.25,1.5,,3
*115  DATA1,,,,,1.67,,2,,,,1.5,,,,,OATMEAL,.5,1.
    5,1,1,,.5,1,,,,,2,,1,.75,.25,,1.25
*117  DATA1,RED VELVET,.5,2.75,1.5,,.5,1.5,,,,,1,2
    ,,6,,2,1,,1,,SAUERKRAUT,.67,2.25,1.25
*119  DATAØ,1,.25,,,,.67,,,3,,,8,1,,,,1.25,,SPICE
    ,,75,2.25,1,,1,1,1,,,,,1,3,,,,,1.5,,Ø
*121  DATAØ,,TWO-EGG,.5,2.25,1.5,,2.5,1,,,,,1,,2,,
    ,,1,,,,,WACKY,,2.5,1.5,,1,1,,,,,Ø
123  DATAØ,,6,1,,1.5,.75,1.5,,WHITE,.75,2.25,1.5,,
    3,1,,,,,1,,5,,1.5,,,,,Z
150  FORI=ØTO9:GETE$:NEXTI:RETURN
200  PRINT"{CLR}{BLU}":END

```

Cookie File

Computer: VIC-20

The DATA method used in "Bake a Cake" can be used for a variety of things — recipes of any kind, craft instructions, and even a type of word processor. "Cookie File" illustrates how the program can be adapted for cookie recipes.

The general baking instructions are as follows:

- Almond cookies — roll into balls, flatten slightly, and place blanched almond on top. Brush with egg if desired.
- Ball cookies — drop cookies onto sheet, then flatten with ice cube or moist cloth; sprinkle colored cake decors on top; and bake just until golden brown around the edges.
- Brownies — melt the cocoa with the shortening first and bake in square pan.
- Butterscotch bars — melt shortening (or butter) with brown sugar; cool, then add other ingredients; and bake in rectangular glass baking dish.
- Chocolate chip cookies — make as drop cookies.
- Chocolate drop cookies — make as drop cookies.
- Mexican Wedding Cookies — roll into balls, bake about 20 minutes, and roll in powdered sugar while still warm, then again when cool.

- Oatmeal chocolate chips — make as drop cookies.
- Oatmeal crisps (refrigerator cookie) — form into long roll and slice, then bake.
- Snickerdoodles — roll dough into balls, and then roll in cinnamon and sugar mixture before baking.
- Toffee bars — press into 9 x 13 pan or on cookie sheet (about ½ inch thick).

Program 9-7. Cookie File

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```

1 DIMG$(17),V$(17,1),M(9),F$(9):POKE36879,30
2 FORI=0TO17:READA$,V$(I,0):G$(I)=A$+" "+V$(I,0):N
  EXT
* 3 PRINT"{CLR}{DOWN}{5 SPACES}COOKIE FILE":PRINT"
  {DOWN}CHOOSE{2 DOWN}":PRINT" 1 NEED TO KNOW":PRI
  NT"{3 SPACES}WHAT CAN BE MADE"
4 PRINT"{2 DOWN} 2 WANT TO SEE RECIPE":PRINT"
  {2 DOWN} 3 END PROGRAM"
5 GETR$:IFR$="3"THEN62
6 IFR$="2"THEN33
7 IFR$<>"1"THEN5
* 8 PRINT"{CLR}{BLU}IN THE FOLLOWING LIST PRESS 'Y'
  {SPACE}IF YOU HAVE THE INGREDIENT":PRINT"'N' IF
  {SPACE}YOU DO NOT."
9 PRINT"PRESS 'S' TO START{4 SPACES}OVER.{2 DOWN}"
  :YS=0:FORK=0TO17:PRINTV$(K,0);"{2 SPACES}{RED}
  {2 P}{BLK}";
10 GETR$:IFR$=""THEN10
12 IFR$="S"THEN8
13 IFR$="N"THENPRINT"N":GOTO16
14 IFR$<>"Y"THEN10
15 PRINT"Y":YS=YS+1
16 V$(K,1)=R$:PRINT"{BLU}";:NEXT:C=0:PRINT"
  {3 DOWN}YOU CAN MAKE:"
17 IFV$(0,1)="N"ORV$(6,1)="N"THEN19
18 IFYS>4THEN24
19 PRINT"NOTHING TODAY.":PRINT"YOU NEED MORE SUPPL
  IES"
20 PRINT"{3 DOWN}{GRN}PRESS RETURN.{BLU}"
21 GETR$:IFR$=""THEN21
22 IFASC(R$)<>13THEN21
23 GOTO3
24 RESTORE:FORJ=1TO36:READB$:NEXT:READA$
25 FORJ=0TO17:READB$:IFB$=""THEN27
26 IFV$(J,1)="N"THEN28
27 NEXT:PRINTA$:C=C+1

```


Fun in Less Than 4K

```
28 READA$: IFA$="ZZZ"THEN31
29 IFLEN(A$)<6THEN28
30 GOTO25
31 IFC=ØTHEN19
32 PRINT"{DOWN}GO AHEAD AND BAKE!":GOTO2Ø
33 PRINT"{CLR}CHOOSE:":PRINT"{DOWN}A ALMOND COOKIE
S{6 SPACES}B BALL COOKIES":PRINT"C BROWNIES"
34 PRINT"D BUTTERSCOTCH BARS{3 SPACES}E CHOCO CHIP
COOKIES{2 SPACES}F CHOCO DROP COOKIES"
*35 PRINT"G MEX WEDDING COOKIES H OATMEAL CHOCO CHI
PS I OATMEAL CRISPS":PRINT"J SNICKERDOODLES"
36 PRINT"K SUGAR COOKIES":PRINT"L TOFFEE BARS"
37 GETR$:IFR$=""THEN37
38 A=ASC(R$):IFA<65ORA>76THEN37
39 RESTORE:PRINT"{CLR}{DOWN}":FORI=1TO36:READB$:NE
XT:IFA=65THEN41
4Ø FORI=1TOA-65:READA$:FORK=ØTO18:READR$:NEXTK,I
41 READA$:PRINTA$:K=Ø:PRINT
42 FORJ=ØTO17:READB$:IFB$=""ORVAL(B$)=ØTHEN44
43 M(K)=VAL(B$):F$(K)=G$(J):PRINTM(K);TAB(6);F$(K)
:K=K+1
44 NEXTJ:READT:PRINT"{DOWN}BAKE AT";T
45 PRINT"{DOWN}{RED}CONVERT RECIPE? (Y/N)"
46 GETR$:IFR$="N"THEN2Ø
47 IFR$<>"Y"THEN46
48 PRINT"{DOWN}MULTIPLY BY WHAT":PRINT"NUMBER OR D
ECIMAL?":PRINT"[%2 P]{BLU}";
49 INPUTF:IFF<=ØTHENPRINT"SORRY, F>Ø":GOTO48
5Ø F=(INT(F*1ØØ)/1ØØ):PRINT"{CLR}":PRINTF;"TIMES O
RIGINAL":PRINT:PRINTA$:PRINT
51 FORI=ØTOK-1:PRINTF*M(I);F$(I):NEXT:GOTO45
52Ø DATAC.,SHORTENING,C.,SUGAR,C.,BROWN SUGAR,C.
525 DATAPDRD SUGAR,,EGGS,TSP.,VANILLA,C.,FLOUR,TSP
53Ø DATABAKING PDR,TSP.,BAKING SODA,TSP.,SALT,TSP.
,CINNAMON,TBSP.,COCOA,TSP.
535 DATAALMOND EXT,C.
54Ø DATAMILK,C.,OATMEAL,OZ.,CHOCO CHIPS,DOZ.,ALMON
DS,TSP.
545 DATACAKE DECORS,ALMOND COOKIES,2,2,,,2
55Ø DATA",,4,2,,,,,2,,,,,4,,375,BALL COOKIES,,.5,.33
,,,1,.5,.75,,,,,,,,,,,,,2,375
555 DATABROWNIES,.5
56Ø DATA1,,,2,1,.75,.5,.5,,6,,,,,,35Ø
565 DATABUTTERSCOTCH BARS,.5,2,,2,1,1.75,2,,.25,,
,,,,,,35Ø
57Ø DATACHOCO CHIP COOKIES,.5,.25,.5,,1,.5,1,,.5,,
5,,,,,6,,,375
575 DATACHOCO DROP COOKIES,.5,,1,Ø
58Ø DATAØ,1,1.67,,,5,.5,,6,,.5,,,,,35Ø
```

```

585 DATAMEXICAN WEDDING,.75,,,,.67,,1,1.5,,,,.25,1,,
    ,,.75,,,Ø
59Ø DATA325
595 DATAOATMEAL CHOCO,1,1,.5,,2,1,2,,1,1,,,,,2,6,,
    ,35Ø,OATMEAL CRISPS,1,1,1,,2,1,1.5
6ØØ DATAØ,1,1,,,,,3,,,,,35Ø,SNICKERDOODLES,1,1.5,,,
    2,,2.75,3,,,5,,,,,4ØØ
6Ø5 DATASUGAR COOKIES
61Ø DATA.67,.75,,,1,.5,2,1.5,,,25,,,,.25,,,,,375
615 DATATOFFEE BARS,1,1,,,,,1,2,,,,,6,,,35Ø,ZZZ
62Ø PRINT"{CLR}":END

```

States and Capitals

Computers: VIC-20; MC-10; TRS-80 Model I; TI-99/4A

Can you name the capital of each state? This program gives you practice in just that. States are chosen in a random order. You have two chances to get the capital city correct, and spelling counts. If the city is missed twice, you'll see the right answer, and that state will appear again. The drill continues until all 50 states' capitals have been correctly entered. The number of total guesses is then shown. The closer your score is to 50, the better your knowledge is of states and their capitals.

All three versions work about the same.

Program control first branches past subroutines. PRESS RETURN or PRESS ENTER will display, and the computer will wait for the response. The VIC-20 version has a delay subroutine and also initializes sound parameters.

The screen clears and the title is printed. The DIMension statement allows for state names S\$ and capitals C\$ in elements zero to 49. The states and corresponding capitals are READ in from DATA. Next the instructions are printed.

G, the number of guesses, is initialized to zero. N is the number of states that have had the capitals named correctly. F is a flag which ordinarily is zero but is 1 if the capital has been missed once. After two misses, the capital is given and that state remains in the list. B\$ is the string variable that keeps the letters the user has typed in. If the back arrow is pressed to erase, B\$ is reduced and a space is printed. After the capital has been named correctly, S\$(R) is set equal to null, "", so that state will not be chosen again.

The last three lines print the total number of guesses and end the program.

Program 9-8. States and Capitals

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```
2 T=36876:POKE36878,15:GOTO14
4 PRINT"{DOWN}{GRN}PRESS <RETURN>"
6 GETES:IFE$=""THEN6
8 IFASC(E$)<>13THEN6
10 PRINT"{CLR}{BLU}":RETURN
12 FORD=1TO100:NEXT:RETURN
14 PRINT"{CLR}{DOWN} STATES AND CAPITALS"
16 DIMS$(49),C$(49)
18 FORI=0TO49:READS$(I),C$(I):NEXT
20 DATAALABAMA,MONTGOMERY,ALASKA,JUNEAU,ARIZONA,PH
   OENIX,ARKANSAS,LITTLE ROCK,CALIFORNIA
22 DATASACRAMENTO,COLORADO,DENVER,CONNECTICUT,HART
   FORD
23 DATADELAWARE,DOVER,FLORIDA,TALLAHASSEE
24 DATAGEORGIA,ATLANTA,HAWAII,HONOLULU,IDAHO,BOISE
   ,ILLINOIS,SPRINGFIELD,INDIANA
26 DATAINDIANAPOLIS,IOWA,DES MOINES,KANSAS,TOPEKA
27 DATAKENTUCKY,FRANKFORT,LOUISIANA,BATON ROUGE
28 DATAMAINE,AUGUSTA,MARYLAND,ANNAPOLIS,MASSACHUSE
   TTS,BOSTON,MICHIGAN,LANSING,MINNESOTA
30 DATAST. PAUL,MISSISSIPPI,JACKSON,MISSOURI,JEFFE
   RSON CITY,MONTANA,HELENA,NEBRASKA
32 DATALINCOLN,NEVADA,CARSON CITY,NEW HAMPSHIRE,CO
   NCORD,NEW JERSEY,TRENTON,NEW MEXICO
34 DATASANTA FE,NEW YORK,ALBANY,NORTH CAROLINA,RAL
   EIGH
35 DATANORTH DAKOTA,BISMARCK,OHIO,COLUMBUS
36 DATAOKLAHOMA,OKLAHOMA CITY,OREGON,SALEM,PENNSYL
   VANIA,HARRISBURG
37 DATARHODE ISLAND,PROVIDENCE
38 DATASOUTH CAROLINA,COLUMBIA,SOUTH DAKOTA,PIERRE
39 DATATENNESSEE,NASHVILLE,TEXAS,AUSTIN,UTAH
40 DATASALT LAKE CITY,VERMONT,MONTPELIER,VIRGINIA,
   RICHMOND,WASHINGTON,OLYMPIA
42 DATAWEST VIRGINIA,CHARLESTON,WISCONSIN,MADISON,
   WYOMING,CHEYENNE
44 PRINT"{DOWN}TYPE IN THE NAME OF{3 SPACES}THE CA
   PITAL AS THE{4 SPACES}STATE APPEARS."
46 PRINT"{DOWN}CITIES MUST BE SPELLEDCORRECTLY TO
   {SPACE}BE":PRINT"COUNTED AS CORRECT.":GOSUB4
48 G=0:M=7879
50 FORN=1TO50
52 F=0:R=INT(50*RND(0)):IFS$(R)=""THEN52
54 PRINT"{CLR}";N:PRINT"{3 DOWN}STATE: ";S$(R)
```

```

56 PRINT"{DOWN}CAPITAL: ":B$="":POKET,225:GOSUB12:
   POKET,Ø:G=G+1
58 FORI=1TO2Ø:POKE3Ø72Ø+M+I,Ø
60 GETES$:IFE$=""THEN6Ø
62 IFASC(E$)=13THEN8Ø
64 IFASC(E$)<>2ØTHEN7Ø
66 I=I-1:IFI=ØTHENI=1:GOTO6Ø
68 B$=LEFT$(B$,LEN(B$)-1):POKEM+I,32:GOTO6Ø
7Ø IFASC(E$)=32THENE=32:GOTO78
72 IFASC(E$)=46THENE=46:GOTO78
74 IFASC(E$)<65OR ASC(E$)>9ØTHEN6Ø
76 E=ASC(E$)-64
78 POKEM+I,E:B$=B$+E$:NEXTI
8Ø IFB$=C$(R)THEN88
82 POKET,159:GOSUB12:POKET,135:GOSUB12:POKET,Ø
84 F=F+1:IFF=1THENPRINT"{3 UP}":FORJ=M TOM+2Ø:POKE
   J,32:NEXTJ:GOTO56
86 PRINT"{4 DOWN}{2 SPACES}"C$(R):GOSUB4:N=N-1:GOT
   O94
88 PRINT"{5 DOWN}{RED}CORRECT!{BLU}"
9Ø POKET,195:GOSUB12:POKET,2Ø7:GOSUB12:POKET,215:G
   OSUB12:POKET,225:GOSUB12:POKET,Ø
92 S$(R)=""
94 NEXTN:PRINT"{CLR}{DOWN}YOU ANSWERED ALL 5Ø."
96 PRINT"{DOWN}NUMBER OF GUESSES:";G:PRINT
98 END

```

Program 9-9. States and Capitals

MC-10

```

1 GOTO6
2 PRINT@495,"PRESS <ENTER>";
3 E$=INKEY$:IFE$=""THEN3
4 IFASC(E$)<>13THEN3
5 CLS:RETURN
6 CLS:PRINT@165,"**STATES AND CAPITALS**"
7 DIMS$(49),C$(49)
8 FORI=ØTO49:READS$(I),C$(I):NEXT
9 DATAALABAMA,MONTGOMERY,ALASKA,JUNEAU,ARIZONA,PHO
   ENIX,ARKANSAS,LITTLE ROCK,CALIFORNIA,SACRAMENTO,
   COLORADO,DENVER
1Ø DATACONNECTICUT,HARTFORD,DELAWARE,DOVER,FLORIDA
   ,TALLHASSEE,GEORGIA,ATLANTA,HAWAII,HONOLULU,ID
   AHO,BOISE,ILLINOIS,SPRINGFIELD
11 DATAINDIANA,INDIANAPOLIS,IOWA,DES MOINES,KANSAS
   ,TOPEKA,KENTUCKY,FRANKFORT,LOUISIANA,BATON ROUG
   E,MAINE,AUGUSTA,MARYLAND
12 DATAANNAPOLIS,MASSACHUSETTS,BOSTON,MICHIGAN,LAN
   SING,MINNESOTA,ST. PAUL,MISSISSIPPI,JACKSON,MIS
   SOURI,JEFFERSON CITY,MONTANA

```

Fun in Less Than 4K

```
13 DATAHELENA,NEBRASKA,LINCOLN,NEVADA,CARSON CITY,
    NEW HAMPSHIRE,CONCORD,NEW JERSEY,TRENTON,NEW ME
    XICO,SANTA FE,NEW YORK
14 DATAALBANY,NORTH CAROLINA,RALEIGH,NORTH DAKOTA,
    BISMARCK,OHIO,COLUMBUS,OKLAHOMA,OKLAHOMA CITY,O
    REGON,SALEM,PENNSYLVANIA
15 DATAHARRISBURG,RHODE ISLAND,PROVIDENCE,SOUTH CA
    ROLINA,COLUMBIA,SOUTH DAKOTA,PIERRE,TENNESSEE,N
    ASHVILLE,TEXAS,AUSTIN,UTAH
16 DATASALT LAKE CITY,VERMONT,MONTPELIER,VIRGINIA,
    RICHMOND,WASHINGTON,OLYMPIA,WEST VIRGINIA,CHARL
    ESTON
17 DATAWISCONSIN,MADISON,WYOMING,CHEYENNE
18 PRINT@288,"TYPE IN THE NAME OF THE CAPITAL AS T
    HE STATE APPEARS."
19 PRINT"CITIES MUST BE SPELLED CORRECTLYTO BE COU
    NTED AS CORRECT.":GOSUB2
20 G=0
21 FORN=1TO50
22 F=0:R=RND(50)-1:IFS$(R)="THEN22
23 PRINT@28,N
24 PRINT@160,"STATE: ";S$(R)
25 PRINT@256,"CAPITAL:":B$="":SOUND227,1:G=G+1
26 FORI=1TO20
27 E$=INKEY$:IFE$="THEN27
28 IFASC(E$)=13THEN33
29 IFASC(E$)<>8THEN32
30 I=I-1:IFI=0THENI=1:GOTO27
31 B$=LEFT$(B$,LEN(B$)-1):PRINT@265+I," ";:GOTO27
32 PRINT@265+I,E$;:B$=B$+E$:NEXT
33 IFB$=C$(R) THEN37
34 SOUND125,2:SOUND89,2:PRINT@266,"{20 SPACES}"
35 F=F+1:IFF=1THEN25
36 PRINT@266,C$(R):GOSUB2:N=N-1:GOTO39
37 PRINT@394,"CORRECT!"
38 SOUND89,1:SOUND125,1:SOUND147,1:SOUND176,3:S$(R
    )=""
39 CLS:NEXTN
40 CLS:PRINT@128,"YOU ANSWERED ALL 50."
41 PRINT@224,"NUMBER OF GUESSES: ";G:PRINT@448
42 END
```

Program 9-10. States and Capitals

TRS-80 Model I

```
1 GOTO6
2 PRINT@990,"PRESS <ENTER>";
3 E$=INKEY$:IFE$="THEN3
4 IFASC(E$)<>13THEN3
5 CLS:RETURN
```

```

6 CLS:PRINT@205,"**STATES AND CAPITALS**
7 DIMS$(49),C$(49)
8 FORI=0TO49:READS$(I),C$(I):NEXT
9 DATAALABAMA,MONTGOMERY,ALASKA,JUNEAU,ARIZONA,PHO
  ENIX,ARKANSAS,LITTLE ROCK,CALIFORNIA,SACRAMENTO,
  COLORADO,DENVER
10 DATACONNECTICUT,HARTFORD,DELAWARE,DOVER,FLORIDA
  ,TALLHASSEE,GEORGIA,ATLANTA,HAWAII,HONOLULU,IDA
  HO,BOISE,ILLINOIS,SPRINGFIELD
11 DATAINDIANA,INDIANAPOLIS,IOWA,DES MOINES,KANSAS
  ,TOPEKA,KENTUCKY,FRANKFORT,LOUISIANA,BATON ROUG
  E,MAINE,AUGUSTA,MARYLAND
12 DATAANNAPOLIS,MASSACHUSETTS,BOSTON,MICHIGAN,LAN
  SING,MINNESOTA,ST. PAUL,MISSISSIPPI,JACKSON,MIS
  SOURI,JEFFERSON CITY,MONTANA
13 DATAHELENA,NEBRASKA,LINCOLN,NEVADA,CARSON CITY,
  NEW HAMPSHIRE,CONCORD,NEW JERSEY,TRENTON,NEW ME
  XICO,SANTA FE,NEW YORK
14 DATAALBANY,NORTH CAROLINA,RALEIGH,NORTH DAKOTA,
  BISMARCK,OHIO,COLUMBUS,OKLAHOMA,OKLAHOMA CITY,O
  REGON,SALEM,PENNSYLVANIA
15 DATAHARRISBURG,RHODE ISLAND,PROVIDENCE,SOUTH CA
  ROLINA,COLUMBIA,SOUTH DAKOTA,PIERRE,TENNESSEE,N
  ASHVILLE,TEXAS,AUSTIN,UTAH
16 DATASALT LAKE CITY,VERMONT,MONTPELIER,VIRGINIA,
  RICHMOND,WASHINGTON,OLYMPIA,WEST VIRGINIA,CHARL
  ESTON
17 DATAWISCONSIN,MADISON,WYOMING,CHEYENNE
18 PRINT@512,"TYPE IN THE NAME OF THE CAPITAL AS T
  HE STATE APPEARS."
19 PRINT"CITIES MUST BE SPELLED CORRECTLYTO BE COU
  NTED AS CORRECT.":GOSUB2
20 G=0
21 FORN=1TO50
22 F=0:R=RND(50)-1:IFS$(R)=""THEN22
23 PRINT@56,N
24 PRINT@320,"STATE: ";S$(R)
25 PRINT@512,"CAPITAL:":B$=""":G=G+1:L=521
26 FORI=1TO20
27 E$=INKEY$:IFE$=""THEN27
28 IFASC(E$)=13THEN33
29 IFASC(E$)<>8THEN32
30 I=I-1:IFI=0THENI=1:GOTO27
31 B$=LEFT$(B$,LEN(B$)-1):PRINT@L+I," ";:GOTO27
32 PRINT@L+I,E$;:B$=B$+E$:NEXT
33 IFB$=C$(R) THEN37
34 PRINT@788,"INCORRECT, TRY AGAIN":FORD=1TO500:NE
  XT:PRINT@788,"{20 SPACES}":PRINT@L,"{21 SPACES}"
35 F=F+1:IFF=1THEN25

```

Fun in Less Than 4K

```
36 PRINT@532,C$(R):GOSUB2:N=N-1:GOTO39
37 PRINT@788,"CORRECT!"
38 FORD=1TO500:NEXT:S$(R)="
39 CLS:NEXTN
40 CLS:PRINT@256,"YOU ANSWERED ALL 50."
41 PRINT@448,"NUMBER OF GUESSES: ";G:PRINT@896,""
42 END
```

Program 9-11. States and Capitals

TI-99/4A

```
100 GOTO 160
110 PRINT :::"PRESS <ENTER>";
120 CALL KEY(0,K,S)
130 IF K<>13 THEN 120
140 CALL CLEAR
150 RETURN
160 CALL CLEAR
170 PRINT "*** STATES AND CAPITALS ***"::::
180 PRINT "TYPE IN THE NAME OF THE"
190 PRINT "CAPITAL AS THE STATE NAME"
200 PRINT "APPEARS."
210 PRINT : "CITIES MUST BE SPELLED": "CORRECTLY TO
    BE COUNTED": "AS CORRECT."
220 PRINT : "PRESS LEFT ARROW TO ERASE.":
230 DIM S$(49),C$(49)
240 FOR I=0 TO 49
250 READ S$(I),C$(I)
260 NEXT I
270 DATA ALABAMA,MONTGOMERY,ALASKA,JUNEAU,ARIZONA,
    PHOENIX,ARKANSAS,LITTLE ROCK,CALIFORNIA,SACRAM
    ENTO,COLORADO
280 DATA DENVER,CONNECTICUT,HARTFORD,DELAWARE,DOVE
    R,FLORIDA,TALLAHASSEE,GEORGIA,ATLANTA,HAWAII,H
    ONOLULU,IDAHO
290 DATA BOISE,ILLINOIS,SPRINGFIELD,INDIANA,INDIAN
    APOLIS,IOWA,DES MOINES,KANSAS,TOPEKA,KENTUCKY,
    FRANKFORT
300 DATA LOUISIANA,BATON ROUGE,MAINE,AUGUSTA,MARYL
    AND,ANNAPOLIS,MASSACHUSETTS,BOSTON,MICHIGAN,LA
    NSING
310 DATA MINNESOTA,ST. PAUL,MISSISSIPPI,JACKSON,MI
    SSOURI,JEFFERSON CITY,MONTANA,HELENA,NEBRASKA,
    LINCOLN
320 DATA NEVADA,CARSON CITY,NEW HAMPSHIRE,CONCORD,
    NEW JERSEY,TRENTON,NEW MEXICO,SANTA FE,NEW YOR
    K,ALBANY
330 DATA NORTH CAROLINA,RALEIGH,NORTH DAKOTA,BISMA
    RCK,OHIO,COLUMBUS,OKLAHOMA,OKLAHOMA CITY,OREGO
    N,SALEM
```

```
340 DATA PENNSYLVANIA,HARRISBURG,RHODE ISLAND,PROV
IDENCE,SOUTH CAROLINA,COLUMBIA,SOUTH DAKOTA,PI
ERRE
350 DATA TENNESSEE,NASHVILLE,TEXAS,AUSTIN,UTAH,SAL
T LAKE CITY,VERMONT,MONTPELIER,VIRGINIA,RICHMO
ND
360 DATA WASHINGTON,OLYMPIA,WEST VIRGINIA,CHARLEST
ON,WISCONSIN,MADISON,WYOMING,CHEYENNE
370 G=0
380 P=19
390 GOSUB 110
400 FOR N=1 TO 50
410 CALL CLEAR
420 F=0
430 RANDOMIZE
440 R=INT(50*RND)
450 IF S$(R)="" THEN 440
460 PRINT N
470 PRINT :::"STATE: ";S$(R)
480 PRINT :::"CAPITAL: ":::::
490 B$=""
500 G=G+1
510 CALL SOUND(100,1497,2)
520 FOR I=1 TO 20
530 CALL KEY(0,K,S)
540 IF K=13 THEN 670
550 IF S<1 THEN 530
560 IF K<>8 THEN 640
570 I=I-1
580 IF I>0 THEN 610
590 I=1
600 GOTO 530
610 B$=SEG$(B$,1,LEN(B$)-1)
620 CALL HCHAR(P,11+I,32)
630 GOTO 530
640 CALL HCHAR(P,11+I,K)
650 B$=B$&CHR$(K)
660 NEXT I
670 IF B$=C$(R)THEN 780
680 CALL SOUND(100,330,4)
690 CALL SOUND(100,262,4)
700 CALL HCHAR(P,12,32,20)
710 F=F+1
720 IF F=1 THEN 490
730 FOR I=1 TO LEN(C$(R))
740 CALL HCHAR(P,11+I,ASC(SEG$(C$(R),I,1)))
750 NEXT I
760 GOSUB 110
770 GOTO 410
```


Fun in Less Than 4K

```
780 PRINT "CORRECT!"
790 CALL SOUND(100,262,2)
800 CALL SOUND(100,330,2)
810 CALL SOUND(100,392,2)
820 CALL SOUND(150,523,2)
830 S$(R)=" "
840 NEXT N
850 CALL CLEAR
860 PRINT "YOU ANSWERED ALL 50."
870 PRINT "::::NUMBER OF GUESSES: ";G:::::
880 END
```

South America

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A

Although shorter than "States and Capitals," this program, which is a drill for the capitals of the countries of South America, is similar. A list of the capital cities is printed on the screen. The computer randomly chooses a country and prints the country, and you then press the letter corresponding to the capital city of that country.

All the versions work like this:

The countries and capitals are READ in from DATA, alphabetically by capital. The country is S\$ and the capital is C\$. After a country has been chosen, S\$() is set equal to " " so it will not be chosen again.

(Note: The names and spellings of the countries and capitals change periodically. You may need to adjust this list to correspond with your sources of information.)

Program 9-12. South America

VIC-20

```
110 DIMS$(11),C$(11)
120 PRINT"{CLR}** SOUTH AMERICA **"
130 PRINT"{DOWN}MATCH THE COUNTRY WITHTHE CAPITAL
{SPACE}CITY.":PRINT"PRESS THE LETTER."
140 FORI=0TO11:READS$(I),C$(I):NEXT
150 DATA COLOMBIA, BOGOTA, BRAZIL, BRASILIA, ARGENTINA,
BUENOS AIRES, VENEZUELA, CARACAS
155 DATA FRENCH GUIANA, CAYENNE, GUYANA, GEORGETOWN, PE
RU, LIMA, URUGUAY, MONTEVIDEO
160 DATA SURINAME, PARAMARIBO, ECUADOR, QUITO, CHILE, SA
NTIAGO, BOLIVIA, SUCRE
170 PRINT"{3 DOWN}PRESS A KEY TO START."
180 GETA$:IFA$=""THEN180
190 S=0
200 FORT=1TO12:PRINT"{CLR}"
210 R=INT(12*RND(0)):IFS$(R)=" "THEN210
220 FORI=0TO11:PRINTCHR$(65+I);" ";C$(I):NEXTI:PRI
NT
```

```

230 PRINTS$(R);"{3 SPACES}???" ;
240 GETA$:IFA$=""THEN240
250 IFASC(A$)<65THEN240
260 IFASC(A$)>76THEN240
270 PRINTA$:IFASC(A$)-65=R THEN290
280 PRINT"{DOWN}THE CAPITAL IS":PRINTC$(R):GOTO300
290 PRINT"CORRECT!":S=S+1
300 S$(R)="" :PRINT"{3 SPACES}PRESS <RETURN>";
310 GETA$:IFA$<>CHR$(13)THEN310
320 NEXTT
330 PRINT"{CLR}{DOWN}SOUTH AMERICA":PRINT"{DOWN}YO
    UR SCORE WAS ";S:PRINT"OUT OF 12 COUNTRIES
    {DOWN}"
340 END

```

Program 9-13. South America

TRS-80 Color Computer; TRS-80 Model I

```

110 DIMS$(11),C$(11)
120 CLS:PRINTTAB(5);"** SOUTH AMERICA **"
130 PRINT:PRINT"MATCH THE COUNTRY WITH THE
    {6 SPACES}CAPITAL CITY.":PRINT"PRESS THE CORRE
    CT LETTER."
140 FORI=0TO11:READS$(I),C$(I):NEXT
150 DATACOLOMBIA,BOGOTA,BRAZIL,BRASILIA,ARGENTINA,
    BUENOS AIRES,VENEZUELA,CARACAS,FRENCH GUIANA,C
    AYENNE,GUYANA,GEORGETOWN
160 DATAPERU,LIMA,URUGUAY,MONTEVIDEO,SURINAME,PARA
    MARIBO,ECUADOR,QUITO,CHILE,SANTIAGO,BOLIVIA,SU
    CRE
170 PRINT:PRINT:PRINT"PRESS ANY KEY TO START."
180 A$=INKEY$:IFA$=""THEN180
190 S=0
200 FORT=1TO12:CLS
210 R=RND(12)-1:IFS$(R)=""THEN210
220 FORI=0TO11:PRINTTAB(4);CHR$(65+I);" ";C$(I):NE
    XTI:PRINT
230 PRINTS$(R);"{3 SPACES}???" ;
240 A$=INKEY$:IFA$=""THEN240
250 IFASC(A$)<65THEN240
260 IFASC(A$)>76THEN240
270 PRINTA$;:IFASC(A$)-65=R THEN290
280 PRINT:PRINT"THE CAPITAL IS ";C$(R):GOTO300
290 PRINT" CORRECT!":S=S+1
300 S$(R)="" :PRINT"{5 SPACES}PRESS <ENTER>";
310 A$=INKEY$:IFA$<>CHR$(13)THEN310
320 NEXTT
330 CLS:PRINT"SOUTH AMERICA":PRINT:PRINT"YOUR SCOR
    E WAS ";S:PRINT"OUT OF 12 COUNTRIES.":PRINT
340 END

```

Program 9-14. South America

TI-99/4A

```
110 DIM S$(11),C$(11)
120 CALL CLEAR
130 PRINT TAB(4); "*** SOUTH AMERICA ***"
140 PRINT :: "MATCH THE COUNTRY WITH THE": "CAPITAL
    CITY.": "PRESS THE CORRECT LETTER.": ::::
150 FOR I=0 TO 11
160 READ S$(I),C$(I)
170 NEXT I
180 DATA COLOMBIA,BOGOTA,BRAZIL,BRASILIA,ARGENTINA
    ,BUENOS AIRES,VENEZUELA,CARACAS,FRENCH GUIANA,
    CAYENNE
190 DATA GUYANA,GEORGETOWN,PERU,LIMA,URUGUAY,MONTE
    VIDEO,SURINAME,PARAMARIBO,ECUADOR,QUITO
200 DATA CHILE,SANTIAGO,BOLIVIA,SUCRE
210 PRINT :: "PRESS ANY KEY TO START."
220 CALL KEY(0,K,S)
230 IF S<1 THEN 220
240 SC=0
250 FOR T=1 TO 12
260 CALL CLEAR
270 RANDOMIZE
280 R=INT(12*RND)
290 IF S$(R)="" THEN 280
300 FOR I=0 TO 11
310 PRINT TAB(4);CHR$(65+I);" ";C$(I)
320 NEXT I
330 PRINT ::S$(R);"   ???"
340 CALL KEY(0,K,S)
350 IF (K<65)+(K>76)THEN 340
360 CALL HCHAR(23,26,K)
370 IF K-65=R THEN 400
380 PRINT : "THE CAPITAL IS ";C$(R)
390 GOTO 420
400 PRINT : "CORRECT!"
410 SC=SC+1
420 S$(R)=""
430 PRINT :: "PRESS <ENTER>";
440 CALL KEY(0,K,S)
450 IF K<>13 THEN 440
460 NEXT T
470 CALL CLEAR
480 PRINT "SOUTH AMERICA": :::: "YOUR SCORE WAS";SC: :
    "OUT OF 12 COUNTRIES.": ::::
490 END
```

Countries

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A
A quiz-style game, "Countries" can be used while you study elementary geography. The seven continents are listed on the screen by the computer, which then prints a country's name. You must press the number of the continent where the country is located. There are ten countries for each quiz. Fifty countries are listed in the program, so it is possible to go through the quiz four times without repeating. If the country is located correctly, it will not appear again, but if it is located incorrectly it could be chosen again.

Countries works like this:

The first DATA statements contain the name of the country and a corresponding continent number, READ in as C\$() and C(). The continents are A\$(). If the continent is named correctly, C\$(R) is set to "" so it will not be chosen again. T is the number of times the test has been given. After four times, the DATA is RESTORED for more quizzes.

(Note: Check periodically to make sure the country names and spellings are current.)

Program 9-15. Countries

VIC-20

```
4 DIMA$(7),C$(50),C(50)
6 PRINT"{CLR}YOU WILL BE GIVEN THE NAME OF A COUNTRY.":PRINT"WHERE IS IT LOCATED?"
8 PRINT"{DOWN}SEVEN CONTINENTS ARE{2 SPACES}LISTED .{2 SPACES}CHOOSE THE{3 SPACES}RIGHT NUMBER."
10 FORI=0TO50:READC$(I),C(I):NEXT
12 DATAFRANCE,3,BRAZIL,2,LIBYA,5,UNITED STATES,1,CANADA,1,MEXICO,1,PERU,2,NORWAY,3
14 DATAU.S.S.R,4,CHILE,2,CHINA,4,JAPAN,4,SPAIN,3,DENMARK,3,ITALY,3,PORTUGAL,3,ENGLAND,3
16 DATAINDIA,4,IRAN,4,GERMANY,3,ARGENTINA,2,URUGUAY,2,PARAGUAY,2,ECUADOR,2,BOLIVIA,2
18 DATACOLOMBIA,2,VENEZUELA,2,KOREA,4,IRELAND,3,QUEENSLAND,6,NEW SOUTH WALES,6
20 DATAWESTERN AUSTRALIA,6,MONGOLIA,4,NIGER,5,CHAD,5,ZAIRE,5,HONDURAS,1,EGYPT,5
22 DATASWITZERLAND,3,CAMBODIA,4,VIETNAM,4,FINLAND,3,SWEDEN,3,GUATEMALA,1,IRAQ,4
24 DATAZAMBIA,5,BANGLADESH,4,AUSTRIA,3,POLAND,3,CZECHOSLOVAKIA,3,GUYANA,2
26 FORI=1TO7:READA$(I):NEXT
28 DATANORTH AMERICA,SOUTH AMERICA,EUROPE,ASIA,AFRICA,AUSTRALIA,ANTARCTICA
```

Fun in Less Than 4K

```
30 PRINT"{DOWN}PRESS F1 TO BEGIN."
32 GETES:IFE$<>CHR$(133)THEN32
34 T=0
36 SC=0
38 FORI=1TO10:PRINT"{CLR}"
40 FORJ=1TO7:PRINTJ;A$(J):NEXTJ:PRINT"{DOWN}"
42 R=INT(51*RND(0)):IFC$(R)=" "THEN42
44 PRINTC$(R):PRINT"{DOWN}WHICH CONTINENT? ";
46 GETES:IFE$=" "THEN46
48 IFASC(E$)<49ORASC(E$)>55THEN46
50 PRINTES:IFVAL(E$)=C(R)THEN54
52 PRINT"{DOWN}SORRY, ";C$(R):PRINT"IS IN
  {SHIFT-SPACE}";A$(C(R)):GOTO56
54 PRINT"{DOWN}CORRECT!":SC=SC+1:C$(R)=" "
56 PRINT"{DOWN}PRESS <RETURN>";
58 GETES:IFE$<>CHR$(13)THEN58
60 NEXTI
62 PRINT"{CLR}{DOWN}OUT OF 10 COUNTRIES":PRINT"
  {DOWN}YOUR SCORE IS";SC
64 PRINT"{2 DOWN}TRY AGAIN? (Y/N)"
66 GETES:IFE$="N"THEN74
68 IFE$<>"Y"THEN66
70 T=T+1:IFT<5THEN36
72 RESTORE:FORI=0TO50:READC$(I),C(I):NEXT:GOTO34
74 END
```

Program 9-16. Countries

TRS-80 Color Computer; TRS-80 Model I

```
2 DIMA$(7),C$(50),C(50)
3 CLS:PRINT"YOU WILL BE GIVEN THE NAME OF A COUNTR
  Y. WHERE IS IT LOCATED?"
4 PRINT:PRINT"SEVEN CONTINENTS ARE LISTED.
  {4 SPACES}CHOOSE THE RIGHT NUMBER."
5 FORI=0TO50:READC$(I),C(I):NEXT
6 DATA FRANCE,3,BRAZIL,2,LIBYA,5,UNITED STATES,1,C
  ANADA,1,MEXICO,1,PERU,2,NORWAY,3,U.S.S.R,4,CHILE
  ,2,CHINA,4,JAPAN,4,SPAIN,3
7 DATA DENMARK,3,ITALY,3,PORTUGAL,3,ENGLAND,3,INDI
  A,4,IRAN,4,GERMANY,3,ARGENTINA,2,URUGUAY,2,PARAG
  UAY,2,ECUADOR,2,BOLIVIA,2
8 DATA COLOMBIA,2,VENEZUELA,2,KOREA,4,IRELAND,3,QU
  EENSLAND,6,NEW SOUTH WALES,6,WESTERN AUSTRALIA,6
  ,MONGOLIA,4,NIGER,5
9 DATA CHAD,5,ZAIRE,5,HONDURAS,1,EGYPT,5,SWITZERLA
  ND,3,CAMBODIA,4,VIETNAM,4,FINLAND,3,SWEDEN,3,GUA
  TEMALA,1,IRAQ,4,ZAMBIA,5
10 DATABANGLADESH,4,AUSTRIA,3,POLAND,3,CZECHOSLAVA
  KIA,3,GUYANA,2
11 FORI=1TO7:READA$(I):NEXT
```

```
12 DATANORTH AMERICA,SOUTH AMERICA,EUROPE,ASIA,AFR
ICA,AUSTRALIA,ANTARCTICA
13 PRINT:PRINT"PRESS <ENTER> TO BEGIN."
14 E$=INKEY$:IFE$<>CHR$(13)THEN14
15 T=0
16 SC=0
17 FORI=1TO10:CLS:FORJ=1TO7:PRINTJ;A$(J):NEXTJ:PRI
NT
18 R=RND(51)-1:IFC$(R)=" "THEN18
19 PRINTC$(R):PRINT:PRINT"WHICH CONTINENT? ";
20 E$=INKEY$:IFE$=" "THEN20
21 IFASC(E$)<49THEN20
22 IFASC(E$)>55THEN20
23 PRINTE$:IFVAL(E$)=C(R)THEN25
24 PRINT:PRINT"SORRY, ";C$(R);" IS IN":PRINTA$(C(R
)):GOTO26
25 PRINT:PRINT"CORRECT!":SC=SC+1:C$(R)=" "
26 PRINT:PRINT"PRESS <ENTER>";
27 E$=INKEY$:IFE$<>CHR$(13)THEN27
28 NEXTI
29 CLS:PRINT"OUT OF 10 COUNTRIES,":PRINT"YOUR SCOR
E IS";SC:PRINT:PRINT"TRY AGAIN? (Y/N)"
30 E$=INKEY$:IFE$="N"THEN34
31 IFE$<>"Y"THEN30
32 T=T+1:IFT<5THEN16
33 RESTORE:FORI=0TO50:READC$(I),C(I):NEXT:GOTO15
34 END
```

Program 9-17. Countries

TI-99/4A

```
110 DIM A$(7),C$(50),C(50)
120 CALL CLEAR
130 PRINT "YOU WILL BE GIVEN THE"::"NAME OF A COUN
TRY."
140 PRINT : "WHERE IS IT LOCATED?"
150 PRINT :: "SEVEN CONTINENTS ARE LISTED."::"CHOOS
E THE RIGHT NUMBER."
160 FOR I=0 TO 50
170 READ C$(I),C(I)
180 NEXT I
190 DATA FRANCE,3,BRAZIL,2,LIBYA,5,UNITED STATES,1
,CANADA,1,MEXICO,1,PERU,2,NORWAY,3,U.S.S.R.,4,
CHILE,2
200 DATA CHINA,4,JAPAN,4,SPAIN,3,DENMARK,3,ITALY,3
,PORTUGAL,3,ENGLAND,3,INDIA,4,IRAN,4,GERMANY,3
210 DATA ARGENTINA,2,URUGUAY,2,PARAGUAY,2,ECUADOR,
2,BOLIVIA,2,COLOMBIA,2,VENEZUELA,2,KOREA,4,IRE
LAND,3
```

Fun in Less Than 4K

```
220 DATA QUEENSLAND,6,NEW SOUTH WALES,6,WESTERN AU
STRALIA,6,MONGOLIA,4,NIGER,5,CHAD,5,ZAIRE,5,HO
NDURAS,1
230 DATA EGYPT,5,SWITZERLAND,3,CAMBODIA,4,VIETNAM,
4,FINLAND,3,SWEDEN,3,GUATEMALA,1,IRAQ,4
240 DATA ZAMBIA,5,BANGLADESH,4,AUSTRIA,3,POLAND,3,
CZECHOSLAVAKIA,3,GUYANA,2
250 FOR I=1 TO 7
260 READ A$(I)
270 NEXT I
280 DATA NORTH AMERICA,SOUTH AMERICA,EUROPE,ASIA,A
FRICA,AUSTRALIA,ANTARCTICA
290 PRINT :: "PRESS ANY KEY TO BEGIN."
300 CALL KEY(0,K,S)
310 IF S<1 THEN 300
320 T=1
330 SC=0
340 FOR I=1 TO 10
350 CALL CLEAR
360 PRINT "1 NORTH AMERICA": "2 SOUTH AMERICA": "3 E
UROPE": "4 ASIA"
370 PRINT "5 AFRICA": "6 AUSTRALIA": "7 ANTARCTICA":
::
380 RANDOMIZE
390 R=INT(51*RND)
400 IF C$(R)="" THEN 390
410 PRINT C$(R)
420 PRINT :: "WHICH CONTINENT?"
430 CALL KEY(0,K,S)
440 IF (K<49)+(K>55) THEN 430
450 CALL HCHAR(23,21,K)
460 IF K=C(R)+48 THEN 490
470 PRINT :: "SORRY, ";C$(R); " IS IN":A$(C(R))
480 GOTO 520
490 PRINT : "CORRECT!"
500 SC=SC+1
510 C$(R)=""
520 PRINT :: "PRESS <ENTER>";
530 CALL KEY(0,K,S)
540 IF K<>13 THEN 530
550 NEXT I
560 CALL CLEAR
570 PRINT "OUT OF 10 PROBLEMS,"
580 PRINT : "YOUR SCORE IS";SC
590 PRINT :::: "TRY AGAIN? (Y/N)"
600 CALL KEY(0,K,S)
610 IF K=78 THEN 710
620 IF K<>89 THEN 600
630 T=T+1
```

```
640 IF T<5 THEN 330
650 PRINT : : "ONE MOMENT PLEASE."
660 RESTORE
670 FOR I=0 TO 50
680 READ C$(I),C(I)
690 NEXT I
700 GOTO 330
```

Multiplication Invasion

Computers: VIC-20; TRS-80 Color Computer; TRS-80 Model I; TI-99/4A

In game format, this program presents a drill of the multiplication tables. Three random multiplication problems start at the bottom of the screen and move upward. Use the arrow keys to control your position at the top of the screen. Get above a particular problem and answer the problem by pressing number keys. If the answer is only one digit, you must press the ENTER or RETURN key. If the answer is correct, that problem disappears, and another problem appears at the bottom. You'll score ten points for every correct answer. If the answer is incorrect, the problem keeps moving upward, and your score is reduced by five points. The game is over when a problem reaches the top of the screen before you can answer it.

In all program versions, L is the counter limit that adjusts the speed of the game. The counter increments during loops when no key is pressed. After the counter gets to L, the computer PRINTs a blank line, which scrolls the problems upward. SC is the score, and T is the number of problems answered correctly.

Y is the position of your marker — 1, 2, or 3, for left, center, or right sections of the screen. C(Y) and P(Y) are coordinates or positions to draw your marker and PRINT the problem in a particular column.

M and N are random numbers from 0 to 9 for the problems, and A(Y) is the answer for the problem in a particular column. R(Y) counts which row the problem is on so it can be erased when the answer is correct. R is decreased by one for each PRINT statement used in scrolling. R also detects when a problem has reached the top of the screen.

Program 9-18. Multiplication Invasion

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```
20 DIMA(3),C(3),R(3),P(3)
30 PRINT"{CLR}* MULTIPLY INVASION *":PRINT"{DOWN}U
  SE F1 TO MOVE LEFT{7 SPACES}F3 TO MOVE RIGHT."
40 PRINT"{DOWN}GET ABOVE THE PROBLEM THEN TYPE THE
  ANSWER{2 SPACES}AND PRESS <RETURN>."
50 PRINT"{DOWN}DON'T LET THE INVADERSFLOAT TO THE
  {SPACE}TOP OR{3 SPACES}THE GAME ENDS!"
60 PRINT"{DOWN}CORRECT = 10 POINTS{3 SPACES}WRONG
  {3 SPACES}=-5"
70 PRINT"{DOWN}CHOOSE: 1 BEGINNER":PRINTTAB(8);"2
  {SPACE}MEDIUM":PRINTTAB(8);"3 FAST";
80 GETES$:IFES$=""THEN80
90 IFES$="1"THENL=80:GOTO130
100 IFES$="2"THENL=20:GOTO130
110 IFES$<>"3"THEN80
120 L=7
130 PRINT"{CLR}{21 DOWN}":SC=0:T=0:S=36876:POKES+2
  ,15
140 X=7691:Y=2
150 C(1)=7682:C(2)=X:C(3)=7699
160 P(1)=8164:P(2)=8173:P(3)=8181:FORI=8164TO8185:
  POKEI+30720,0:NEXT
170 FORI=1TO3:R(I)=21:FORJ=C(I)-2TOC(I)+2:POKEJ,10
  2:POKEJ+30720,6
175 POKEC(I)+30742,2:POKEC(I)+30743,2:NEXTJ,I
180 POKEX+22,90
190 FORI=1TO3
200 M=INT(10*RND(0)):N=INT(10*RND(0))
210 A(I)=M*N:B=P(I):POKEB,48+M:POKEB+1,42:POKEB+2,
  48+N:POKEB+3,61:POKEB+4,63
220 NEXTI
230 K=0
240 GETES$:IFES$=""THEN450
250 IFES$<>CHR$(133)THEN280
260 POKEC(Y),32:Y=Y-1:IFY<1THENY=1
270 GOTO460
280 IFES$<>CHR$(134)THEN310
290 POKEC(Y),32:Y=Y+1:IFY>3THENY=3
300 GOTO460
310 IFASC(ES$)<48THEN470
320 IFASC(ES$)>57THEN470
330 POKEC(Y)+22,ASC(ES$)
340 GETFS$:IFFS$=""THEN340
350 IFFS$<>CHR$(13)THEN370
```

```

360 Z=VAL(E$):GOTO400
370 IFASC(F$)<48THEN340
380 IFASC(F$)>57THEN340
390 POKEC(Y)+23,ASC(F$):POKEC(Y)+30743,2:Z=VAL(E$+
F$)
400 IFZ=A(Y)THEN420
410 POKES,159:GOSUB540:POKES,135:GOSUB540:SC=SC-5:
GOTO470
420 FORI=225TO231:POKES,I:GOSUB540:NEXTI:SC=SC+10
430 FORI=0TO4:POKEP(Y)-(21-R(Y))*22+I,32:NEXTI:POK
EC(Y)+22,32:POKEC(Y)+23,32
440 M=INT(10*RND(0)):N=INT(10*RND(0)):A(Y)=M*N:R(Y
)=21:B=P(Y):POKEB,48+M:POKEB+1,42
445 POKEB+2,48+N:POKEB+3,61:POKEB+4,63:FORJ=BTOB+4
:POKEJ+30720,0:NEXT:T=T+1
450 K=K+1:IFK<L THEN240
460 POKEC(Y)+22,90
* 470 PRINT:FORI=1TO3:FORJ=C(I)-2TOC(I)+2:POKEJ,102:
NEXTJ,I:POKEC(Y)+22,90:POKEC(Y)+30742,2
475 FORJ=38400TO38421:POKEJ,6:NEXTJ
480 FORI=1TO3:R(I)=R(I)-1:IFR(I)=-1THEN510
490 NEXTI
500 GOTO230
510 POKES,159:FORD=1TO200:NEXT:POKES,135:FORD=1TO2
00:NEXT:POKES,0
520 PRINT"{CLR}GAME OVER":PRINT"{2 DOWN}SCORE = ";
SC:PRINT"{DOWN}PROBLEMS = ";T:PRINT:GOTO550
540 FORD=1TO99:NEXT:POKES,0:RETURN
550 END

```

Program 9-19. Multiplication Invasion

TRS-80 Color Computer

```

20 DIMA(3),C(3),R(3),P(3)
30 CLS:PRINT"**MULTIPLICATION INVASION**":PRINT:PR
INT"USE THE ARROW KEYS TO MOVE":PRINT"LEFT OR R
IGHT."
40 PRINT"GET ABOVE THE PROBLEM THEN TYPE THE ANSWE
R AND PRESS <ENTER>."
50 PRINT:PRINT"DON'T LET THE INVADERS FLOAT
{4 SPACES}TO THE TOP OR THE GAME ENDS!"
60 PRINT:PRINT"CORRECT=10 POINTS; WRONG=-5"
70 PRINT:PRINT"CHOOSE: 1 BEGINNER":PRINTTAB(8);"2
MEDIUM":PRINTTAB(8);"3 FAST";
80 E$=INKEY$:IFE$=""THEN80
90 IFE$="1"THEN L=80:GOTO130
100 IFE$="2"THEN L=50:GOTO130
110 IFE$<>"3"THEN80
120 L=25

```

Fun in Less Than 4K

```
130 CLS:SC=0:T=0
140 X=32:Y=2
150 C(1)=8:C(2)=30:C(3)=52
160 P(1)=449:P(2)=460:P(3)=471
170 FORI=1TO3:R(I)=15:FORJ=0TO5:SET(C(I)+J,0,4):SE
    T(C(I)+J,1,4):NEXTJ,I
180 SET(X,1,2):SET(X+1,1,2)
190 FORI=1TO3
200 M=RND(10)-1:N=RND(10)-1
210 A(I)=M*N:PRINT@P(I),M;"*";N;"=?";
220 NEXTI
230 K=0
240 E$=INKEY$:IFE$=""THEN 450
250 IFE$<>CHR$(8)THEN280
260 SET(X,1,4):SET(X+1,1,4):X=X-22:Y=Y-1:IFX<10THE
    NX=10:Y=1
270 GOTO460
280 IFE$<>CHR$(9)THEN310
290 SET(X,1,4):SET(X+1,1,4):X=X+22:Y=Y+1:IFX>54THE
    NX=54:Y=3
300 GOTO460
310 IFASC(E$)<48 THEN 470
320 IFASC(E$)>57 THEN 470
330 PRINT@X/2+32,E$;
340 F$=INKEY$:IFF$=""THEN340
350 IFF$<>CHR$(13)THEN370
360 Z=VAL(E$):GOTO400
370 IFASC(F$)<48THEN340
380 IFASC(F$)>57THEN340
390 PRINT@X/2+33,F$;:Z=VAL(E$+F$)
400 IFZ=A(Y)THEN420
410 SOUND125,2:SOUND89,2:SC=SC-5:GOTO470
420 FORI=176TO193:SOUNDI,1:NEXTI:SC=SC+10
430 PRINT@P(Y)-(15-R(Y))*32,"{9 SPACES}";
440 M=RND(10)-1:N=RND(10)-1:A(Y)=M*N:R(Y)=15:PRINT
    @P(Y),M;"*";N;"=?";:T=T+1
450 K=K+1:IFK<L THEN 240
460 SET(X,1,2):SET(X+1,1,2)
470 PRINT@480," ":FORI=1TO3:FORJ=0TO5:SET(C(I)+J,0
    ,4):SET(C(I)+J,1,4):NEXTJ,I:SET(X,1,2):SET(X+1
    ,1,2)
480 FORI=1TO3:R(I)=R(I)-1:IFR(I)=1THEN510
490 NEXTI
500 GOTO230
510 SOUND58,5:SOUND5,8
520 CLS:PRINT"GAME OVER":PRINT:PRINT"SCORE = ";SC:
    PRINT:PRINT"NUMBER OF PROBLEMS = ";T:PRINT
530 END
```

Program 9-20. Multiplication Invasion

TRS-80 Model I

```

15 CLEAR300
20 DIMA(3),R(3),P(3)
30 CLS:PRINTTAB(5); "***MULTIPLICATION INVASION***":P
  RINT:PRINT"USE THE ARROW KEYS TO MOVE LEFT OR R
  IGH T."
40 PRINT"GET ABOVE THE PROBLEM THEN TYPE THE ANSWE
  R AND PRESS <ENTER>."
50 PRINT:PRINT"DON'T LET THE INVADERS FLOAT TO THE
  TOP OR THE GAME ENDS!"
60 PRINT:PRINT"10 POINTS FOR CORRECT ANSWERS; -5 F
  OR INCORRECT ONES"
70 PRINT:PRINT"CHOOSE: 1 BEGINNER":PRINTTAB(8);"2
  MEDIUM":PRINTTAB(8);"3 FAST";
80 E$=INKEY$:IFE$=""THEN80
90 IFE$="1"THEN L=80:GOTO130
100 IFE$="2"THEN L=50:GOTO130
110 IFE$<>"3"THEN80
120 L=25
130 CLS:SC=0:T=0:H$=CHR$(191):G$=H$+H$+H$+"
  {18 SPACES}" +H$+H$+H$+"{18 SPACES}" +H$+H$+H$
140 Y=2:FORI=1TO3:READX(I),P(I):NEXT
150 DATA75,904,96,924,117,944
170 PRINT@10,G$:PRINT@X(Y),H$
190 FORI=1TO3:R(I)=15
200 M=RND(10)-1:N=RND(10)-1
210 A(I)=M*N:PRINT@P(I),M;"*";N;"=?";
220 NEXTI
230 K=0
240 E$=INKEY$:IFE$=""THEN 450
250 IFE$<>CHR$(8)THEN280
260 PRINT@X(Y), " ":Y=Y-1:IFY<1THENY=1
270 GOTO460
280 IFE$<>CHR$(9)THEN310
290 PRINT@X(Y), " ":Y=Y+1:IFY>3THENY=3
300 GOTO460
310 IFASC(E$)<48 THEN 470
320 IFASC(E$)>57 THEN 470
330 PRINT@X(Y),E$;
340 F$=INKEY$:IFF$=""THEN340
350 IFF$<>CHR$(13)THEN370
360 Z=VAL(E$):GOTO400
370 IFASC(F$)<48THEN340
380 IFASC(F$)>57THEN340
390 PRINT@X(Y)+1,F$;:Z=VAL(E$+F$)
400 IFZ=A(Y)THEN420
410 FORI=1TO20:PRINT@X(Y)-3,"-----";:PRINT@X(Y)
  -3,"{8 SPACES}";:NEXT:SC=SC-5:GOTO470

```

Fun in Less Than 4K

```
420 FORI=1TO35:PRINT@X(Y)-64,"*";:PRINT@X(Y)-64,H$
;:NEXTI:SC=SC+10:PRINT@X(Y)," ";
430 PRINT@P(Y)-(15-R(Y))*64,"{10 SPACES}";
440 M=RND(10)-1:N=RND(10)-1:A(Y)=M*N:R(Y)=15:PRINT
@P(Y),M;"*";N;"= ?";:T=T+1
450 K=K+1:IFK<L THEN 240
460 PRINT@X(Y),H$
470 PRINT@960," ":PRINT@10,G$:PRINT@X(Y),H$
480 FORI=1TO3:R(I)=R(I)-1:IFR(I)=2THEN510
490 NEXTI
500 GOTO230
510 FORI=1TO20:PRINT@X(Y)," ":PRINT@X(Y),H$:NEXT
520 CLS:PRINT"GAME OVER":PRINT:PRINT"SCORE = ";SC:
PRINT:PRINT"NUMBER OF PROBLEMS = ";T:PRINT
530 END
```

Program 9-21. Multiplication Invasion

TI-99/4A

```
110 DIM A(3),C(3),R(3),P(3)
120 CALL CLEAR
130 PRINT "* MULTIPLICATION INVASION *"
140 PRINT ::"USE THE ARROW KEYS TO MOVE LEFT OR R
IGHT."
150 PRINT ::"GET ABOVE THE PROBLEM THEN TYPE THE A
NSWER AND": "PRESS <ENTER>"
160 PRINT ::"DON'T LET THE INVADERS FLOAT TO THE TOP
OR THE GAME ENDS!"
170 PRINT ::"CORRECT = 10 POINTS": "WRONG
{3 SPACES}= -5 POINTS"
180 CALL CHAR(96,"FFFFFFFFFFFFFF")
190 CALL COLOR(9,7,1)
200 CALL CHAR(97,"08080808087F3E1C08")
210 PRINT ::"CHOOSE: 1 BEGINNER":TAB(9);"2 MEDIUM"
:TAB(9);"3 FAST";
220 CALL KEY(0,K,S)
230 IF (K<49)+(K>51)THEN 220
240 L=(4-(K-48))*2
250 CALL CLEAR
260 SC=0
270 T=0
280 C(1)=3
290 C(2)=14
300 C(3)=25
310 FOR I=1 TO 3
320 CALL HCHAR(1,C(I),96,5)
330 NEXT I
340 FOR Y=1 TO 3
350 RANDOMIZE
360 GOSUB 880
```

```
370 NEXT Y
380 Y=2
390 CALL HCHAR(2,C(Y)+2,97)
400 KL=0
410 CALL KEY(0,K,S)
420 IF S<1 THEN 770
430 IF K<>83 THEN 490
440 CALL HCHAR(2,C(Y)+2,32)
450 Y=Y-1
460 IF Y>0 THEN 770
470 Y=1
480 GOTO 770
490 IF K<>68 THEN 550
500 CALL HCHAR(2,C(Y)+2,32)
510 Y=Y+1
520 IF Y<4 THEN 770
530 Y=3
540 GOTO 770
550 IF (K<48)+(K>57)THEN 770
560 CALL HCHAR(2,C(Y)+2,K)
570 B=K-48
580 CALL KEY(0,K,S)
590 IF S<1 THEN 580
600 IF K=13 THEN 640
610 IF (K<48)+(K>57)THEN 580
620 CALL HCHAR(2,C(Y)+3,K)
630 B=10*B+(K-48)
640 IF B<>A(Y)THEN 740
650 FOR I=262 TO 392 STEP 12
660 CALL SOUND(-99,I,2)
670 NEXT I
680 CALL HCHAR(R(Y),C(Y),32,5)
690 CALL HCHAR(2,C(Y)+2,32,2)
700 GOSUB 880
710 SC=SC+10
720 T=T+1
730 GOTO 760
740 CALL SOUND(100,-6,4)
750 SC=SC-5
760 CALL HCHAR(2,C(Y)+2,32,2)
770 KL=KL+1
780 CALL HCHAR(2,C(Y)+2,97)
790 IF KL<L THEN 410
800 PRINT
810 FOR I=1 TO 3
820 CALL HCHAR(1,C(I),96,5)
830 R(I)=R(I)-1
840 IF R(I)<=2 THEN 980
850 NEXT I
860 CALL HCHAR(2,C(Y)+2,97)
```

```
870 GOTO 400
880 M=INT(10*RND)
890 N=INT(10*RND)
900 A(Y)=M*N
910 CALL HCHAR(23,C(Y),M+48)
920 CALL HCHAR(23,C(Y)+1,42)
930 CALL HCHAR(23,C(Y)+2,N+48)
940 CALL HCHAR(23,C(Y)+3,61)
950 CALL HCHAR(23,C(Y)+4,63)
960 R(Y)=23
970 RETURN
980 FOR I=1 TO 16
990 CALL SCREEN(I)
1000 NEXT I
1010 CALL CLEAR
1020 PRINT "GAME OVER"
1030 PRINT ::"SCORE = ";SC::"NUMBER OF PROBLEMS =
";T:
1040 END
```

Defend

Computer: VIC-20

This is a simple intercept, shooting game for one person. You are positioned at the left of the screen and need to defend your territory — don't let the invaders coming from the right of the screen get to your border. Line up horizontally with an invader by pressing the up arrow to move up and the down arrow to go down, then shoot with the space bar or the f7 key. You score ten points for each invader you successfully shoot, but you lose five points if you miss.

After you have played this game once or twice, adapt it for your own game — use different graphics and screen colors. Change the motion to vertical instead of horizontal. Adapt the scoring. Perhaps you could change the shapes of the invaders after certain scores are reached — and vary the speed they travel.

Here's how "Defend" works:

Line	Function
1	Turn on volume for sound and initialize TS for the top score.
2	Define function R(X) to calculate the location number for a random row and branch to line 200.
10	Clear screen and set screen and border color. Initialize variables — N is the location of your ship, SC is the score, and D is difficulty level.

- 20 Place defending ship on screen.
- 22-25 Randomly place invaders, making sure they are not on the same row.
- 30 Detect your key press. If it is one of the firing keys (space bar or f7), branch to line 60.
- 32-34 If arrow keys are pressed, move up or down.
- 35 Increment L to determine speed of invaders.
- 36 Increment invaders' positions — move one spot to left.
- 37-42 If an invader reaches left side of screen, branch to line 100 to end game.
- 44-50 Move invaders and branch back to receive next key press.
- 60 Make firing noise and blink defender.
- 62-68 Check positions of invaders to see if one was shot.
- 70 Decrease score by 5 if shot missed.
- 72-78 Procedure if invader is shot and new invader position chosen.
- 80 Score is increased by 10, clear invader, and stop noise.
- 82-84 PRINT score and branch back for next key press.
- 90-94 Check boundary positions of defender, then draw defender on screen in new position.
- 100-110 Procedure at game's end.
- 120-160 PRINT ending message, score, and high score.
- 170-190 PRINT option to try again and branch appropriately.
- 200-280 PRINT instruction screen.
- 290 END.

Program 9-22. Defend

VIC-20

(In this program, make sure to use abbreviations for all BASIC keywords in the lines marked with an asterisk (*). If you don't, the program will not run correctly. Refer to Appendix C, "BASIC Keyword Abbreviations," for the proper abbreviations.)

```


1 POKE36878,15:TS=0
2 DEF FNR(X)=7745+22*(INT(RND(0)*20)):GOTO 200
10 PRINT"{CLR}":POKE36879,107:N=7922:SC=0:D=10
20 POKEN,90
22 I=FNR(X):J=FNR(X):K=FNR(X):H=FNR(X)
24 IFH=IORH=JORH=KORI=JORI=KORJ=K THEN 22
25 POKEH,42:POKEI,42:POKEJ,42:POKEK,42
30 GETA$:IFA$=CHR$(136)OR A$=CHR$(32)THEN 60
32 IFA$=CHR$(94)THENPOKEN,32:N=N-22:GOTO90
34 IFA$=CHR$(17)THENPOKEN,32:N=N+22:GOTO90
35 L=L+1:IFL<DTHEN30
36 H=H-1:I=I-1:J=J-1:K=K-1:L=0
    
```


Fun in Less Than 4K


```
37 IF(H-7680)/22=INT((H-7680)/22)THEN100
38 IF(I-7680)/22=INT((I-7680)/22)THEN100
40 IF(J-7680)/22=INT((J-7680)/22)THEN100
42 IF(K-7680)/22=INT((K-7680)/22)THEN100
* 44 POKE36876,240:POKEH+1,32:POKEI+1,32:POKEJ+1,32:
    POKEK+1,32:POKEH,42:POKEI,42:POKEJ,42:POKEK,42:
    POKE36876,0
50 GOTO30
60 FORM=200TO220:POKE36876,M:POKEN+30720,2:POKEN+3
    0720,1:NEXT:POKE36876,0
62 IFH>NANDH<N+22THEN72
64 IFI>NANDI<N+22THEN74
66 IFJ>NANDJ<N+22THEN76
68 IFK>NANDK<N+22THEN78
70 SC=SC-5:GOTO82
72 POKE36877,160:POKEH,102:B=H:H=FNR(X):GOTO80
74 POKE36877,150:POKEI,102:B=I:I=FNR(X):GOTO80
76 POKE36877,180:POKEJ,102:B=J:J=FNR(X):GOTO80
78 POKE36877,190:POKEK,102:B=K:K=FNR(X)
80 SC=SC+10:POKEB,32:POKE36877,0
82 PRINT"{HOME}{BLU}{RVS}{22 SPACES}{OFF}":PRINT"
    {HOME}{WHT}SCORE =";SC:IFSC>500THEND=5
84 GOTO30
90 IFN<7724THENN=7724
92 IFN>8164THENN=8164
94 POKEN,90:GOTO30
100 POKE36877,220:FORC=38445TO38885STEP22:POKEC,2:
    NEXTC:FORC=1TO100:NEXTC
110 FORC=38445TO38885STEP22:POKEC,1:NEXTC:POKE3687
    7,0
120 PRINT"{WHT}GAME OVER"
130 FORC=1TO1000:NEXT:POKE36879,27
140 PRINT"{CLR}{BLU}{2 DOWN}YOUR FINAL SCORE WAS
    {3 SPACES}":PRINT"{RED}";SC:PRINT"{BLU}
    {2 DOWN}"
150 IF SC>TS THEN TS=SC
160 PRINT"HIGH SCORE = ";TS
170 PRINT"{GRN}{3 DOWN}TRY AGAIN? (Y/N)"
180 GETA$:IF A$="Y"THEN 10
185 IF A$<>"N" THEN 180
190 PRINT"{BLU}":END
200 POKE36879,186:PRINT"{CLR}{BLK}":PRINTTAB(5);"*
    * DEFEND **{2 DOWN}"
220 PRINT"{2 DOWN}PRESS ↑ TO MOVE UP":PRINT"PRESS
    {SPACE}CRSR TO GO DOWN"
230 PRINT"{DOWN}PRESS F7 OR SPACE":PRINT"TO FIRE.
    {3 DOWN}"
240 PRINT"KEEP THE INVADERS FROMREACHING YOUR SIDE
    OF THE SCREEN."
```

Fun in Less Than 4K

```
250 PRINT"{2 DOWN}{RED}PRESS RETURN TO START";
260 GETA$:IFA$=""THEN 260
270 IF ASC(A$)=13 THEN 10
280 GOTO260
290 END
```



Appendices



A Beginner's Guide to Typing In Programs

What Is a Program?

A computer cannot perform any task by itself. Like a car without gas, a computer has *potential*, but without a program, it isn't going anywhere. Most of the programs published in this book are written in a computer language called BASIC. BASIC is easy to learn and is built into most computers (on some computers, you have to purchase an optional BASIC cartridge).

BASIC Programs

To start out, type in only programs written for your machine. Later, when you gain experience with your computer's BASIC, you can try typing in and converting certain programs from one computer to yours.

Computers can be picky. Unlike the English language, which is full of ambiguities, BASIC usually has only one right way of stating something. Every letter, character, or number is significant. A common mistake is substituting a letter such as O for the numeral 0, a lowercase l for the numeral 1, or an uppercase B for the numeral 8. Also, you must enter all punctuation such as colons and commas just as they appear in the book. Spacing can be important. To be safe, type in the listings *exactly* as they appear.

Braces and Special Characters

The exception to this typing rule is when you see the braces, such as {DOWN}. Anything within a set of braces is a special character or characters that cannot easily be listed on a printer. When you come across such a special statement, refer to the appropriate key for your computer. For example, if you have a VIC-20, refer to the VIC's section in "How To Type In Programs."

About DATA Statements

Some programs contain a section or sections of DATA statements. These lines provide information needed by the program. Some DATA statements contain actual programs (called machine language); others contain graphics codes. These lines are especially sensitive to errors.

If a single number in any one DATA statement is mistyped, your machine could lock up, or crash. The keyboard, break key,

APPENDIX A

and RESET (or STOP) keys may all seem dead, and the screen may go blank. Don't panic — no damage is done. To regain control, you have to turn off your computer, then turn it back on. This will erase whatever program was in memory, so always SAVE a copy of your program before you RUN it. If your computer crashes, you can LOAD the program and look for your mistake.

Sometimes a mistyped DATA statement will cause an error message when the program is RUN. The error message may refer to the program line that READs the data. *The error is still in the DATA statements, though.*

Get to Know Your Machine

You should familiarize yourself with your computer before attempting to type in a program. Learn the statements you use to store and retrieve programs from tape or disk. You'll want to save a copy of your program, so that you won't have to type it in every time you want to use it. Learn to use your machine's editing functions. How do you change a line if you made a mistake? You can always retype the line, but you at least need to know how to backspace. Do you know how to enter reverse video, lowercase, and control characters? It's all explained in your computer's manuals.

A Quick Review

- 1) Type in the program a line at a time, in order. Press RETURN or ENTER at the end of each line. Use backspace or the back arrow to correct mistakes.
- 2) Check the line you've typed against the line in the book. You can check the entire program again if you get an error when you RUN the program.
- 3) Make sure you've entered statements in brackets as the appropriate control key (see Appendix B, "How To Type In Programs").

How to Type In Programs

To make it easy to tell exactly what to type when entering one of these programs into your computer, we have established the following listing conventions. There is a separate key for each computer. Refer to the appropriate tables when you come across an unusual symbol in a program listing. If you are unsure how to actually enter a control character, consult your computer's manuals.

VIC-20

Generally, any VIC-20 program listings will contain words within braces which spell out any special characters: {DOWN} would mean to press the cursor down key. {5 SPACES} would mean to press the space bar five times.

To indicate that a key should be *shifted* (hold down the SHIFT key while pressing the other key), the key would be underlined in our listings. For example, S would mean to type the S key while holding the Shift key. This would appear on your screen as a "heart" symbol. If you find an underlined key enclosed in braces (e.g., {10 N}), you should type the key as many times as indicated (in our example, you would enter ten shifted N's).

If a key is enclosed in special brackets, [>], you should hold down the *Commodore key* while pressing the key inside the special brackets. (The Commodore key is the key in the lower-left corner of the keyboard.) Again, if the key is preceded by a number, you should press the key as many times as necessary.

Rarely, you'll see a solitary letter of the alphabet enclosed in braces. You should never have to enter such a character on the VIC-20, but if you do, you would have to leave the quote mode (press RETURN and cursor back up to the position where the control character should go), press CTRL-9 (RVS ON), the letter in braces, and then CTRL-0 (RVS OFF).

About the *quote mode*: you know that you can move the cursor around the screen with the CRSR keys. Sometimes a programmer will want to move the cursor under program control. That's why you see all the {LEFT}'s, {HOME}'s, and {BLU}'s in our programs. The only way the computer can tell the difference between direct and programmed cursor control is the quote mode.

Once you press the quote (the double quote, SHIFT-2), you are in the quote mode. If you type something and then try to change it by moving the cursor left, you'll only get a bunch of reverse-

APPENDIX B

video lines. These are the symbols for cursor left. The only editing key that isn't programmable is the DEL key; you can still use DEL to back up and edit the line. Once you type another quote, you are out of quote mode.

You also go into quote mode when you INsErT spaces into a line. In any case, the easiest way to get out of quote mode is to just press RETURN. You'll then be out of quote mode and you can cursor up to the mistyped line and fix it.

Use the following table when entering cursor and color control keys:

When You Read:	Press:	See:	When You Read:	Press:	See:
{ CLR }	SHIFT CLR/HOME		{ GRN }	CTRL 6	
{ HOME }	CLR/HOME		{ BLU }	CTRL 7	
{ UP }	SHIFT CRSR		{ YEL }	CTRL 8	
{ DOWN }	CRSR		{ F1 }	F1	
{ LEFT }	SHIFT CRSR		{ F2 }	SHIFT F1	
{ RIGHT }	CRSR		{ F3 }	F3	
{ RVS }	CTRL 9		{ F4 }	SHIFT F3	
{ OFF }	CTRL 0		{ F5 }	F5	
{ BLK }	CTRL 1		{ F6 }	SHIFT F5	
{ WHT }	CTRL 2		{ F7 }	F7	
{ RED }	CTRL 3		{ F8 }	SHIFT F7	
{ CYN }	CTRL 4				
{ PUR }	CTRL 5			SHIFT	

Texas Instruments 99/4

The only special characters used are in PRINT statements to indicate where two or more spaces should be left between words. For example, ENERGY { 10 SPACES } MANAGEMENT means that ten spaces should be left between the words ENERGY and MANAGEMENT. Do not type in the braces or the words 10 SPACES. Enter all programs with the ALPHA LOCK on (in the down position). Release the ALPHA LOCK to enter lowercase text.

TRS-80 Color Computer

As with the TI-99/4A, some special characters are used in the PRINT statements. To indicate where two or more spaces should

be left between words, the program listing will show something like `LOCATING {5 SPACES} POINTS`. This means that five spaces should be left between the words `LOCATING` and `POINTS`. Don't type in the braces or the actual words `5 SPACES`. All the programs should be entered in uppercase on the TRS-80 Color Computer.

Another character you'll see in the program listing is the up caret (`^`). This represents the exponential symbol you'll see on your screen as an up arrow (`↑`). Simply press the up arrow key on the left side of the keyboard when you see the `^` symbol in a program.

BASIC Keyword Abbreviations

Most keywords in VIC BASIC can be abbreviated by typing the first one or two letters of the word followed by the SHIFTed next letter. The use of abbreviations will not only save time when entering programs or commands but will reduce the total number of characters per line. Abbreviated keywords will LIST in the full, unabbreviated form when used in a program line.

Most Commonly Used Keyword Abbreviations

Keyword:	Press:	See:	Keyword:	Press:	See:
AND	A SHIFT N	A <input checked="" type="checkbox"/>	ON	NONE	ON
CLR	C SHIFT L	C <input type="checkbox"/>	OR	NONE	OR
DATA	D SHIFT A	D <input type="checkbox"/>	PEEK	P SHIFT E	P <input type="checkbox"/>
DIM	D SHIFT I	D <input type="checkbox"/>	POKE	P SHIFT O	P <input type="checkbox"/>
FOR	F SHIFT O	F <input type="checkbox"/>	PRINT	?	?
GET	G SHIFT E	G <input type="checkbox"/>	READ	R SHIFT E	R <input type="checkbox"/>
GOSUB	GO SHIFT S	GO <input type="checkbox"/>	REM	NONE	REM
GOTO	G SHIFT O	G <input type="checkbox"/>	RESTORE	RE SHIFT S	RE <input type="checkbox"/>
IF	NONE	IF	RETURN	RE SHIFT T	RE <input type="checkbox"/>
INPUT	NONE	INPUT	RND	R SHIFT N	R <input checked="" type="checkbox"/>
INT	NONE	INT	RUN	R SHIFT U	R <input checked="" type="checkbox"/>
LET	L SHIFT E	L <input type="checkbox"/>	SAVE	S SHIFT A	S <input type="checkbox"/>
LIST	L SHIFT I	L <input type="checkbox"/>	STEP	ST SHIFT E	ST <input type="checkbox"/>
LOAD	L SHIFT O	L <input type="checkbox"/>	TAB(T SHIFT A	T <input type="checkbox"/>
NEXT	N SHIFT E	N <input type="checkbox"/>	THEN	T SHIFT H	T <input type="checkbox"/>

Other Abbreviations

Keyword:	Press:	See:	Keyword:	Press:	See:
ABS	A <input type="checkbox"/> SHIFT <input type="checkbox"/> B	A <input type="checkbox"/>	MID\$	M <input type="checkbox"/> SHIFT <input type="checkbox"/> I	M <input type="checkbox"/>
ASC	A <input type="checkbox"/> SHIFT <input type="checkbox"/> S	A <input type="checkbox"/>	NOT	N <input type="checkbox"/> SHIFT <input type="checkbox"/> O	N <input type="checkbox"/>
ATN	A <input type="checkbox"/> SHIFT <input type="checkbox"/> T	A <input type="checkbox"/>	OPEN	O <input type="checkbox"/> SHIFT <input type="checkbox"/> P	O <input type="checkbox"/>
CHR\$	C <input type="checkbox"/> SHIFT <input type="checkbox"/> H	C <input type="checkbox"/>	POS	NONE	POS
CLOSE	CL <input type="checkbox"/> SHIFT <input type="checkbox"/> O	CL <input type="checkbox"/>	PRINT#	P <input type="checkbox"/> SHIFT <input type="checkbox"/> R	P <input type="checkbox"/>
CMD	C <input type="checkbox"/> SHIFT <input type="checkbox"/> M	C <input type="checkbox"/>	RIGHT\$	R <input type="checkbox"/> SHIFT <input type="checkbox"/> I	R <input type="checkbox"/>
CONT	C <input type="checkbox"/> SHIFT <input type="checkbox"/> O	C <input type="checkbox"/>	SGN	S <input type="checkbox"/> SHIFT <input type="checkbox"/> G	S <input type="checkbox"/>
COS	NONE	COS	SIN	S <input type="checkbox"/> SHIFT <input type="checkbox"/> I	S <input type="checkbox"/>
DEF	D <input type="checkbox"/> SHIFT <input type="checkbox"/> E	D <input type="checkbox"/>	SPC(S <input type="checkbox"/> SHIFT <input type="checkbox"/> P	S <input type="checkbox"/>
END	E <input type="checkbox"/> SHIFT <input type="checkbox"/> N	E <input type="checkbox"/>	SQR	S <input type="checkbox"/> SHIFT <input type="checkbox"/> Q	S <input type="checkbox"/>
EXP	E <input type="checkbox"/> SHIFT <input type="checkbox"/> X	E <input type="checkbox"/>	STOP	S <input type="checkbox"/> SHIFT <input type="checkbox"/> T	S <input type="checkbox"/>
FN	NONE	FN	STR\$	ST <input type="checkbox"/> SHIFT <input type="checkbox"/> R	ST <input type="checkbox"/>
FRE	F <input type="checkbox"/> SHIFT <input type="checkbox"/> R	F <input type="checkbox"/>	SYS	S <input type="checkbox"/> SHIFT <input type="checkbox"/> Y	S <input type="checkbox"/>
GET#	NONE	GET#	TAN	NONE	TAN
INPUT#	I <input type="checkbox"/> SHIFT <input type="checkbox"/> N	I <input type="checkbox"/>	USR	U <input type="checkbox"/> SHIFT <input type="checkbox"/> S	U <input type="checkbox"/>
LEFT\$	LE <input type="checkbox"/> SHIFT <input type="checkbox"/> F	LE <input type="checkbox"/>	VAL	V <input type="checkbox"/> SHIFT <input type="checkbox"/> A	V <input type="checkbox"/>
LEN	NONE	LEN	VERIFY	V <input type="checkbox"/> SHIFT <input type="checkbox"/> E	V <input type="checkbox"/>
LOG	NONE	LOG	WAIT	W <input type="checkbox"/> SHIFT <input type="checkbox"/> A	W <input type="checkbox"/>

Index

- abbreviations, VIC-20 264-5
- "Adding Fractions" program 179-85
- "Adverbs" program (translatable) 18-20
- "All Factors" program 151-53
- amortization 23
- "Amortization" program (translatable) 43-44
- Apple computers 23
- "Arithmetic Mean" program (translatable) 36-37
- "Bake a Cake" program (VIC-20) 209, 223-28
- "Bass Clef" program 104-9
- braces 259
- bubble sort 195-96
- "Buying Items" program 113-22
- cakes, how to bake 225-26
- "Circles" program (translatable) 31-34
- clearing screen 13
- "Color Code" program 53-59
- Commodore computers 23
- "Compound Interest" program (translatable) 38-39
- conserving memory 3-10
 - in "Nouns" program 5-6
- "Conversions" program (translatable) 24-27
- "Cookie File" program (VIC-20) 209, 228-31
- "Counting Shapes" program 68-75
- "Countries" program 241-45
- DATA statements 259-60
 - memory conservation and 4, 5
- "Defend" program (VIC-20) 209, 252-55
- DEF FN statement 23-24
- degrees 23
- DIM statement, memory conservation and 3-4
- "Distance" program (translatable) 27-28
- "Dividing Fractions" program 175-79
- "Division" program 149-51
- "Doodle" program 48-53
- DRAW command, memory conservation and 3
- "Earning Money" program 127-36
- "Equivalent Fractions" program 163-68
 - exponentiation 38, 39
- "Find All Factors" program 151-53
- FOR-NEXT, use to save memory 3
- function definitions 23
- GOSUB statement 5
- GOTO statement 5
 - memory conservation and 3
- graphics 47-88
 - computer-specific 47
- "Graphics Demo" program 47-48
- "Greatest Common Factor" program 155-58
- IF-THEN statements 5
 - not always memory-efficient 4
- key, detect which being pressed 14
- "Least Common Multiple" program 159-63
- line numbers
 - low ones save memory 4-5
- "Loan Payments" program (translatable) 42-43
- "Locating Points" program 76-85
- "Matching Shapes" program 59-68
- math competency programs 113-46
- maximum and minimum sort 198-200
- MC-10 computer ix, x, 5, 18, 47
- "Multiplication Invasion" program 245-52
- "Multiplying Fractions" program 171-75
- multistatement lines
 - illegal in TI BASIC 5
 - memory conservation and 4, 5-6
- "Musical keyboard" program 89-96
- "Nouns" program 5-10
- null string 224
- ON-GOTO statements
 - take less space than IF-THEN 4
- ON-GOSUB statements
 - take less space than IF-THEN 4
- "Parallel Resistance" program (translatable) 35-36
- "Prime Factors" program 153-55
- program listing conventions
 - TI 262
 - TRS-80 262
 - VIC-20 261-62
- "Quick Sort" program 200-202
- radians 23
- random numbers 13
- READ statements, memory conservation and 4
- "Rectangles" program (translatable) 29-31
- REM statements, delete to conserve memory 3
- repetition 149

RESTORE statement 225
 "Roman Numerals" program 15-18
 "Sales Tax" program 122-27
 "Saving Money" program 142-46
 scrolling 245
 shell sort 196-98
 "Simple Interest" program (translatable)
 37
 "Simplifying Fractions" program 168-71
 "Simultaneous Equations" program
 186-91
 "Single Payment" program (translatable)
 39
 "Sinking Fund" program (translatable) 41
 sorting 193-206
 sound 87-109
 computer-specific 47, 89
 "South America" program 238-40
 spaces, omit to save memory 4
 "Square Root" program (translatable)
 34-35
 "Squares" program (translatable) 28-29
 "States and Capitals" program 209, 231-38
 "Teeth Wisdom" program 217-23
 "Test Scores" program 202-6

TI BASIC
 function definitions in 24
 multistatement lines not allowed 5
 TimeX/Sinclair computer vii, 13, 14
 TI-99/4 computer 5, 15
 TI-99/4A computer viii, 5, 15, 47, 89
 touch-typing 210
 translatable programs, features of 13
 translating BASIC 13-20
 "Treble Clef" program 96-103
 TRS-80 Color Computer iii, viii, 15, 18,
 38, 39, 47, 89
 TRS-80 computers 23
 TRS-80 Extended Color Basic 38, 39, 40
 TRS-80 4K Color Computer 14, 31, 34
 TRS-80 16K Color Computer 5, 14
 TRS-80 Model I computer vii, viii, 5, 15,
 18, 47
 TRS-80 Model IV computer vii
 "Typing" program 209-17
 "Uniform Series" program (translatable)
 40
 variable names 3
 VIC-20 computer iii, vii, viii, 5, 14, 15, 38,
 47, 89, 209
 "Weekly Expenses" program 136-42

Notes



Notes

Notes



Notes

Notes



If you've enjoyed the articles in this book, you'll find the same style and quality in every monthly issue of **COMPUTE!** Magazine. Use this form to order your subscription to **COMPUTE!**.

For Fastest Service,
Call Our **Toll-Free** US Order Line
800-334-0868
In NC call 919-275-9809

COMPUTE!

P.O. Box 5406
Greensboro, NC 27403

My Computer is:

- Commodore 64 TI-99/4A Timex/Sinclair VIC-20 PET
 Radio Shack Color Computer Apple Atari Other _____
 Don't yet have one...

- \$24 One Year US Subscription
 \$45 Two Year US Subscription
 \$65 Three Year US Subscription

Subscription rates outside the US:

- \$30 Canada
 \$42 Europe, Australia, New Zealand/Air Delivery
 \$52 Middle East, North Africa, Central America/Air Mail
 \$72 Elsewhere/Air Mail
 \$30 International Surface Mail (lengthy, unreliable delivery)

Name _____

Address _____

City _____

State _____

Zip _____

Country _____

Payment must be in US Funds drawn on a US Bank; International Money Order, or charge card.

Payment Enclosed


VISA

MasterCard

American Express

Acc t. No. _____

Expires _____ / _____



If you've enjoyed the articles in this book, you'll find the same style and quality in every monthly issue of **COMPUTE!'s Gazette** for Commodore.

For Fastest Service
Call Our **Toll-Free** US Order Line
800-334-0868
In NC call **919-275-9809**

COMPUTE!'s GAZETTE

P.O. Box 5406
Greensboro, NC 27403

My computer is:

Commodore 64 VIC-20 Other _____
01 02 03

- \$20 One Year US Subscription
 \$36 Two Year US Subscription
 \$54 Three Year US Subscription

Subscription rates outside the US:

- \$25 Canada
 \$45 Air Mail Delivery
 \$25 International Surface Mail

Name _____

Address _____

City _____

State _____

Zip _____

Country _____


Payment must be in US Funds drawn on a US Bank, International Money Order, or charge card. Your subscription will begin with the next available issue. Please allow 4-6 weeks for delivery of first issue. Subscription prices subject to change at any time.

- Payment Enclosed VISA
 MasterCard American Express

Acct. No. _____

Expires _____

The *COMPUTE!'s Gazette* subscriber list is made available to carefully screened organizations with a product or service which may be of interest to our readers. If you prefer not to receive such mailings, please check this box .



COMPUTE! Books

P.O. Box 5406 Greensboro, NC 27403

Ask your retailer for these **COMPUTE! Books**. If he or she has sold out, order directly from **COMPUTE!**

For Fastest Service
Call Our **TOLL FREE US Order Line**

800-334-0868
In NC call 919-275-9809

Quantity	Title	Price	Total
_____	Machine Language for Beginners	\$14.95*	_____
_____	Home Energy Applications	\$14.95*	_____
_____	COMPUTE!'s First Book of VIC	\$12.95*	_____
_____	COMPUTE!'s Second Book of VIC	\$12.95*	_____
_____	COMPUTE!'s First Book of VIC Games	\$12.95*	_____
_____	COMPUTE!'s First Book of 64	\$12.95*	_____
_____	COMPUTE!'s First Book of Atari	\$12.95*	_____
_____	COMPUTE!'s Second Book of Atari	\$12.95*	_____
_____	COMPUTE!'s First Book of Atari Graphics	\$12.95*	_____
_____	COMPUTE!'s First Book of Atari Games	\$12.95*	_____
_____	Mapping The Atari	\$14.95*	_____
_____	Inside Atari DOS	\$19.95*	_____
_____	The Atari BASIC Sourcebook	\$12.95*	_____
_____	Programmer's Reference Guide for TI-99/4A	\$14.95*	_____
_____	COMPUTE!'s First Book of TI Games	\$12.95*	_____
_____	Every Kid's First Book of Robots and Computers	\$ 4.95†	_____
_____	The Beginner's Guide to Buying A Personal Computer	\$ 3.95†	_____

* Add \$2 shipping and handling Outside US add \$5 air mail. \$2 surface mail

† Add \$1 shipping and handling Outside US add \$5 air mail. \$2 surface mail

Please add shipping and handling for each book ordered. _____

Total enclosed or to be charged. _____

All orders must be prepaid (money order, check, or charge). All payments must be in US funds. NC residents add 4% sales tax.

Payment enclosed Please charge my: VISA MasterCard
 American Express Acc't. No. _____ Expires ____/____


Name _____

Address _____

City _____ State _____ Zip _____

Country _____

Allow 4-5 weeks for delivery.



Your computer may have limited memory, but it's still a powerful tool — with the right programs. In less than 4K of memory, you can play games, check homework, calculate a loan, or create dazzling graphic displays. When you try out the dozens of different programs in this book, you'll discover the power of your computer.

C. Regena, columnist for **COMPUTE! Magazine** and **COMPUTE!'s Gazette** and author of the best-selling **Programmer's Reference Guide to the TI-99/4A**, knows how to get the most out of a small computer. Her programs for the VIC-20, the TRS-80 Color Computer, and the TI-99/4A are not only excellent software, but also examples of effective, memory-saving programming. You'll see how to translate programs from one computer to another, as well as the way that the programs work. The programs and information include:

- How to conserve memory and get more into your computer
- Graphics and sound
- Games that are fun to play — while they teach
- Homework helpers
- Computer sorting
- Typing, geography, and mathematics aids